

# Users' manual for the **Sollya** tool

Release 2.0 alpha

Équipe-projet Arénaire  
Laboratoire de l'Informatique du Parallélisme

UMR CNRS - ENS Lyon - UCB Lyon 1 - INRIA 5668

Sylvain Chevillard                      Christoph Lauter  
`sylvain.chevillard@ens-lyon.org`      `christoph.lauter@ens-lyon.org`

Mioara Joldes  
`mioara.joldes@ens-lyon.fr`

March 16, 2010

## License

The **Sollya** tool is copyright © 2008-2010 Laboratoire de l'Informatique du Parallélisme - UMR CNRS - ENS Lyon - UCB Lyon 1 - INRIA 5668.

The **Sollya** tool is open software. It is distributed and can be used, modified and redistributed under the terms of the CeCILL-C licence available at <http://www.cecill.info/> and reproduced in the **COPYING** file of the distribution. The distribution contains parts of other libraries as a support for but not integral part of **Sollya**. These libraries are reigned by the GNU Lesser General Public License that is available at <http://www.gnu.org/licenses/> and reproduced in the **COPYING** file of the distribution.

## Contents

|          |  |           |
|----------|--|-----------|
| <b>1</b> | <b>Compilation and installation of the Sollya tool</b> | <b>5</b>  |
| 1.1      | Compilation dependencies . . . . .                     | 5         |
| 1.2      | Sollya command line options . . . . .                  | 5         |
| <b>2</b> | <b>Introduction</b>                                    | <b>6</b>  |
| <b>3</b> | <b>General principles</b>                              | <b>8</b>  |
| <b>4</b> | <b>Variables</b>                                       | <b>10</b> |
| <b>5</b> | <b>Data types</b>                                      | <b>11</b> |
| 5.1      | Booleans . . . . .                                     | 11        |
| 5.2      | Numbers . . . . .                                      | 12        |
| 5.3      | Rational numbers and rational arithmetic . . . . .     | 14        |
| 5.4      | Intervals and interval arithmetic . . . . .            | 15        |
| 5.5      | Functions . . . . .                                    | 19        |
| 5.6      | Strings . . . . .                                      | 20        |
| 5.7      | Particular values . . . . .                            | 20        |
| 5.8      | Lists . . . . .  | 20        |

|          |   |           |
|----------|---|-----------|
| <b>6</b> | <b>Iterative language elements: assignments, conditional statements and loops</b> | <b>22</b> |
| 6.1      | Blocks . . . . .  | 22        |
| 6.2      | Assignments . . . . .   | 22        |
| 6.3      | Conditional statements . . . . .  | 23        |
| 6.4      | Loops . . . . .   | 23        |
| <b>7</b> | <b>Functional language elements: procedures</b>                                   | <b>25</b> |
| <b>8</b> | <b>Commands and functions</b>   | <b>26</b> |
| 8.1      | abs . . . . .   | 26        |
| 8.2      | absolute . . . . .  | 27        |
| 8.3      | accurateinfnorm . . . . .   | 27        |
| 8.4      | acos . . . . .  | 28        |
| 8.5      | acosh . . . . .   | 28        |
| 8.6      | && . . . . .  | 28        |
| 8.7      | ∴ . . . . .   | 29        |
| 8.8      | ~ . . . . .   | 29        |
| 8.9      | asciipLOT . . . . .   | 31        |
| 8.10     | asin . . . . .  | 32        |
| 8.11     | asinh . . . . .   | 32        |
| 8.12     | atan . . . . .  | 33        |
| 8.13     | atanh . . . . .   | 33        |
| 8.14     | autosimplify . . . . .  | 33        |
| 8.15     | bashexecute . . . . .   | 34        |
| 8.16     | binary . . . . .  | 34        |
| 8.17     | boolean . . . . .   | 35        |
| 8.18     | canonical . . . . .   | 35        |
| 8.19     | ceil . . . . .  | 36        |
| 8.20     | checkinfnorm . . . . .  | 36        |
| 8.21     | coeff . . . . .   | 37        |
| 8.22     | @ . . . . .   | 38        |
| 8.23     | constant . . . . .  | 39        |
| 8.24     | cos . . . . .   | 39        |
| 8.25     | cosh . . . . .  | 39        |
| 8.26     | decimal . . . . .   | 40        |
| 8.27     | default . . . . .   | 40        |
| 8.28     | degree . . . . .  | 40        |
| 8.29     | denominator . . . . .   | 41        |
| 8.30     | diam . . . . .  | 42        |
| 8.31     | diff . . . . .  | 42        |
| 8.32     | dirtyfindzeros . . . . .  | 43        |
| 8.33     | dirtyinfnorm . . . . .  | 43        |
| 8.34     | dirtyintegral . . . . .   | 44        |
| 8.35     | display . . . . .   | 45        |
| 8.36     | . . . . .   | 46        |
| 8.37     | double . . . . .  | 47        |
| 8.38     | doubledouble . . . . .  | 47        |
| 8.39     | doubleextended . . . . .  | 48        |
| 8.40     | dyadic . . . . .  | 49        |
| 8.41     | == . . . . .  | 49        |
| 8.42     | erf . . . . .   | 50        |
| 8.43     | erfc . . . . .  | 50        |
| 8.44     | error . . . . .   | 51        |
| 8.45     | evaluate . . . . .  | 51        |
| 8.46     | execute . . . . .   | 53        |
| 8.47     | exp . . . . .   | 54        |

|       |                   |    |
|-------|-------------------|----|
| 8.48  | expand            | 54 |
| 8.49  | expm1             | 54 |
| 8.50  | exponent          | 55 |
| 8.51  | externalplot      | 55 |
| 8.52  | externalproc      | 56 |
| 8.53  | false             | 58 |
| 8.54  | file              | 58 |
| 8.55  | findzeros         | 58 |
| 8.56  | fixed             | 59 |
| 8.57  | floating          | 59 |
| 8.58  | floor             | 60 |
| 8.59  | fpminimax         | 60 |
| 8.60  | fullparentheses   | 63 |
| 8.61  | function          | 64 |
| 8.62  | $\geq$            | 64 |
| 8.63  | $>$               | 65 |
| 8.64  | guessdegree       | 65 |
| 8.65  | head              | 66 |
| 8.66  | hexadecimal       | 67 |
| 8.67  | honorcoeffprec    | 67 |
| 8.68  | hopitalrecursions | 67 |
| 8.69  | horner            | 68 |
| 8.70  | implementpoly     | 69 |
| 8.71  | inf               | 71 |
| 8.72  | infnorm           | 72 |
| 8.73  | integer           | 74 |
| 8.74  | integral          | 74 |
| 8.75  | isbound           | 75 |
| 8.76  | isevaluable       | 76 |
| 8.77  | $\leq$            | 77 |
| 8.78  | length            | 78 |
| 8.79  | library           | 78 |
| 8.80  | list of           | 79 |
| 8.81  | log               | 79 |
| 8.82  | log10             | 80 |
| 8.83  | log1p             | 80 |
| 8.84  | log2              | 80 |
| 8.85  | $<$               | 80 |
| 8.86  | mantissa          | 81 |
| 8.87  | mid               | 81 |
| 8.88  | midpointmode      | 82 |
| 8.89  | $-$               | 82 |
| 8.90  | $*$               | 83 |
| 8.91  | nearestint        | 84 |
| 8.92  | $\neq$            | 84 |
| 8.93  | nop               | 85 |
| 8.94  | !                 | 86 |
| 8.95  | numerator         | 86 |
| 8.96  | off               | 87 |
| 8.97  | on                | 87 |
| 8.98  |                   | 88 |
| 8.99  | parse             | 88 |
| 8.100 | perturb           | 89 |
| 8.101 | pi                | 89 |
| 8.102 | plot              | 90 |
| 8.103 | +                 | 92 |

|       |                   |     |
|-------|-------------------|-----|
| 8.104 | points            | 92  |
| 8.105 | postscript        | 93  |
| 8.106 | postscriptfile    | 93  |
| 8.107 | $\wedge$          | 94  |
| 8.108 | powers            | 95  |
| 8.109 | prec              | 95  |
| 8.110 | precision         | 95  |
| 8.111 | .                 | 96  |
| 8.112 | print             | 96  |
| 8.113 | printexpansion    | 99  |
| 8.114 | printfloat        | 100 |
| 8.115 | printhexa         | 100 |
| 8.116 | printxml          | 101 |
| 8.117 | proc              | 102 |
| 8.118 | procedure         | 107 |
| 8.119 | quit              | 108 |
| 8.120 | range             | 108 |
| 8.121 | rationalapprox    | 109 |
| 8.122 | rationalmode      | 110 |
| 8.123 | RD                | 111 |
| 8.124 | readfile          | 111 |
| 8.125 | readxml           | 112 |
| 8.126 | relative          | 112 |
| 8.127 | remez             | 113 |
| 8.128 | rename            | 114 |
| 8.129 | restart           | 115 |
| 8.130 | return            | 116 |
| 8.131 | revert            | 117 |
| 8.132 | RN                | 118 |
| 8.133 | round             | 118 |
| 8.134 | roundcoefficients | 119 |
| 8.135 | roundcorrectly    | 120 |
| 8.136 | roundingwarnings  | 121 |
| 8.137 | RU                | 122 |
| 8.138 | RZ                | 122 |
| 8.139 | searchgal         | 122 |
| 8.140 | simplify          | 124 |
| 8.141 | simplifysafe      | 124 |
| 8.142 | sin               | 125 |
| 8.143 | single            | 125 |
| 8.144 | sinh              | 126 |
| 8.145 | sort              | 126 |
| 8.146 | sqrt              | 126 |
| 8.147 | string            | 126 |
| 8.148 | subpoly           | 127 |
| 8.149 | substitute        | 127 |
| 8.150 | sup               | 128 |
| 8.151 | tail              | 129 |
| 8.152 | tan               | 129 |
| 8.153 | tanh              | 130 |
| 8.154 | taylor            | 130 |
| 8.155 | taylorrecursions  | 131 |
| 8.156 | time              | 131 |
| 8.157 | timing            | 132 |
| 8.158 | tripledouble      | 133 |
| 8.159 | true              | 134 |

|                          |     |
|--------------------------|-----|
| 8.160var . . . . .       | 134 |
| 8.161verbosity . . . . . | 134 |
| 8.162void . . . . .      | 135 |
| 8.163worstcase . . . . . | 137 |
| 8.164write . . . . .     | 137 |

# 1 Compilation and installation of the Sollya tool

## 1.1 Compilation dependencies

The Sollya distribution can be compiled and installed using the usual `./configure`, `make`, `make install` procedure. Besides a C compiler, Sollya needs the following software libraries and tools to be installed. The `./configure` script checks for the installation of the libraries. However Sollya will build without error if some of its external tools are not installed. In this case an error will be displayed at runtime.

- GMP
- MPFR
- MPFI
- fp111
- libxml2
- gnuplot

The use of the external tool `rlwrap` is highly recommended but not required. Use the `-A` options of `rlwrap` for correctly displayed ANSI X3.64/ ISO/IEC 6429 colored prompts (see below).

## 1.2 Sollya command line options

Sollya can read input on standard input or in a file whose name is given as an argument when Sollya is invoked. The tool will always produce its output on standard output, unless specifically instructed by a particular Sollya command that writes to a file. The following lines are valid invocations of Sollya, assuming that `bash` is used as a shell:

```
~/ % sollya
...
~/ % sollya myfile.sollya
...
~/ % sollya < myfile.sollya
```

If a file given as an input does not exist, an error message is displayed.

All configurations of the internal state of the tool are done by commands given on the Sollya prompt or in Sollya scripts. Nevertheless, some command line options are supported; they work at a very basic I/O-level and can therefore not be implemented as commands.

The following options are supported when calling Sollya:

- `--nocolor`: Sollya supports coloring of the output using ANSI X3.64/ ISO/IEC 6429 escape sequences. Coloring is deactivated when Sollya is connected on standard input to a file that is not a terminal. This option forces the deactivation of ANSI coloring. This might be necessary on very old grey-scale terminals or when encountering problems with old versions of `rlwrap`.
- `--noprompt`: Sollya prints a prompt symbol when connected on standard input to a pseudo-file that is a terminal. The option deactivates the prompt.

- `--oldrlwrapcompatible`: This option is deprecated. It makes **Sollya** emit a non ANSI X3.64 compliant coloring escape sequence for making it compatible with versions of **rlwrap** that do not support the `-A` option. The option is considered as a hack since it is known to garble the output of the tool under some particular circumstances.
- `--donotmodifystacksize`: When invoked, **Sollya** tries to increase the stack size that is available to a user process to the maximum size supported by the kernel. On some systems, the correspondent `ioctl` does not work properly. Use the option to prevent **Sollya** from changing the stack size.
- `--oldautoprint`: The behaviour of an undocumented feature for displaying values has changed in **Sollya** from version 1.1 to version 2.0. The old feature is deprecated. If you wish to use it nevertheless, use this deprecated option.
- `--help`: Prints help on the usage of the tool and quits.

## 2 Introduction

**Sollya** is an interactive tool for handling numerical functions and working with arbitrary precision. It can evaluate functions accurately, compute polynomial approximations of functions, automatically implement polynomials for use in math libraries, plot functions, compute infinity norms, etc. The language **Sollya** comes with is a full-featured script programming language with support for procedures etc.

Let us begin this manual with an example. **Sollya** does not allow command line edition; since that may quickly become uncomfortable, we highly suggest to use the **rlwrap** tool with **Sollya**:

```
~/ % rlwrap -A sollya
>
```

**Sollya** manipulates only functions in one variable. The first time that an unbound variable is used, this name is fixed. It will be used to refer to the free variable. For instance, try

```
> f = sin(x)/x;
> g = cos(y)-1;
Warning: the identifier "y" is neither assigned to, nor bound to a library function nor external procedure, nor equal to the current free variable.
Will interpret "y" as "x".
> g;
cos(x) - 1
```

Now, the name  $x$  can only be used to refer to the free variable:

```
> x = 3;
Warning: the identifier "x" is already bound to the free variable, to a library function or to an external procedure.
The command will have no effect.
Warning: the last assignment will have no effect.
```

If you really want to unbound  $x$ , you can use the **rename** command and change the name of the free variable:

```
> rename(x,y);
Information: the free variable has been renamed from "x" to "y".
> g;
cos(y) - 1
> x=3;
> x;
3
```

As you have seen, you can name functions and easily work with them. The basic thing to do with a function is to evaluate it at some point:

```
> f(-2);
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.45464871341284084769800993295587242135112748572394
> evaluate(f,-2);
0.45464871341284084769800993295587242135112748572394
```

The printed value is generally a faithful rounding of the exact value at the working precision. The working precision is controlled by the global variable `prec`:

```
> prec;
165
> prec=200;
The precision has been set to 200 bits.
> prec;
200
> f(-2);
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.4546487134128408476980099329558724213511274857239451341894865
```

Sometimes a faithful rounding cannot easily be computed. In such a case, an approximated value is printed:

```
> log2(5)/log2(17) - log(5)/log(17);
Warning: rounding may have happened.
If there is rounding, the displayed value is *NOT* guaranteed to be a faithful rounding of the true result.
0
```

The philosophy of Sollya is: *Whenever something is not exact, print a warning.* This explains the warnings in the previous examples. If the result can be shown to be exact, there is no warning:

```
> sin(0);
0
```

Let us finish this Section with a small complete example that shows a bit of what can be done with Sollya:

```

> restart;
The tool has been restarted.
> prec=50;
The precision has been set to 50 bits.
> f=cos(2*exp(x));
> d=[-1/8;1/8];
> p=remez(f,2,d);
> derivativeZeros = dirtyfindzeros(diff(p-f),d);
> derivativeZeros = inf(d)::derivativeZeros::sup(d);
> max=0;
> for t in derivativeZeros do {
>   r = evaluate(abs(p-f), t);
>   if r > max then { max=r; argmax=t; };
> };
> print("The infinity norm of", p-f, "is", max, "and is reached at", argmax);
The infinity norm of -0.416265572875373 + x * (-1.798067209218835 + x * (-3.8971
0727747639e-2)) - cos(2 * exp(x)) is 8.630659443624325e-4 and is reached at -5.8
01672331417684e-2

```

In this example, we define a function  $f$ , an interval  $d$  and we compute the best degree-4 polynomial approximation of  $f$  on  $d$  with respect to the infinity norm. In other words,  $\max_{x \in d} \{|p(x) - f(x)|\}$  is minimal amongst polynomials with degree not greater than 4. Then, we compute the list of the zeros of the derivative of  $p - f$  and add the bounds of  $d$  to this list. Finally, we evaluate  $|p - f|$  for each point in the list and store the maximum and the point where it is reached. We conclude by printing the result in a formatted way.

Note that you do not really need to use such a script for computing infinity norm; as we will see, the command `dirtyinfnorm` does this for you.

### 3 General principles

The first goal of **Sollya** is to help people to use numerical functions and numerical algorithms in a safe way. It is first designed to be used interactively but it can also be used in scripts<sup>1</sup>.

One of the particularities of **Sollya** is to work with multi-precision arithmetic (it uses the **MPFR** library). For safety purposes, **Sollya** knows how to use interval arithmetic. It uses interval arithmetic to produce tight and safe results with the precision required by the user.

The general philosophy of **Sollya** is: *When you can make a computation exactly and sufficiently quickly, do it; when you cannot, do not, unless you have been explicitly asked for.*

The precision of the tool is set by the global variable `prec`. In general, the variable `prec` determines the precision of the outputs of commands: more precisely, the command will internally determine what precision should be used during the computations in order to ensure that the output is a faithfully rounded result with `prec` bits.

For decidability and efficiency reasons, this general principle cannot be applied every time, so be careful. Moreover certain commands are known to be unsafe: they give in general excellent results and give almost `prec` correct bits in output for everyday examples. However they are merely based on heuristics and should not be used when the result must be safe. See the documentation of each command to know precisely how confident you can be with their result.

A second principle (that comes together with the first one) is: *When a computation leads to inexact results, inform the user with a warning.* This can be quite irritating in some circumstances: in particular if you are using **Sollya** within other scripts. The global variable `verbosity` lets you change the level of verbosity of **Sollya**. When the variable is set to 0, **Sollya** becomes completely silent on standard output and prints only very important messages on standard error. Increase `verbosity` if you want more information about what **Sollya** is doing. Note that when you affect a value to a global variable, a message is always printed even if `verbosity` is set to 0. In order to silently affect a global variable, use `!`:

<sup>1</sup>Remark: some of the behaviors of **Sollya** slightly change when it is used in scripts. For example, no prompt is printed.



```
> prec=30;
The precision has been set to 30 bits.
> prec=30!;
>
```

For conviviality reasons, values are displayed in decimal by default. This lets a normal human being understand the numbers they manipulate. But since constants are internally represented in binary, this causes permanent conversions that are sources of roundings. Thus you are loosing in accuracy and **Sollya** is always complaining about inexact results. If you just want to store or communicate your results (to another tools for instance) you can use bit-exact representations available in **Sollya**. The global variable **display** defines the way constants are displayed. Here is an example of the five available modes:

```
> prec=30!;
> a = 17.25;
> display=decimal;
Display mode is decimal numbers.
> a;
1.725e1
> display=binary;
Display mode is binary numbers.
> a;
1.000101_2 * 2^(4)
> display=powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
> a;
69 * 2^(-2)
> display=dyadic;
Display mode is dyadic numbers.
> a;
69b-2
> display=hexadecimal;
Display mode is hexadecimal numbers.
> a;
0x1.14p4
```

Please note that it is possible to maintain the general verbosity level at some higher setting while deactivating all warnings on roundings. This feature is controlled using the **roundingwarnings** global variable. It may be set to **on** or **off**. By default, the warnings are activated (**roundingwarnings = on**) when **Sollya** is connected on standard input to a pseudo-file that represents a terminal. They are deactivated when **Sollya** is connected on standard input to a real file. See 8.136 for further details; the behavior is illustrated with examples there.

As always, the symbol **e** means  $\times 10^\square$ . The same way the symbol **b** means  $\times 2^\square$ . The symbol **p** means  $\times 16^\square$  and is used only with the **0x** prefix. The prefix **0x** indicates that the digits of the following number until a symbol **p** or white-space are hexadecimal. The suffix **\_2** indicates to **Sollya** that the previous number has been written in binary. **Sollya** can parse these notations even if you are not in the corresponding **display** mode, so you can always use them.

You can also use memory-dump hexadecimal notation frequently used to represent IEEE 754 **double** and **single** precision numbers. Since this notation does not allow for exactly representing numbers with arbitrary precision, there is no corresponding **display** mode. However, the commands **printhexa** respectively **printfloat** round the value to the nearest **double** respectively **single**. The number is then printed in hexadecimal as the integer number corresponding to the memory representation of the IEEE 754 **double** or **single** number:

```
> printhexa(a);
0x4031400000000000
> printfloat(a);
0x418a0000
```

**Sollya** can parse these memory-dump hexadecimal notation back in any **display** mode. The difference of this memory-dump notation with the hexadecimal notation (as defined above) is made by the presence or absence of a **p** indicator.

## 4 Variables

As already explained, **Sollya** can manipulate variate functional expressions in one variable. These expressions contain a unique free variable the name of which is fixed by its first usage in an expression that is not a left-hand-side of an assignment. This global and unique free variable is a variable in the mathematical sense of the term.

**Sollya** also provides variables in the sense programming languages give to the term. These variables, which must be different in their name from the global free variable, may be global or declared and attached to a block of statements, i.e. a begin-end-block. These programming language variables may hold any object of the **Sollya** language, as for example functional expressions, strings, intervals, constant values, procedures, external functions and procedures, etc.

Global variables need not to be declared. They start existing, i.e. can be correctly used in expressions that are not left-hand-sides of assignments, when they are assigned a value in an assignment. Since they are global, this kind of variables is recommended only for small **Sollya** scripts. Larger scripts with code reuse should use declared variables in order to avoid name clashes for example in loop variables.

Declared variables are attached to a begin-end-block. The block structure builds scopes for declared variables. Declared variables in inner scopes shadow variables (global and declared) of outer scopes. The global free variable, i.e. the mathematical variable for variate functional expressions in one variable, cannot be shadowed. Variables are declared using **var** keyword. See section 8.160 for details on its usage and semantic.

The following code examples illustrate the usage of variables.

```

> f = exp(x);
> f;
exp(x)
> a = "Hello world";
> a;
Hello world
> b = 5;
> f(b);
Warning: rounding has happened. The value displayed is a faithful rounding of the
true result.
1.48413159102576603421115580040552279623487667593878e2
> {var b; b = 4; f(b); };
Warning: rounding has happened. The value displayed is a faithful rounding of the
true result.
5.45981500331442390781102612028608784027907370386137e1
> {var x; x = 3; };
Warning: the identifier "x" is already bound to the current free variable.
It cannot be declared as a local variable. The declaration of "x" will have no effect.
Warning: the identifier "x" is already bound to the free variable, to a library
function or to an external procedure.
The command will have no effect.
Warning: the last assignment will have no effect.
> {var a, b; a=5; b=3; {var a; var b; b = true; a = 1; a; b;}; a; b; };
1
true
5
3
> a;
Hello world

```

Let us state that a variable identifier, just as every identifier in **Sollya**, contains at least one character, starting with a ASCII letter or one of the special symbols “\_”, “\$” or “\$” and continuing with ASCII letters, special symbols or numerical digits.

## 5 Data types

**Sollya** has a (very) basic system of types. If you try to perform an illicit operation (such as adding a number and a string, for instance), you will get a typing error. Let us see the available data types.

### 5.1 Booleans

There are two special values **true** and **false**. Boolean expressions can be constructed using the boolean connectors **&&** (and), **||** (or), **!** (not), and comparisons.

The comparison operators **<**, **<=**, **>** and **>=** can only be used between two numbers or constant expressions.

The comparison operators **==** and **!=** are polymorphic. You can use it to compare any two objects, like two strings, two intervals, etc. Note that testing the equality between two functions will return **true** if and only if the expression trees representing the two functions are exactly the same. See 8.44 for an exception concerning the special object **error**. Example:

```

> 1+x==1+x;
true

```

## 5.2 Numbers

Sollya represents numbers as binary multi-precision floating-point values. For integer values and values in dyadic, binary, hexadecimal or memory dump notation, it automatically uses a precision needed for representing the value exactly. Additionally, automatic precision adaption takes place for all integer values (even in decimal notation) written without the exponent sign `e` or with the exponent sign `e` and an exponent sufficiently small that they are less than  $10^{999}$ . Otherwise the values are represented with the current precision `prec`. A number in an expression is rounded to the precision `prec` when the expression gets evaluated:

```
> prec=12!;
> 4097.1;
Warning: Rounding occurred when converting the constant "4097.1" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
4098
> 4097.1+1;
Warning: Rounding occurred when converting the constant "4097.1" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
4099
```

Note that each variable has its own precision that corresponds to its intrinsic precision or, if it cannot be represented, to the value of `prec` when the variable was set. Thus you can work with variables having a precision bigger than the current precision.

The same way, if you define a function that refers to some constant, this constant is stored in the function with the current precision and will keep this value in the future, even if `prec` becomes smaller.

If you define a function that refers to some variable, the precision of the variable is kept, independently of the current precision:

```
> prec = 50!;
> a = 4097.1;
Warning: Rounding occurred when converting the constant "4097.1" to floating-point with 50 bits.
If safe computation is needed, try to increase the precision.
> prec = 12!;
> f = x + a;
> g = x + 4097.1;
Warning: Rounding occurred when converting the constant "4097.1" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
> prec = 120;
The precision has been set to 120 bits.
> f;
4.097099999999998544808477163314819335e3 + x
> g;
4098 + x
```

Sollya is an environment that uses floating-point arithmetic. The IEEE 754-2008 standard on floating-point arithmetic does not only define floating-point numbers that represent real numbers but also floating-point data representing infinities and Not-a-Numbers (NaNs). Sollya also supports infinities and NaNs in the spirit of the IEEE 754-2008 standard without taking the standard's choices literally.

- Signed infinities are available through the Sollya objects `infty`, `-infty`, `@Inf@` and `-@Inf@`.

- Not-a-Numbers are supported through the **Sollya** objects **NaN** and **@NaN@**. **Sollya** does not have support for NaN payloads, signaling or quiet NaNs or signs of NaNs. Signaling NaNs are supported for input for single and double precision memory notation (see section 3). However, they immediately get converted to plain **Sollya** NaNs.

The evaluation of an expression involving a NaN or the evaluation of a function at a point being NaN always results in a NaN.

Infinities are considered to be the limits of expressions tending to infinity. They are supported as bounds of intervals in some cases. However, particular commands might prohibit their use even though there might be a mathematical meaning attached to such expressions. For example, while **Sollya** will evaluate expressions such as  $\lim_{x \rightarrow -\infty} e^x$ , expressed e.g. through `evaluate(exp(x),[-infty;0])`, it will not accept to compute the (finite) value of

$$\int_{-\infty}^0 e^x dx.$$

The following examples give an idea of what can be done with **Sollya** infinities and NaNs. Here is what can be done with infinities:

```
> f = exp(x) + 5;
> f(-infty);
5
> evaluate(f,[-infty;infty]);
[5;@Inf@]
> f(infty);
Warning: the given expression is undefined or numerically unstable.
@NaN@
> [-infty;5] * [3;4];
[-@Inf@;20]
> -infty < 5;
true
> log(0);
Warning: the given expression is undefined or numerically unstable.
@NaN@
> [log(0);17];
Warning: the given expression is not a constant but an expression to evaluate and
a faithful evaluation is not possible.
Will use a plain floating-point evaluation, which might yield a completely wrong
value.
Warning: inclusion property is satisfied but the diameter may be greater than the
least possible.
[-@Inf@;17]
>
```

And the following example illustrates NaN behavior.

```

> 3/0;
Warning: the given expression is undefined or numerically unstable.
@NaN@
> (-3)/0;
Warning: the given expression is undefined or numerically unstable.
@NaN@
> infty/infty;
Warning: the given expression is undefined or numerically unstable.
@NaN@
> infty + infty;
Warning: the given expression is undefined or numerically unstable.
@Inf@
> infty - infty;
Warning: the given expression is undefined or numerically unstable.
@NaN@
> f = exp(x) + 5;
> f(NaN);
Warning: the given expression is undefined or numerically unstable.
@NaN@
> NaN == 5;
false
> NaN == NaN;
false
> NaN != NaN;
false
> X = "Vive la Republique!";
> !(X == X);
false
> X = 5;
> !(X == X);
false
> X = NaN;
> !(X == X);
true
>

```

### 5.3 Rational numbers and rational arithmetic

The **Sollya** tool is mainly based on floating-point arithmetic: wherever possible, floating-point algorithms, including algorithms using interval arithmetic, are used to produce approximate but safe results. For some particular cases, floating-point arithmetic is not sufficient: some algorithms just require natural and rational numbers to be handled exactly. More importantly, for these applications, it is required that rational numbers be displayed as such.

**Sollya** implements a particular mode that offers a lightweight support for rational arithmetic. When needed, it can be enabled by assigning **on** to the global variable **rationalmode**. It is disabled by assigning **off**; the default is **off**.

When the mode for rational arithmetic is enabled, **Sollya**'s behavior will change as follows:

- When a constant expression is given at the **Sollya** prompt, **Sollya** will first try to simplify the expression to a rational number. If such an evaluation to a rational number is possible, **Sollya** will display that number as an integer or a fraction of two integers. Only if **Sollya** is not able to simplify the constant expression to a rational number, it will launch the default behavior of evaluating constant expressions to floating-point numbers that are generally faithful roundings of the expressions.
- When the global mode **autosimplify** is **on**, which is the default, **Sollya** will additionally use rational arithmetic while trying to simplify expressions given in argument of commands.

Even when `rationalmode` is on, Sollya will not be able to exhibit integer ratios between transcendental quantities. For example, Sollya will not display  $\frac{1}{6}$  for  $\arcsin(\frac{1}{2})/\pi$  but 0.16666.... Sollya's evaluator for rational arithmetic is only able to simplify rational expressions based on addition, subtraction, multiplication, division, negation, perfect squares (for square root) and integer powers.

The following example illustrates what can and what cannot be done with Sollya's mode for rational arithmetic:

```
> 1/3 - 1/7;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.19047619047619047619047619047619047619047619047619
> rationalmode = on;
Rational mode has been activated.
> 1/3 - 1/7;
4 / 21
> (2 + 1/7)^2 + (6/7)^2 + 2 * (2 + 1/7) * 6/7;
9
> rationalmode = off;
Rational mode has been deactivated.
> (2 + 1/7)^2 + (6/7)^2 + 2 * (2 + 1/7) * 6/7;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
9
> rationalmode = on;
Rational mode has been activated.
> asin(1)/pi;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.5
> sin(1/6 * pi);
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.5
> exp(1/7 - 3/21) / 7;
1 / 7
> rationalmode = off;
Rational mode has been deactivated.
> exp(1/7 - 3/21) / 7;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.142857142857142857142857142857142857142857142857145
> print(1/7 - 3/21);
1 / 7 - 3 / 21
> rationalmode = on;
Rational mode has been activated.
> print(1/7 - 3/21);
0
```

## 5.4 Intervals and interval arithmetic

Sollya intervals are generally closed subsets of the real numbers. Several ways of defining intervals exist in Sollya. There is the most common way where intervals are composed of two numbers or constant expressions representing the lower and the upper bound. These values are separated either by commas or semi-colons:

```

> d=[1;2];
> d2=[1,1+1];
> d==d2;
true

```

Sollya has a mode for printing intervals that are that thin that their bounds have a number of decimal digits in common when printed. That mode is called `midpointmode`; see section 8.88 for details. As Sollya must be able to parse back its own output, a syntax is provided to input intervals in midpoint mode:

```

> midpointmode = on!;
> 1.7~25/5~e4;
0.17~2/5~e5
> midpointmode = off!;
> 1.7~25/5~e4;
[17250;17500]

```

In some cases, intervals become infinitely thin in theory, in which case one tends to think of point intervals even if their floating-point representation is not infinitely thin. Sollya provides a very convenient way for input of such point intervals. Instead of writing `[a;a]`, it is possible to just write `[a]`. Sollya will expand the notation while making sure that the inclusion property is satisfied:

```

> [3];
[3;3]
> [1/7];
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
[0.14285713;0.14285716]
> [exp(8)];
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
[2.9809577e3;2.9809582e3]

```

When the mode `midpointmode` is set to `on` (see 8.88), Sollya will display intervals that are provably reduced to one point in this extended interval syntax. It will use `midpointmode` syntax for intervals that are sufficiently thin but not reduced to one point (see section 8.88 for details):



```

> midpointmode = off;
Midpoint mode has been deactivated.
> [17;17];
[17;17]
> [exp(pi);exp(pi)];
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
[2.31406926327792690057290863679485473802661062425994e1;2.3140692632779269005729
0863679485473802661062426008e1]
> midpointmode = on;
Midpoint mode has been activated.
> [17;17];
[17]
> [exp(pi);exp(pi)];
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
0.23140692632779269005729086367948547380266106242~5/7~e2
>

```

Sollya intervals are internally represented with floating-point numbers as bounds; rational numbers are not supported here. If bounds are defined by constant expressions, these are evaluated to floating-point numbers using the current precision. Numbers or variables containing numbers keep their precision for the interval bounds. Interval bound evaluation is performed in a way that ensures the inclusion property: all points in the original, unevaluated interval will be contained in the interval with its bounds evaluated to floating-point numbers.

Constant expressions get evaluated to floating-point values immediately; this includes  $\pi$  and rational numbers, even in when `rationalmode` is on (see section 5.3 for this mode).

```

> prec = 300!;
> a = 4097.1;
Warning: Rounding occurred when converting the constant "4097.1" to floating-poi
nt with 300 bits.
If safe computation is needed, try to increase the precision.
> prec = 12!;
> d = [4097.1; a];
Warning: Rounding occurred when converting the constant "4097.1" to floating-poi
nt with 12 bits.
If safe computation is needed, try to increase the precision.
> prec = 300!;
> d;
[4096;4.0971e3]
> prec = 30!;
> [-pi;pi];
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
Warning: the given expression is not a constant but an expression to evaluate. A
faithful evaluation will be used.
[-3.141592659;3.141592659]

```

You can get the upper-bound (respectively the lower-bound) of an interval with the command `sup` (respectively `inf`). The middle of the interval can be computed with the command `mid`. Note that these

commands can also be used on numbers (in that case, the number is interpreted as an interval containing only one single point. In that case the commands `inf`, `mid` and `sup` are just the identity):

```
> d=[1;3];
> inf(d);
1
> mid(d);
2
> sup(4);
4
```

Let us mention that the `mid` operator never provokes a rounding. It is rewritten as an unevaluated expression in terms of `inf` and `sup`.

**Sollya** permits intervals to also have non-real bounds, such as infinities or NaNs. When evaluating certain expressions, in particular given as interval bounds, **Sollya** will itself generate intervals containing infinities or NaNs. When evaluation yield an interval with a NaN bound, the given expression is most likely undefined or numerically unstable. Such results should not be trusted; a warning is displayed.

While computations on intervals with bounds being NaN will always fail, **Sollya** will try to interpret infinities in the common way as limits. However, this is not guaranteed to work, even if it is guaranteed that no unsafe results will be produced. See also section 5.2 for more detail on infinities in **Sollya**.

```
> evaluate(exp(x),[-infty;0]);
[0;1]
> dirtyinfnorm(exp(x),[-infty;0]);
Warning: a bound of the interval is infinite or NaN.
This command cannot handle such intervals.
@NaN@
>
> f = log(x);
> [f(0); f(1)];
Warning: the given expression is not a constant but an expression to evaluate and
a faithful evaluation is not possible.
Will use a plain floating-point evaluation, which might yield a completely wrong
value.
Warning: inclusion property is satisfied but the diameter may be greater than the
least possible.
[-@Inf@;0]
>
```

**Sollya** internally uses interval arithmetic extensively to provide safe answers. In order to provide for algorithms written in the **Sollya** language being able to use interval arithmetic, **Sollya** offers native support of interval arithmetic. Intervals can be added, subtracted, multiplied, divided, raised to powers, for short, given in argument to any **Sollya** function. The tool will apply the rules of interval arithmetic in order to compute output intervals that safely encompass the hull of the image of the function on the given interval:

```
> [1;2] + [3;4];
[4;6]
> [1;2] * [3;4];
[3;8]
> sqrt([9;25]);
[3;5]
> exp(sin([10;100]));
[0.36787942;2.7182819]
```

When such expressions involving intervals are given, **Sollya** will follow the rules of interval arithmetic in precision **prec** for immediately evaluating them to interval enclosures. While **Sollya**'s evaluator always guarantees the inclusion property, it also applies some optimisations in some cases in order to make the image interval as thin as possible. For example, **Sollya** will use a Taylor expansion based evaluation if a composed function, call it  $f$ , is applied to an interval. In other words, in this case **Sollya** will behave as if the **evaluate** command (see section 8.45) were implicitly used. In most cases, the result will be different from the one obtained by replacing all occurrences of the free variable of a function by the interval the function is to be evaluated on:

```
> f = x - sin(x);
> [-1b-10;1b-10] - sin([-1b-10;1b-10]);
[-1.95312484477957829894e-3;1.95312484477957829894e-3]
> f([-1b-10;1b-10]);
[-4.656612503003139269e-10;4.656612503003139269e-10]
> evaluate(f, [-1b-10;1b-10]);
[-4.656612503003139269e-10;4.656612503003139269e-10]
```

## 5.5 Functions

**Sollya** knows only about functions with one single variable. The first time in a session that an unbound name is used (without being assigned) it determines the name used to refer to the free variable.

The basic functions available in **Sollya** are the following:

- $+$ ,  $-$ ,  $*$ ,  $/$ ,  $^$
- **sqrt**
- **abs**
- **sin**, **cos**, **tan**, **sinh**, **cosh**, **tanh**
- **asin**, **acos**, **atan**, **asinh**, **atanh**
- **exp**, **expm1** (defined as  $\text{expm1}(x) = \exp(x) - 1$ )
- **log** (natural logarithm), **log2** (binary logarithm), **log10** (decimal logarithm), **log1p** (defined as  $\text{log1p}(x) = \log(1 + x)$ )
- **erf**, **erfc**
- **single**, **double**, **doubleextended**, **doubledouble**, **tripledouble** (see sections 8.143, 8.37, 8.39, 8.38 and 8.158)
- **SG**, **D**, **DE**, **DD**, **TD** (see sections 8.143, 8.37, 8.39, 8.38 and 8.158)
- **floor**, **ceil**, **nearestint**.

The constant  $\pi$  is available through the keyword **pi** as a 0-ary function:

```
> display=binary!;
> prec=12!;
> a=pi;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
1.10010010001_2 * 2^(1)
> prec=30!;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
1.10010010000111111011010101001_2 * 2^(1)
```

## 5.6 Strings

Anything written between quotes is interpreted as a string. The infix operator `@` concatenates two strings. To get the length of a string, use the `length` function. You can access the  $i$ -th character of a string using brackets (see the example below). There is no character type in Sollya: the  $i$ -th character of a string is returned as a string itself.

```
> s1 = "Hello "; s2 = "World!";
> s = s1@s2;
> length(s);
12
> s[0];
H
> s[11];
!
```

Strings may contain the following escape sequences: `\\`, `\`, `\?`, `\``, `\n`, `\t`, `\a`, `\b`, `\f`, `\r`, `\v`, `\x[hexadecimal number]` and `\[octal number]`. Refer to the C99 standard for their meaning.

## 5.7 Particular values

Sollya knows about some particular values. These values do not really have a type but they can be stored in variables and in lists. A (possibly not exhaustive) list of such values is the following:

- `on`, `off` (see sections 8.97 and 8.96)
- `dyadic`, `powers`, `binary`, `decimal`, `hexadecimal` (see sections 8.40, 8.108, 8.16, 8.26 and 8.66)
- `file`, `postscript`, `postscriptfile` (see sections 8.54, 8.105 and 8.106)
- `RU`, `RD`, `RN`, `RZ` (see sections 8.137, 8.123, 8.132 and 8.138)
- `absolute`, `relative` (see sections 8.2 and 8.126)
- `floating`, `fixed` (see sections 8.57 and 8.56)
- `single`, `double`, `doubleextended`, `doubledouble`, `tripledouble` (see sections 8.143, 8.37, 8.39, 8.38 and 8.158)
- `SG`, `D`, `DE`, `DD`, `TD` (see sections 8.143, 8.37, 8.39, 8.38 and 8.158)
- `perturb` (see section 8.100)
- `honorcoeffprec` (see section 8.67)
- `default` (see section 8.27)
- `error` (see section 8.44)
- `void` (see section 8.162)

## 5.8 Lists

Objects can be grouped into lists. A list can contain elements with different types. As for strings, you can concatenate two lists with `@`. The function `length` also gives the length of a list.

You can prepend an element to a list using `.` and you can append an element to a list using `...`. The following example illustrates some features:

```

> l = [| "foo" |];
> l = 1:.1;
> l = "bar"::.1;
> l;
[| "bar", "foo", 1|]
> l[1];
foo
> l@1;
[| "bar", "foo", 1, "bar", "foo", 1|]

```

Lists can be considered arrays and elements of lists can be referenced using brackets. Possible indices start at 0. The following example illustrates this point:

```

> l = [|1,2,3,4,5|];
> l;
[|1, 2, 3, 4, 5|]
> l[3];
4

```

Please be aware of the fact that the complexity for accessing an element of the list using indices is  $\mathcal{O}(n)$ .

Lists may contain ellipses indicated by `,...`, between elements that are constant and evaluate to integers that are incrementally ordered. `Sollya` translates such ellipses to the full list upon evaluation. Using ellipses between elements that are not constants is not allowed. This feature is provided for ease of programming; remark that the complexity for expanding such lists is high. For illustration, see the following example:

```

> [|1,...,5|];
[|1, 2, 3, 4, 5|]
> [| -5,...,5|];
[|-5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5|]
> [|3,...,1|];
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error
> [|true,...,false|];
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error

```

Lists may be continued to infinity by means of the `...` indicator after the last element given. At least one element must explicitly be given. If the last element given is a constant expression that evaluates to an integer, the list is considered as continued to infinity by all integers greater than that last element. If the last element is another object, the list is considered as continued to infinity by re-duplicating this last element. Let us remark that bracket notation is supported for such end-elliptic lists even for implicitly given elements. However, evaluation complexity is high. Combinations of ellipses inside a list and in its end are possible. The usage of lists described here is best illustrated by the following examples:

```

> l = [|1,2,true,3...|];
> l;
[|1, 2, true, 3...|]
> l[2];
true
> l[3];
3
> l[4];
4
> l[1200];
1200
> l = [|1,...,5,true...|];
> l;
[|1, 2, 3, 4, 5, true...|]
> l[1200];
true

```

## 6 Iterative language elements: assignments, conditional statements and loops

### 6.1 Blocks

Statements in **Sollya** can be grouped in blocks, so-called begin-end-blocks. This can be done using the keywords **begin** and **end** or their shorter variants **{** and **}**. Blocks declared this way are considered to be one single statement. As already explained in section 4, using begin-end-blocks also opens the possibility of declaring variables through the keyword **var**.

### 6.2 Assignments

**Sollya** has two different assignment operators, **=** and **:=**. The assignment operator **=** assigns its right-hand-object “as is”, i.e. without evaluating functional expressions. For instance, **i = i + 1;** will dereference the identifier **i** with some content, notate it *y*, build up the expression (function) *y* + 1 and assign this expression back to **i**. In the example, if **i** stood for the value 1000, the statement **i = i + 1;** would assign 1000 + 1 – and not 1001 – to **i**. The assignment operator **:=** evaluates constant functional expressions before assigning them. On other expressions it behaves like **=**. Still in the example, the statement **i := i + 1;** really assigns 1001 to **i**.

Both **Sollya** assignment operators support indexing of lists or strings elements using brackets on the left-hand-side of the assignment operator. The indexed element of the list or string gets replaced by the right-hand-side of the assignment operator. When indexing strings this way, that right-hand side must evaluate to a string of length 1. End-elliptic lists are supported with their usual semantic for this kind of assignment. When referencing and assigning a value in the implicit part of the end-elliptic list, the list gets expanded to the corresponding length. The indexing of lists on left-hand sides of assignments is reduced to the first order. Multiple indexing of lists of lists is not supported for complexity reasons.

The following examples well illustrate the behavior of assignment statements:

```

> autosimplify = off;
Automatic pure tree simplification has been deactivated.
> i = 1000;
> i = i + 1;
> print(i);
1000 + 1
> i := i + 1;
> print(i);
1002
> l = [|1,...,5|];
> print(l);
[|1, 2, 3, 4, 5|]
> l[3] = l[3] + 1;
> l[4] := l[4] + 1;
> print(l);
[|1, 2, 3, 4 + 1, 6|]
> l[5] = true;
> l;
[|1, 2, 3, 5, 6, true|]
> s = "Hello world";
> s;
Hello world
> s[1] = "a";
> s;
Hallo world
> l = [|true,1,...,5,9...|];
> l;
[|true, 1, 2, 3, 4, 5, 9...|]
> l[13] = "Hello";
> l;
[|true, 1, 2, 3, 4, 5, 9, 10, 11, 12, 13, 14, 15, "Hello"...|]

```

### 6.3 Conditional statements

Sollya supports conditional statements expressed with the keywords **if**, **then** and optionally **else**. Let us mention that only conditional statements are supported and not conditional expressions.

The following examples illustrate both syntax and semantic of conditional statements in Sollya. Concerning syntax, be aware that there must not be any semicolon before the **else** keyword.

```

> a = 3;
> b = 4;
> if (a == b) then print("Hello world");
> b = 3;
> if (a == b) then print("Hello world");
Hello world
> if (a == b) then print("You are telling the truth") else print("Liar!");
You are telling the truth

```

### 6.4 Loops

Sollya supports three kinds of loops. General *while-condition* loops can be expressed using the keywords **while** and **do**. One has to be aware of the fact that the condition test is executed always before the loop, there is no *do-until-condition* loop. Consider the following examples for both syntax and semantic:

```

> verbosity = 0!;
> prec = 30!;
> i = 5;
> while (expm1(i) > 0) do { expm1(i); i := i - 1; };
1.474131591e2
5.359815e1
1.908553692e1
6.3890561
1.718281827
> print(i);
0

```

The second kind of loops are loops on a variable ranging from a numerical start value and a end value. These kind of loops can be expressed using the keywords **for**, **from**, **to**, **do** and optionally **by**. The **by** statement indicates the width of the steps on the variable from the start value to the end value. Once again, syntax and semantic are best explained with an example:

```

> for i from 1 to 5 do print ("Hello world",i);
Hello world 1
Hello world 2
Hello world 3
Hello world 4
Hello world 5
> for i from 2 to 1 by -0.5 do print("Hello world",i);
Hello world 2
Hello world 1.5
Hello world 1

```

The third kind of loops are loops on a variable ranging on values contained in a list. In order to ensure the termination of the loop, that list must not be end-elliptic. The loop is expressed using the keywords **for**, **in** and **do** as in the following examples:

```

> l = [true, false, 1,...,4, "Hello", exp(x)];
> for i in l do i;
true
false
1
2
3
4
Hello
exp(x)

```

For both types of **for** loops, assigning the loop variable is allowed and possible. When the loop terminates, the loop variable will contain the value that made the loop condition fail. Consider the following examples:

```

> for i from 1 to 5 do { if (i == 3) then i = 4 else i; };
1
2
5
> i;
6

```



## 7 Functional language elements: procedures

Sollya has some elements of functional languages. In order to avoid confusion with mathematical functions, the associated programming objects are called *procedures* in Sollya.

Sollya procedures are common objects that can be, for example, assigned to variables or stored in lists. Procedures are declared by the `proc` keyword; see section 8.117 for details. The returned procedure object must then be assigned to a variable. It can hence be applied to arguments with common application syntax. The `procedure` keyword provides an abbreviation for declaring and assigning a procedure; see section 8.118 for details.

Sollya procedures can return objects using the `return` keyword at the end of the begin-end-block of the procedure. Section 8.130 gives details on the usage of `return`. Procedures further can take any type of object in argument, in particular also other procedures that are then applied to arguments. Procedures can be declared inside other procedures.

Common Sollya procedures are declared with a certain number of formal parameters. When the procedure is applied to actual parameters, a check is performed if the right number of actual parameters is given. Then the actual parameters are applied to the formal parameters. In some cases, it is required that the number of parameters of a procedure be variable. Sollya provides support for the case with procedures with an arbitrary number of actual arguments. When the procedure is called, those actual arguments are gathered in a list which is applied to the only formal list parameter of a procedure with an arbitrary number of arguments. See section 8.118 for the exact syntax and details; an example is given just below.

Let us remark that declaring a procedure does not involve any evaluation or other interpretation of the procedure body. In particular, this means that constants are evaluated to floating-point values inside Sollya when the procedure is applied to actual parameters and the global precision valid at this moment.

Sollya procedures are well illustrated with the following examples:

```
> succ = proc(n) { return n + 1; };
> succ(5);
6
> 3 + succ(0);
4
> succ;
proc(n)
begin
  nop;
  return (n) + (1);
end
```

```
> add = proc(m,n) { var res; res := m + n; return res; };
> add(5,6);
11
> hey = proc() { print("Hello world."); };
> hey();
Hello world.
> print(hey());
Hello world.
void
> hey;
proc()
begin
  print("Hello world.");
  return void;
end
```

```

> fac = proc(n) { var res; if (n == 0) then res := 1 else res := n * fac(n - 1);
  return res; };
> fac(5);
120
> fac(11);
39916800
> fac;
proc(n)
begin
var res;
if (n) == (0) then
res := 1
else
res := (n) * (fac((n) - (1)));
return res;
end

```

```

> sumall = proc(args = ...) { var i, acc; acc = 0; for i in args do acc = acc +
i; return acc; };
> sumall;
proc(args = ...)
begin
var i, acc;
acc = 0;
for i in args do
acc = (acc) + (i);
return acc;
end
> sumall();
0
> sumall(1);
1
> sumall(1,5);
6
> sumall(1,5,9);
15
> sumall @ [|1,5,9,4,8|];
27
>

```

Sollya also supports external procedures, i.e. procedures written in C (or some other language) and dynamically bound to Sollya identifiers. See 8.52 for details.

## 8 Commands and functions

### 8.1 abs

Name: **abs**

the absolute value.

Description:

- **abs** is the absolute value function.  $\text{abs}(x) = \begin{cases} x & x > 0 \\ -x & x \leq 0 \end{cases}$ .

## 8.2 absolute

Name: **absolute**

indicates an absolute error for **externalplot** or **fpminimax**

Usage:

**absolute** : absolute|relative

Description:

- The use of **absolute** in the command **externalplot** indicates that during plotting in **externalplot** an absolute error is to be considered.  
See **externalplot** for details.
- Used with **fpminimax**, **absolute** indicates that **fpminimax** must try to minimize the absolute error.  
See **fpminimax** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");  
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l  
mpfr");  
> externalplot("./externalplotexample",absolute,exp(x),[-1/2;1/2],12,perturb);
```

See also: **externalplot** (8.51), **fpminimax** (8.59), **relative** (8.126), **bashexecute** (8.15)

## 8.3 accurateinfnorm

Name: **accurateinfnorm**

computes a faithful rounding of the infinity norm of a function

Usage:

**accurateinfnorm**(*function,range,constant*) : (function, range, constant) → constant  
**accurateinfnorm**(*function,range,constant,exclusion range 1,...,exclusion range n*) : (function, range, constant, range, ..., range) → constant

Parameters:

- *function* represents the function whose infinity norm is to be computed
- *range* represents the infinity norm is to be considered on
- *constant* represents the number of bits in the significant of the result
- *exclusion range 1* through *exclusion range n* represent ranges to be excluded

Description:

- The command **accurateinfnorm** computes an upper bound to the infinity norm of function *function* in *range*. This upper bound is the least floating-point number greater than the value of the infinity norm that lies in the set of dyadic floating point numbers having *constant* significant mantissa bits. This means the value **accurateinfnorm** evaluates to is at the time an upper bound and a faithful rounding to *constant* bits of the infinity norm of function *function* on range *range*.  
If given, the fourth and further arguments of the command **accurateinfnorm**, *exclusion range 1* through *exclusion range n* the infinity norm of the function *function* is not to be considered on.

Example 1:

```

> p = remez(exp(x), 5, [-1;1]);
> accurateinfnorm(p - exp(x), [-1;1], 20);
4.52055246569216251373291015625e-5
> accurateinfnorm(p - exp(x), [-1;1], 30);
4.520552107578623690642416477203369140625e-5
> accurateinfnorm(p - exp(x), [-1;1], 40);
4.5205521043867324948450914234854280948638916015625e-5

```

Example 2:

```

> p = remez(exp(x), 5, [-1;1]);
> midpointmode = on!;
> infnorm(p - exp(x), [-1;1]);
0.45205~5/7~e-4
> accurateinfnorm(p - exp(x), [-1;1], 40);
4.5205521043867324948450914234854280948638916015625e-5

```

See also: **infnorm** (8.72), **dirtyinfnorm** (8.33), **checkinfnorm** (8.20), **remez** (8.127), **diam** (8.30)

## 8.4 acos

Name: **acos**

the arccosine function.

Description:

- **acos** is the inverse of the function **cos**: **acos**( $y$ ) is the unique number  $x \in [0; \pi]$  such that **cos**( $x$ )= $y$ .
- It is defined only for  $y \in [-1; 1]$ .

See also: **cos** (8.24)

## 8.5 acosh

Name: **acosh**

the arg-hyperbolic cosine function.

Description:

- **acosh** is the inverse of the function **cosh**: **acosh**( $y$ ) is the unique number  $x \in [0; +\infty]$  such that **cosh**( $x$ )= $y$ .
- It is defined only for  $y \in [0; +\infty]$ .

See also: **cosh** (8.25)

## 8.6 &&

Name: **&&**

boolean AND operator

Usage:

$$expr1 \ \&\& \ expr2 : (\text{boolean}, \text{boolean}) \rightarrow \text{boolean}$$

Parameters:

- $expr1$  and  $expr2$  represent boolean expressions

Description:

- **&&** evaluates to the boolean AND of the two boolean expressions  $expr1$  and  $expr2$ . **&&** evaluates to true iff both  $expr1$  and  $expr2$  evaluate to true.

Example 1:

```
> true && false;
false
```

Example 2:

```
> (1 == exp(0)) && (0 == log(1));
true
```

See also: `||` (8.98), `!` (8.94)

## 8.7 `::`

Name: `::`

add an element at the end of a list.

Usage:

$$L::x : (\text{list, any type}) \rightarrow \text{list}$$

Parameters:

- $L$  is a list (possibly empty).
- $x$  is an object of any type.

Description:

- `::` adds the element  $x$  at the end of the list  $L$ .
- Note that since  $x$  may be of any type, it can in particular be a list.

Example 1:

```
> [|2,3,4|]::5;
[|2, 3, 4, 5|]
```

Example 2:

```
> [|1,2,3|]::[|4,5,6|];
[|1, 2, 3, [|4, 5, 6|]|]
```

Example 3:

```
> [|]|::1;
[|1|]
```

See also: `::` (8.111), `@` (8.22)

## 8.8 `~`

Name: `~`

floating-point evaluation of a constant expression

Usage:

$$\begin{aligned} \sim \text{expression} &: \text{function} \rightarrow \text{constant} \\ \sim \text{something} &: \text{any type} \rightarrow \text{any type} \end{aligned}$$

Parameters:

- *expression* stands for an expression that is a constant

- *something* stands for some language element that is not a constant expression

Description:

- $\sim$  *expression* evaluates the *expression* that is a constant term to a floating-point constant. The evaluation may involve a rounding. If *expression* is not a constant, the evaluated constant is a faithful rounding of *expression* with **precision** bits, unless the *expression* is exactly 0 as a result of cancellation. In the latter case, a floating-point approximation of some (unknown) accuracy is returned.
- $\sim$  does not do anything on all language elements that are not a constant expression. In other words, it behaves like the identity function on any type that is not a constant expression. It can hence be used in any place where one wants to be sure that expressions are simplified using floating-point computations to constants of a known precision, regardless of the type of actual language elements.
- $\sim$  **error** evaluates to error and provokes a warning.
- $\sim$  is a prefix operator not requiring parentheses. Its precedence is the same as for the unary  $+$  and  $-$  operators. It cannot be repeatedly used without brackets.

Example 1:

```
> print(exp(5));
exp(5)
> print(~ exp(5));
1.48413159102576603421115580040552279623487667593878e2
```

Example 2:

```
> autosimplify = off!;
```

Example 3:

```
> print(~sin(5 * pi));
0
```

Example 4:

```
> print(~exp(x));
exp(x)
> print(~ "Hello");
Hello
```

Example 5:

```
> print(~exp(x*5*Pi));
exp((pi) * 5 * x)
> print(exp(x* ~(5*Pi)));
exp(x * 1.57079632679489661923132169163975144209858469968757e1)
```

Example 6:

```
> print(~exp(5)*x);
1.48413159102576603421115580040552279623487667593878e2 * x
> print( (~exp(5))*x);
1.48413159102576603421115580040552279623487667593878e2 * x
> print(~(exp(5)*x));
exp(5) * x
```

See also: **evaluate** (8.45), **prec** (8.109), **error** (8.44)

## 8.9 asciiplot

Name: **asciiplot**

plots a function in a range using ASCII characters

Usage:

**asciiplot**(*function*, *range*) : (function, range)  $\rightarrow$  void

Parameters:


- *function* represents a function to be plotted
- *range* represents a range the function is to be plotted in

Description:


- **asciiplot** plots the function *function* in range *range* using ASCII characters. On systems that provide the necessary `TIOCGWINSZ ioctl`, **Sollya** determines the size of the terminal for the plot size if connected to a terminal. If it is not connected to a terminal or if the test is not possible, the plot is of fixed size  $77 \times 25$  characters. The function is evaluated on a number of points equal to the number of columns available. Its value is rounded to the next integer in the range of lines available. A letter **x** is written at this place. If zero is in the hull of the image domain of the function, an x-axis is displayed. If zero is in range, a y-axis is displayed. If the function is constant or if the range is reduced to one point, the function is evaluated to a constant and the constant is displayed instead of a plot.

Example 1:

```
> asciiplot(exp(x),[1;2]);
```



Example 2:



Example 3:

```
> asciiplot(5, [-1;1]);
5
```

Example 4:

```
> asciiplot(exp(x), [1;1]);
2.71828182845904523536028747135266249775724709369998
```

See also: **plot** (8.102)

## 8.10 asin

Name: **asin**

the arcsine function.

Description:

- **asin** is the inverse of the function **sin**: **asin**( $y$ ) is the unique number  $x \in [-\pi/2; \pi/2]$  such that **sin**( $x$ )= $y$ .
- It is defined only for  $y \in [-1; 1]$ .

See also: **sin** (8.142)

## 8.11 asinh

Name: **asinh**

the arg-hyperbolic sine function.

Description:

- **asinh** is the inverse of the function **sinh**: **asinh**( $y$ ) is the unique number  $x \in [-\infty; +\infty]$  such that **sinh**( $x$ )= $y$ .



- It is defined for every real number  $y$ .

See also: **sinh** (8.144)

## 8.12 atan

Name: **atan**

the arctangent function.

Description:

- **atan** is the inverse of the function **tan**: **atan**( $y$ ) is the unique number  $x \in [-\pi/2; +\pi/2]$  such that **tan**( $x$ )= $y$ .
- It is defined for every real number  $y$ .

See also: **tan** (8.152)

## 8.13 atanh

Name: **atanh**

the hyperbolic arctangent function.

Description:

- **atanh** is the inverse of the function **tanh**: **atanh**( $y$ ) is the unique number  $x \in [-\infty; +\infty]$  such that **tanh**( $x$ )= $y$ .
- It is defined only for  $y \in [-1; 1]$ .

See also: **tanh** (8.153)

## 8.14 autosimplify

Name: **autosimplify**

activates, deactivates or inspects the value of the automatic simplification state variable

Usage:

$$\begin{aligned} \mathbf{autosimplify} &= \textit{activation value} : \mathbf{on|off} \rightarrow \mathbf{void} \\ \mathbf{autosimplify} &= \textit{activation value} ! : \mathbf{on|off} \rightarrow \mathbf{void} \\ \mathbf{autosimplify} &: \mathbf{on|off} \end{aligned}$$

Parameters:

- *activation value* represents **on** or **off**, i.e. activation or deactivation

Description:

- An assignment **autosimplify** = *activation value*, where *activation value* is one of **on** or **off**, activates respectively deactivates the automatic safe simplification of expressions of functions generated by the evaluation of commands or in argument of other commands.

**Sollya** commands like **remez**, **taylor** or **rationalapprox** sometimes produce expressions that can be simplified. Constant subexpressions can be evaluated to dyadic floating-point numbers, monomials with coefficients 0 can be eliminated. Further, expressions indicated by the user perform better in many commands when simplified before being passed in argument to a command. When the automatic simplification of expressions is activated, **Sollya** automatically performs a safe (not value changing) simplification process on such expressions.

The automatic generation of subexpressions can be annoying, in particular if it takes too much time for not enough benefit. Further the user might want to inspect the structure of the expression tree returned by a command. In this case, the automatic simplification should be deactivated.

If the assignment **autosimplify** = *activation value* is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

Example 1:

```
> autosimplify = on !;
> print(x - x);
0
> autosimplify = off ;
Automatic pure tree simplification has been deactivated.
> print(x - x);
x - x
```

Example 2:

```
> autosimplify = on !;
> print(rationalapprox(sin(pi/5.9),7));
0.5
> autosimplify = off !;
> print(rationalapprox(sin(pi/5.9),7));
1 / 2
```

See also: **print** (8.112), **prec** (8.109), **points** (8.104), **diam** (8.30), **display** (8.35), **verbosity** (8.161), **canonical** (8.18), **taylorrecursions** (8.155), **timing** (8.157), **fullparentheses** (8.60), **midpointmode** (8.88), **hopitalrecursions** (8.68), **remez** (8.127), **rationalapprox** (8.121), **taylor** (8.154)

## 8.15 bashexecute

Name: **bashexecute**

executes a shell command.

Usage:

**bashexecute**(*command*) : string  $\rightarrow$  void

Parameters:

- *command* is a command to be interpreted by the shell.

Description:

- **bashexecute**(*command*) lets the shell interpret *command*. It is useful to execute some external code within Sollya.
- **bashexecute** does not return anything. It just executes its argument. However, if *command* produces an output in a file, this result can be imported in Sollya with help of commands like **execute**, **readfile** and **parse**.

Example 1:

```
> bashexecute("LANG=C date");
Tue Mar 16 16:03:12 CET 2010
```

See also: **execute** (8.46), **readfile** (8.124), **parse** (8.99)

## 8.16 binary

Name: **binary**

special value for global state **display**

Description:

- **binary** is a special value used for the global state **display**. If the global state **display** is equal to **binary**, all data will be output in binary notation.

As any value it can be affected to a variable and stored in lists.

See also: **decimal** (8.26), **dyadic** (8.40), **powers** (8.108), **hexadecimal** (8.66)

## 8.17 boolean

Name: **boolean**

keyword representing a boolean type

Usage:

**boolean** : type type

Description:

- **boolean** represents the boolean type for declarations of external procedures by means of **externalproc**.

Remark that in contrast to other indicators, type indicators like **boolean** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **constant** (8.23), **function** (8.61), **integer** (8.73), **list of** (8.80), **range** (8.120), **string** (8.147)

## 8.18 canonical

Name: **canonical**

brings all polynomial subexpressions of an expression to canonical form or activates, deactivates or checks canonical form printing

Usage:

**canonical**(*function*) : function → function  
**canonical** = *activation value* : on|off → void  
**canonical** = *activation value* ! : on|off → void

Parameters:

- *function* represents the expression to be rewritten in canonical form
- *activation value* represents **on** or **off**, i.e. activation or deactivation

Description:

- The command **canonical** rewrites the expression representing the function *function* in a way such that all polynomial subexpressions (or the whole expression itself, if it is a polynomial) are written in canonical form, i.e. as a sum of monomials in the canonical base. The canonical base is the base of the integer powers of the global free variable. The command **canonical** does not endanger the safety of computations even in **Sollya**'s floating-point environment: the function returned is mathematically equal to the function *function*.
- An assignment **canonical** = *activation value*, where *activation value* is one of **on** or **off**, activates respectively deactivates the automatic printing of polynomial expressions in canonical form, i.e. as a sum of monomials in the canonical base. If automatic printing in canonical form is deactivated, automatic printing yields to displaying polynomial subexpressions in Horner form.

If the assignment **canonical** = *activation value* is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

Example 1:

```
> print(canonical(1 + x * (x + 3 * x^2)));  
1 + x^2 + 3 * x^3  
> print(canonical((x + 1)^7));  
1 + 7 * x + 21 * x^2 + 35 * x^3 + 35 * x^4 + 21 * x^5 + 7 * x^6 + x^7
```

Example 2:

```
> print(canonical(exp((x + 1)^5) - log(asin(((x + 2) + x)^4 * (x + 1)) + x)));
exp(1 + 5 * x + 10 * x^2 + 10 * x^3 + 5 * x^4 + x^5) - log(asin(16 + 80 * x + 16
0 * x^2 + 160 * x^3 + 80 * x^4 + 16 * x^5) + x)
```

Example 3:

```
> canonical;
off
> (x + 2)^9;
512 + x * (2304 + x * (4608 + x * (5376 + x * (4032 + x * (2016 + x * (672 + x *
(144 + x * (18 + x))))))))))
> canonical = on;
Canonical automatic printing output has been activated.
> (x + 2)^9;
512 + 2304 * x + 4608 * x^2 + 5376 * x^3 + 4032 * x^4 + 2016 * x^5 + 672 * x^6 +
144 * x^7 + 18 * x^8 + x^9
> canonical;
on
> canonical = off!;
> (x + 2)^9;
512 + x * (2304 + x * (4608 + x * (5376 + x * (4032 + x * (2016 + x * (672 + x *
(144 + x * (18 + x))))))))))
```

See also: **horner** (8.69), **print** (8.112)

## 8.19 ceil

Name: **ceil**

the usual function ceil.

Description:

- **ceil** is defined as usual: **ceil**( $x$ ) is the smallest integer  $y$  such that  $y \geq x$ .
- It is defined for every real number  $x$ .

See also: **floor** (8.58), **nearestint** (8.91)

## 8.20 checkinfnorm

Name: **checkinfnorm**

checks whether the infinity norm of a function is bounded by a value

Usage:

**checkinfnorm**(*function*, *range*, *constant*) : (function, range, constant)  $\rightarrow$  boolean

Parameters:

- *function* represents the function whose infinity norm is to be checked
- *range* represents the infinity norm is to be considered on
- *constant* represents the upper bound the infinity norm is to be checked to

Description:

- The command **checkinfnorm** checks whether the infinity norm of the given function *function* in the range *range* can be proven (by **Sollya**) to be less than the given bound *bound*. This means, if **checkinfnorm** evaluates to **true**, the infinity norm has been proven (by **Sollya**'s interval arithmetic) to be less than the bound. If **checkinfnorm** evaluates to **false**, there are two possibilities:

either the bound is less than or equal to the infinity norm of the function or the bound is greater than the infinity norm but **Sollya** could not conclude using its internal interval arithmetic.

**checkinfnorm** is sensitive to the global variable **diam**. The smaller **diam**, the more time **Sollya** will spend on the evaluation of **checkinfnorm** in order to prove the bound before returning **false** although the infinity norm is bounded by the bound. If **diam** is equal to 0, **Sollya** will eventually spend infinite time on instances where the given bound *bound* is less or equal to the infinity norm of the function *function* in range *range*. In contrast, with **diam** being zero, **checkinfnorm** evaluates to **true** iff the infinity norm of the function in the range is bounded by the given bound.

Example 1:

```
> checkinfnorm(sin(x), [0;1.75], 1);
true
> checkinfnorm(sin(x), [0;1.75], 1/2); checkinfnorm(sin(x), [0;20/39], 1/2);
false
true
```

Example 2:

```
> p = remez(exp(x), 5, [-1;1]);
> b = dirtyinfnorm(p - exp(x), [-1;1]);
> checkinfnorm(p - exp(x), [-1;1], b);
false
> b1 = round(b, 15, RU);
> checkinfnorm(p - exp(x), [-1;1], b1);
true
> b2 = round(b, 25, RU);
> checkinfnorm(p - exp(x), [-1;1], b2);
false
> diam = 1b-20!;
> checkinfnorm(p - exp(x), [-1;1], b2);
true
```

See also: **infnorm** (8.72), **dirtyinfnorm** (8.33), **accurateinfnorm** (8.3), **remez** (8.127), **diam** (8.30)

## 8.21 coeff

Name: **coeff**

gives the coefficient of degree  $n$  of a polynomial

Usage:

$$\mathbf{coeff}(f, n) : (\text{function}, \text{integer}) \rightarrow \text{constant}$$

Parameters:

- $f$  is a function (usually a polynomial).
- $n$  is an integer

Description:

- If  $f$  is a polynomial, **coeff**( $f, n$ ) returns the coefficient of degree  $n$  in  $f$ .
- If  $f$  is a function that is not a polynomial, **coeff**( $f, n$ ) returns 0.

Example 1:

```
> coeff((1+x)^5, 3);
10
```

Example 2:

```
> coeff(sin(x),0);  
0
```

See also: **degree** (8.28)

## 8.22 @

Name: @

concatenates two lists or strings or applies a list as arguments to a procedure

Usage:

$$\begin{aligned} L1@L2 &: (\text{list}, \text{list}) \rightarrow \text{list} \\ string1@string2 &: (\text{string}, \text{string}) \rightarrow \text{string} \\ proc@L1 &: (\text{procedure}, \text{list}) \rightarrow \text{any type} \end{aligned}$$

Parameters:

- $L1$  and  $L2$  are two lists.
- $string1$  and  $string2$  are two strings.
- $proc$  is a procedure.

Description:

- In its first usage form, @ concatenates two lists or strings.
- In its second usage form, @ applies the elements of a list as arguments to a procedure. In the case when  $proc$  is a procedure with a fixed number of arguments, a check is done if the number of elements in the list corresponds to the number of formal parameters of the procedure. An empty list can therefore applied only to a procedure that does not take any argument. In the case of a procedure with an arbitrary number of arguments, no such check is performed.

Example 1:

```
> [[1,...,3]]@[[7,8,9]];  
[[1, 2, 3, 7, 8, 9]]
```

Example 2:

```
> "Hello "@World!";  
Hello World!
```

Example 3:

```
> procedure cool(a,b,c) {  
> write(a," ", b," and ",c," are cool guys.\n");  
> };  
> cool @ [| "Christoph", "Mioara", "Sylvain" |];  
Christoph, Mioara and Sylvain are cool guys.
```

Example 4:

```
> procedure sayhello() {  
> "Hello! how are you?";  
> };  
> sayhello();  
Hello! how are you?  
> sayhello @ [||];  
Hello! how are you?
```

Example 5:

```
> procedure add(l = ...) {  
> var acc, i;  
> acc = 0;  
> for i in l do acc = i + acc;  
> return acc;  
> };  
> add(1,2);  
3  
> add(1,2,3);  
6  
> add @ [|1, 2|];  
3  
> add @ [|1, 2, 3|];  
6  
> add @ [| |];  
0
```

See also: `∴` (8.111), `∴` (8.7), **procedure** (8.118), **proc** (8.117)

### 8.23 constant

Name: **constant**

keyword representing a **constant** type

Usage:

**constant** : type type

Description:

- **constant** represents the **constant** type for declarations of external procedures **externalproc**.

Remark that in contrast to other indicators, type indicators like **constant** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **boolean** (8.17), **function** (8.61), **integer** (8.73), **list of** (8.80), **range** (8.120), **string** (8.147)

### 8.24 cos

Name: **cos**

the cosine function.

Description:

- **cos** is the usual cosine function.
- It is defined for every real number  $x$ .

See also: **acos** (8.4), **sin** (8.142), **tan** (8.152)

### 8.25 cosh

Name: **cosh**

the hyperbolic cosine function.

Description:

- **cosh** is the usual hyperbolic function:  $\cosh(x) = \frac{e^x + e^{-x}}{2}$ .
- It is defined for every real number  $x$ .

See also: **acosh** (8.5), **sinh** (8.144), **tanh** (8.153), **exp** (8.47)

## 8.26 decimal

Name: **decimal**

special value for global state **display**

Description:

- **decimal** is a special value used for the global state **display**. If the global state **display** is equal to **decimal**, all data will be output in decimal notation.

As any value it can be affected to a variable and stored in lists.

See also: **dyadic** (8.40), **powers** (8.108), **hexadecimal** (8.66), **binary** (8.16)

## 8.27 default

Name: **default**

default value for some commands.

Description:

- **default** is a special value and is replaced by something depending on the context where it is used. It can often be used as a joker, when you want to specify one of the optional parameters of a command and not the others: set the value of uninteresting parameters to **default**.
- Global variables can be reset by affecting them the special value **default**.

Example 1:

```
> p = remez(exp(x),5,[0;1],default,1e-5);
> q = remez(exp(x),5,[0;1],1,1e-5);
> p==q;
true
```

Example 2:

```
> prec;
165
> prec=200;
The precision has been set to 200 bits.
> prec=default;
The precision has been set to 165 bits.
```

## 8.28 degree

Name: **degree**

gives the degree of a polynomial.

Usage:

**degree**( $f$ ) : function  $\rightarrow$  integer

Parameters:

- $f$  is a function (usually a polynomial).

Description:

- If  $f$  is a polynomial, **degree**( $f$ ) returns the degree of  $f$ .
- Contrary to the usage, **Sollya** considers that the degree of the null polynomial is 0.
- If  $f$  is a function that is not a polynomial, **degree**( $f$ ) returns -1.

Example 1:



```
> degree((1+x)*(2+5*x^2));
3
> degree(0);
0
```

Example 2:

```
> degree(sin(x));
-1
```

See also: **coeff** (8.21)

## 8.29 denominator

Name: **denominator**

gives the denominator of an expression

Usage:

**denominator**(*expr*) : function  $\rightarrow$  function

Parameters:

- *expr* represents an expression

Description:

- If *expr* represents a fraction *expr1*/*expr2*, **denominator**(*expr*) returns the denominator of this fraction, i.e. *expr2*.

If *expr* represents something else, **denominator**(*expr*) returns 1.

Note that for all expressions *expr*, **numerator**(*expr*) / **denominator**(*expr*) is equal to *expr*.

Example 1:

```
> denominator(5/3);
3
```

Example 2:

```
> denominator(exp(x));
1
```

Example 3:

```
> a = 5/3;
> b = numerator(a)/denominator(a);
> print(a);
5 / 3
> print(b);
5 / 3
```

Example 4:

```
> a = exp(x/3);
> b = numerator(a)/denominator(a);
> print(a);
exp(x / 3)
> print(b);
exp(x / 3)
```

See also: **numerator** (8.95)

### 8.30 **diam**

Name: **diam**

parameter used in safe algorithms of **Sollya** and controlling the maximal length of the involved intervals.

Usage:

```
diam = width : constant → void
diam = width ! : constant → void
diam : constant
```

Parameters:

- *width* represents the maximal relative width of the intervals used

Description:

- **diam** is a global variable. Its value represents the maximal width allowed for intervals involved in safe algorithms of **Sollya** (namely **infnorm**, **checkinfnorm**, **accurateinfnorm**, **integral**, **findzeros**).
- More precisely, **diam** is relative to the width of the input interval of the command. For instance, suppose that **diam**=1e-5: if **infnorm** is called on interval  $[0, 1]$ , the maximal width of an interval will be 1e-5. But if it is called on interval  $[0, 1e-3]$ , the maximal width will be 1e-8.

See also: **infnorm** (8.72), **checkinfnorm** (8.20), **accurateinfnorm** (8.3), **integral** (8.74), **findzeros** (8.55)

### 8.31 **diff**

Name: **diff**

differentiation operator

Usage:

```
diff(function) : function → function
```

Parameters:

- *function* represents a function

Description:

- **diff**(*function*) returns the symbolic derivative of the function *function* by the global free variable. If *function* represents a function symbol that is externally bound to some code by **library**, the derivative is performed as a symbolic annotation to the returned expression tree.

Example 1:

```
> diff(sin(x));
cos(x)
```

Example 2:

```
> diff(x);
1
```

Example 3:

```
> diff(x^x);
x^x * (1 + log(x))
```

See also: **library** (8.79)

### 8.32 **dirtyfindzeros**

Name: **dirtyfindzeros**

gives a list of numerical values listing the zeros of a function on an interval.

Usage:

$$\mathbf{dirtyfindzeros}(f,I) : (\text{function}, \text{range}) \rightarrow \text{list}$$

Parameters:

- $f$  is a function.
- $I$  is an interval.

Description:

- **dirtyfindzeros**( $f,I$ ) returns a list containing some zeros of  $f$  in the interval  $I$ . The values in the list are numerical approximation of the exact zeros. The precision of these approximations is approximately the precision stored in **prec**. If  $f$  does not have two zeros very close to each other, it can be expected that all zeros are listed. However, some zeros may be forgotten. This command should be considered as a numerical algorithm and should not be used if safety is critical.
- More precisely, the algorithm relies on global variables **prec** and **points** and it performs the following steps: let  $n$  be the value of variable **points** and  $t$  be the value of variable **prec**.
  - Evaluate  $|f|$  at  $n$  evenly distributed points in the interval  $I$ . The working precision to be used is automatically chosen in order to ensure that the sign is correct.
  - Whenever  $f$  changes its sign for two consecutive points, find an approximation  $x$  of its zero with precision  $t$  using Newton's algorithm. The number of steps in Newton's iteration depends on  $t$ : the precision of the approximation is supposed to be doubled at each step.
  - Add this value to the list.

Example 1:

```
> dirtyfindzeros(sin(x), [-5;5]);  
[|-3.14159265358979323846264338327950288419716939937508, 0, 3.141592653589793238  
46264338327950288419716939937508|]
```

Example 2:

```
> L1=dirtyfindzeros(x^2*sin(1/x), [0;1]);  
> points=1000!;  
> L2=dirtyfindzeros(x^2*sin(1/x), [0;1]);  
> length(L1); length(L2);  
18  
25
```

See also: **prec** (8.109), **points** (8.104), **findzeros** (8.55)

### 8.33 **dirtyinfnorm**

Name: **dirtyinfnorm**

computes a numerical approximation of the infinity norm of a function on an interval.

Usage:

$$\mathbf{dirtyinfnorm}(f,I) : (\text{function}, \text{range}) \rightarrow \text{constant}$$

Parameters:

- $f$  is a function.

- $I$  is an interval.

Description:

- **dirtyinfnorm**( $f, I$ ) computes an approximation of the infinity norm of the given function  $f$  on the interval  $I$ , e.g.  $\max_{x \in I} \{|f(x)|\}$ .
- The interval must be bound. If the interval contains one of  $-\text{Inf}$  or  $+\text{Inf}$ , the result of **dirtyinfnorm** is NaN.
- The result of this command depends on the global variables **prec** and **points**. Therefore, the returned result is generally a good approximation of the exact infinity norm, with precision **prec**. However, the result is generally underestimated and should not be used when safety is critical. Use **infnorm** instead.
- The following algorithm is used: let  $n$  be the value of variable **points** and  $t$  be the value of variable **prec**.
  - Evaluate  $|f|$  at  $n$  evenly distributed points in the interval  $I$ . The evaluation are faithful roundings of the exact results at precision  $t$ .
  - Whenever the derivative of  $f$  changes its sign for two consecutive points, find an approximation  $x$  of its zero with precision  $t$ . Then compute a faithful rounding of  $|f(x)|$  at precision  $t$ .
  - Return the maximum of all computed values.

Example 1:

```
> dirtyinfnorm(sin(x), [-10;10]);
1
```

Example 2:

```
> prec=15!;
> dirtyinfnorm(exp(cos(x))*sin(x), [0;5]);
1.45856
> prec=40!;
> dirtyinfnorm(exp(cos(x))*sin(x), [0;5]);
1.458528537135
> prec=100!;
> dirtyinfnorm(exp(cos(x))*sin(x), [0;5]);
1.458528537136237644438147455024
> prec=200!;
> dirtyinfnorm(exp(cos(x))*sin(x), [0;5]);
1.458528537136237644438147455023841718299214087993682374094153
```

Example 3:

```
> dirtyinfnorm(x^2, [log(0);log(1)]);
@NaN@
```

See also: **prec** (8.109), **points** (8.104), **infnorm** (8.72), **checkinfnorm** (8.20)

## 8.34 dirtyintegral

Name: **dirtyintegral**

computes a numerical approximation of the integral of a function on an interval.

Usage:

**dirtyintegral**( $f, I$ ) : (function, range)  $\rightarrow$  constant

Parameters:

- $f$  is a function.
- $I$  is an interval.

Description:

- **dirtyintegral**( $f, I$ ) computes an approximation of the integral of  $f$  on  $I$ .
- The interval must be bound. If the interval contains one of  $-\text{Inf}$  or  $+\text{Inf}$ , the result of **dirtyintegral** is NaN, even if the integral has a meaning.
- The result of this command depends on the global variables **prec** and **points**. The method used is the trapezium rule applied at  $n$  evenly distributed points in the interval, where  $n$  is the value of global variable **points**.
- This command computes a numerical approximation of the exact value of the integral. It should not be used if safety is critical. In this case, use command **integral** instead.
- Warning: this command is currently known to be unsatisfactory. If you really need to compute integrals, think of using an other tool or report a feature request to [sylvain.chevillard@ens-lyon.org](mailto:sylvain.chevillard@ens-lyon.org).

Example 1:

```
> sin(10);
-0.54402111088936981340474766185137728168364301291621
> dirtyintegral(cos(x), [0;10]);
-0.54400304905152629822448058882475382036536298356281
> points=2000!;
> dirtyintegral(cos(x), [0;10]);
-0.5440199775115832197222697312583199035995837926892
```

See also: **prec** (8.109), **points** (8.104), **integral** (8.74)

## 8.35 display

Name: **display**

sets or inspects the global variable specifying number notation

Usage:

**display** = *notation value* : decimal|binary|dyadic|powers|hexadecimal  $\rightarrow$  void  
**display** = *notation value* ! : decimal|binary|dyadic|powers|hexadecimal  $\rightarrow$  void  
**display** : decimal|binary|dyadic|powers|hexadecimal

Parameters:

- *notation value* represents a variable of type decimal|binary|dyadic|powers|hexadecimal

Description:

- An assignment **display** = *notation value*, where *notation value* is one of **decimal**, **dyadic**, **powers**, **binary** or **hexadecimal**, activates the corresponding notation for output of values in **print**, **write** or at the Sollya prompt.

If the global notation variable **display** is **decimal**, all numbers will be output in scientific decimal notation. If the global notation variable **display** is **dyadic**, all numbers will be output as dyadic numbers with Gappa notation. If the global notation variable **display** is **powers**, all numbers will be output as dyadic numbers with a notation compatible with Maple and PARI/GP. If the global notation variable **display** is **binary**, all numbers will be output in binary notation. If the global notation variable **display** is **hexadecimal**, all numbers will be output in C99/ IEEE754-2008

notation. All output notations can be parsed back by **Sollya**, inducing no error if the input and output precisions are the same (see **prec**).

If the assignment **display** = *notation value* is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

Example 1:

```
> display = decimal;
Display mode is decimal numbers.
> a = evaluate(sin(pi * x), 0.25);
> a;
0.70710678118654752440084436210484903928483593768847
> display = binary;
Display mode is binary numbers.
> a;
1.011010100000100111100110011001111111001110111100110010010000100010110010111110
11000100110110011011101010100101010111110100111110001110101101111011000001011101
010001_2 * 2^(-1)
> display = hexadecimal;
Display mode is hexadecimal numbers.
> a;
0xb.504f333f9de6484597d89b3754abe9fd6f60ba88p-4
> display = dyadic;
Display mode is dyadic numbers.
> a;
33070006991101558613323983488220944360067107133265b-165
> display = powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
> a;
33070006991101558613323983488220944360067107133265 * 2^(-165)
```

See also: **print** (8.112), **write** (8.164), **decimal** (8.26), **dyadic** (8.40), **powers** (8.108), **binary** (8.16), **hexadecimal** (8.66), **prec** (8.109)

## 8.36

Name: /  
division function  
Usage:

$function1 / function2 : (function, function) \rightarrow function$

Parameters:

- *function1* and *function2* represent functions

Description:

- / represents the division (function) on reals. The expression *function1* / *function2* stands for the function composed of the division function and the two functions *function1* and *function2*, where *function1* is the numerator and *function2* the denominator.

Example 1:

```
> 5 / 2;
2.5
```

Example 2:

```
> x / 2;  
x * 0.5
```

Example 3:

```
> x / x;  
1
```

Example 4:

```
> 3 / 0;  
@NaN@
```

Example 5:

```
> diff(sin(x) / exp(x));  
(exp(x) * cos(x) - sin(x) * exp(x)) / exp(x)^2
```

See also: `+` (8.103), `-` (8.89), `*` (8.90), `^` (8.107)

## 8.37 double

Names: **double**, **D**

rounding to the nearest IEEE 754 double (binary64).

Description:

- **double** is both a function and a constant.
- As a function, it rounds its argument to the nearest IEEE 754 double precision (i.e. IEEE754-2008 binary64) number. Subnormal numbers are supported as well as standard numbers: it is the real rounding described in the standard.
- As a constant, it symbolizes the double precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **round**, **roundcoefficients** and **implementpoly**. See the corresponding help pages for examples.

Example 1:

```
> display=binary!;  
> D(0.1);  
1.100110011001100110011001100110011001100110011001101_2 * 2^(-4)  
> D(4.17);  
1.000010101110000101000111101011100001010001111010111_2 * 2^(2)  
> D(1.011_2 * 2^(-1073));  
1.1_2 * 2^(-1073)
```

See also: **single** (8.143), **printhexa** (8.115), **doubleextended** (8.39), **doubledouble** (8.38), **triple-double** (8.158), **roundcoefficients** (8.134), **implementpoly** (8.70), **round** (8.133)

## 8.38 doubledouble

Names: **doubledouble**, **DD**

represents a number as the sum of two IEEE doubles.

Description:

- **doubledouble** is both a function and a constant.
- As a function, it rounds its argument to the nearest number that can be written as the sum of two double precision numbers.





Example 3:

```
> verbosity=1!;  
> f = sin(DE(x));  
> f(pi);  
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.  
-5.0165576126683320235573270803307570138315616702549e-20  
> g = sin(round(x,64,RN));  
Warning: at least one of the given expressions or a subexpression is not correctly typed  
or its evaluation has failed because of some error on a side-effect.
```

See also: **roundcoefficients** (8.134), **single** (8.143), **double** (8.37), **doubledouble** (8.38), **tripledouble** (8.158), **round** (8.133)

## 8.40 dyadic

Name: **dyadic**

special value for global state **display**

Description:

- **dyadic** is a special value used for the global state **display**. If the global state **display** is equal to **dyadic**, all data will be output in dyadic notation with numbers displayed in Gappa format.

As any value it can be affected to a variable and stored in lists.

See also: **decimal** (8.26), **powers** (8.108), **hexadecimal** (8.66), **binary** (8.16)

## 8.41 ==

Name: ==

equality test operator

Usage:

$$expr1 == expr2 : (\text{any type}, \text{any type}) \rightarrow \text{boolean}$$

Parameters:

- *expr1* and *expr2* represent expressions

Description:

- The operator == evaluates to true iff its operands *expr1* and *expr2* are syntactically equal and different from **error** or constant expressions that are not constants and that evaluate to the same floating-point number with the global precision **prec**. The user should be aware of the fact that because of floating-point evaluation, the operator == is not exactly the same as the mathematical equality. Further remark that according to IEEE 754-2008 floating-point rules, which **Sollya** emulates, floating-point data which are NaN do not compare equal to any other floating-point datum, including NaN.

Example 1:

```
> "Hello" == "Hello";  
true  
> "Hello" == "Salut";  
false  
> "Hello" == 5;  
false  
> 5 + x == 5 + x;  
true
```

Example 2:

```
> 1 == exp(0);
true
> asin(1) * 2 == pi;
true
> exp(5) == log(4);
false
```

Example 3:

```
> sin(pi/6) == 1/2 * sqrt(3);
false
```

Example 4:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384.1 == 16385.1;
true
```

Example 5:

```
> error == error;
false
```

Example 6:

```
> a = "Biba";
> b = NaN;
> a == a;
true
> b == b;
false
```

See also: `!=` (8.92), `>` (8.63), `>=` (8.62), `<=` (8.77), `<` (8.85), `!` (8.94), `&&` (8.6), `||` (8.98), `error` (8.44), `prec` (8.109)

## 8.42 erf

Name: **erf**

the error function.

Description:

- **erf** is the error function defined by:

$$\operatorname{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt.$$

- It is defined for every real number  $x$ .

See also: **erfc** (8.43), **exp** (8.47)

## 8.43 erfc

Name: **erfc**

the complementary error function.

Description:

- **erfc** is the complementary error function defined by  $\operatorname{erfc}(x) = 1 - \operatorname{erf}(x)$ .
- It is defined for every real number  $x$ .

See also: **erf** (8.42)

## 8.44 error

Name: **error**

expression representing an input that is wrongly typed or that cannot be executed

Usage:

**error** : error

Description:

- The variable **error** represents an input during the evaluation of which a type or execution error has been detected or is to be detected. Inputs that are syntactically correct but wrongly typed evaluate to **error** at some stage. Inputs that are correctly typed but containing commands that depend on side-effects that cannot be performed or inputs that are wrongly typed at meta-level (cf. **parse**), evaluate to **error**.

Remark that in contrast to all other elements of the Sollya language, **error** compares neither equal nor unequal to itself. This provides a means of detecting syntax errors inside the Sollya language itself without introducing issues of two different wrongly typed inputs being equal.

Example 1:

```
> print(5 + "foo");
error
```

Example 2:

```
> error;
error
```

Example 3:

```
> error == error;
false
> error != error;
false
```

Example 4:

```
> correct = 5 + 6;
> incorrect = 5 + "foo";
> (correct == error || correct != error);
true
> (incorrect == error || incorrect != error);
false
```

See also: **void** (8.162), **parse** (8.99)

## 8.45 evaluate

Name: **evaluate**

evaluates a function at a constant point or in a range

Usage:

**evaluate**(*function*, *constant*) : (function, constant) → constant | range  
**evaluate**(*function*, *range*) : (function, range) → range  
**evaluate**(*function*, *function2*) : (function, function) → function

Parameters:

- *function* represents a function

- *constant* represents a constant point
- *range* represents a range
- *function2* represents a function that is not constant

Description:

- If its second argument is a constant *constant*, **evaluate** evaluates its first argument *function* at the point indicated by *constant*. This evaluation is performed in a way that the result is a faithful rounding of the real value of the *function* at *constant* to the current global precision. If such a faithful rounding is not possible, **evaluate** returns a range surely encompassing the real value of the function *function* at *constant*. If even interval evaluation is not possible because the expression is undefined or numerically unstable, NaN will be produced.
- If its second argument is a range *range*, **evaluate** evaluates its first argument *function* by interval evaluation on this range *range*. This ensures that the image domain of the function *function* on the preimage domain *range* is surely enclosed in the returned range.
- In the case when the second argument is a range that is reduced to a single point (such that [1; 1] for instance), the evaluation is performed in the same way as when the second argument is a constant but it produces a range as a result: **evaluate** automatically adjusts the precision of the intern computations and returns a range that contains at most three floating-point consecutive numbers in precision **prec**. This corresponds to the same accuracy as a faithful rounding of the actual result. If such a faithful rounding is not possible, **evaluate** has the same behavior as in the case when the second argument is a constant.
- If its second argument is a function *function2* that is not a constant, **evaluate** replaces all occurrences of the free variable in function *function* by function *function2*.

Example 1:

```
> midpointmode=on!;
> print(evaluate(sin(pi * x), 2.25));
0.70710678118654752440084436210484903928483593768847
> print(evaluate(sin(pi * x), [2.25; 2.25]));
0.707106781186547524400844362104849039284835937688~4/5~
```

Example 2:

```
> print(evaluate(sin(pi * x), 2));
[-1.72986452514381269516508615031098129542836767991679e-12715;7.5941198201187963
145069564314525661706039084390067e-12716]
```

Example 3:

```
> print(evaluate(sin(pi * x), [2, 2.25]));
[-5.143390272677254630046998919961912407349224165421e-50;0.707106781186547524400
84436210484903928483593768866]
```

Example 4:

```
> print(evaluate(sin(pi * x), 2 + 0.25 * x));
sin((pi) * (2 + 0.25 * x))
```

Example 5:

```
> print(evaluate(sin(pi * 1/x), 0));
[@NaN@;@NaN@]
```

See also: **isevaluable** (8.76)

## 8.46 execute

Name: **execute**

executes the content of a file

Usage:

**execute**(*filename*) : string → void

Parameters:

- *filename* is a string representing a file name

Description:

- **execute** opens the file indicated by *filename*, and executes the sequence of commands it contains. This command is evaluated at execution time: this way you can modify the file *filename* (for instance using **bashexecute**) and execute it just after.
- If *filename* contains a command **execute**, it will be executed recursively.
- If *filename* contains a call to **restart**, it will be neglected.
- If *filename* contains a call to **quit**, the commands following **quit** in *filename* will be neglected.

Example 1:

```
> a=2;
> a;
2
> print("a=1;") > "example.sollya";
> execute("example.sollya");
> a;
1
```

Example 2:

```
> verbosity=1!;
> print("a=1; restart; a=2;") > "example.sollya";
> execute("example.sollya");
Warning: a restart command has been used in a file read into another.
This restart command will be neglected.
> a;
2
```

Example 3:

```
> verbosity=1!;
> print("a=1; quit; a=2;") > "example.sollya";
> execute("example.sollya");
Warning: the execution of a file read by execute demanded stopping the interpretation but it is not stopped.
> a;
1
```

See also: **parse** (8.99), **readfile** (8.124), **write** (8.164), **print** (8.112), **bashexecute** (8.15)

## 8.47 exp

Name: **exp**

the exponential function.

Description:

- **exp** is the usual exponential function defined as the solution of the ordinary differential equation  $y' = y$  with  $y(0) = 1$ .
- **exp**( $x$ ) is defined for every real number  $x$ .

See also: **exp** (8.47), **log** (8.81)

## 8.48 expand

Name: **expand**

expands polynomial subexpressions

Usage:

**expand**(*function*) : function  $\rightarrow$  function

Parameters:

- *function* represents a function

Description:

- **expand**(*function*) expands all polynomial subexpressions in function *function* as far as possible. Factors of sums are multiplied out, power operators with constant positive integer exponents are replaced by multiplications and divisions are multiplied out, i.e. denominators are brought at the most interior point of expressions.

Example 1:

```
> print(expand(x^3));  
x * x * x
```

Example 2:

```
> print(expand((x + 2)^3 + 2 * x));  
8 + 12 * x + 6 * x * x + x * x * x + 2 * x
```

Example 3:

```
> print(expand(exp((x + (x + 3))^5)));  
exp(243 + 405 * x + 270 * x * x + 90 * x * x * x + 15 * x * x * x * x + x * x *  
x * x * x + x * 405 + 108 * x * 5 * x + 54 * x * x * 5 * x + 12 * x * x * x * 5  
* x + x * x * x * x * 5 * x + x * x * 270 + 27 * x * x * x * 10 + 9 * x * x * x  
* x * 10 + x * x * x * x * x * 10 + x * x * x * 90 + 6 * x * x * x * x * 10 + x  
* x * x * x * x * 10 + x * x * x * x * 5 * x + 15 * x * x * x * x + x * x * x *  
x * x)
```

See also: **simplify** (8.140), **simplifysafe** (8.141), **horner** (8.69)

## 8.49 expm1

Name: **expm1**

shifted exponential function.

Description:

- **expm1** is defined by  $\text{expm1}(x) = \exp(x) - 1$ .
- It is defined for every real number  $x$ .

See also: **exp** (8.47)

## 8.50 exponent

Name: **exponent**

returns the scaled binary exponent of a number.

Usage:

**exponent**( $x$ ) : constant  $\rightarrow$  integer

Parameters:

- $x$  is a dyadic number.

Description:

- **exponent**( $x$ ) is by definition 0 if  $x$  is one of 0, NaN, or Inf.
- If  $x$  is not zero, it can be uniquely written as  $x = m \cdot 2^e$  where  $m$  is an odd integer and  $e$  is an integer. **exponent**( $x$ ) returns  $e$ .

Example 1:

```
> a=round(Pi,20,RN);
> e=exponent(a);
> e;
-17
> m=mantissa(a);
> a-m*2^e;
0
```

See also: **mantissa** (8.86), **precision** (8.110)

## 8.51 externalplot

Name: **externalplot**

plots the error of an external code with regard to a function

Usage:

**externalplot**(*filename, mode, function, range, precision*) : (string, absolute|relative, function, range, integer)  $\rightarrow$  void

**externalplot**(*filename, mode, function, range, precision, perturb*) : (string, absolute|relative, function, range, integer, perturb)  $\rightarrow$  void

**externalplot**(*filename, mode, function, range, precision, plot mode, result filename*) : (string, absolute|relative, function, range, integer, file|postscript|postscriptfile, string)  $\rightarrow$  void

**externalplot**(*filename, mode, function, range, precision, perturb, plot mode, result filename*) : (string, absolute|relative, function, range, integer, perturb, file|postscript|postscriptfile, string)  $\rightarrow$  void

Description:

- The command **externalplot** plots the error of an external function evaluation code sequence implemented in the object file named *filename* with regard to the function *function*. If *mode* evaluates to *absolute*, the difference of both functions is considered as an error function; if *mode* evaluates to *relative*, the difference is divided by the function *function*. The resulting error function is plotted on all floating-point numbers with *precision* significant mantissa bits in the range *range*.

If the sixth argument of the command **externalplot** is given and evaluates to **perturb**, each of the floating-point numbers the function is evaluated at gets perturbed by a random value that is uniformly distributed in  $\pm 1$  ulp around the original *precision* bit floating-point variable.

If a sixth and seventh argument, respectively a seventh and eighth argument in the presence of **perturb** as a sixth argument, are given that evaluate to a variable of type file|postscript|postscriptfile respectively to a character sequence of type string, **externalplot** will plot (additionally) to a file in the same way as the command **plot** does. See **plot** for details.

The external function evaluation code given in the object file name *filename* is supposed to define a function name *f* as follows (here in C syntax): `void f(mpfr_t rop, mpfr_op)`. This function is supposed to evaluate *op* with an accuracy corresponding to the precision of *rop* and assign this value to *rop*.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -lmpfr");
> externalplot("./externalplotexample",relative,exp(x),[-1/2;1/2],12,perturb);
```

See also: **plot** (8.102), **asciipLOT** (8.9), **perturb** (8.100), **absolute** (8.2), **relative** (8.126), **file** (8.54), **postscript** (8.105), **postscriptfile** (8.106), **bashexecute** (8.15), **externalproc** (8.52), **library** (8.79)

## 8.52 externalproc

Name: **externalproc**

binds an external code to a **Sollya** procedure

Usage:

**externalproc**(*identifier*, *filename*, *argumenttype*  $\rightarrow$  *resulttype*) : (identifier type, string, type type, type type)  $\rightarrow$  void

Parameters:

- *identifier* represents the identifier the code is to be bound to
- *filename* of type **string** represents the name of the object file where the code of procedure can be found
- *argumenttype* represents a definition of the types of the arguments of the **Sollya** procedure and the external code
- *resulttype* represents a definition of the result type of the external code

Description:

- **externalproc** allows for binding the **Sollya** identifier *identifier* to an external code. After this binding, when **Sollya** encounters *identifier* applied to a list of actual parameters, it will evaluate these parameters and call the external code with these parameters. If the external code indicated success, it will receive the result produced by the external code, transform it to **Sollya**'s internal representation and return it.

In order to allow correct evaluation and typing of the data in parameter and in result to be passed to and received from the external code, **externalproc** has a third parameter *argumenttype*  $\rightarrow$  *resulttype*. Both *argumenttype* and *resulttype* are one of **void**, **constant**, **function**, **range**, **integer**, **string**, **boolean**, **list of constant**, **list of function**, **list of range**, **list of integer**, **list of string**, **list of boolean**.

If upon a usage of a procedure bound to an external procedure the type of the actual parameters given or its number is not correct, **Sollya** produces a type error. An external function not applied to arguments represents itself and prints out with its argument and result types.

The external function is supposed to return an integer indicating success. It returns its result depending on its **Sollya** result type as follows. Here, the external procedure is assumed to be implemented as a C function.

If the **Sollya** result type is **void**, the C function has no pointer argument for the result. If the **Sollya** result type is **constant**, the first argument of the C function is of C type `mpfr_t *`, the result is returned by affecting the MPFR variable. If the **Sollya** result type is **function**, the first argument of the C function is of C type `node **`, the result is returned by the `node *` pointed



with a new node \*. If the Sollya result type is **range**, the first argument of the C function is of C type `mpfi_t *`, the result is returned by affecting the MPFI variable. If the Sollya result type is **integer**, the first argument of the C function is of C type `int *`, the result is returned by affecting the int variable. If the Sollya result type is **string**, the first argument of the C function is of C type `char **`, the result is returned by the `char *` pointed with a new `char *`. If the Sollya result type is **boolean**, the first argument of the C function is of C type `int *`, the result is returned by affecting the int variable with a boolean value. If the Sollya result type is **list of** type, the first argument of the C function is of C type `chain **`, the result is returned by the `chain *` pointed with a new `chain *`. This chain contains for Sollya type **constant** pointers `mpfr_t *` to new MPFR variables, for Sollya type **function** pointers `node *` to new nodes, for Sollya type **range** pointers `mpfi_t *` to new MPFI variables, for Sollya type **integer** pointers `int *` to new int variables for Sollya type **string** pointers `char *` to new `char *` variables and for Sollya type **boolean** pointers `int *` to new int variables representing boolean values.

The external procedure affects its possible pointer argument if and only if it succeeds. This means, if the function returns an integer indicating failure, it does not leak any memory to the encompassing environment.

The external procedure receives its arguments as follows: If the Sollya argument type is **void**, no argument array is given. Otherwise the C function receives a C `void **` argument representing an array of size equal to the arity of the function where each entry (of C type `void *`) represents a value with a C type depending on the corresponding Sollya type. If the Sollya type is **constant**, the C type the `void *` is to be casted to is `mpfr_t *`. If the Sollya type is **function**, the C type the `void *` is to be casted to is `node *`. If the Sollya type is **range**, the C type the `void *` is to be casted to is `mpfi_t *`. If the Sollya type is **integer**, the C type the `void *` is to be casted to is `int *`. If the Sollya type is **string**, the C type the `void *` is to be casted to is `char *`. If the Sollya type is **boolean**, the C type the `void *` is to be casted to is `int *`. If the Sollya type is **list of** type, the C type the `void *` is to be casted to is `chain *`. Here depending on type, the values in the chain are to be casted to `mpfr_t *` for Sollya type **constant**, `node *` for Sollya type **function**, `mpfi_t *` for Sollya type **range**, `int *` for Sollya type **integer**, `char *` for Sollya type **string** and `int *` for Sollya type **boolean**.

The external procedure is not supposed to alter the memory pointed by its array argument `void **`.

In both directions (argument and result values), empty lists are represented by `chain * NULL` pointers.

In contrast to internal procedures, externally bounded procedures can be considered to be objects inside Sollya that can be assigned to other variables, stored in list etc.

Example 1:

```
> bashexecute("gcc -fPIC -Wall -c externalprocexample.c");
> bashexecute("gcc -fPIC -shared -o externalprocexample externalprocexample.o");

> externalproc(foo, "./externalprocexample", (integer, integer) -> integer);
> foo;
foo(integer, integer) -> integer
> foo(5, 6);
11
> verbosity = 1!;
> foo();
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error
> a = foo;
> a(5,6);
11
```

See also: **library** (8.79), **externalplot** (8.51), **bashexecute** (8.15), **void** (8.162), **constant** (8.23), **function** (8.61), **range** (8.120), **integer** (8.73), **string** (8.147), **boolean** (8.17), **list of** (8.80)

### 8.53 false

Name: **false**

the boolean value representing the false.

Description:

- **false** is the usual boolean value.

Example 1:

```
> true && false;
false
> 2<1;
false
```

See also: **true** (8.159), **&&** (8.6), **||** (8.98)

### 8.54 file

Name: **file**

special value for commands **plot** and **externalplot**

Description:

- **file** is a special value used in commands **plot** and **externalplot** to save the result of the command in a data file.
- As any value it can be affected to a variable and stored in lists.

Example 1:

```
> savemode=file;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);
```

See also: **externalplot** (8.51), **plot** (8.102), **postscript** (8.105), **postscriptfile** (8.106)

### 8.55 findzeros

Name: **findzeros**

gives a list of intervals containing all zeros of a function on an interval.

Usage:

**findzeros**( $f, I$ ) : (function, range)  $\rightarrow$  list

Parameters:

- $f$  is a function.
- $I$  is an interval.

Description:

- **findzeros**( $f, I$ ) returns a list of intervals  $I_1, \dots, I_n$  such that, for every zero  $z$  of  $f$ , there exists some  $k$  such that  $z \in I_k$ .
- The list may contain intervals  $I_k$  that do not contain any zero of  $f$ . An interval  $I_k$  may contain many zeros of  $f$ .

- This command is meant for cases when safety is critical. If you want to be sure not to forget any zero, use **findzeros**. However, if you just want to know numerical values for the zeros of  $f$ , **dirtyfindzeros** should be quite satisfactory and a lot faster.
- If  $\delta$  denotes the value of global variable **diam**, the algorithm ensures that for each  $k$ ,  $|I_k| \leq \delta \cdot |I|$ .
- The algorithm used is basically a bisection algorithm. It is the same algorithm that the one used for **infnorm**. See the help page of this command for more details. In short, the behavior of the algorithm depends on global variables **prec**, **diam**, **taylorrecursions** and **hopitalrecursions**.

Example 1:

```
> findzeros(sin(x),[-5;5]);
[|-3.14208984375;-3.140869140625], [-1.220703125e-3;1.220703125e-3], [3.140869140625;3.14208984375]]
> diam=1e-10!;
> findzeros(sin(x),[-5;5]);
[|-3.14159265370108187198638916015625;-3.141592652536928653717041015625], [-1.16415321826934814453125e-9;1.16415321826934814453125e-9], [3.141592652536928653717041015625;3.14159265370108187198638916015625]]
```

See also: **dirtyfindzeros** (8.32), **infnorm** (8.72), **prec** (8.109), **diam** (8.30), **taylorrecursions** (8.155), **hopitalrecursions** (8.68)

## 8.56 fixed

Name: **fixed**

indicates that fixed-point formats should be used for **fpminimax**

Usage:

**fixed** : fixed|floating

Description:

- The use of **fixed** in the command **fpminimax** indicates that the list of formats given as argument is to be considered to be a list of fixed-point formats. See **fpminimax** for details.

Example 1:

```
> fpminimax(cos(x),6,[|32,32,32,32,32,32,32|],[-1;1],fixed);
0.9999997480772435665130615234375 + x^2 * (-0.4999928693287074565887451171875 +
x^2 * (4.163351492024958133697509765625e-2 + x^2 * (-1.338223926723003387451171875e-3)))
```

See also: **fpminimax** (8.59), **floating** (8.57)

## 8.57 floating

Name: **floating**

indicates that floating-point formats should be used for **fpminimax**

Usage:

**floating** : fixed|floating

Description:

- The use of **floating** in the command **fpminimax** indicates that the list of formats given as argument is to be considered to be a list of floating-point formats. See **fpminimax** for details.

Example 1:

```
> fpminimax(cos(x),6,[|D...|],[-1;1],floating);
0.99999974816012948686250183527590706944465637207031 + x * (5.521004406122249513
1782035802443168321913900126185e-14 + x * (-0.4999928698019768802396356477402150
630950927734375 + x * (-3.95371609372064761555136192612768146546591008227978e-13
+ x * (4.16335155285858099505347240665287245064973831176758e-2 + x * (5.2492670
395835122748014980938834327670386437070249e-13 + x * (-1.33822408807599468535953
768366653093835338950157166e-3))))))
```

See also: **fpminimax** (8.59), **fixed** (8.56)

## 8.58 floor

Name: **floor**

the usual function floor.

Description:

- **floor** is defined as usual: **floor**( $x$ ) is the greatest integer  $y$  such that  $y \leq x$ .
- It is defined for every real number  $x$ .

See also: **ceil** (8.19), **floor** (8.58), **nearestint** (8.91)

## 8.59 fpminimax

Name: **fpminimax**

computes a good polynomial approximation with fixed-point or floating-point coefficients

Usage:

**fpminimax**( $f, n, \text{formats}, \text{range}, \text{indic1}, \text{indic2}, \text{indic3}, P$ ) : (function, integer, list, range, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, function)  $\rightarrow$  function

**fpminimax**( $f, \text{monomials}, \text{formats}, \text{range}, \text{indic1}, \text{indic2}, \text{indic3}, P$ ) : (function, list, list, range, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, function)  $\rightarrow$  function

**fpminimax**( $f, n, \text{formats}, L, \text{indic1}, \text{indic2}, \text{indic3}, P$ ) : (function, integer, list, list, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, function)  $\rightarrow$  function

**fpminimax**( $f, \text{monomials}, \text{formats}, L, \text{indic1}, \text{indic2}, \text{indic3}, P$ ) : (function, list, list, list, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, absolute|relative | fixed|floating | function, function)  $\rightarrow$  function

Parameters:

- $f$  is the function to be approximated
- $n$  is the degree of the polynomial that must approximate  $f$
- *monomials* is the list of monomials that must be used to represent the polynomial that approximates  $f$
- *formats* is a list indicating the formats that the coefficients of the polynomial must have
- *range* is the interval where the function must be approximated
- $L$  is a list of interpolation points used by the method
- *indic1* (optional) is one of the optional indication parameters. See the detailed description below.
- *indic2* (optional) is one of the optional indication parameters. See the detailed description below.
- *indic3* (optional) is one of the optional indication parameters. See the detailed description below.

- $P$  (optional) is the minimax polynomial to be considered for solving the problem.

Description:

- **fpminimax** uses a heuristic (but practically efficient) method to find a good polynomial approximation of a function  $f$  on an interval  $range$ . It implements the method published in the article:  
Efficient polynomial  $L^\infty$ -approximations  
Nicolas Brisebarre and Sylvain Chevillard  
Proceedings of the 18th IEEE Symposium on Computer Arithmetic (ARITH 18)  
pp. 169-176
- The basic usage of this command is **fpminimax**( $f, n, formats, range$ ). It computes a polynomial approximation of  $f$  with degree at most  $n$  on the interval  $range$ .  $formats$  is a list of integers or format types (such as **double**, **doubledouble**, etc.). The polynomial returned by the command has its coefficients that fit the formats indications. For instance, if  $formats[0]$  is 35, the coefficient of degree 0 of the polynomial will fit a floating-point format of 35 bits. If  $formats[1]$  is D, the coefficient of degree 1 will be representable by a floating-point number with a precision of 53 bits (which is not necessarily an IEEE 754 double precision number. See the remark below), etc.
- The second argument may be either an integer or a list of integers interpreted as the list of desired monomials. For instance, the list  $[0, 2, 4, 6]$  indicates that the polynomial must be even and of degree at most 6. Giving an integer  $n$  as second argument is equivalent as giving  $[0, \dots, n]$ . The list of formats is interpreted with respect to the list of monomials. For instance, if the list of monomials is  $[0, 2, 4, 6]$  and the list of formats is  $[161, 107, 53, 24]$ , the coefficients of degree 0 is searched as a floating-point number with precision 161, the coefficient of degree 2 is searched as a number of precision 107, and so on.
- The list of formats may contain either integers or format types (**double**, **doubledouble**, **triple-double** and **doubleextended**). The list may be too large or even infinite. Only the first indications will be considered. For instance, for a degree  $n$  polynomial,  $formats[n+1]$  and above will be discarded. This lets one use elliptical indications for the last coefficients.
- The floating-point coefficients considered by **fpminimax** do not have an exponent range. In particular, in the format list, **double** or 53 does not imply that the corresponding coefficient is an IEEE-754 double.
- By default, the list of formats is interpreted as a list of floating-point formats. This may be changed by passing **fixed** as an optional argument (see below). Let us take an example: **fpminimax**( $f, 2, [107, DD, 53], [0;1]$ ). Here the optional argument is missing (we could have set it to **floating**). Thus, **fpminimax** will search for a polynomial of degree 2 with a constant coefficient that is a 107 bits floating-point number, etc.  
Currently, **doubledouble** is just a synonym for 107 and **tripleddouble** a synonym for 161. This behavior may change in the future (taking into account the fact that some double-doubles are not representable with 107 bits).  
Second example: **fpminimax**( $f, 2, [25, 18, 30], [0;1], \text{fixed}$ ). In this case, **fpminimax** will search for a polynomial of degree 2 with a constant coefficient of the form  $m/2^{25}$  where  $m$  is an integer. In other words, it is a fixed-point number with 25 bits after the point. Note that even with argument **fixed**, the formats list is allowed to contain **double**, **doubledouble** or **tripleddouble**. In this case, it is just a synonym for 53, 107 or 161. This is deprecated and may change in the future.
- The fourth argument may be a range or a list. Lists are for advanced users that know what they are doing. The core of the method is a kind of approximated interpolation. The list given here is a list of points that must be considered for the interpolation. It must contain at least as many points as unknown coefficients. If you give a list, it is also recommended that you provide the minimax polynomial as last argument. If you give a range, the list of points will be automatically computed.
- The fifth, sixth and seventh arguments are optional. By default, **fpminimax** will approximate  $f$  while optimizing the relative error, and interpreting the list of formats as a list of floating-point formats.

This default behavior may be changed with these optional arguments. You may provide zero, one, two or three of the arguments in any order. This lets the user indicate only the non-default arguments.

The three possible arguments are:

- **relative** or **absolute**: the error to be optimized;
- **floating** or **fixed**: formats of the coefficients;
- a constrained part  $q$ .

The constrained part lets the user assign in advance some of the coefficients. For instance, for approximating  $\exp(x)$ , it may be interesting to search for a polynomial  $p$  of the form

$$p = 1 + x + \frac{x^2}{2} + a_3x^3 + a_4x^4.$$

Thus, there is a constrained part  $q = 1 + x + x^2/2$  and the unknown polynomial should be considered in the monomial basis  $[[3, 4]]$ . Calling **fpminimax** with monomial basis  $[[3, 4]]$  and constrained part  $q$ , will return a polynomial with the right form.

- The last argument is for advanced users. It is the minimax polynomial that approximates the function  $f$  in the monomial basis. If it is not given this polynomial will be automatically computed by **fpminimax**.

This minimax polynomial is used to compute the list of interpolation points required by the method. In general, you do not have to provide this argument. But if you want to obtain several polynomials of the same degree that approximate the same function on the same range, just changing the formats, you should probably consider computing only once the minimax polynomial and the list of points instead of letting **fpminimax** recompute them each time.

Note that in the case when a constrained part is given, the minimax polynomial must take that into account. For instance, in the previous example, the minimax would be obtained by the following command:

```
P = remez(1-(1+x+x^2/2)/exp(x), [[3,4]], range, 1/exp(x));
```

Note that the constrained part is not to be added to  $P$ .

- Note that **fpminimax** internally computes a minimax polynomial (using the same algorithm as **remez** command). Thus **fpminimax** may encounter the same problems as **remez**. In particular, it may be very slow when Haar condition is not fulfilled. Another consequence is that currently **fpminimax** has to be run with a sufficiently high working precision.

Example 1:

```
> P = fpminimax(cos(x),6,[|DD, DD, D...|],[-1b-5;1b-5]);
> printexpansion(P);
(0x3ff0000000000000 + 0xbc09fda20235c100) + x * ((0x3b29ecd485d34781 + 0xb7c1cbc
97120359a) + x * (0xbfdffffffffffff98 + x * (0xbbfa6e0b3183cb0d + x * (0x3fa55555
55145337 + x * (0x3ca3540480618939 + x * 0xbf56c138142d8c3b))))))
```

Example 2:

```
> P = fpminimax(sin(x),6,[|32...|],[-1b-5;1b-5], fixed, absolute);
> display = powers!;
> P;
x * (1 + x^2 * (-357913941 * 2^(-31) + x^2 * (35789873 * 2^(-32))))
```

Example 3:

```

> P = fpminimax(exp(x), [|3,4|], [|D,24|], [-1/256; 1/246], 1+x+x^2/2);
> display = powers!;
> P;
1 + x * (1 + x * (1 * 2^(-1) + x * (375300225001191 * 2^(-51) + x * (5592621 * 2
^(-27)))))

```

Example 4:

```

> f = cos(exp(x));
> pstar = remez(f, 5, [-1b-7;1b-7]);
> listpoints = dirtyfindzeros(f-pstar, [-1b-7; 1b-7]);
> P1 = fpminimax(f, 5, [|DD...|], listpoints, absolute, default, default, pstar)
;
> P2 = fpminimax(f, 5, [|D...|], listpoints, absolute, default, default, pstar);

> P3 = fpminimax(f, 5, [|D, D, D, 24...|], listpoints, absolute, default, default, pstar);
> print("Error of pstar: ", dirtyinfnorm(f-pstar, [-1b-7; 1b-7]));
Error of pstar: 7.9048441305459735102879831325718745399379329453102e-16
> print("Error of P1:      ", dirtyinfnorm(f-P1, [-1b-7; 1b-7]));
Error of P1:      7.9048441305459735159848647089192667442047469404883e-16
> print("Error of P2:      ", dirtyinfnorm(f-P2, [-1b-7; 1b-7]));
Error of P2:      8.2477144579950871061147021597406077993657714575238e-16
> print("Error of P3:      ", dirtyinfnorm(f-P3, [-1b-7; 1b-7]));
Error of P3:      1.08454277156993282593701156841863009789063333951055e-15

```

See also: **remez** (8.127), **dirtyfindzeros** (8.32), **absolute** (8.2), **relative** (8.126), **fixed** (8.56), **floating** (8.57), **default** (8.27)

## 8.60 fullparentheses

Name: **fullparentheses**

activates, deactivates or inspects the state variable controlling output with full parenthesising

Usage:

```

fullparentheses = activation value : on|off → void
fullparentheses = activation value ! : on|off → void

```

Parameters:

- *activation value* represents **on** or **off**, i.e. activation or deactivation

Description:

- An assignment **fullparentheses** = *activation value*, where *activation value* is one of **on** or **off**, activates respectively deactivates the output of expressions with full parenthesising. In full parenthesising mode, **Sollya** commands like **print**, **write** and the implicit command when an expression is given at the prompt will output expressions with parenthesising at all places where it is necessary for expressions containing infix operators to be parsed back with the same result. Otherwise parentheses around associative operators are omitted.

If the assignment **fullparentheses** = *activation value* is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

Example 1:

```

> autosimplify = off!;
> fullparentheses = off;
Full parentheses mode has been deactivated.
> print(1 + 2 + 3);
1 + 2 + 3
> fullparentheses = on;
Full parentheses mode has been activated.
> print(1 + 2 + 3);
(1 + 2) + 3

```

See also: **print** (8.112), **write** (8.164), **autosimplify** (8.14)

## 8.61 function

Name: **function**

keyword representing a function type

Usage:

**function** : type type

Description:

- **function** represents the function type for declarations of external procedures by means of **externalproc**.

Remark that in contrast to other indicators, type indicators like **function** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **boolean** (8.17), **constant** (8.23), **integer** (8.73), **list of** (8.80), **range** (8.120), **string** (8.147)

## 8.62 >=

Name: >=

greater-than-or-equal-to operator

Usage:

$expr1 \geq expr2 : (\text{constant}, \text{constant}) \rightarrow \text{boolean}$

Parameters:

- *expr1* and *expr2* represent constant expressions

Description:

- The operator >= evaluates to true iff its operands *expr1* and *expr2* evaluate to two floating-point numbers  $a_1$  respectively  $a_2$  with the global precision **prec** and  $a_1$  is greater than or equal to  $a_2$ . The user should be aware of the fact that because of floating-point evaluation, the operator >= is not exactly the same as the mathematical operation *greater-than-or-equal-to*.

Example 1:

```

> 5 >= 4;
true
> 5 >= 5;
true
> 5 >= 6;
false
> exp(2) >= exp(1);
true
> log(1) >= exp(2);
false

```



Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384.1 >= 16385.1;
true
```

See also: `==` (8.41), `!=` (8.92), `>` (8.63), `<=` (8.77), `<` (8.85), `!` (8.94), `&&` (8.6), `||` (8.98), `prec` (8.109)

### 8.63 `>`

Name: `>`

greater-than operator

Usage:

$$expr1 > expr2 : (\text{constant}, \text{constant}) \rightarrow \text{boolean}$$

Parameters:

- *expr1* and *expr2* represent constant expressions

Description:

- The operator `>` evaluates to true iff its operands *expr1* and *expr2* evaluate to two floating-point numbers  $a_1$  respectively  $a_2$  with the global precision `prec` and  $a_1$  is greater than  $a_2$ . The user should be aware of the fact that because of floating-point evaluation, the operator `>` is not exactly the same as the mathematical operation *greater-than*.

Example 1:

```
> 5 > 4;
true
> 5 > 5;
false
> 5 > 6;
false
> exp(2) > exp(1);
true
> log(1) > exp(2);
false
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16385.1 > 16384.1;
false
```

See also: `==` (8.41), `!=` (8.92), `>=` (8.62), `<=` (8.77), `<` (8.85), `!` (8.94), `&&` (8.6), `||` (8.98), `prec` (8.109)

### 8.64 `guessdegree`

Name: `guessdegree`

returns the minimal degree needed for a polynomial to approximate a function with a certain error on an interval.

Usage:

$$\text{guessdegree}(f, I, \text{eps}, w) : (\text{function}, \text{range}, \text{constant}, \text{function}) \rightarrow \text{range}$$

Parameters:

- $f$  is the function to be approximated.
- $I$  is the interval where the function must be approximated.
- $eps$  is the maximal acceptable error.
- $w$  (optional) is a weight function. Default is 1.

Description:

- **guessdegree** tries to find the minimal degree needed to approximate  $f$  on  $I$  by a polynomial with an error  $\epsilon = pw - f$  whose infinity norm not greater than  $eps$ . More precisely, it finds  $n$  minimal such that there exists a polynomial  $p$  of degree  $n$  such that  $\|pw - f\|_\infty < eps$ .
- **guessdegree** returns an interval: for common cases, this interval is reduced to a single number (e.g. the minimal degree). But in certain cases, **guessdegree** does not succeed in finding the minimal degree. In such cases the returned interval is of the form  $[n, p]$  such that:
  - no polynomial of degree  $n - 1$  gives an error less than  $eps$ .
  - there exists a polynomial of degree  $p$  giving an error less than  $eps$ .

Example 1:

```
> guessdegree(exp(x), [-1;1], 1e-10);  
[10;10]
```

Example 2:

```
> guessdegree(1, [-1;1], 1e-8, 1/exp(x));  
[8;9]
```

See also: **dirtyinfnorm** (8.33), **remez** (8.127)

## 8.65 head

Name: **head**

gives the first element of a list.

Usage:

**head**( $L$ ) : list  $\rightarrow$  any type

Parameters:

- $L$  is a list.

Description:

- **head**( $L$ ) returns the first element of the list  $L$ . It is equivalent to  $L[0]$ .
- If  $L$  is empty, the command will fail with an error.

Example 1:

```
> head([|1,2,3|]);  
1  
> head([|1,2...|]);  
1
```

See also: **tail** (8.151)

## 8.66 hexadecimal

Name: **hexadecimal**

special value for global state **display**

Description:

- **hexadecimal** is a special value used for the global state **display**. If the global state **display** is equal to **hexadecimal**, all data will be output in hexadecimal C99/ IEEE 754-2008 notation.

As any value it can be affected to a variable and stored in lists.

See also: **decimal** (8.26), **dyadic** (8.40), **powers** (8.108), **binary** (8.16)

## 8.67 honorcoeffprec

Name: **honorcoeffprec**

indicates the (forced) honoring the precision of the coefficients in **implementpoly**

Usage:

**honorcoeffprec** : honorcoeffprec

Description:

- Used with command **implementpoly**, **honorcoeffprec** makes **implementpoly** honor the precision of the given polynomial. This means if a coefficient needs a double-double or a triple-double to be exactly stored, **implementpoly** will allocate appropriate space and use a double-double or triple-double operation even if the automatic (heuristic) determination implemented in command **implementpoly** indicates that the coefficient could be stored on less precision or, respectively, the operation could be performed with less precision. See **implementpoly** for details.

Example 1:

```
> verbosity = 1!;
> q = implementpoly(1 - simplify(TD(1/6)) * x^2, [-1b-10;1b-10], 1b-60, DD, "p", "implementation.c");
Warning: at least one of the coefficients of the given polynomial has been rounded in a way
that the target precision can be achieved at lower cost. Nevertheless, the implemented polynomial
is different from the given one.
> printexpansion(q);
0x3ff0000000000000 + x^2 * 0xbfc5555555555555
> r = implementpoly(1 - simplify(TD(1/6)) * x^2, [-1b-10;1b-10], 1b-60, DD, "p", "implementation.c", honorcoeffprec);
Warning: the inferred precision of the 2th coefficient of the polynomial is greater than
the necessary precision computed for this step. This may make the automatic determination
of precisions useless.
> printexpansion(r);
0x3ff0000000000000 + x^2 * (0xbfc5555555555555 + 0xbc65555555555555 + 0xb905555555555555)
```

See also: **implementpoly** (8.70), **printexpansion** (8.113)

## 8.68 hopitalrecursions

Name: **hopitalrecursions**

controls the number of recursion steps when applying L'Hopital's rule.

Usage:

**hopitalrecursions** =  $n$  : integer  $\rightarrow$  void  
**hopitalrecursions** =  $n$  ! : integer  $\rightarrow$  void  
**hopitalrecursions** : integer

Parameters:

- $n$  represents the number of recursions

Description:

- **hopitalrecursions** is a global variable. Its value represents the number of steps of recursion that are tried when applying L'Hopital's rule. This rule is applied by the interval evaluator present in the core of **Sollya** (and particularly visible in commands like **infnorm**).
- If an expression of the form  $f/g$  has to be evaluated by interval arithmetic on an interval  $I$  and if  $f$  and  $g$  have a common zero in  $I$ , a direct evaluation leads to NaN. **Sollya** implements a safe heuristic to avoid this, based on L'Hopital's rule: in such a case, it can be shown that  $(f/g)(I) \subseteq (f'/g')(I)$ . Since the same problem may exist for  $f'/g'$ , the rule is applied recursively. The number of step in this recursion process is controlled by **hopitalrecursions**.
- Setting **hopitalrecursions** to 0 makes **Sollya** use this rule only once; setting it to 1 makes **Sollya** use the rule twice, and so on. In particular: the rule is always applied at least once, if necessary.

Example 1:

```

> hopitalrecursions=0;
The number of recursions for Hopital's rule has been set to 0.
> evaluate(log(1+x)^2/x^2,[-1/2; 1]);
[-@Inf@;@Inf@]
> hopitalrecursions=1;
The number of recursions for Hopital's rule has been set to 1.
> evaluate(log(1+x)^2/x^2,[-1/2; 1]);
[-2.52258872223978123766892848583270627230200053744108;6.77258872223978123766892
84858327062723020005374411]
  
```

## 8.69 horner

Name: **horner**

brings all polynomial subexpressions of an expression to Horner form

Usage:

**horner**(*function*) : function  $\rightarrow$  function

Parameters:

- *function* represents the expression to be rewritten in Horner form

Description:

- The command **horner** rewrites the expression representing the function *function* in a way such that all polynomial subexpressions (or the whole expression itself, if it is a polynomial) are written in Horner form. The command **horner** does not endanger the safety of computations even in **Sollya**'s floating-point environment: the function returned is mathematically equal to the function *function*.

Example 1:

```

> print(horner(1 + 2 * x + 3 * x^2));
1 + x * (2 + x * 3)
> print(horner((x + 1)^7));
1 + x * (7 + x * (21 + x * (35 + x * (35 + x * (21 + x * (7 + x))))))
  
```

Example 2:

```
> print(horner(exp((x + 1)^5) - log(asin(x + x^3) + x)));
exp(1 + x * (5 + x * (10 + x * (10 + x * (5 + x)))) - log(asin(x * (1 + x^2)) +
x)
```

See also: **canonical** (8.18), **print** (8.112)

## 8.70 implementpoly

Name: **implementpoly**

implements a polynomial using double, double-double and triple-double arithmetic and generates a Gappa proof

Usage:

**implementpoly**(*polynomial*, *range*, *error bound*, *format*, *functionname*, *filename*) : (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string) → function

**implementpoly**(*polynomial*, *range*, *error bound*, *format*, *functionname*, *filename*, *honor coefficient precisions*) : (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string, honorcoeffprec) → function

**implementpoly**(*polynomial*, *range*, *error bound*, *format*, *functionname*, *filename*, *proof filename*) : (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string, string) → function

**implementpoly**(*polynomial*, *range*, *error bound*, *format*, *functionname*, *filename*, *honor coefficient precisions*, *proof filename*) : (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string, honorcoeffprec, string) → function

Description:

- The command **implementpoly** implements the polynomial *polynomial* in range *range* as a function called *functionname* in C code using double, double-double and triple-double arithmetic in a way that the rounding error (estimated at its first order) is bounded by *error bound*. The produced code is output in a file named *filename*. The argument *format* indicates the double, double-double or triple-double format of the variable in which the polynomial varies, influencing also in the signature of the C function.

If a seventh or eighth argument *proof filename* is given and if this argument evaluates to a variable of type string, the command **implementpoly** will produce a Gappa proof that the rounding error is less than the given bound. This proof will be output in Gappa syntax in a file name *proof filename*.

The command **implementpoly** returns the polynomial that has been implemented. As the command **implementpoly** tries to adapt the precision needed in each evaluation step to its strict minimum and as it applies renormalization to double-double and triple-double precision coefficients to bring them to a round-to-nearest expansion form, the returned polynomial may differ from the polynomial *polynomial*. Nevertheless the difference will be small enough that the rounding error bound with regard to the polynomial *polynomial* (estimated at its first order) will be less than the given error bound.

If a seventh argument *honor coefficient precisions* is given and evaluates to a variable **honorcoeffprec** of type honorcoeffprec, **implementpoly** will honor the precision of the given polynomial *polynomials*. This means if a coefficient needs a double-double or a triple-double to be exactly stored, **implementpoly** will allocate appropriate space and use a double-double or triple-double operation even if the automatic (heuristic) determination implemented in command **implementpoly** indicates that the coefficient could be stored on less precision or, respectively, the operation could be performed with less precision. The use of **honorcoeffprec** has advantages and disadvantages. If the polynomial *polynomial* given has not been determined by a process considering directly polynomials with floating-point coefficients, **honorcoeffprec** should not be indicated. The **implementpoly** command can then determine the needed precision using the same error estimation as used for the determination of the precisions of the operations. Generally, the coefficients will get rounded to double, double-double and triple-double precision in a way that minimizes their

The implementer behind the **implementpoly** command makes some assumptions on its input and verifies them. If some assumption cannot be verified, the implementation will not succeed and **implementpoly** will evaluate to a variable **error** of type `error`. The same behaviour is observed if some file is not writable or some other side-effect fails, e.g. if the implementer runs out of memory.

Example 1:

70

Example 2:

```
> implementpoly(1 - 1/6 * x^2 + 1/120 * x^4, [-1b-10;1b-10], 1b-30, D, "p","impl
ementation.c","implementation.gappa");
1 + x^2 * (-0.166666666666666657414808128123695496469736099243164 + x^2 * 8.3333
33333333332176851016015461937058717012405395e-3)
```

Example 3:

```
> verbosity = 1!;
> q = implementpoly(1 - simplify(TD(1/6)) * x^2, [-1b-10;1b-10], 1b-60, DD, "p", "imp
lementation.c");
Warning: at least one of the coefficients of the given polynomial has been round
ed in a way
that the target precision can be achieved at lower cost. Nevertheless, the imple
mented polynomial
is different from the given one.
> printexpansion(q);
0x3ff0000000000000 + x^2 * 0xbfc5555555555555
> r = implementpoly(1 - simplify(TD(1/6)) * x^2, [-1b-10;1b-10], 1b-60, DD, "p", "imp
lementation.c", honorcoeffprec);
Warning: the inferred precision of the 2th coefficient of the polynomial is great
er than
the necessary precision computed for this step. This may make the automatic dete
rmination
of precisions useless.
> printexpansion(r);
0x3ff0000000000000 + x^2 * (0xbfc5555555555555 + 0xbc65555555555555 + 0xb9055555
55555555)
```

Example 4:

```
> p = 0x3ff0000000000000 + x * (0x3ff0000000000000 + x * (0x3fe0000000000000 + x
* (0x3fc5555555555555 + x * (0x3fa5555555555555 + x * (0x3f8111111111106e2 + x
* (0x3f56c16c16bf5eb7 + x * (0x3f2a01a01a292dcd + x * (0x3efa01a0218a016a + x
* (0x3ec71de360331aad + x * (0x3e927e42e3823bf3 + x * (0x3e5ae6b2710c2c9a + x
* (0x3e2203730c0a7c1d + x * 0x3de5da557e0781df)))))))));
> q = implementpoly(p, [-1/2;1/2], 1b-60, D, "p", "implementation.c", honorcoeffprec, "
implementation.gappa");
> if (q != p) then print("During implementation, rounding has happened.") else p
rint("Polynomial implemented as given.");
Polynomial implemented as given.
```

See also: **honorcoeffprec** (8.67), **roundcoefficients** (8.134), **double** (8.37), **doubledouble** (8.38), **tripleddouble** (8.158), **readfile** (8.124), **printexpansion** (8.113), **error** (8.44)

## 8.71 inf

Name: **inf**

gives the lower bound of an interval.

Usage:

**inf**( $I$ ) : range  $\rightarrow$  constant  
**inf**( $x$ ) : constant  $\rightarrow$  constant

Parameters:

- $I$  is an interval.

- $x$  is a real number.

Description:

- Returns the lower bound of the interval  $I$ . Each bound of an interval has its own precision, so this command is exact, even if the current precision is too small to represent the bound.
- When called on a real number  $x$ , **inf** considers it as an interval formed of a single point:  $[x, x]$ . In other words, **inf** behaves like the identity.

Example 1:

```
> inf([1;3]);
1
> inf(0);
0
```

Example 2:

```
> display=binary!;
> I=[0.111110000011111_2; 1];
> inf(I);
1.11110000011111_2 * 2^(-1)
> prec=12!;
> inf(I);
1.11110000011111_2 * 2^(-1)
```

See also: **mid** (8.87), **sup** (8.150)

## 8.72 infnorm

Name: **infnorm**

computes an interval bounding the infinity norm of a function on an interval.

Usage:

**infnorm**( $f, I, filename, Ilist$ ) : (function, range, string, list)  $\rightarrow$  range

Parameters:

- $f$  is a function.
- $I$  is an interval.
- $filename$  (optional) is the name of the file into a proof will be saved.
- $Ilist$  (optional) is a list of intervals to be excluded.

Description:

- **infnorm**( $f, range$ ) computes an interval bounding the infinity norm of the given function  $f$  on the interval  $I$ , e.g. computes an interval  $J$  such that  $\max_{x \in I} \{|f(x)|\} \subseteq J$ .
- If  $filename$  is given, a proof in English will be produced (and stored in file called  $filename$ ) proving that  $\max_{x \in I} \{|f(x)|\} \subseteq J$ .
- If a list  $Ilist$  of intervals  $I_1, \dots, I_n$  is given, the infinity norm will be computed on  $I \setminus (I_1 \cup \dots \cup I_n)$ .
- The function  $f$  is assumed to be at least twice continuous on  $I$ . More generally, if  $f$  is  $\mathcal{C}^k$ , global variables **hopitalrecursions** and **taylorrecursions** must have values not greater than  $k$ .
- If the interval is reduced to a single point, the result of **infnorm** is an interval containing the exact absolute value of  $f$  at this point.



- If the interval is not bound, the result will be  $[0, +\infty]$  which is correct but perfectly useless. **infnorm** is not meant to be used with infinite intervals.
- The result of this command depends on the global variables **prec**, **diam**, **taylorrecursions** and **hopitalrecursions**. The contribution of each variable is not easy even to analyse.
  - The algorithm uses interval arithmetic with precision **prec**. The precision should thus be set high enough to ensure that no critical cancellation will occur.
  - When an evaluation is performed on an interval  $[a, b]$ , if the result is considered being too large, the interval is split into  $[a, \frac{a+b}{2}]$  and  $[\frac{a+b}{2}, b]$  and so on recursively. This recursion step is not performed if the  $(b - a) < \delta \cdot |I|$  where  $\delta$  is the value of variable **diam**. In other words, **diam** controls the minimum length of an interval during the algorithm.
  - To perform the evaluation of a function on an interval, Taylor's rule is applied, e.g.  $f([a, b]) \subseteq f(m) + [a - m, b - m] \cdot f'([a, b])$  where  $m = \frac{a+b}{2}$ . This rule is recursively applied  $n$  times where  $n$  is the value of variable **taylorrecursions**. Roughly speaking, the evaluations will avoid decorrelation up to order  $n$ .
  - When a function of the form  $\frac{g}{h}$  has to be evaluated on an interval  $[a, b]$  and when  $g$  and  $h$  vanish at a same point  $z$  of the interval, the ratio may be defined even if the expression  $\frac{g(z)}{h(z)} = \frac{0}{0}$  does not make any sense. In this case, L'Hopital's rule may be used and  $(\frac{g}{h})([a, b]) \subseteq (\frac{g'}{h'})([a, b])$ . Since the same can occur with the ratio  $\frac{g'}{h'}$ , the rule is applied recursively. The variable **hopitalrecursions** controls the number of recursion steps.
- The algorithm used for this command is quite complex to be explained here. Please find a complete description in the following article:  
 S. Chevillard and C. Lauter  
 A certified infinity norm for the implementation of elementary functions  
 LIP Research Report number RR2007-26  
<http://prunel.ccsd.cnrs.fr/ensl-00119810>

Example 1:

```
> infnorm(exp(x), [-2;3]);
[2.00855369231876677409285296545817178969879078385537e1;2.0085536923187667740928
5296545817178969879078385544e1]
```

Example 2:

```
> infnorm(exp(x), [-2;3], "proof.txt");
[2.00855369231876677409285296545817178969879078385537e1;2.0085536923187667740928
5296545817178969879078385544e1]
```

Example 3:

```
> infnorm(exp(x), [-2;3], [| [0;1], [2;2.5] |]);
[2.00855369231876677409285296545817178969879078385537e1;2.0085536923187667740928
5296545817178969879078385544e1]
```

Example 4:

```
> infnorm(exp(x), [-2;3], "proof.txt", [| [0;1], [2;2.5] |]);
[2.00855369231876677409285296545817178969879078385537e1;2.0085536923187667740928
5296545817178969879078385544e1]
```

Example 5:

```
> infnorm(exp(x), [1;1]);
[2.71828182845904523536028747135266249775724709369989;2.718281828459045235360287
47135266249775724709369998]
```

Example 6:

```
> infnorm(exp(x), [log(0);log(1)]);
[0;@Inf@]
```

See also: **prec** (8.109), **diam** (8.30), **hopitalrecursions** (8.68), **dirtyinfnorm** (8.33), **checkinfnorm** (8.20)

### 8.73 integer

Name: **integer**

keyword representing a machine integer type

Usage:

**integer** : type type

Description:

- **integer** represents the machine integer type for declarations of external procedures **externalproc**.  
Remark that in contrast to other indicators, type indicators like **integer** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **boolean** (8.17), **constant** (8.23), **function** (8.61), **list of** (8.80), **range** (8.120), **string** (8.147)

### 8.74 integral

Name: **integral**

computes an interval bounding the integral of a function on an interval.

Usage:

**integral**( $f, I$ ) : (function, range)  $\rightarrow$  range

Parameters:

- $f$  is a function.
- $I$  is an interval.

Description:

- **integral**( $f, I$ ) returns an interval  $J$  such that the exact value of the integral of  $f$  on  $I$  lies in  $J$ .
- This command is safe but very inefficient. Use **dirtyintegral** if you just want an approximate value.
- The result of this command depends on the global variable **diam**. The method used is the following:  $I$  is cut into intervals of length not greater than  $\delta \cdot |I|$  where  $\delta$  is the value of global variable **diam**. On each small interval  $J$ , an evaluation of  $f$  by interval is performed. The result is multiplied by the length of  $J$ . Finally all values are summed.

Example 1:

```

> sin(10);
-0.54402111088936981340474766185137728168364301291621
> integral(cos(x), [0;10]);
[-0.54710197983579690224097637163525943075698599257332;-0.5409401513001318384815
0540881373370744053741191728]
> diam=1e-5!;
> integral(cos(x), [0;10]);
[-0.54432915685955427101857780295936956775293876382777;-0.5437130640124996950803
9644221927489010425803173555]

```

See also: **diam** (8.30), **dirtyintegral** (8.34)

## 8.75 isbound

Name: **isbound**

indicates whether a variable is bound or not.

Usage:

**isbound**(*ident*) : boolean

Parameters:

- *ident* is a name.

Description:

- **isbound**(*ident*) returns a boolean value indicating whether the name *ident* is used or not to represent a variable. It returns true when *ident* is the name used to represent the global variable or if the name is currently used to refer to a (possibly local) variable.
- When a variable is defined in a block and has not been defined outside, **isbound** returns true when called inside the block, and false outside. Note that **isbound** returns true as soon as a variable has been declared with **var**, even if no value is actually stored in it.
- If *ident1* is bound to a variable and if *ident2* refers to the global variable, the command **re-name**(*ident2*, *ident1*) hides the value of *ident1* which becomes the global variable. However, if the global variable is again renamed, *ident1* gets its value back. In this case, **isbound**(*ident1*) returns true. If *ident1* was not bound before, **isbound**(*ident1*) returns false after that *ident1* has been renamed.

Example 1:

```

> isbound(x);
false
> isbound(f);
false
> isbound(g);
false
> f=sin(x);
> isbound(x);
true
> isbound(f);
true
> isbound(g);
false

```

Example 2:

```

> isbound(a);
false
> { var a; isbound(a); };
true
> isbound(a);
false

```

Example 3:

```

> f=sin(x);
> isbound(x);
true
> rename(x,y);
> isbound(x);
false

```

Example 4:

```

> x=1;
> f=sin(y);
> rename(y,x);
> f;
sin(x)
> x;
x
> isbound(x);
true
> rename(x,y);
> isbound(x);
true
> x;
1

```

See also: **rename** (8.128)

## 8.76 isevaluable

Name: **isevaluable**

tests whether a function can be evaluated at a point

Usage:

$$\mathbf{isevaluable}(\mathit{function}, \mathit{constant}) : (\mathit{function}, \mathit{constant}) \rightarrow \mathbf{boolean}$$

Parameters:

- *function* represents a function
- *constant* represents a constant point

Description:

- **isevaluable** applied to function *function* and a constant *constant* returns a boolean indicating whether or not a subsequent call to **evaluate** on the same function *function* and constant *constant* will produce a numerical result or NaN. This means **isevaluable** returns false iff **evaluate** will return NaN.

Example 1:

```
> isevaluable(sin(pi * 1/x), 0.75);
true
> print(evaluate(sin(pi * 1/x), 0.75));
-0.86602540378443864676372317075293618347140262690518
```

Example 2:

```
> isevaluable(sin(pi * 1/x), 0.5);
true
> print(evaluate(sin(pi * 1/x), 0.5));
[-1.72986452514381269516508615031098129542836767991679e-12715;7.5941198201187963
145069564314525661706039084390067e-12716]
```

Example 3:

```
> isevaluable(sin(pi * 1/x), 0);
false
> print(evaluate(sin(pi * 1/x), 0));
[@NaN@;@NaN@]
```

See also: **evaluate** (8.45)

## 8.77 <=

Name: <=

less-than-or-equal-to operator

Usage:

$$expr1 \leq expr2 : (\text{constant}, \text{constant}) \rightarrow \text{boolean}$$

Parameters:

- *expr1* and *expr2* represent constant expressions

Description:

- The operator <= evaluates to true iff its operands *expr1* and *expr2* evaluate to two floating-point numbers  $a_1$  respectively  $a_2$  with the global precision **prec** and  $a_1$  is less than or equal to  $a_2$ . The user should be aware of the fact that because of floating-point evaluation, the operator <= is not exactly the same as the mathematical operation *less-than-or-equal-to*.

Example 1:

```
> 5 <= 4;
false
> 5 <= 5;
true
> 5 <= 6;
true
> exp(2) <= exp(1);
false
> log(1) <= exp(2);
true
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16385.1 <= 16384.1;
true
```

See also: == (8.41), != (8.92), >= (8.62), > (8.63), < (8.85), ! (8.94), && (8.6), || (8.98), **prec** (8.109)

## 8.78 length

Name: **length**

computes the length of a list or string.

Usage:

$$\begin{aligned}\text{length}(L) &: \text{list} \rightarrow \text{integer} \\ \text{length}(s) &: \text{string} \rightarrow \text{integer}\end{aligned}$$

Parameters:

- $L$  is a list.
- $s$  is a string.

Description:

- **length** returns the length of a list or a string, e.g. the number of elements or letters.
- The empty list or string have length 0. If  $L$  is an end-elliptic list, **length** returns +Inf.

Example 1:

```
> length("Hello World!");  
12
```

Example 2:

```
> length([1,...,5]);  
5
```

Example 3:

```
> length([| |]);  
0
```

Example 4:

```
> length([1,2...|]);  
@Inf@
```

## 8.79 library

Name: **library**

binds an external mathematical function to a variable in **Sollya**

Usage:

$$\text{library}(\text{path}) : \text{string} \rightarrow \text{function}$$

Description:

- The command **library** lets you extend the set of mathematical functions known to **Sollya**. By default, **Sollya** knows the most common mathematical functions such as **exp**, **sin**, **erf**, etc. Within **Sollya**, these functions may be composed. This way, **Sollya** should satisfy the needs of a lot of users. However, for particular applications, one may want to manipulate other functions such as Bessel functions, or functions defined by an integral or even a particular solution of an ODE.
- **library** makes it possible to let **Sollya** know about new functions. In order to let it know, you have to provide an implementation of the function you are interested in. This implementation is a C file containing a function of the form:

```
int my_ident(mpfi_t result, mpfi_t op, int n)
```

The semantic of this function is the following: it is an implementation of the function and its derivatives in interval arithmetic. `my_ident(result, I, n)` shall store in `result` an enclosure of the image set of the  $n$ -th derivative of the function  $f$  over  $I$ :  $f^{(n)}(I) \subseteq \text{result}$ .

- The integer value returned by the function implementation currently has no meaning.
- You do not need to provide a working implementation for any  $n$ . Most functions of **Sollya** requires a relevant implementation only for  $f$ ,  $f'$  and  $f''$ . For higher derivatives, its is not so critical and the implementation may just store  $[-\infty, +\infty]$  in result whenever  $n > 2$ .
- Note that you should respect somehow MPFI standards in your implementation: `result` has its own precision and you should perform the intermediate computations so that `result` is as tight as possible.
- You can include `sollya.h` in your implementation and use library fonctionnalités of **Sollya** for your implementation. However, this requires to have compiled **Sollya** with `-fPIC` in order to make the **Sollya** executable code position independent and to use a system on with programs, using `dlopen` to open dynamic routines can dynamically open themselves.
- To bind your function into **Sollya**, you must use the same identifier as the function name used in your implementation file (`my_ident` in the previous example). Once the function code has been bound to an identifier, you can use a simple assignment to assign the bound identifier to yet another identifier. This way, you may use convenient names inside **Sollya** even if your implementation environment requires you to use a less convenient name.

Example 1:

```
> bashexecute("gcc -fPIC -Wall -c libraryexample.c -I$HOME/.local/include");
> bashexecute("gcc -shared -o libraryexample libraryexample.o -lgmp -lmpfr");
> myownlog = library("./libraryexample");
> evaluate(log(x), 2);
0.69314718055994530941723212145817656807550013436024
> evaluate(myownlog(x), 2);
0.69314718055994530941723212145817656807550013436024
```

See also: **bashexecute** (8.15), **externalproc** (8.52), **externalplot** (8.51)

## 8.80 list of

Name: **list of**

keyword used in combination with a type keyword

Description:

- **list of** is used in combination with one of the following keywords for indicating lists of the respective type in declarations of external procedures using **externalproc**: **boolean**, **constant**, **function**, **integer**, **range** and **string**.

See also: **externalproc** (8.52), **boolean** (8.17), **constant** (8.23), **function** (8.61), **integer** (8.73), **range** (8.120), **string** (8.147)

## 8.81 log

Name: **log**

natural logarithm.

Description:

- **log** is the natural logarithm defined as the inverse of the exponential function:  $\log(y)$  is the unique real number  $x$  such that  $\exp(x) = y$ .
- It is defined only for  $y \in [0; +\infty]$ .

See also: **exp** (8.47), **log2** (8.84), **log10** (8.82)

## 8.82 **log10**

Name: **log10**

decimal logarithm.

Description:

- **log10** is the decimal logarithm defined by:  $\log_{10}(x) = \log(x)/\log(10)$ .
- It is defined only for  $x \in [0; +\infty]$ .

See also: **log** (8.81), **log2** (8.84)

## 8.83 **log1p**

Name: **log1p**

translated logarithm.

Description:

- **log1p** is the function defined by  $\log_{1p}(x) = \log(1 + x)$ .
- It is defined only for  $x \in [-1; +\infty]$ .

See also: **log** (8.81)

## 8.84 **log2**

Name: **log2**

binary logarithm.

Description:

- **log2** is the binary logarithm defined by:  $\log_2(x) = \log(x)/\log(2)$ .
- It is defined only for  $x \in [0; +\infty]$ .

See also: **log** (8.81), **log10** (8.82)

## 8.85 **<**

Name: **<**

less-than operator

Usage:

$$expr1 < expr2 : (\text{constant}, \text{constant}) \rightarrow \text{boolean}$$

Parameters:

- *expr1* and *expr2* represent constant expressions

Description:

- The operator **<** evaluates to true iff its operands *expr1* and *expr2* evaluate to two floating-point numbers  $a_1$  respectively  $a_2$  with the global precision **prec** and  $a_1$  is less than  $a_2$ . The user should be aware of the fact that because of floating-point evaluation, the operator **<** is not exactly the same as the mathematical operation *less-than*.

Example 1:



```

> 5 < 4;
false
> 5 < 5;
false
> 5 < 6;
true
> exp(2) < exp(1);
false
> log(1) < exp(2);
true

```

Example 2:

```

> prec = 12;
The precision has been set to 12 bits.
> 16384.1 < 16385.1;
false

```

See also: `==` (8.41), `!=` (8.92), `>=` (8.62), `>` (8.63), `<=` (8.77), `!` (8.94), `&&` (8.6), `||` (8.98), `prec` (8.109)

## 8.86 mantissa

Name: **`mantissa`**

returns the integer mantissa of a number.

Usage:

**`mantissa( $x$ )`** : constant  $\rightarrow$  integer

Parameters:

- $x$  is a dyadic number.

Description:

- **`mantissa( $x$ )`** is by definition  $x$  if  $x$  equals 0, NaN, or Inf.
- If  $x$  is not zero, it can be uniquely written as  $x = m \cdot 2^e$  where  $m$  is an odd integer and  $e$  is an integer. **`mantissa( $x$ )`** returns  $m$ .

Example 1:

```

> a=round(Pi,20,RN);
> e=exponent(a);
> m=mantissa(a);
> m;
411775
> a-m*2^e;
0

```

See also: **`exponent`** (8.50), **`precision`** (8.110)

## 8.87 mid

Name: **`mid`**

gives the middle of an interval.

Usage:

**`mid( $I$ )`** : range  $\rightarrow$  constant  
**`mid( $x$ )`** : constant  $\rightarrow$  constant

Parameters:

- $I$  is an interval.
- $x$  is a real number.

Description:

- Returns the middle of the interval  $I$ . If the middle is not exactly representable at the current precision, the value is returned as an unevaluated expression.
- When called on a real number  $x$ , **mid** considers it as an interval formed of a single point:  $[x, x]$ . In other words, **mid** behaves like the identity.

Example 1:

```
> mid([1;3]);  
2  
> mid(17);  
17
```

See also: **inf** (8.71), **sup** (8.150)

## 8.88 midpointmode

Name: **midpointmode**

global variable controlling the way intervals are displayed.

Description:

- **midpointmode** is a global variable. When its value is **off**, intervals are displayed as usual (in the form  $[a; b]$ ). When its value is **on**, and if  $a$  and  $b$  have the same first significant digits, the interval is displayed in a way that lets one immediately see the common digits of the two bounds.
- This mode is supported only with **display** set to **decimal**. In other modes of display, **midpointmode** value is simply ignored.

Example 1:

```
> a = round(Pi,30,RD);  
> b = round(Pi,30,RU);  
> d = [a,b];  
> d;  
[3.1415926516056060791015625;3.1415926553308963775634765625]  
> midpointmode=on!;  
> d;  
0.314159265~1/6~e1
```

See also: **on** (8.97), **off** (8.96), **roundingwarnings** (8.136)

## 8.89 —

Name: —

subtraction function

Usage:

$$function1 - function2 : (function, function) \rightarrow function$$

Parameters:

- $function1$  and  $function2$  represent functions

Description:

- $-$  represents the subtraction (function) on reals. The expression  $function1 - function2$  stands for the function composed of the subtraction function and the two functions  $function1$  and  $function2$ , where  $function1$  is the subtrahend and  $function2$  the subtrahend.

Example 1:

```
> 5 - 2;  
3
```

Example 2:

```
> x - 2;  
-2 + x
```

Example 3:

```
> x - x;  
0
```

Example 4:

```
> diff(sin(x) - exp(x));  
cos(x) - exp(x)
```

See also:  $+$  (8.103),  $*$  (8.90),  $/$  (8.36),  $^$  (8.107)

## 8.90 $*$

Name:  $*$

multiplication function

Usage:

$function1 * function2 : (function, function) \rightarrow function$

Parameters:

- $function1$  and  $function2$  represent functions

Description:

- $*$  represents the multiplication (function) on reals. The expression  $function1 * function2$  stands for the function composed of the multiplication function and the two functions  $function1$  and  $function2$ .

Example 1:

```
> 5 * 2;  
10
```

Example 2:

```
> x * 2;  
x * 2
```

Example 3:

```
> x * x;  
x^2
```

Example 4:

```
> diff(sin(x) * exp(x));  
sin(x) * exp(x) + exp(x) * cos(x)
```

See also:  $+$  (8.103),  $-$  (8.89),  $/$  (8.36),  $^$  (8.107)

## 8.91 nearestint

Name: **nearestint**

the function mapping the reals to the integers nearest to them.

Description:

- **nearestint** is defined as usual: **nearestint**( $x$ ) is the integer nearest to  $x$ , with the special rule that the even integer is chosen if there exist two integers equally near to  $x$ .
- It is defined for every real number  $x$ .

See also: **ceil** (8.19), **floor** (8.58)

## 8.92 !=

Name: **!=**

negated equality test operator

Usage:

$$expr1 \text{ != } expr2 : (\text{any type}, \text{any type}) \rightarrow \text{boolean}$$

Parameters:

- $expr1$  and  $expr2$  represent expressions

Description:

- The operator **!=** evaluates to true iff its operands  $expr1$  and  $expr2$  are syntactically unequal and both different from **error** or constant expressions that are not constants and that evaluate to two different floating-point number with the global precision **prec**. The user should be aware of the fact that because of floating-point evaluation, the operator **!=** is not exactly the same as the negation of the mathematical equality.

Note that the expressions  $!(expr1 \text{ != } expr2)$  and  $expr1 == expr2$  do not evaluate to the same boolean value. See **error** for details.

Example 1:

```
> "Hello" != "Hello";
false
> "Hello" != "Salut";
true
> "Hello" != 5;
true
> 5 + x != 5 + x;
false
```

Example 2:

```
> 1 != exp(0);
false
> asin(1) * 2 != pi;
false
> exp(5) != log(4);
true
```

Example 3:

```
> sin(pi/6) != 1/2 * sqrt(3);
true
```

Example 4:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384.1 != 16385.1;
false
```

Example 5:

```
> error != error;
false
```

See also: `==` (8.41), `>` (8.63), `>=` (8.62), `<=` (8.77), `<` (8.85), `!` (8.94), `&&` (8.6), `||` (8.98), **error** (8.44), **prec** (8.109)

### 8.93 **nop**

Name: **nop**  
no operation  
Usage:

**nop** : void  $\rightarrow$  void  
**nop()** : void  $\rightarrow$  void  
**nop(*n*)** : integer  $\rightarrow$  void

Description:

- The command **nop** does nothing. This means it is an explicit parse element in the **Sollya** language that finally does not produce any result or side-effect.
- The command **nop** may take an optional positive integer argument *n*. The argument controls how much (useless) integer additions **Sollya** performs while doing nothing. With this behaviour, **nop** can be used for calibration of timing tests.
- The keyword **nop** is implicit in some procedure definitions. Procedures without imperative body get parsed as if they had an imperative body containing one **nop** statement.

Example 1:

```
> nop;
```

Example 2:

```
> nop(100);
```

Example 3:

```
> succ = proc(n) { return n + 1; };
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
> succ(5);
6
```

See also: **proc** (8.117)

## 8.94 !

Name: !

boolean NOT operator

Usage:

$! \text{ expr} : \text{boolean} \rightarrow \text{boolean}$

Parameters:

- *expr* represents a boolean expression

Description:

- ! evaluates to the boolean NOT of the boolean expression *expr*. ! *expr* evaluates to true iff *expr* does not evaluate to true.

Example 1:

```
> ! false;  
true
```

Example 2:

```
> ! (1 == exp(0));  
false
```

See also: && (8.6), || (8.98)

## 8.95 numerator

Name: **numerator**

gives the numerator of an expression

Usage:

$\text{numerator}(\text{expr}) : \text{function} \rightarrow \text{function}$

Parameters:

- *expr* represents an expression

Description:

- If *expr* represents a fraction *expr1* / *expr2*, **numerator**(*expr*) returns the numerator of this fraction, i.e. *expr1*.

If *expr* represents something else, **numerator**(*expr*) returns the expression itself, i.e. *expr*.

Note that for all expressions *expr*, **numerator**(*expr*) / **denominator**(*expr*) is equal to *expr*.

Example 1:

```
> numerator(5/3);  
5
```

Example 2:

```
> numerator(exp(x));  
exp(x)
```

Example 3:

```

> a = 5/3;
> b = numerator(a)/denominator(a);
> print(a);
5 / 3
> print(b);
5 / 3

```

Example 4:

```

> a = exp(x/3);
> b = numerator(a)/denominator(a);
> print(a);
exp(x / 3)
> print(b);
exp(x / 3)

```

See also: **denominator** (8.29)

## 8.96 off

Name: **off**

special value for certain global variables.

Description:

- **off** is a special value used to deactivate certain fonctionnalities of Sollya (namely **canonical**, **timing**, **fullparentheses**, **midpointmode**).
- As any value it can be affected to a variable and stored in lists.

Example 1:

```

> canonical=on;
Canonical automatic printing output has been activated.
> p=1+x+x^2;
> mode=off;
> p;
1 + x + x^2
> canonical=mode;
Canonical automatic printing output has been deactivated.
> p;
1 + x * (1 + x)

```

See also: **on** (8.97), **canonical** (8.18), **timing** (8.157), **fullparentheses** (8.60), **midpointmode** (8.88)

## 8.97 on

Name: **on**

special value for certain global variables.

Description:

- **on** is a special value used to activate certain fonctionnalities of Sollya (namely **canonical**, **timing**, **fullparentheses**, **midpointmode**).
- As any value it can be affected to a variable and stored in lists.

Example 1:

```

> p=1+x+x^2;
> mode=on;
> p;
1 + x * (1 + x)
> canonical=mode;
Canonical automatic printing output has been activated.
> p;
1 + x + x^2

```

See also: **off** (8.96), **canonical** (8.18), **timing** (8.157), **fullparentheses** (8.60), **midpointmode** (8.88)

## 8.98 ||

Name: ||

boolean OR operator

Usage:

$$expr1 \parallel expr2 : (\text{boolean}, \text{boolean}) \rightarrow \text{boolean}$$

Parameters:

- *expr1* and *expr2* represent boolean expressions

Description:

- || evaluates to the boolean OR of the two boolean expressions *expr1* and *expr2*. || evaluates to true iff at least one of *expr1* or *expr2* evaluates to true.

Example 1:

```

> false || false;
false

```

Example 2:

```

> (1 == exp(0)) || (0 == log(1));
true

```

See also: && (8.6), ! (8.94)

## 8.99 parse

Name: **parse**

parses an expression contained in a string

Usage:

$$\text{parse}(\text{string}) : \text{string} \rightarrow \text{function} \mid \text{error}$$

Parameters:

- *string* represents a character sequence

Description:

- **parse**(*string*) parses the character sequence *string* containing an expression built on constants and base functions.  
If the character sequence does not contain a well-defined expression, a warning is displayed indicating a syntax error and **parse** returns a **error** of type **error**.
- The character sequence to be parsed by **parse** may contain commands that return expressions, including **parse** itself. Those commands get executed after the string has been parsed. **parse**(*string*) will return the expression computed by the commands contained in the character sequence *string*.



Example 1:

```
> parse("exp(x)");
exp(x)
```

Example 2:

```
> text = "remez(exp(x),5,[-1;1])";
> print("The string", text, "gives", parse(text));
The string remez(exp(x),5,[-1;1]) gives 1.00004475029055070643077052482053398765
426158966754 + x * (1.00003834652983970735244541124504033817544233075356 + x * (
0.49919698262882986492168824494240374771969012861297 + x * (0.166424656075155194
415920597322727380932279602909199 + x * (4.3793696387328047027125756620718349665
9575464236489e-2 + x * 8.7381910388065551140158420278330960479960476713376e-3)))
)
```

Example 3:

```
> verbosity = 1!;
> parse("5 + * 3");
Warning: syntax error, unexpected MULTOKEN. Will try to continue parsing (expect
ing ";"). May leak memory.
Warning: the string "5 + * 3" could not be parsed by the miniparser.
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
error
```

See also: **execute** (8.46), **readfile** (8.124), **print** (8.112)

## 8.100 perturb

Name: **perturb**

indicates random perturbation of sampling points for **externalplot**

Usage:

**perturb** : perturb

Description:

- The use of **perturb** in the command **externalplot** enables the addition of some random noise around each sampling point in **externalplot**.

See **externalplot** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l
mpfr");
> externalplot("./externalplotexample",relative,exp(x),[-1/2;1/2],12,perturb);
```

See also: **externalplot** (8.51), **absolute** (8.2), **relative** (8.126), **bashexecute** (8.15)

## 8.101 pi

Name: **pi**

the constant  $\pi$ .

Description:

- **pi** is the constant  $\pi$ , defined as half the period of sine and cosine.
- In Sollya, **pi** is considered a 0-ary function. This way, the constant is not evaluated at the time of its definition but at the time of its use. For instance, when you define a constant or a function relating to  $\pi$ , the current precision at the time of the definition does not matter. What is important is the current precision when you evaluate the function or the constant value.
- Remark that when you define an interval, the bounds are first evaluated and then the interval is defined. In this case, **pi** will be evaluated as any other constant value at the definition time of the interval, thus using the current precision at this time.

Example 1:

```
> verbosity=1!; prec=12!;
> a = 2*pi;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
6.283
> prec=20!;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
6.283187
```

Example 2:

```
> display=binary;
Display mode is binary numbers.
> prec=12!;
> d = [pi; 5];
> d;
[1.1001001_2 * 2^(1); 1.01_2 * 2^(2)]
> prec=20!;
> d;
[1.1001001_2 * 2^(1); 1.01_2 * 2^(2)]
```

See also: **cos** (8.24), **sin** (8.142)

## 8.102 plot

Name: **plot**

plots one or several functions

Usage:

```
plot(f1, ... ,fn, I) : (function, ... ,function, range) → void
plot(f1, ... ,fn, I, file, name) : (function, ... ,function, range, file, string) → void
plot(f1, ... ,fn, I, postscript, name) : (function, ... ,function, range, postscript, string) → void
plot(f1, ... ,fn, I, postscriptfile, name) : (function, ... ,function, range, postscriptfile, string) → void
plot(L, I) : (list, range) → void
plot(L, I, file, name) : (list, range, file, string) → void
plot(L, I, postscript, name) : (list, range, postscript, string) → void
plot(L, I, postscriptfile, name) : (list, range, postscriptfile, string) → void
```

Parameters:

- $f1, \dots, fn$  are functions to be plotted.
- $L$  is a list of functions to be plotted.

- $I$  is the interval where the functions have to be plotted.
- *name* is a string representing the name of a file.

Description:

- This command plots one or several functions  $f_1, \dots, f_n$  on an interval  $I$ . Functions can be either given as parameters of **plot** or as a list  $L$  which elements are functions. The functions are drawn on the same plot with different colors.
- If  $L$  contains an element that is not a function (or a constant), an error occurs.
- **plot** relies on the value of global variable **points**. Let  $n$  be the value of this variable. The algorithm is the following: each function is evaluated at  $n$  evenly distributed points in  $I$ . At each point, the computed value is a faithful rounding of the exact value with a sufficiently high precision. Each point is finally plotted. This should avoid numerical artefacts such as critical cancellations.
- You can save the function plot either as a data file or as a postscript file.
- If you use argument **file** with a string *name*, **Sollya** will save a data file called *name.dat* and a gnuplot directives file called *name.p*. Invoking gnuplot on *name.p* will plot the data stored in *name.dat*.
- If you use argument **postscript** with a string *name*, **Sollya** will save a postscript file called *name.eps* representing your plot.
- If you use argument **postscriptfile** with a string *name*, **Sollya** will produce the corresponding *name.dat*, *name.p* and *name.eps*.
- This command uses gnuplot to produce the final plot. If your terminal is not graphic (typically if you use **Sollya** through ssh without -X) gnuplot should be able to detect that and produce an ASCII-art version on the standard output. If it is not the case, you can either store the plot in a postscript file to view it locally, or use **asciipLOT** command.
- If every function is constant, **plot** will not plot them but just display their value.
- If the interval is reduced to a single point, **plot** will just display the value of the functions at this point.

Example 1:

```
> plot(sin(x),0,cos(x),[-Pi,Pi]);
```

Example 2:

```
> plot(sin(x),0,cos(x),[-Pi,Pi],postscriptfile,"plotSinCos");
```

Example 3:

```
> plot(exp(0), sin(1), [0;1]);
1
0.84147098480789650665250232163029899962256306079837
```

Example 4:

```
> plot(sin(x), cos(x), [1;1]);
0.84147098480789650665250232163029899962256306079837
0.54030230586813971740093660744297660373231042061792
```

See also: **externalplot** (8.51), **asciipLOT** (8.9), **file** (8.54), **postscript** (8.105), **postscriptfile** (8.106), **points** (8.104)

### 8.103 +

Name: +  
addition function  
Usage:

$$function1 + function2 : (function, function) \rightarrow function$$

Parameters:

- *function1* and *function2* represent functions

Description:

- + represents the addition (function) on reals. The expression *function1* + *function2* stands for the function composed of the addition function and the two functions *function1* and *function2*.

Example 1:

```
> 1 + 2;  
3
```

Example 2:

```
> x + 2;  
2 + x
```

Example 3:

```
> x + x;  
x * 2
```

Example 4:

```
> diff(sin(x) + exp(x));  
cos(x) + exp(x)
```

See also: − (8.89), \* (8.90), / (8.36), ^ (8.107)

### 8.104 points

Name: **points**  
controls the number of points chosen by **Sollya** in certain commands.  
Usage:

$$\begin{aligned} \mathbf{points} = n : \text{integer} &\rightarrow \text{void} \\ \mathbf{points} = n ! : \text{integer} &\rightarrow \text{void} \\ \mathbf{points} : \text{constant} \end{aligned}$$

Parameters:

- *n* represents the number of points

Description:

- **points** is a global variable. Its value represents the number of points used in numerical algorithms of **Sollya** (namely **dirtyinfnorm**, **dirtyintegral**, **dirtyfindzeros**, **plot**).

Example 1:

```

> f=x^2*sin(1/x);
> points=10;
The number of points has been set to 10.
> dirtyfindzeros(f, [0;1]);
[|0, 0.318309886183790671537767526745028724068919291480918|]
> points=100;
The number of points has been set to 100.
> dirtyfindzeros(f, [0;1]);
[|0, 2.4485375860291590118289809749617594159147637806224e-2, 3.97887357729738339
422209408431285905086149114351147e-2, 4.5472840883398667362538218106432674866988
4702115589e-2, 5.3051647697298445256294587790838120678153215246819e-2, 6.3661977
236758134307553505349005744813783858296183e-2, 7.9577471545947667884441881686257
181017229822870229e-2, 0.106103295394596890512589175581676241356306430493638, 0.
159154943091895335768883763372514362034459645740459, 0.3183098861837906715377675
26745028724068919291480918|]

```

See also: **dirtyinfnorm** (8.33), **dirtyintegral** (8.34), **dirtyfindzeros** (8.32), **plot** (8.102)

## 8.105 postscript

Name: **postscript**

special value for commands **plot** and **externalplot**

Description:

- **postscript** is a special value used in commands **plot** and **externalplot** to save the result of the command in a postscript file.
- As any value it can be affected to a variable and stored in lists.

Example 1:

```

> savemode=postscript;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);

```

See also: **externalplot** (8.51), **plot** (8.102), **file** (8.54), **postscriptfile** (8.106)

## 8.106 postscriptfile

Name: **postscriptfile**

special value for commands **plot** and **externalplot**

Description:

- **postscriptfile** is a special value used in commands **plot** and **externalplot** to save the result of the command in a data file and a postscript file.
- As any value it can be affected to a variable and stored in lists.

Example 1:

```

> savemode=postscriptfile;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);

```

See also: **externalplot** (8.51), **plot** (8.102), **file** (8.54), **postscript** (8.105)

## 8.107 ^

Name: ^  
power function  
Usage:

$$function1 \wedge function2 : (function, function) \rightarrow function$$

Parameters:

- *function1* and *function2* represent functions

Description:

- ^ represents the power (function) on reals. The expression *function1* ^ *function2* stands for the function composed of the power function and the two functions *function1* and *function2*, where *function1* is the base and *function2* the exponent. If *function2* is a constant integer, ^ is defined on negative values of *function1*. Otherwise ^ is defined as  $e^{y \cdot \ln x}$ .
- Note that whenever several ^ are composed, the priority goes to the last ^. This corresponds to the natural way of thinking when a tower of powers is written on a paper. Thus,  $2^{3^5}$  is read as  $2^{3^5}$  and is interpreted as  $2^{(3^5)}$ .

Example 1:

```
> 5 ^ 2;  
25
```

Example 2:

```
> x ^ 2;  
x^2
```

Example 3:

```
> 3 ^ (-5);  
4.1152263374485596707818930041152263374485596707818e-3
```

Example 4:

```
> (-3) ^ (-2.5);  
@NaN@
```

Example 5:

```
> diff(sin(x) ^ exp(x));  
sin(x)^exp(x) * ((cos(x) * exp(x)) / sin(x) + exp(x) * log(sin(x)))
```

Example 6:

```
> 2^3^5;  
1.4134776518227074636666380005943348126619871175005e73  
> (2^3)^5;  
32768  
> 2^(3^5);  
1.4134776518227074636666380005943348126619871175005e73
```

See also: + (8.103), - (8.89), \* (8.90), / (8.36)

## 8.108 powers

Name: **powers**

special value for global state **display**

Description:

- **powers** is a special value used for the global state **display**. If the global state **display** is equal to **powers**, all data will be output in dyadic notation with numbers displayed in a Maple and PARI/GP compatible format.

As any value it can be affected to a variable and stored in lists.

See also: **decimal** (8.26), **dyadic** (8.40), **hexadecimal** (8.66), **binary** (8.16)

## 8.109 prec

Name: **prec**

controls the precision used in numerical computations.

Description:

- **prec** is a global variable. Its value represents the precision of the floating-point format used in numerical computations.
- Many commands try to adapt their working precision in order to have approximately  $n$  correct bits in output, where  $n$  is the value of **prec**.

Example 1:

```
> display=binary!;  
> prec=50;  
The precision has been set to 50 bits.  
> dirtyinfnorm(exp(x),[1;2]);  
1.110110001110011001001011100011010100110111011011_2 * 2^(2)  
> prec=100;  
The precision has been set to 100 bits.  
> dirtyinfnorm(exp(x),[1;2]);  
1.110110001110011001001011100011010100110111011010110111001100001100111010001110  
11101000100000011011_2 * 2^(2)
```

## 8.110 precision

Name: **precision**

returns the precision necessary to represent a number.

Usage:

**precision**( $x$ ) : constant  $\rightarrow$  integer

Parameters:

- $x$  is a dyadic number.

Description:

- **precision**( $x$ ) is by definition  $|x|$  if  $x$  equals 0, NaN, or Inf.
- If  $x$  is not zero, it can be uniquely written as  $x = m \cdot 2^e$  where  $m$  is an odd integer and  $e$  is an integer. **precision**( $x$ ) returns the number of bits necessary to write  $m$  in binary (i.e.  $\lceil \log_2(m) \rceil$ ).

Example 1:

```

> a=round(Pi,20,RN);
> precision(a);
19
> m=mantissa(a);
> ceil(log2(m));
19

```

See also: **`mantissa`** (8.86), **`exponent`** (8.50)

## 8.111 `..`:

Name: `..`:

add an element at the beginning of a list.

Usage:

$$x::L : (\text{any type, list}) \rightarrow \text{list}$$

Parameters:

- $x$  is an object of any type.
- $L$  is a list (possibly empty).

Description:

- `..` adds the element  $x$  at the beginning of the list  $L$ .
- Note that since  $x$  may be of any type, it can be in particular a list.

Example 1:

```

> 1.: [|2,3,4|];
[|1, 2, 3, 4|]

```

Example 2:

```

> [|1,2,3|].. [|4,5,6|];
[| [|1, 2, 3|], 4, 5, 6|]

```

Example 3:

```

> 1.: [| |];
[|1|]

```

See also: `..` (8.7), `@` (8.22)

## 8.112 `print`

Name: **`print`**

prints an expression

Usage:

$$\begin{aligned} \mathbf{print}(expr1, \dots, exprn) &: (\text{any type}, \dots, \text{any type}) \rightarrow \text{void} \\ \mathbf{print}(expr1, \dots, exprn) > filename &: (\text{any type}, \dots, \text{any type}, \text{string}) \rightarrow \text{void} \\ \mathbf{print}(expr1, \dots, exprn) >> filename &: (\text{any type}, \dots, \text{any type}, \text{string}) \rightarrow \text{void} \end{aligned}$$

Parameters:

- $expr$  represents an expression
- $filename$  represents a character sequence indicating a file name



Description:

- **print**(*expr1*,...,*exprn*) prints the expressions *expr1* through *exprn* separated by spaces and followed by a newline.

If a second argument *filename* is given after a single ">", the displaying is not output on the standard output of **Sollya** but if in the file *filename* that get newly created or overwritten. If a double ">>" is given, the output will be appended to the file *filename*.

The global variables **display**, **midpointmode** and **fullparentheses** have some influence on the formatting of the output (see **display**, **midpointmode** and **fullparentheses**).

Remark that if one of the expressions *expri* given in argument is of type **string**, the character sequence *expri* evaluates to is displayed. However, if *expri* is of type **list** and this list contains a variable of type **string**, the expression for the list is displayed, i.e. all character sequences get displayed surrounded by double quotes ("). Nevertheless, escape sequences used upon defining character sequences are interpreted immediately.

Example 1:

```
> print(x + 2 + exp(sin(x)));
x + 2 + exp(sin(x))
> print("Hello","world");
Hello world
> print("Hello","you", 4 + 3, "other persons.");
Hello you 7 other persons.
```

Example 2:

```
> print("Hello");
Hello
> print(["Hello"]);
["Hello"]
> s = "Hello";
> print(s,[s]);
Hello ["Hello"]
> t = "Hello\tyou";
> print(t,[t]);
Hello    you ["Hello    you"]
```

Example 3:

```
> print(x + 2 + exp(sin(x))) > "foo.sol";
> readfile("foo.sol");
x + 2 + exp(sin(x))
```

Example 4:

```
> print(x + 2 + exp(sin(x))) >> "foo.sol";
```

Example 5:

```

> display = decimal;
Display mode is decimal numbers.
> a = evaluate(sin(pi * x), 0.25);
> b = evaluate(sin(pi * x), [0.25; 0.25 + 1b-50]);
> print(a);
0.70710678118654752440084436210484903928483593768847
> display = binary;
Display mode is binary numbers.
> print(a);
1.011010100000100111100110011001111111001110111100110010010000100010110010111110
1100010011011001101110101010010101011110100111110001110101101111011000001011101
010001_2 * 2(-1)
> display = hexadecimal;
Display mode is hexadecimal numbers.
> print(a);
0xb.504f333f9de6484597d89b3754abe9f1d6f60ba88p-4
> display = dyadic;
Display mode is dyadic numbers.
> print(a);
33070006991101558613323983488220944360067107133265b-165
> display = powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
> print(a);
33070006991101558613323983488220944360067107133265 * 2(-165)
> display = decimal;
Display mode is decimal numbers.
> midpointmode = off;
Midpoint mode has been deactivated.
> print(b);
[0.70710678118654752440084436210484903928483593768844;0.707106781186549497437217
82517557347782646274417048]
> midpointmode = on;
Midpoint mode has been activated.
> print(b);
0.7071067811865~4/5~
> display = dyadic;
Display mode is dyadic numbers.
> print(b);
[2066875436943847413332748968013809022504194195829b-161;165350034955508254441962
37019385936414432675156571b-164]
> display = decimal;
Display mode is decimal numbers.
> autosimplify = off;
Automatic pure tree simplification has been deactivated.
> fullparentheses = off;
Full parentheses mode has been deactivated.
> print(x + x * ((x + 1) + 1));
x + x * (x + 1 + 1)
> fullparentheses = on;
Full parentheses mode has been activated.
> print(x + x * ((x + 1) + 1));
x + (x * ((x + 1) + 1))

```

See also: **write** (8.164), **printexpansion** (8.113), **printhea** (8.115), **printfloat** (8.114), **printxml** (8.116), **readfile** (8.124), **autosimplify** (8.14), **display** (8.35), **midpointmode** (8.88), **fullparentheses** (8.60), **evaluate** (8.45)

### 8.113 printexpansion

Name: **printexpansion**

prints a polynomial in Horner form with its coefficients written as a expansions of double precision numbers

Usage:

**printexpansion**(*polynomial*) : function  $\rightarrow$  void

Parameters:

- *polynomial* represents the polynomial to be printed

Description:

- The command **printexpansion** prints the polynomial *polynomial* in Horner form writing its coefficients as expansions of double precision numbers. The double precision numbers themselves are displayed in hexadecimal memory notation (see **printhexa**).

If some of the coefficients of the polynomial *polynomial* are not floating-point constants but constant expressions, they are evaluated to floating-point constants using the global precision **prec**. If a rounding occurs in this evaluation, a warning is displayed.

If the exponent range of double precision is not sufficient to display all the mantissa bits of a coefficient, the coefficient is displayed rounded and a warning is displayed.

If the argument *polynomial* does not a polynomial, nothing but a warning or a newline is displayed. Constants can be displayed using **printexpansion** since they are polynomials of degree 0.

Example 1:

```
> printexpansion(roundcoefficients(taylor(exp(x),5,0),[|DD...|]));
0x3ff0000000000000 + x * (0x3ff0000000000000 + x * (0x3fe0000000000000 + x * ((0
x3fc5555555555555 + 0x3c65555555555555) + x * ((0x3fa5555555555555 + 0x3c4555555
5555555) + x * (0x3f81111111111111 + 0x3c01111111111111))))))
```

Example 2:

```
> printexpansion(remez(exp(x),5,[-1;1]));
(0x3ff0002eec908ce9 + 0xbc7df99eb225af5b + 0xb8d55834b08b1f18) + x * ((0x3ff0002
835917719 + 0x3c6d82c073b25ebf + 0xb902cf062b54b7b6 + 0x35b0000000000000) + x *
((0x3fdff2d7e6a9c5e9 + 0xbc7b09a95b0d520f + 0xb915b639add55731 + 0x35a8000000000
000) + x * ((0x3fc54d67338ba09f + 0x3c4867596d0631cf + 0xb8ef0756bdb4af6e) + x *
((0x3fa66c209b825167 + 0x3c45ec5b6655b076 + 0xb8d8c125286400bc) + x * (0x3f81e5
5425e72ab4 + 0x3c263b25a1bf597b + 0xb8c843e0401dadd0 + 0x3540000000000000))))))
```

Example 3:

```
> verbosity = 1!;
> prec = 3500!;
> printexpansion(pi);
(0x400921fb54442d18 + 0x3ca1a62633145c07 + 0xb92f1976b7ed8fbc + 0x35c4cf98e80417
7d + 0x32631d89cd9128a5 + 0x2ec0f31c6809bbdf + 0x2b5519b3cd3a431b + 0x27e8158536
f92f8a + 0x246ba7f09ab6b6a9 + 0xa0eedd0dbd2544cf + 0x1d779fb1bd1310ba + 0x1a1a63
7ed6b0bfff6 + 0x96aa485fca40908e + 0x933e501295d98169 + 0x8fd160dbee83b4e0 + 0x8c
59b6d799ae131c + 0x08f6cf70801f2e28 + 0x05963bf0598da483 + 0x023871574e69a459 +
0x8000000005702db3 + 0x8000000000000000)
Warning: the expansion is not complete because of the limited exponent range of
double precision.
Warning: rounding occurred while printing.
```

See also: **printhexa** (8.115), **horner** (8.69), **print** (8.112), **prec** (8.109), **remez** (8.127), **taylor** (8.154), **roundcoefficients** (8.134)

### 8.114 printfloat

Name: **printfloat**

prints a constant value as a hexadecimal single precision number

Usage:

**printfloat**(*constant*) : constant  $\rightarrow$  void

Parameters:

- *constant* represents a constant

Description:

- Prints a constant value as a hexadecimal number on 8 hexadecimal digits. The hexadecimal number represents the integer equivalent to the 32 bit memory representation of the constant considered as a single precision number.

If the constant value does not hold on a single precision number, it is first rounded to the nearest single precision number before it is displayed. A warning is displayed in this case.

Example 1:

```
> printfloat(3);  
0x40400000
```

Example 2:

```
> prec=100!;  
> verbosity = 1!;  
> printfloat(exp(5));  
Warning: the given expression is not a constant but an expression to evaluate. A  
faithful evaluation will be used.  
Warning: rounding down occurred before printing a value as a simple.  
0x431469c5
```

See also: **printhexa** (8.115)

### 8.115 printhexa

Name: **printhexa**

prints a constant value as a hexadecimal double precision number

Usage:

**printhexa**(*constant*) : constant  $\rightarrow$  void

Parameters:

- *constant* represents a constant

Description:

- Prints a constant value as a hexadecimal number on 16 hexadecimal digits. The hexadecimal number represents the integer equivalent to the 64 bit memory representation of the constant considered as a double precision number.

If the constant value does not hold on a double precision number, it is first rounded to the nearest double precision number before displayed. A warning is displayed in this case.

Example 1:

```
> printhexa(3);  
0x4008000000000000
```

Example 2:

```
> prec=100!;  
> verbosity = 1!;  
> printhexa(exp(5));  
Warning: the given expression is not a constant but an expression to evaluate. A  
faithful evaluation will be used.  
Warning: rounding down occurred before printing a value as a double.  
0x40628d389970338f
```

See also: **printfloat** (8.114), **printexpansion** (8.113)

## 8.116 printxml

Name: **printxml**

prints an expression as an MathML-Content-Tree

Usage:

```
printxml(expr) : function → void  
printxml(expr) > filename : (function, string) → void  
printxml(expr) > > filename : (function, string) → void
```

Parameters:

- *expr* represents a functional expression
- *filename* represents a character sequence indicating a file name

Description:

- **printxml**(*expr*) prints the functional expression *expr* as a tree of MathML Content Definition Markups. This XML tree can be re-read in external tools or by usage of the **readxml** command. If a second argument *filename* is given after a single >, the MathML tree is not output on the standard output of **Sollya** but if in the file *filename* that get newly created or overwritten. If a double > > is given, the output will be appended to the file *filename*.

Example 1:

```

> printxml(x + 2 + exp(sin(x)));

<?xml version="1.0" encoding="UTF-8"?>
<!-- generated by sollya: http://sollya.gforge.inria.fr/ -->
<!-- syntax: printxml(...);   example: printxml(x^2-2*x+5); -->
<?xml-stylesheet type="text/xsl" href="http://perso.ens-lyon.fr/nicolas.jourdan/
mathmlc2p-web.xsl"?>
<?xml-stylesheet type="text/xsl" href="mathmlc2p-web.xsl"?>
<!-- This stylesheet allows direct web browsing of MathML-c XML files (http:// o
r file://) -->

<math xmlns="http://www.w3.org/1998/Math/MathML">
<semantics>
<annotation-xml encoding="MathML-Content">
<lambda>
<bvar><ci> x </ci></bvar>
<apply>
<apply>
<plus/>
<apply>
<plus/>
<ci> x </ci>
<cn type="integer" base="10"> 2 </cn>
</apply>
<apply>
<exp/>
<apply>
<sin/>
<ci> x </ci>
</apply>
</apply>
</apply>
</apply>
</lambda>
</annotation-xml>
<annotation encoding="sollya/text">(x + 1b1) + exp(sin(x))</annotation>
</semantics>
</math>

```

Example 2:

```

> printxml(x + 2 + exp(sin(x))) > "foo.xml";

```

Example 3:

```

> printxml(x + 2 + exp(sin(x))) >> "foo.xml";

```

See also: **readxml** (8.125), **print** (8.112), **write** (8.164)

## 8.117 **proc**

Name: **proc**

defines a Sollya procedure

Usage:

```

proc(formal parameter1, formal parameter2,..., formal parameter n) begin procedure body end : void
    → procedure
proc(formal parameter1, formal parameter2,..., formal parameter n) begin procedure body return
    expression; end : void → procedure
    proc(formal list parameter = ...) begin procedure body end : void → procedure
proc(formal list parameter = ...) begin procedure body return expression; end : void → procedure

```

Parameters:

- *formal parameter1, formal parameter2* through *formal parameter n* represent identifiers used as formal parameters
- *formal list parameter* represents an identifier used as a formal parameter for the list of an arbitrary number of parameters
- *procedure body* represents the imperative statements in the body of the procedure
- *expression* represents the expression **proc** shall evaluate to

Description:

- The **proc** keyword allows for defining procedures in the Sollya language. These procedures are common Sollya objects that can be applied to actual parameters after definition. Upon such an application, the Sollya interpreter applies the actual parameters to the formal parameters *formal parameter1* through *formal parameter n* (resp. builds up the list of arguments and applies it to the list *formal list parameter*) and executes the *procedure body*. The procedure applied to actual parameters evaluates then to the expression *expression* in the **return** statement after the *procedure body*, or to **void**, if no return statement is given (i.e. a **return void** statement is implicitly given).
- Sollya procedures defined by **proc** have no name. They can be bound to an identifier by assigning the procedure object a **proc** expression produces to an identifier. However, it is possible to use procedures without giving them any name. For instance, Sollya procedures, i.e. procedure objects, can be elements of lists. They can even be given as an argument to other internal Sollya procedures. See also **procedure** on this subject.
- Upon definition of a Sollya procedure using **proc**, no type check is performed. More precisely, the statements in *procedure body* are merely parsed but not interpreted upon procedure definition with **proc**. Type checks are performed once the procedure is applied to actual parameters or to **void**. At this time, if the procedure was defined using several different formal parameters *formal parameter 1* through *formal parameter n*, it is checked whether the number of actual parameters corresponds to the number of formal parameters. If the procedure was defined using the syntax for a procedure with an arbitrary number of parameters by giving a *formal list parameter*, the number of actual arguments is not checked but only a list *[formal list parameter]* of appropriate length is built up. Type checks are further performed upon execution of each statement in *procedure body* and upon evaluation of the expression *expression* to be returned.

Procedures defined by **proc** containing a **quit** or **restart** command cannot be executed (i.e. applied). Upon application of a procedure, the Sollya interpreter checks beforehand for such a statement. If one is found, the application of the procedure to its arguments evaluates to **error**. A warning is displayed. Remark that in contrast to other type or semantic correctness checks, this check is really performed before interpreting any other statement in the body of the procedure.

- Through the **var** keyword it is possible to declare local variables and thus to have full support of recursive procedures. This means a procedure defined using **proc** may contain in its *procedure body* an application of itself to some actual parameters: it suffices to assign the procedure (object) to an identifier with an appropriate name.
- Sollya procedures defined using **proc** may return other procedures. Further *procedure body* may contain assignments of locally defined procedure objects to identifiers. See **var** for the particular behaviour of local and global variables.

- The expression *expression* returned by a procedure is evaluated with regard to **Sollya** commands, procedures and external procedures. Simplification may be performed. However, an application of a procedure defined by **proc** to actual parameters evaluates to the expression *expression* that may contain the free global variable or that may be composed.

Example 1:

```
> succ = proc(n) { return n + 1; };
> succ(5);
6
> 3 + succ(0);
4
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
```

Example 2:

```
> add = proc(m,n) { var res; res := m + n; return res; };
> add(5,6);
11
> add;
proc(m, n)
begin
var res;
res := (m) + (n);
return res;
end
> verbosity = 1!;
> add(3);
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error
> add(true,false);
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
Warning: the given expression or command could not be handled.
error
```

Example 3:

```
> succ = proc(n) { return n + 1; };
> succ(5);
6
> succ(x);
1 + x
```

Example 4:



```

> hey = proc() { print("Hello world."); };
> hey();
Hello world.
> print(hey());
Hello world.
void
> hey;
proc()
begin
print("Hello world.");
return void;
end

```

Example 5:

```

> fac = proc(n) { var res; if (n == 0) then res := 1 else res := n * fac(n - 1);
  return res; };
> fac(5);
120
> fac(11);
39916800
> fac;
proc(n)
begin
var res;
if (n) == (0) then
res := 1
else
res := (n) * (fac((n) - (1)));
return res;
end

```

Example 6:

```

> myprocs = [| proc(m,n) { return m + n; }, proc(m,n) { return m - n; } |];
> (myprocs[0])(5,6);
11
> (myprocs[1])(5,6);
-1
> succ = proc(n) { return n + 1; };
> pred = proc(n) { return n - 1; };
> applier = proc(p,n) { return p(n); };
> applier(succ,5);
6
> applier(pred,5);
4

```

Example 7:

```

> verbosity = 1!;
> myquit = proc(n) { print(n); quit; };
> myquit;
proc(n)
begin
print(n);
quit;
return void;
end
> myquit(5);
Warning: a quit or restart command may not be part of a procedure body.
The procedure will not be executed.
Warning: an error occurred while executing a procedure.
Warning: the given expression or command could not be handled.
error

```

Example 8:

```

> printsucc = proc(n) { var succ; succ = proc(n) { return n + 1; }; print("Succe
ssor of",n,"is",succ(n)); };
> printsucc(5);
Successor of 5 is 6

```

Example 9:

```

> makeadd = proc(n) { var add; print("n =",n); add = proc(m,n) { return n + m; }
; return add; };
> makeadd(4);
n = 4
proc(m, n)
begin
nop;
return (n) + (m);
end
> (makeadd(4))(5,6);
n = 4
11

```

Example 10:

```

> sumall = proc(l = ...) { var acc, i; acc = 0; for i in l do acc = acc + i; ret
urn acc; };
> sumall;
proc(l = ...)
begin
var acc, i;
acc = 0;
for i in l do
acc = (acc) + (i);
return acc;
end
> sumall();
0
> sumall(2);
2
> sumall(2,5);
7
> sumall(2,5,7,9,16);
39
> sumall @ [|1,...,10|];
55

```

See also: **return** (8.130), **externalproc** (8.52), **void** (8.162), **quit** (8.119), **restart** (8.129), **var** (8.160), **@** (8.22)

## 8.118 procedure

Name: **procedure**

defines and assigns a Sollya procedure

Usage:

**procedure** *identifier*(*formal parameter1*, *formal parameter2*,..., *formal parameter n*) **begin** *procedure body* **end** : void → void

**procedure** *identifier*(*formal parameter1*, *formal parameter2*,..., *formal parameter n*) **begin** *procedure body* **return** *expression*; **end** : void → void

**procedure** *identifier*(*formal list parameter* = ...) **begin** *procedure body* **end** : void → void

**procedure** *identifier*(*formal list parameter* = ...) **begin** *procedure body* **return** *expression*; **end** : void → void

Parameters:

- *identifier* represents the name of the procedure to be defined and assigned
- *formal parameter1*, *formal parameter2* through *formal parameter n* represent identifiers used as formal parameters
- *formal list parameter* represents an identifier used as a formal parameter for the list of an arbitrary number of parameters
- *procedure body* represents the imperative statements in the body of the procedure
- *expression* represents the expression **procedure** shall evaluate to

Description:

- The **procedure** keyword allows for defining and assigning procedures in the Sollya language. It is an abbreviation to a procedure definition using **proc** with the same formal parameters, procedure body and return-expression followed by an assignment of the procedure (object) to the identifier *identifier*. In particular, all rules concerning local variables declared using the **var** keyword apply for **procedure**.

Example 1:

```
> procedure succ(n) { return n + 1; };
> succ(5);
6
> 3 + succ(0);
4
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
```

Example 2:

```
> procedure myprint(l = ...) { var i; for i in l do i; };
> myprint("Lyon","Nancy","Beaverton","Coye-la-Forêt","Amberg","Nizhny Novgorod",
"Cluj-Napoca");
Lyon
Nancy
Beaverton
Coye-la-Forêt
Amberg
Nizhny Novgorod
Cluj-Napoca
```

See also: **proc** (8.117), **var** (8.160)

## 8.119 quit

Name: **quit**

quits Sollya

Usage:

**quit** : void → void

Description:

- The command **quit**, when executed, stops the execution of a **Sollya** script and leaves the **Sollya** interpreter unless the **quit** command is executed in a **Sollya** script read into a main **Sollya** script by **execute** or **#include**.

Upon exiting the **Sollya** interpreter, all state is thrown away, all memory is deallocated, all bound libraries are unbound and the temporary files produced by **plot** and **externalplot** are deleted.

If the **quit** command does not lead to a halt of the **Sollya** interpreter, a warning is displayed.

Example 1:

```
> quit;
```

See also: **restart** (8.129), **execute** (8.46), **plot** (8.102), **externalplot** (8.51)

## 8.120 range

Name: **range**

keyword representing a **range** type

Usage:

**range** : type type

Description:

- **range** represents the **range** type for declarations of external procedures by means of **externalproc**.

Remark that in contrast to other indicators, type indicators like **range** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **boolean** (8.17), **constant** (8.23), **function** (8.61), **integer** (8.73), **list of** (8.80), **string** (8.147)

## 8.121 rationalapprox

Name: **rationalapprox**

returns a fraction close to a given number.

Usage:

**rationalapprox**( $x, n$ ) : (constant, integer)  $\rightarrow$  function

Parameters:

- $x$  is a number to approximate.
- $n$  is a integer (representing a format).

Description:

- **rationalapprox**( $x, n$ ) returns a constant function of the form  $a/b$  where  $a$  and  $b$  are integers. The value  $a/b$  is an approximation of  $x$ . The quality of this approximation is determined by the parameter  $n$  that indicates the number of correct bits that  $a/b$  should have.
- The command is not safe in the sense that it is not ensured that the error between  $a/b$  and  $x$  is less than  $2^{-n}$ .
- The following algorithm is used:  $x$  is first rounded downwards and upwards to a format of  $n$  bits, thus obtaining an interval  $[x_l, x_u]$ . This interval is then developped into a continued fraction as far as the representation is the same for every elements of  $[x_l, x_u]$ . The corresponding fraction is returned.
- Since rational numbers are not a primitive object of **Sollya**, the fraction is returned as a constant function. This can be quite amazing, because **Sollya** immediately simplifies a constant function by evaluating it when the constant has to be displayed. To avoid this, you can use **print** (that displays the expression representing the constant and not the constant itself) or the commands **numerator** and **denominator**.

Example 1:

```
> pi10 = rationalapprox(Pi,10);
> pi50 = rationalapprox(Pi,50);
> pi100 = rationalapprox(Pi,100);
> print( pi10, ":", simplify(floor(-log2(abs(pi10-Pi)/Pi))), "bits." );
22 / 7 : 11 bits.
> print( pi50, ":", simplify(floor(-log2(abs(pi50-Pi)/Pi))), "bits." );
90982559 / 28960648 : 50 bits.
> print( pi100, ":", simplify(floor(-log2(abs(pi100-Pi)/Pi))), "bits." );
4850225745369133 / 1543874804974140 : 101 bits.
```

Example 2:

```

> a=0.1;
> b=rationalapprox(a,4);
> numerator(b); denominator(b);
1
10
> print(simplify(floor(-log2(abs((b-a)/a)))), "bits.");
166 bits.

```

See also: **print** (8.112), **numerator** (8.95), **denominator** (8.29)

## 8.122 rationalmode

Name: **rationalmode**

global variable controlling if rational arithmetic is used or not.

Usage:

```

rationalmode = activation value : on|off → void
rationalmode = activation value ! : on|off → void
rationalmode : on|off

```

Parameters:

- *activation value* controls if rational arithmetic should be used or not

Description:

- **rationalmode** is a global variable. When its value is **off**, which is the default, **Sollya** will not use rational arithmetic to simplify expressions. All computations, including the evaluation of constant expressions given on the **Sollya** prompt, will be performed using floating-point and interval arithmetic. Constant expressions will be approximated by floating-point numbers, which are in most cases faithful roundings of the expressions, when shown at the prompt.
- When the value of the global variable **rationalmode** is **on**, **Sollya** will use rational arithmetic when simplifying expressions. Constant expressions, given at the **Sollya** prompt, will be simplified to rational numbers and displayed as such when they are in the set of the rational numbers. Otherwise, floating-point and interval arithmetic will be used to compute a floating-point approximation, which is in most cases a faithful rounding of the constant expression.

Example 1:

```

> rationalmode=off!;
> 19/17 + 3/94;
1.1495619524405506883604505632040050062578222778473
> rationalmode=on!;
> 19/17 + 3/94;
1837 / 1598

```

Example 2:

```

> rationalmode=off!;
> exp(19/17 + 3/94);
3.15680977395514136754709208944824276340328162814418
> rationalmode=on!;
> exp(19/17 + 3/94);
3.15680977395514136754709208944824276340328162814418

```

See also: **on** (8.97), **off** (8.96), **numerator** (8.95), **denominator** (8.29), **simplifysafe** (8.141)

## 8.123 RD

Name: **RD**

constant representing rounding-downwards mode.

Description:

- **RD** is used in command **round** to specify that the value  $x$  must be rounded to the greatest floating-point number  $y$  such that  $y \leq x$ .

Example 1:

```
> display=binary!;  
> round(Pi,20,RD);  
1.1001001000011111101_2 * 2^(1)
```

See also: **RZ** (8.138), **RU** (8.137), **RN** (8.132), **round** (8.133)

## 8.124 readfile

Name: **readfile**

reads the content of a file into a string variable

Usage:

**readfile**(*filename*) : string → string

Parameters:

- *filename* represents a character sequence indicating a file name

Description:

- **readfile** opens the file indicated by *filename*, reads it and puts its contents in a character sequence of type **string** that is returned.  
If the file indicated by *filename* cannot be opened for reading, a warning is displayed and **readfile** evaluates to an **error** variable of type **error**.

Example 1:

```
> print("Hello world") > "myfile.txt";  
> t = readfile("myfile.txt");  
> t;  
Hello world
```

Example 2:

```
> verbosity=1!;  
> readfile("afile.txt");  
Warning: the file "afile.txt" could not be opened for reading.  
Warning: at least one of the given expressions or a subexpression is not correctly typed  
or its evaluation has failed because of some error on a side-effect.  
error
```

See also: **parse** (8.99), **execute** (8.46), **write** (8.164), **print** (8.112)

## 8.125 readxml

Name: **readxml**

reads an expression written as a MathML-Content-Tree in a file

Usage:

$$\text{readxml}(\text{filename}) : \text{string} \rightarrow \text{function} \mid \text{error}$$

Parameters:

- *filename* represents a character sequence indicating a file name

Description:

- **readxml**(*filename*) reads the first occurrence of a lambda application with one bounded variable on applications of the supported basic functions in file *filename* and returns it as a **Sollya** functional expression.

If the file *filename* does not contain a valid MathML-Content tree, **readxml** tries to find an "annotation encoding" markup of type "sollya/text". If this annotation contains a character sequence that can be parsed by **parse**, **readxml** returns that expression. Otherwise **readxml** displays a warning and returns an **error** variable of type **error**.

Example 1:

```
> readxml("readxmlexample.xml");  
2 + x + exp(sin(x))
```

See also: **printxml** (8.116), **readfile** (8.124), **parse** (8.99)

## 8.126 relative

Name: **relative**

indicates a relative error for **externalplot** or **fpminimax**

Usage:

$$\text{relative} : \text{absolute} \mid \text{relative}$$

Description:

- The use of **relative** in the command **externalplot** indicates that during plotting in **externalplot** a relative error is to be considered.

See **externalplot** for details. Used with **fpminimax**, **relative** indicates that **fpminimax** must try to minimize the relative error.

See **fpminimax** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");  
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l  
mpfr");  
> externalplot("./externalplotexample",absolute,exp(x),[-1/2;1/2],12,perturb);
```

See also: **externalplot** (8.51), **fpminimax** (8.59), **absolute** (8.2), **bashexecute** (8.15)



## 8.127 `remez`

Name: **remez**

computes the minimax of a function on an interval.

Usage:

**remez**(*f*, *n*, *range*, *w*, *quality*) : (function, integer, range, function, constant) → function  
**remez**(*f*, *L*, *range*, *w*, *quality*) : (function, list, range, function, constant) → function

Parameters:

- *f* is the function to be approximated
- *n* is the degree of the polynomial that must approximate *f*
- *L* is a list of monomials that can be used to represent the polynomial that must approximate *f*
- *range* is the interval where the function must be approximated
- *w* (optional) is a weight function. Default is 1.
- *quality* (optional) is a parameter that controls the quality of the returned polynomial *p*, with respect to the exact minimax  $p^*$ . Default is 1e-5.

Description:

- **remez** computes an approximation of the function *f* with respect to the weight function *w* on the interval *range*. More precisely, it searches a polynomial *p* such that  $\|pw - f\|_\infty$  is (almost) minimal among all polynomials *p* of a certain form. The norm is the infinity norm, e.g.  $\|g\|_\infty = \max\{|g(x)|, x \in \text{range}\}$ .
- If *w* = 1 (the default case), it consists in searching the best polynomial approximation of *f* with respect to the absolute error. If *f* = 1 and *w* is of the form 1/*g*, it consists in searching the best polynomial approximation of *g* with respect to the relative error.
- If *n* is given, the polynomial *p* is searched among the polynomials with degree not greater than *n*. If *L* is given, the polynomial *p* is searched as a linear combination of monomials  $X^k$  where *k* belongs to *L*. *L* may contain ellipses but cannot be end-elliptic.
- The polynomial is obtained by a convergent iteration called Remez' algorithm. The algorithm computes a sequence  $p_1, \dots, p_k, \dots$  such that  $e_k = \|p_k w - f\|_\infty$  converges towards the optimal value *e*. The algorithm is stopped when the relative error between  $e_k$  and *e* is less than *quality*.
- Note: the algorithm may not converge in certain cases. Moreover, it may converge towards a polynomial that is not optimal. These cases correspond to the cases when the Haar condition is not fulfilled. See [Cheney - Approximation theory] for details.

Example 1:

```
> p = remez(exp(x),5,[0;1]);
> degree(p);
5
> dirtyinfnorm(p-exp(x),[0;1]);
1.12956984638214536849843017679626063762687501534126e-6
```

Example 2:

```
> p = remez(1,[|0,2,4,6,8|],[0,Pi/4],1/cos(x));
> canonical=on!;
> p;
0.9999999994393749280444571988532724907643631727381 + -0.4999999957155746773720
4931630836834563663039748203 * x^2 + 4.16666132335010905188253972212748718651775
241902969e-2 * x^4 + -1.38865291475286141707180658383176799662601691348739e-3 *
x^6 + 2.437267919111162694221738667927916761689966804242e-5 * x^8
```

Example 3:

```
> p1 = remez(exp(x),5,[0;1],default,1e-5);
> p2 = remez(exp(x),5,[0;1],default,1e-10);
> p3 = remez(exp(x),5,[0;1],default,1e-15);
> dirtyinfnorm(p1-exp(x),[0;1]);
1.12956984638214536849843017679626063762687501534126e-6
> dirtyinfnorm(p2-exp(x),[0;1]);
1.12956980227478687332174207517728389861926659249056e-6
> dirtyinfnorm(p3-exp(x),[0;1]);
1.12956980227478687332174207517728389861926659249056e-6
```

See also: **dirtyinfnorm** (8.33), **infnorm** (8.72)

## 8.128 rename

Name: **rename**

rename the free variable.

Usage:

**rename**(*ident1*,*ident2*) : void

Parameters:

- *ident1* is the current name of the free variable.
- *ident2* is a fresh name.

Description:

- **rename** permits a change of the name of the free variable. **Sollya** can handle only one free variable at a time. The first time in a session that an unbound name is used in a context where it can be interpreted as a free variable, the name is used to represent the free variable of **Sollya**. In the following, this name can be changed using **rename**.
- Be careful: if *ident2* has been set before, its value will be lost. Use the command **isbound** to know if *ident2* is already used or not.
- If *ident1* is not the current name of the free variable, an error occurs.
- If **rename** is used at a time when the name of the free variable has not been defined, *ident1* is just ignored and the name of the free variable is set to *ident2*.

Example 1:

```
> f=sin(x);
> f;
sin(x)
> rename(x,y);
> f;
sin(y)
```

Example 2:

```
> a=1;
> f=sin(x);
> rename(x,a);
> a;
a
> f;
sin(a)
```

Example 3:

```
> verbosity=1!;  
> f=sin(x);  
> rename(y,z);  
Warning: the current free variable is named "x" and not "y". Can only rename the  
free variable.  
The last command will have no effect.
```

Example 4:

```
> rename(x,y);  
> isbound(x);  
false  
> isbound(y);  
true
```

See also: **isbound** (8.75)

## 8.129 restart

Name: **restart**

brings Sollya back to its initial state

Usage:

**restart** : void  $\rightarrow$  void

Description:

- The command **restart** brings Sollya back to its initial state. All current state is abandoned, all libraries unbound and all memory freed.

The **restart** command has no effect when executed inside a Sollya script read into a main Sollya script using **execute**. It is executed in a Sollya script included by a **#include** macro.

Using the **restart** command in nested elements of imperative programming like for or while loops is possible. Since in most cases abandoning the current state of Sollya means altering a loop invariant, warnings for the impossibility of continuing a loop may follow unless the state is rebuilt.

Example 1:

```
> print(exp(x));  
exp(x)  
> a = 3;  
> restart;  
The tool has been restarted.  
> print(x);  
x  
> a;  
Warning: the identifier "a" is neither assigned to, nor bound to a library function nor external procedure, nor equal to the current free variable.  
Will interpret "a" as "x".  
x
```

Example 2:

```

> print(exp(x));
exp(x)
> for i from 1 to 10 do {
>   print(i);
>   if (i == 5) then restart;
> };
1
2
3
4
5
The tool has been restarted.
Warning: the tool has been restarted inside a for loop.
The for loop will no longer be executed.

```

Example 3:

```

> print(exp(x));
exp(x)
> a = 3;
> for i from 1 to 10 do {
>   print(i);
>   if (i == 5) then {
>     restart;
>     i = 7;
>   };
> };
1
2
3
4
5
The tool has been restarted.
8
9
10
> print(x);
x
> a;
Warning: the identifier "a" is neither assigned to, nor bound to a library function nor external procedure, nor equal to the current free variable.
Will interpret "a" as "x".
x

```

See also: **quit** (8.119), **execute** (8.46)

### 8.130 return

Name: **return**

indicates an expression to be returned in a procedure

Usage:

**return** *expression* : void

Parameters:

- *expression* represents the expression to be returned

Description:

- The keyword **return** allows for returning the (evaluated) expression *expression* at the end of a begin-end-block (-block) used as a **Sollya** procedure body. See **proc** for further details concerning **Sollya** procedure definitions.

Statements for returning expressions using **return** are only possible at the end of a begin-end-block used as a **Sollya** procedure body. Only one **return** statement can be given per begin-end-block.

- If at the end of a procedure definition using **proc** no **return** statement is given, a **return void** statement is implicitly added. Procedures, i.e. procedure objects, when printed out in **Sollya** defined with an implicit **return void** statement are displayed with this statement explicitly given.

Example 1:

```
> succ = proc(n) { var res; res := n + 1; return res; };
> succ(5);
6
> succ;
proc(n)
begin
var res;
res := (n) + (1);
return res;
end
```

Example 2:

```
> hey = proc(s) { print("Hello",s); };
> hey("world");
Hello world
> hey;
proc(s)
begin
print("Hello", s);
return void;
end
```

See also: **proc** (8.117), **void** (8.162)

### 8.131 revert

Name: **revert**

reverts a list.

Usage:

$$\mathbf{revert}(L) : \text{list} \rightarrow \text{list}$$

Parameters:

- $L$  is a list.

Description:

- **revert**( $L$ ) returns the same list, but with its elements in reverse order.
- If  $L$  is an end-elliptic list, **revert** will fail with an error.

Example 1:

```
> revert([| |]);
[| |]
```

Example 2:

```
> revert([|2,3,5,2,1,4|]);  
[|4, 1, 2, 5, 3, 2|]
```

### 8.132 RN

Name: **RN**

constant representing rounding-to-nearest mode.

Description:

- **RN** is used in command **round** to specify that the value must be rounded to the nearest representable floating-point number.

Example 1:

```
> display=binary!;  
> round(Pi,20,RN);  
1.100100100001111111_2 * 2^(1)
```

See also: **RD** (8.123), **RU** (8.137), **RZ** (8.138), **round** (8.133)

### 8.133 round

Name: **round**

rounds a number to a floating-point format.

Usage:

$\mathbf{round}(x,n,mode) : (\text{constant, integer, } \mathbf{RD} \mid \mathbf{RU} \mid \mathbf{RN} \mid \mathbf{RZ}) \rightarrow \text{constant}$   
 $\mathbf{round}(x,format,mode) : (\text{constant, D|double|DD|doubledouble|DE|doubleextended|TD|tripleddouble, } \mathbf{RD} \mid \mathbf{RU} \mid \mathbf{RN} \mid \mathbf{RZ}) \rightarrow \text{constant}$

Parameters:

- $x$  is a constant to be rounded.
- $n$  is the precision of the target format.
- $format$  is the name of a supported floating-point format.
- $mode$  is the desired rounding mode.

Description:

- If used with an integer parameter  $n$ , **round**( $x,n,mode$ ) rounds  $x$  to a floating-point number with precision  $n$ , according to rounding-mode  $mode$ .
- If used with a format parameter  $format$ , **round**( $x,format,mode$ ) rounds  $x$  to a floating-point number in the floating-point format  $format$ , according to rounding-mode  $mode$ .
- Subnormal numbers are not handled are handled only if a  $format$  parameter is given that is different from **doubleextended**. The range of possible exponents is the range used for all numbers represented in **Sollya** (e.g. basically the range used in the library **MPFR**).

Example 1:

```
> display=binary!;  
> round(Pi,20,RN);  
1.100100100001111111_2 * 2^(1)
```

Example 2:

```
> printhexa(round(exp(17),53,RU));
0x417709348c0ea4f9
> printhexa(D(exp(17)));
0x417709348c0ea4f9
```

Example 3:

```
> display=binary!;
> a=2^(-1100);
> round(a,53,RN);
1_2 * 2^(-1100)
> round(a,D,RN);
0
> double(a);
0
```

See also: **RN** (8.132), **RD** (8.123), **RU** (8.137), **RZ** (8.138), **single** (8.143), **double** (8.37), **double-extended** (8.39), **doubledouble** (8.38), **tripledouble** (8.158), **roundcoefficients** (8.134), **roundcorrectly** (8.135), **printhexa** (8.115), **printfloat** (8.114)

### 8.134 roundcoefficients

Name: **roundcoefficients**

rounds the coefficients of a polynomial to classical formats.

Usage:

$$\text{roundcoefficients}(p,L) : (\text{function}, \text{list}) \rightarrow \text{function}$$

Parameters:

- $p$  is a function. Usually a polynomial.
- $L$  is a list of formats.

Description:

- If  $p$  is a polynomial and  $L$  a list of floating-point formats, **roundcoefficients**( $p,L$ ) rounds each coefficient of  $p$  to the corresponding format in  $L$ .
- If  $p$  is not a polynomial, **roundcoefficients** does not do anything.
- If  $L$  contains other elements than **D**, **double**, **DD**, **doubledouble**, **TD** and **tripledouble**, an error occurs.
- The coefficients in  $p$  corresponding to  $X^i$  is rounded to the format  $L[i]$ . If  $L$  does not contain enough elements (e.g. if  $\text{length}(L) < \text{degree}(p)+1$ ), a warning is displayed. However, the coefficients corresponding to an element of  $L$  are rounded. The trailing coefficients (that do not have a corresponding element in  $L$ ) are kept with their own precision. If  $L$  contains too much elements, the trailing useless elements are ignored. In particular  $L$  may be end-elliptic in which case **roundcoefficients** has the natural behavior.

Example 1:

```

> p=exp(1) + x*(exp(2) + x*exp(3));
> display=binary!;
> roundcoefficients(p, [|DD,D,D|]);
1.01011011111100001010100010110001010001010111011010010101001101010101111101110
001010110001000000010011101_2 * 2^(1) + x * (1.110110001110011001001011100011010
100110111011010111_2 * 2^(2) + x * (1.01000001010111100101101111110110111101100
010000011_2 * 2^(4)))
> roundcoefficients(p, [|DD,D...|]);
1.01011011111100001010100010110001010001010111011010010101001101010101111101110
001010110001000000010011101_2 * 2^(1) + x * (1.110110001110011001001011100011010
100110111011010111_2 * 2^(2) + x * (1.01000001010111100101101111110110111101100
010000011_2 * 2^(4)))

```

Example 2:

```

> f=sin(exp(1)*x);
> display=binary!;
> f;
sin(x * (1.010110111111000010101000101100010100010101110110100101010011010101011
11110111000101011000100000001001110011110100111100111100011101100010111001110001
01100000111101_2 * 2^(1)))
> roundcoefficients(f, [|D...|]);
sin(x * (1.010110111111000010101000101100010100010101110110100101010011010101011
11110111000101011000100000001001110011110100111100111100011101100010111001110001
01100000111101_2 * 2^(1)))

```

Example 3:

```

> p=exp(1) + x*(exp(2) + x*exp(3));
> verbosity=1!;
> display=binary!;
> roundcoefficients(p, [|DD,D|]);
Warning: the number of the given formats does not correspond to the degree of th
e given polynomial.
Warning: the 0th coefficient of the given polynomial does not evaluate to a floa
ting-point constant without any rounding.
Will evaluate the coefficient in the current precision in floating-point before
rounding to the target format.
Warning: the 1th coefficient of the given polynomial does not evaluate to a floa
ting-point constant without any rounding.
Will evaluate the coefficient in the current precision in floating-point before
rounding to the target format.
Warning: rounding may have happened.
1.01011011111100001010100010110001010001010111011010010101001101010101111101110
001010110001000000010011101_2 * 2^(1) + x * (1.110110001110011001001011100011010
100110111011010111_2 * 2^(2) + x * (1.01000001010111100101101111110110111101100
010000010111110010110101001011101111110001010011011101000100110000111010001110
010000010110000101100000111001011100101001_2 * 2^(4)))

```

See also: **single** (8.143), **double** (8.37), **doubledouble** (8.38), **tripledouble** (8.158)

## 8.135 roundcorrectly

Name: **roundcorrectly**

rounds an approximation range correctly to some precision

Usage:



**roundcorrectly**(*range*) : range → constant

Parameters:

- *range* represents a range in which an exact value lies

Description:

- Let *range* be a range of values, determined by some approximation process, safely bounding an unknown value *v*. The command **roundcorrectly**(*range*) determines a precision such that for this precision, rounding to the nearest any value in *range* yields to the same result, i.e. to the correct rounding of *v*.

If no such precision exists, a warning is displayed and **roundcorrectly** evaluates to NaN.

Example 1:

```
> printbinary(roundcorrectly([1.010001_2; 1.0101_2]));  
1.01_2  
> printbinary(roundcorrectly([1.00001_2; 1.001_2]));  
1_2
```

Example 2:

```
> roundcorrectly([-1; 1]);  
@NaN@
```

See also: **round** (8.133)

## 8.136 roundingwarnings

Name: **roundingwarnings**

global variable controlling whether or not a warning is displayed when roundings occur.

Usage:

**roundingwarnings** = *activation value* : on|off → void  
**roundingwarnings** = *activation value* ! : on|off → void  
**roundingwarnings** : on|off

Parameters:

- *activation value* controls if warnings should be shown or not

Description:

- **roundingwarnings** is a global variable. When its value is **on**, warnings are emitted in appropriate verbosity modes (see **verbosity**) when roundings occur. When its value is **off**, these warnings are suppressed.
- This mode depends on a verbosity of at least 1. See **verbosity** for more details.
- Default is **on** when the standard input is a terminal and **off** when **Sollya** input is read from a file.

Example 1:

```

> verbosity=1!;
> roundingwarnings = on;
Rounding warning mode has been activated.
> exp(0.1);
Warning: Rounding occurred when converting the constant "0.1" to floating-point
with 165 bits.
If safe computation is needed, try to increase the precision.
Warning: rounding has happened. The value displayed is a faithful rounding of th
e true result.
1.1051709180756476248117078264902466682245471947375
> roundingwarnings = off;
Rounding warning mode has been deactivated.
> exp(0.1);
1.1051709180756476248117078264902466682245471947375

```

See also: **on** (8.97), **off** (8.96), **verbosity** (8.161), **midpointmode** (8.88), **rationalmode** (8.122)

### 8.137 RU

Name: **RU**

constant representing rounding-upwards mode.

Description:

- **RU** is used in command **round** to specify that the value  $x$  must be rounded to the smallest floating-point number  $y$  such that  $x \leq y$ .

Example 1:

```

> display=binary!;
> round(Pi,20,RU);
1.100100100001111111_2 * 2^(1)

```

See also: **RZ** (8.138), **RD** (8.123), **RN** (8.132), **round** (8.133)

### 8.138 RZ

Name: **RZ**

constant representing rounding-to-zero mode.

Description:

- **RZ** is used in command **round** to specify that the value must be rounded to the closest floating-point number towards zero. It just consists in truncating the value to the desired format.

Example 1:

```

> display=binary!;
> round(Pi,20,RZ);
1.1001001000011111101_2 * 2^(1)

```

See also: **RD** (8.123), **RU** (8.137), **RN** (8.132), **round** (8.133)

### 8.139 searchgal

Name: **searchgal**

searches for a preimage of a function such that the rounding the image yields an error smaller than a constant

Usage:

**searchgal**(*function*, *start*, *preimage precision*, *steps*, *format*, *error bound*) : (function, constant, integer, integer, D|double|DD|doubledouble|DE|doubleextended|TD|tripledouble, constant) → list  
**searchgal**(*list of functions*, *start*, *preimage precision*, *steps*, *list of format*, *list of error bounds*) : (list, constant, integer, integer, list, list) → list

Parameters:

- *function* represents the function to be considered
- *start* represents a value around which the search is to be performed
- *preimage precision* represents the precision (discretization) for the eligible preimage values
- *steps* represents the binary logarithm ( $\log_2$ ) of the number of search steps to be performed
- *format* represents the format the image of the function is to be rounded to
- *error bound* represents a upper bound on the relative rounding error when rounding the image
- *list of functions* represents the functions to be considered
- *list of formats* represents the respective formats the images of the functions are to be rounded to
- *list of error bounds* represents a upper bound on the relative rounding error when rounding the image

Description:

- The command **searchgal** searches for a preimage  $z$  of function *function* or a list of functions *list of functions* such that  $z$  is a floating-point number with *preimage precision* significant mantissa bits and the image  $y$  of the function, respectively each image  $y_i$  of the functions, rounds to format *format* respectively to the corresponding format in *list of format* with a relative rounding error less than *error bound* respectively the corresponding value in *list of error bounds*. During this search, at most  $2^{\text{steps}}$  attempts are made. The search starts with a preimage value equal to *start*. This value is then increased and decreased by 1 ulp in precision *preimage precision* until a value is found or the step limit is reached.

If the search finds an appropriate preimage  $z$ , **searchgal** evaluates to a list containing this value. Otherwise, **searchgal** evaluates to an empty list.

Example 1:

```
> searchgal(log(x),2,53,15,DD,1b-112);
[] []
> searchgal(log(x),2,53,18,DD,1b-112);
[|2.0000000000384972054234822280704975128173828125|]
```

Example 2:

```
> f = exp(x);
> s = searchgal(f,2,53,18,DD,1b-112);
> if (s != []) then {
>   v = s[0];
>   print("The rounding error is 2^(",evaluate(log2(abs(DD(f)/f - 1)),v),")");
> } else print("No value found");
The rounding error is 2^(-1.12106878438809380148206984258358542322113874177832e
2 )
```

Example 3:

```
> searchgal([|sin(x),cos(x)|],1,53,15,[|D,D|],[|1b-62,1b-60|]);
[|1.00000000000159494639717649988597258925437927246094|]
```

See also: **round** (8.133), **double** (8.37), **doubledouble** (8.38), **tripledouble** (8.158), **evaluate** (8.45), **worstcase** (8.163)

## 8.140 `simplify`

Name: **simplify**

simplifies an expression representing a function

Usage:

$$\text{**simplify**}(\textit{function}) : \text{function} \rightarrow \text{function}$$

Parameters:

- *function* represents the expression to be simplified

Description:

- The command **simplify** simplifies constant subexpressions of the expression given in argument representing the function *function*. Those constant subexpressions are evaluated using floating-point arithmetic with the global precision **prec**.

Example 1:

```
> print(simplify(sin(pi * x)));
sin(3.14159265358979323846264338327950288419716939937508 * x)
> print(simplify(erf(exp(3) + x * log(4))));
erf(2.00855369231876677409285296545817178969879078385544e1 + x * 1.3862943611198
906188344642429163531361510002687205)
```

Example 2:

```
> prec = 20!;
> t = erf(0.5);
> s = simplify(erf(0.5));
> prec = 200!;
> t;
0.5204998778130465376827466538919645287364515757579637000588058
> s;
0.52050018310546875
```

See also: **simplifysafe** (8.141), **autosimplify** (8.14), **prec** (8.109), **evaluate** (8.45)

## 8.141 `simplifysafe`

Name: **simplifysafe**

simplifies an expression representing a function

Usage:

$$\text{**simplifysafe**}(\textit{function}) : \text{function} \rightarrow \text{function}$$

Parameters:

- *function* represents the expression to be simplified

Description:

- The command **simplifysafe** simplifies the expression given in argument representing the function *function*. The command **simplifysafe** does not endanger the safety of computations even in Sollya's floating-point environment: the function returned is mathematically equal to the function *function*.

Remark that the simplification provided by **simplifysafe** is not perfect: they may exist simpler equivalent expressions for expressions returned by **simplifysafe**.

Example 1:

```
> print(simplifysafe((6 + 2) + (5 + exp(0)) * x));
8 + 6 * x
```

Example 2:

```
> print(simplifysafe((log(x - x + 1) + asin(1))));
(pi) / 2
```

Example 3:

```
> print(simplifysafe((log(x - x + 1) + asin(1)) - (atan(1) * 2)));
(pi) / 2 - (pi) / 4 * 2
```

See also: **simplify** (8.140), **autosimplify** (8.14), **rationalmode** (8.122)

## 8.142 sin

Name: **sin**

the sine function.

Description:

- **sin** is the usual sine function.
- It is defined for every real number  $x$ .

See also: **asin** (8.10), **cos** (8.24), **tan** (8.152)

## 8.143 single

Names: **single**, **SG**

rounding to the nearest IEEE 754 single (binary32).

Description:

- **single** is both a function and a constant.
- As a function, it rounds its argument to the nearest IEEE 754 single precision (i.e. IEEE754-2008 binary32) number. Subnormal numbers are supported as well as standard numbers: it is the real rounding described in the standard.
- As a constant, it symbolizes the single precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **round** and **roundcoefficients**. It is not supported for **implementpoly**. See the corresponding help pages for examples.

Example 1:

```
> display=binary!;
> SG(0.1);
1.10011001100110011001101_2 * 2^(-4)
> SG(4.17);
1.000010101110000101001_2 * 2^(2)
> SG(1.011_2 * 2^(-1073));
0
```

See also: **double** (8.37), **doubleextended** (8.39), **doubledouble** (8.38), **tripleddouble** (8.158), **roundcoefficients** (8.134), **implementpoly** (8.70), **round** (8.133), **printfloat** (8.114)

### 8.144 **sinh**

Name: **sinh**

the hyperbolic sine function.

Description:

- **sinh** is the usual hyperbolic sine function:  $\sinh(x) = \frac{e^x - e^{-x}}{2}$ .
- It is defined for every real number  $x$ .

See also: **asinh** (8.11), **cosh** (8.25), **tanh** (8.153)

### 8.145 **sort**

Name: **sort**

sorts a list of real numbers.

Usage:

**sort**( $L$ ) : list  $\rightarrow$  list

Parameters:

- $L$  is a list.

Description:

- If  $L$  contains only constant values, **sort**( $L$ ) returns the same list, but sorted in increasing order.
- If  $L$  contains at least one element that is not a constant, the command fails with a type error.
- If  $L$  is an end-elliptic list, **sort** will fail with an error.

Example 1:

```
> sort([| |]);  
[| |]  
> sort([|2,3,5,2,1,4|]);  
[|1, 2, 2, 3, 4, 5|]
```

### 8.146 **sqrt**

Name: **sqrt**

square root.

Description:

- **sqrt** is the square root, e.g. the inverse of the function square:  $\sqrt{y}$  is the unique positive  $x$  such that  $x^2 = y$ .
- It is defined only for  $x$  in  $[0; +\infty]$ .

### 8.147 **string**

Name: **string**

keyword representing a **string** type

Usage:

**string** : type type

Description:

- **string** represents the **string** type for declarations of external procedures by means of **externalproc**.  
Remark that in contrast to other indicators, type indicators like **string** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: **externalproc** (8.52), **boolean** (8.17), **constant** (8.23), **function** (8.61), **integer** (8.73), **list of** (8.80), **range** (8.120)

## 8.148 subpoly

Name: **subpoly**

restricts the monomial basis of a polynomial to a list of monomials

Usage:

$$\text{subpoly}(\text{polynomial}, \text{list}) : (\text{function}, \text{list}) \rightarrow \text{function}$$

Parameters:

- *polynomial* represents the polynomial the coefficients are taken from
- *list* represents the list of monomials to be taken

Description:

- **subpoly** extracts the coefficients of a polynomial *polynomial* and builds up a new polynomial out of those coefficients associated to monomial degrees figuring in the list *list*.

If *polynomial* represents a function that is not a polynomial, subpoly returns 0.

If *list* is a list that is end-elliptic, let be *j* the last value explicitly specified in the list. All coefficients of the polynomial associated to monomials greater or equal to *j* are taken.

Example 1:

```
> p = taylor(exp(x),5,0);
> s = subpoly(p,[1,3,5]);
> print(p);
1 + x * (1 + x * (0.5 + x * (1 / 6 + x * (1 / 24 + x / 120))))
> print(s);
x * (1 + x^2 * (1 / 6 + x^2 / 120))
```

Example 2:

```
> p = remez(atan(x),10,[-1,1]);
> subpoly(p,[1,3,5...]);
x * (0.99986632946591986997581285958052433296267358727229 + x^2 * (-0.3303047855
04861260596093435534236137298206064685038 + x^2 * (0.180159294636523467997437751
178959039617773054107393 + x * (-1.217048583218660289061758356493901143118773602
60197e-14 + x * (-8.5156350833702702996505336803770858918120961566741e-2 + x * (
1.39681284176342339364451388757935358048374878126733e-14 + x * (2.08451141754345
61643018447784809880955983412532269e-2 + x * (-5.6810131012579436265697622426011
349460288598691964e-15))))))
```

Example 3:

```
> subpoly(exp(x),[1,2,3]);
0
```

See also: **roundcoefficients** (8.134), **taylor** (8.154), **remez** (8.127)

## 8.149 substitute

Name: **substitute**

replace the occurrences of the free variable in an expression.

Usage:

$$\begin{aligned} \text{substitute}(f,g) &: (\text{function}, \text{function}) \rightarrow \text{function} \\ \text{substitute}(f,t) &: (\text{function}, \text{constant}) \rightarrow \text{constant} \end{aligned}$$

Parameters:

- $f$  is a function.
- $g$  is a function.
- $t$  is a real number.

Description:

- **substitute**( $f, g$ ) produces the function  $(f \circ g) : x \mapsto f(g(x))$ .
- **substitute**( $f, t$ ) is the constant  $f(t)$ . Note that the constant is represented by its expression until it has been evaluated (exactly the same way as if you type the expression  $f$  replacing instances of the free variable by  $t$ ).
- If  $f$  is stored in a variable  $F$ , the effect of the commands **substitute**( $F, g$ ) or **substitute**( $F, t$ ) is absolutely equivalent to writing  $F(g)$  resp.  $F(t)$ .

Example 1:

```
> f=sin(x);
> g=cos(x);
> substitute(f,g);
sin(cos(x))
> f(g);
sin(cos(x))
```

Example 2:

```
> a=1;
> f=sin(x);
> substitute(f,a);
0.84147098480789650665250232163029899962256306079837
> f(a);
0.84147098480789650665250232163029899962256306079837
```

## 8.150 sup

Name: **sup**

gives the upper bound of an interval.

Usage:

$$\begin{aligned} \mathbf{sup}(I) &: \text{range} \rightarrow \text{constant} \\ \mathbf{sup}(x) &: \text{constant} \rightarrow \text{constant} \end{aligned}$$

Parameters:

- $I$  is an interval.
- $x$  is a real number.

Description:

- Returns the upper bound of the interval  $I$ . Each bound of an interval has its own precision, so this command is exact, even if the current precision is too small to represent the bound.
- When called on a real number  $x$ , **sup** considers it as an interval formed of a single point:  $[x, x]$ . In other words, **sup** behaves like the identity.

Example 1:



```
> sup([1;3]);
3
> sup(5);
5
```

Example 2:

```
> display=binary!;
> I=[0; 0.111110000011111_2];
> sup(I);
1.11110000011111_2 * 2^(-1)
> prec=12!;
> sup(I);
1.11110000011111_2 * 2^(-1)
```

See also: **inf** (8.71), **mid** (8.87)

### 8.151 tail

Name: **tail**

gives the tail of a list.

Usage:

**tail**( $L$ ) : list  $\rightarrow$  list

Parameters:

- $L$  is a list.

Description:

- **tail**( $L$ ) returns the list  $L$  without its first element.
- If  $L$  is empty, the command will fail with an error.
- **tail** can also be used with end-elliptic lists. In this case, the result of **tail** is also an end-elliptic list.

Example 1:

```
> tail([1,2,3]);
[2, 3]
> tail([1,2...]);
[2...]
```

See also: **head** (8.65)

### 8.152 tan

Name: **tan**

the tangent function.

Description:

- **tan** is the tangent function, defined by  $\tan(x) = \sin(x)/\cos(x)$ .
- It is defined for every real number  $x$  that is not of the form  $n\pi + \pi/2$  where  $n$  is an integer.

See also: **atan** (8.12), **cos** (8.24), **sin** (8.142)

### 8.153 tanh

Name: **tanh**

the hyperbolic tangent function.

Description:

- **tanh** is the hyperbolic tangent function, defined by  $\tanh(x) = \sinh(x) / \cosh(x)$ .
- It is defined for every real number  $x$ .

See also: **atanh** (8.13), **cosh** (8.25), **sinh** (8.144)

### 8.154 taylor

Name: **taylor**

computes a Taylor expansion of a function in a point

Usage:

**taylor**(*function*, *degree*, *point*) : (function, integer, constant) → function

Parameters:

- *function* represents the function to be expanded
- *degree* represents the degree of the expansion to be delivered
- *point* represents the point in which the function is to be developed

Description:

- The command **taylor** returns an expression that is a Taylor expansion of function *function* in point *point* having the degree *degree*.

Let  $f$  be the function *function*,  $t$  be the point *point* and  $n$  be the degree *degree*. Then, **taylor**(*function*,*degree*,*point*) evaluates to an expression mathematically equal to

$$\sum_{i=0}^n \frac{f^{(i)}(t)}{i!} (x - t)^i$$

Remark that **taylor** evaluates to 0 if the degree *degree* is negative.

Example 1:

```
> print(taylor(exp(x),5,0));  
1 + x * (1 + x * (0.5 + x * (1 / 6 + x * (1 / 24 + x / 120))))
```

Example 2:

```
> print(taylor(asin(x),7,0));  
x * (1 + x^2 * (1 / 6 + x^2 * (9 / 120 + x^2 * 225 / 5040)))
```

Example 3:

```
> print(taylor(erf(x),6,0));  
x * (1 / sqrt((pi) / 4) + x^2 * ((sqrt((pi) / 4) * 4 / (pi) * (-2)) / 6 + x^2 *  
(sqrt((pi) / 4) * 4 / (pi) * 12) / 120))
```

See also: **remez** (8.127)

## 8.155 `taylorrecursions`

Name: **taylorrecursions**

controls the number of recursion steps when applying Taylor's rule.

Usage:

```
taylorrecursions =  $n$  : integer  $\rightarrow$  void  
taylorrecursions =  $n$  ! : integer  $\rightarrow$  void  
taylorrecursions : integer
```

Parameters:

- $n$  represents the number of recursions

Description:

- **taylorrecursions** is a global variable. Its value represents the number of steps of recursion that are used when applying Taylor's rule. This rule is applied by the interval evaluator present in the core of **Sollya** (and particularly visible in commands like **infnorm**).
- To improve the quality of an interval evaluation of a function  $f$ , in particular when there are problems of decorrelation), the evaluator of **Sollya** uses Taylor's rule:  $f([a, b]) \subseteq f(m) + [a - m, b - m] \cdot f'([a, b])$  where  $m = \frac{a+b}{2}$ . This rule can be applied recursively. The number of step in this recursion process is controlled by **taylorrecursions**.
- Setting **taylorrecursions** to 0 makes **Sollya** use this rule only once; setting it to 1 makes **Sollya** use the rule twice, and so on. In particular: the rule is always applied at least once.

Example 1:

```
> f=exp(x);  
> p=remez(f,3,[0;1]);  
> taylorrecursions=0;  
The number of recursions for Taylor evaluation has been set to 0.  
> evaluate(f-p, [0;1]);  
[-0.46839364816303627522963565754743169862357620487739;0.46947781754667086491682  
464997088054443583003517779]  
> taylorrecursions=1;  
The number of recursions for Taylor evaluation has been set to 1.  
> evaluate(f-p, [0;1]);  
[-0.13813111495387910066337940912697015317218647208804;0.13921528433751369035056  
840155041899898444030238844]
```

## 8.156 `time`

Name: **time**

procedure for timing **Sollya** code.

Usage:

```
time(code) : code  $\rightarrow$  constant
```

Parameters:

- *code* is the code to be timed.

Description:

- **time** permits timing a **Sollya** instruction, resp. a begin-end block of **Sollya** instructions. The timing value, measured in seconds, is returned as a **Sollya** constant (and not merely displayed as for **timing**). This permits performing computations of the timing measurement value inside **Sollya**.

- The extended **nop** command permits executing a defined number of useless instructions. Taking the ratio of the time needed to execute a certain **Sollya** instruction and the time for executing a **nop** therefore gives a way to abstract from the speed of a particular machine when evaluating an algorithm's performance.

Example 1:

```
> t = time(p=remez(sin(x),10,[-1;1]));
> write(t,"s were spent computing p = ",p,"\n");
0.151346000000000000676889100326150128239532932639s were spent computing p = -
3.3426550293345171908513995127407122194691200059639e-17 + x * (0.999999999736283
59955372011464713121003442988167693 + x * (7.88027518773027866844993437990477324
95568873819693e-16 + x * (-0.1666666661386013037032912982196741385680498698107285
+ x * (-5.3734444911159112186289355138557504839692987221233e-15 + x * (8.333303
7186548537651002133031675072810009327877148e-3 + x * (1.337972213892188158841123
41005509831429347230871284e-14 + x * (-1.983448630182774164932681551541589244220
04290239026e-4 + x * (-1.3789116451286674170531616441916183417598709732816e-14 +
x * (2.6876259495430304684251822024896210963401672262005e-6 + x * 5.02823783500
10211058128384123578805586173782863605e-15)))))))))
```

Example 2:

```
> write(time({ p=remez(sin(x),10,[-1;1]); write("The error is 2^(", log2(dirtyin
fnorm(p-sin(x),[-1;1])), ")\n"); }, "were spent\n");
The error is 2^(log2(2.39602467695631727848641768186659313738474584992648e-11))
0.26773700000000000005350581089302863802004139870405were spent
```

Example 3:

```
> t = time(basheexecute("sleep 10"));
> write(~(t-10),"s of execution overhead.\n");
5.10899999999999989588328475065281963907182216644287e-3s of execution overhead.
```

Example 4:

```
> ratio := time(p=remez(sin(x),10,[-1;1]))/time(nop(10));
> write("This ratio = ", ratio, " should somehow be independent of the type of m
achine.\n");
This ratio = 3.7833151183970856108712196613616522183287118530965 should somehow
be independent of the type of machine.
```

See also: **timing** (8.157), **nop** (8.93)

## 8.157 timing

Name: **timing**

global variable controlling timing measures in **Sollya**.

Usage:

```
timing = activation value : on|off → void
timing = activation value ! : on|off → void
timing : on|off
```

Parameters:

- *activation value* controls if timing should be performed or not

Description:

- **timing** is a global variable. When its value is **on**, the time spent in each command is measured and displayed (for **verbosity** levels higher than 1).

Example 1:

```
> verbosity=1!;
> timing=on;
Timing has been activated.
> p=remez(sin(x),10,[-1;1]);
Information: Remez: computing the matrix spent 1 ms
Information: Remez: computing the quality of approximation spent 8 ms
Information: Remez: computing the matrix spent 1 ms
Information: Remez: solving the system spent 1 ms
Information: Remez: computing the quality of approximation spent 8 ms
Information: Remez: computing the matrix spent 1 ms
Information: Remez: computing the quality of approximation spent 8 ms
Information: computing a minimax approximation spent 152 ms
Information: assignment spent 152 ms
Information: full execution of the last parse chunk spent 152 ms
```

See also: **on** (8.97), **off** (8.96), **time** (8.156)

## 8.158 tripledouble

Names: **tripledouble**, **TD**

represents a number as the sum of three IEEE doubles.

Description:

- **tripledouble** is both a function and a constant.
- As a function, it rounds its argument to the nearest number that can be written as the sum of three double precision numbers.
- The algorithm used to compute **tripledouble**( $x$ ) is the following: let  $x_h = \mathbf{double}(x)$ , let  $x_m = \mathbf{double}(x - x_h)$  and let  $x_l = \mathbf{double}(x - x_h - x_m)$ . Return the number  $x_h + x_m + x_l$ . Note that if the current precision is not sufficient to represent exactly  $x_h + x_m + x_l$ , a rounding will occur and the result of **tripledouble**( $x$ ) will be useless.
- As a constant, it symbolizes the triple-double precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **roundcoefficients** and **implementpoly**. See the corresponding help pages for examples.

Example 1:

```
> verbosity=1!;
> a = 1+ 2^(-55)+2^(-115);
> TD(a);
1.00000000000000002775557561562891353466491600711096
> prec=110!;
> TD(a);
Warning: double rounding occurred on invoking the triple-double rounding operator.
Try to increase the working precision.
1.000000000000000027755575615628913
```

See also: **single** (8.143), **double** (8.37), **doubleextended** (8.39), **doubledouble** (8.38), **roundcoefficients** (8.134), **implementpoly** (8.70)

## 8.159 true

Name: **true**

the boolean value representing the truth.

Description:

- **true** is the usual boolean value.

Example 1:

```
> true && false;
false
> 2>1;
true
```

See also: **false** (8.53), **&&** (8.6), **||** (8.98)

## 8.160 var

Name: **var**

declaration of a local variable in a scope

Usage:

**var** *identifier1, identifier2,... , identifiern* : void

Parameters:

- *identifier1, identifier2,... , identifiern* represent variable identifiers

Description:

- The keyword **var** allows for the declaration of local variables *identifier1* through *identifiern* in a begin-end-block (**{}**-block). Once declared as a local variable, an identifier will shadow identifiers declared in higher scopes and undeclared identifiers available at top-level.

Variable declarations using **var** are only possible in the beginning of a begin-end-block. Several **var** statements can be given. Once another statement is given in a begin-end-block, no more **var** statements can be given.

Variables declared by **var** statements are dereferenced as **error** until they are assigned a value.

Example 1:

```
> exp(x);
exp(x)
> a = 3;
> {var a, b; a=5; b=3; {var a; var b; b = true; a = 1; a; b;}; a; b; };
1
true
5
3
> a;
3
```

See also: **error** (8.44)

## 8.161 verbosity

Name: **verbosity**

global variable controlling the amount of information displayed by commands.

Usage:

```

verbosity = n : integer → void
verbosity = n ! : integer → void
verbosity : integer

```

Parameters:

- $n$  controls the amount of information to be displayed

Description:

- **verbosity** accepts any integer value. At level 0, commands do not display anything on standard output. Note that very critical information may however be displayed on standard error.
- Default level is 1. It displays important information such as warnings when roundings happen.
- For higher levels more information is displayed depending on the command.

Example 1:

[illegible]

See also: **roundingwarnings** (8.136)

## 8.162 void

Name: **void**

the functional result of a side-effect or empty argument resp. the corresponding type

Usage:

```
void : void | type type
```

Description:

- The variable **void** represents the functional result of a side-effect or an empty argument. It is used only in combination with the applications of procedures or identifiers bound through **externalproc** to external procedures.

The **void** result produced by a procedure or an external procedure is not printed at the prompt. However, it is possible to print it out in a print statement or in complex data types such as lists.

The **void** argument is implicit when giving no argument to a procedure or an external procedure when applied. It can nevertheless be given explicitly. For example, suppose that `foo` is a procedure or an external procedure with a void argument. Then `foo()` and `foo(void)` are correct calls to `foo`. Here, a distinction must be made for procedures having an arbitrary number of arguments. In this case, an implicit **void** as the only parameter to a call of such a procedure gets converted into an empty list of arguments, an explicit **void** gets passed as-is in the formal list of parameters the procedure receives.

- **void** is used also as a type identifier for **externalproc**. Typically, an external procedure taking **void** as an argument or returning **void** is bound with a signature **void** – > some type or some type – > **void**. See **externalproc** for more details.

Example 1:

```
> print(void);
void
> void;
```

Example 2:

```
> hey = proc() { print("Hello world."); };
> hey;
proc()
begin
print("Hello world.");
return void;
end
> hey();
Hello world.
> hey(void);
Hello world.
> print(hey());
Hello world.
void
```

Example 3:

```
> bashexecute("gcc -fPIC -Wall -c externalprocvoidexample.c");
> bashexecute("gcc -fPIC -shared -o externalprocvoidexample externalprocvoidexample.o");
> externalproc(foo, "./externalprocvoidexample", void -> void);
> foo;
foo(void) -> void
> foo();
Hello from the external world.
> foo(void);
Hello from the external world.
> print(foo());
Hello from the external world.
void
```

Example 4:



```
> procedure blub(l = ...) { print("Argument list:", l); };
> blub(1);
Argument list: [|1|]
> blub();
Argument list: [| |]
> blub(void);
Argument list: [|void|]
```

See also: **error** (8.44), **proc** (8.117), **externalproc** (8.52)

8.163 worstcase

Name: **worstcase**

searches for hard-to-round cases of a function

Usage:

```
worstcase(function, preimage precision, preimage exponent range, image precision, error bound) :  
          (function, integer, range, integer, constant) → void
```

```
worstcase(function, preimage precision, preimage exponent range, image precision, error bound,
           filename) : (function, integer, range, integer, constant, string) → void
```

Parameters:

- *function* represents the function to be considered
- *preimage precision* represents the precision of the preimages
- *preimage exponent range* represents the exponents in the preimage format
- *image precision* represents the precision of the format the images are to be rounded to
- *error bound* represents the upper bound for the search w.r.t. the relative rounding error
- *filename* represents a character sequence containing a filename

Description:

- The **worstcase** command is deprecated. It searches for hard-to-round cases of a function. The command **searchgal** has a comparable functionality.

Example 1:

```
> worstcase(exp(x),24,[1,2],24,1b-26);
prec = 165
x = 1.99999988079071044921875          f(x) = 7.3890552520751953125          eps = 4
.5998601423446695596184695493764120138001954979037e-9 = 2^(-27.695763)
x = 2          f(x) = 7.38905620574951171875          eps = 1.4456360874967301812222
8379395533417878125150587072e-8 = 2^(-26.043720)
```

See also: **round** (8.133), **searchgal** (8.139), **evaluate** (8.45)

8.164 write

Name: write

prints an expression without separators

Usage:

```

write(expr1,...,exprn) : (any type,..., any type) → void
write(expr1,...,exprn) > filename : (any type,..., any type, string) → void
write(expr1,...,exprn) >> filename : (any type,...,any type, string) → void

```

Parameters:

- *expr* represents an expression
- *filename* represents a character sequence indicating a file name

Description:

- **write**(*expr1*,...,*exprn*) prints the expressions *expr1* through *exprn*. The character sequences corresponding to the expressions are concatenated without any separator. No newline is displayed at the end. In contrast to **print**, **write** expects the user to give all separators and newlines explicitly. If a second argument *filename* is given after a single ">", the displaying is not output on the standard output of **Sollya** but if in the file *filename* that get newly created or overwritten. If a double ">>" is given, the output will be appended to the file *filename*.

The global variables **display**, **midpointmode** and **fullparentheses** have some influence on the formatting of the output (see **display**, **midpointmode** and **fullparentheses**).

Remark that if one of the expressions *expri* given in argument is of type **string**, the character sequence *expri* evaluates to is displayed. However, if *expri* is of type **list** and this list contains a variable of type **string**, the expression for the list is displayed, i.e. all character sequences get displayed surrounded by quotes ("). Nevertheless, escape sequences used upon defining character sequences are interpreted immediately.

Example 1:

```
> write(x + 2 + exp(sin(x)));
> write("Hello\n");
x + 2 + exp(sin(x))Hello
> write("Hello","world\n");
Helloworld
> write("Hello","you", 4 + 3, "other persons.\n");
Helloyou7other persons.
```

Example 2:

```
> write("Hello","\n");
Hello
> write(["Hello"],"\n");
["Hello"]
> s = "Hello";
> write(s,[s],"\n");
Hello["Hello"]
> t = "Hello\tyou";
> write(t,[t],"\n");
Hello    you["Hello    you"]
```

Example 3:

```
> write(x + 2 + exp(sin(x))) > "foo.sol";
> readfile("foo.sol");
x + 2 + exp(sin(x))
```

Example 4:

```
> write(x + 2 + exp(sin(x))) >> "foo.sol";
```

See also: **print** (8.112), **printexpansion** (8.113), **printhexa** (8.115), **printfloat** (8.114), **printxml** (8.116), **readfile** (8.124), **autosimplify** (8.14), **display** (8.35), **midpointmode** (8.88), **fullparentheses** (8.60), **evaluate** (8.45)