Users' manual for the Sollya tool

Release 1.0

Laboratoire de l'Informatique du Parallélisme UMR CNRS - ENS Lyon - UCB Lyon 1 - INRIA 5668

Sylvain Chevillard sylvain.chevillard@ens-lyon.fr

Christoph Lauter sylvain.chevillard@ens-lyon.fr

Nicolas Jourdan
sylvain.chevillard@ens-lyon.fr

December 19, 2007

License

The Sollya tool is copyright © 2007 Laboratoire de l'Informatique du Parallélisme - UMR CNRS - ENS Lyon - UCB Lyon 1 - INRIA 5668.

The Sollya tool is open software. It is distributed and can be used, modified and redistributed under the terms of the CeCILL-C licence available at http://www.cecill.info/ and reproduced in the COPYING file of the distribution. The distribution contains parts of other libraries as a support for but not integral part of Sollya. These libraries are reigned by the GNU Lesser General Public License that is available at http://www.gnu.org/licenses/ and reproduced in the COPYING file of the distribution.

Contents

1	Compilation and installation of the Sollya tool	4
2	Introduction	5
3	General principles	7
4	Data types	8
	4.1 Booleans	8
	4.2 Numbers	
	4.3 Intervals	
	4.4 Functions	
	4.5 Strings	
	4.6 Particular values	
	4.7 Lists	
5	Iterative language elements: assignments, conditional statements and loops	13
	5.1 Assignments	13
	5.2 Conditional statements	
	5.3 Loops	
6	Functional language elements: procedures	14

7	Con		15
	7.1	abs	15
	7.2	absolute	15
	7.3	accurateinfnorm	15
	7.4		16
	7.5		16
	7.6		17
	7.7		$17 \\ 17$
	7.8	1	18
	7.9		19
			19
			20
			20
		1 0	20
	7.14	bashexecute	21
	7.15	binary	22
	7.16	boolean	22
	7.17	canonical	23
			24
			$\frac{1}{24}$
			$\frac{1}{25}$
			$\frac{26}{26}$
			$\frac{26}{26}$
			$\frac{20}{26}$
			$\frac{20}{27}$
			27
			27
			28
			28
			29
			29
	7.31	dirtyfindzeros	30
	7.32	dirtyinfnorm	31
	7.33	dirtyintegral	32
	7.34	display	32
	7.35	divide	33
	7.36	double	34
			$\frac{1}{35}$
			35
			36
			36
		-	37
			37
			38
			38
			39
		1	40
			41
		· I	41
	7.49	exponent	42
	7.50	externalplot	42
			43
			45
			45^{-3}
			45
			46
			-0

	$full parentheses \dots \dots \dots \dots \dots \dots \dots \dots \dots $	
7.57	function	. 47
7.58	ge	. 47
7.59	gt	. 48
7.60	guessdegree	. 49
7.61	head	. 49
7.62	$\label{eq:hexadecimal} \text{hexadecimal} \dots \dots$. 50
7.63	honorcoeffprec	. 50
7.64	hopital recursions	. 51
7.65	horner	. 52
	implementpoly	
7.67	$\inf \ \ldots $. 56
7.68	$infnorm \ \dots $. 57
7.69	integer	. 59
7.70	integral	. 59
7.71	isbound	. 60
7.72	$is evaluable \dots \dots$. 61
	le	
	length	
	library	
	listof	
	log	
	$\log 10 \ldots \ldots \ldots \ldots \ldots$	
	$\log^2 1p \ldots $	
	\log^2	
	lt	
	mantissa	
	mid	
	$midpoint mode \dots \dots$	
	minus	
	mult	
	neq	
	nop	
	not	
7.90	numerator	. 71
	off	
	on	
	or	. 72
7.94	parse	. 73
7.95	perturb	. 73
7.96	pi	. 74
	plot	
7.98	plus	. 76
	points	
	Opostscript	
	postscriptfile	
	2power	
	Bpowers	
	lprec	
	precision	
	Sprepend	
	print	
	Brintexpansion	
	Oprintfloat	
) printhexa \cdots	
	print	
	.pr	. 00

7.112proc
7.113procedure
7.114quit
7.115range
7.116rationalapprox
7.117RD
7.118readfile
7.119readxml
7.120 relative
7.121remez
7.122rename
7.123restart
7.124return
7.125revert
7.126RN
7.127round
7.128roundcoefficients
7.129roundcorrectly
7.130RU
7.131RZ
7.132searchgal
7.133simplify
7.134simplifysafe
7.135sin
7.136sinh
7.137sort
7.138sqrt
7.139string
7.140subpoly
7.141substitute
7.142sup
7.143tail
7.144tan
7.145tanh
7.146taylor
7.147taylorrecursions
7.148timing
7.149tripledouble
7.150true
7.151var
7.152verbosity
7.153void
7.154worstcase
7.155write

8 Grammar of the Sollya language

119

1 Compilation and installation of the Sollya tool

The Sollya distribution can be compiled and installed using the usual ./configure, make, make install procedure. Besides a C compiler, Sollya needs the following software libraries and tools to be installed. The ./configure script checks for the installation of the libraries. However Sollya will build without error if some of its external tools is not installed. In this case an error will be produced at runtime.

• MPFR

- MPFI
- PARI version 2.3.0
- libxml2
- gnuplot

The use of the external tool rlwrap is highly recommended but not indispensable.

2 Introduction

Sollya is an interactive tool for handling numerical functions and working with arbitrary precision. It can evaluate functions accurately, compute polynomial approximations of functions, automatically implement polynomials for use in math libraries, plot functions, compute infinite norms, etc. The language Sollya comes with is a full-featured script programming language with support for procedures etc.

Let us begin this manual with an example. Sollya does not allow command line edition; since that may quickly become uncomfortable, we highly suggest to use the software rlwrap with Sollya:

```
/%rlwrap sollya
>
```

Sollya manipulates only univariate functions. The first time that an unbound variable is used, this name is fixed. It will be used to refer to the free variable. For instance, try

```
/% rlwrap sollya
> f = sin(x)/x;
> g = cos(y)-1;
Warning: the identifier "y" is neither assigned to, nor bound to a library function nor equal to the current free variable.
Will interpret "y" as "x".
> g;
cos(x) - 1
>
```

Now, the name x can only be used to refer to the free variable:

```
> x=3;
Warning: the identifier "x" is already bound to the free variable or to a
library function
The command will have no effect.
Warning: the last assignment will have no effect.
>
```

If you really want to unbound x, you can use the **rename** command and change the name of the free variable:

```
> rename(x,y);
Information: the free variable has been renamed from "x" to "y".
> g;
cos(y) - 1
> x=3;
> x;
3
>
```

As you have seen, you can name functions and easily work with. The basic thing to do with a function is to evaluate it at some point:

```
> evaluate f(-2);
Warning: rounding has happened. The value displayed is a faithful rounding of
the true result.
0.454648713412840847698009932955872421351127485723941
```

The printed value is generally a faithful rounding of the exact value at the working precision. The working precision is controlled by the global variable prec:

```
> prec=?;
165
> prec=200;
The precision has been set to 200 bits.
> prec=?;
200
> f(-2);
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
0.45464871341284084769800993295587242135112748572394513418948652
```

Sometimes, a faithful rounding cannot easily be computed. In such a case, an approximated value is printed:

```
> sin(pi);
Warning: rounding has happened. The value displayed is not a faithful rounding of the true result.
-0.379705991005939815725347821572628308530195421950339e-12715
```

The philosophy of Sollya is: whenever something is not exact, print a warning. This explains the warnings in the previous examples. If the result can be shown to be exact, there is no warning:

```
> sin(0);
0
```

Let us finish this Section with a small complete example that shows a bit of what can be done with Sollya:

```
> restart;
The tool has been restarted.
> prec=50;
The precision has been set to 50 bits.
> f=cos(2*exp(x));
> d=[-1/8;1/8];
> p=remez(f,2,d);
> derivativeZeros = dirtyfindzeros(diff(p-f),d);
> derivativeZeros = inf(d).:derivativeZeros:.sup(d);
> max=0; for t in derivativeZeros do
r = evaluate(abs(p-f), t);
if r > max then max=r; argmax=t;;
> print("The infinite norm of", p-f, "is", max, "and is reached at", argmax);
The infinite norm of (-0.41626557294429078481812212) + x * ((-0.1798067204872539)
9037039096583612263e1) + x * (-0.38971068364047456444865247249254026e-1)) - cos(
2 * \exp(x)) is 0.86306625059183635084725239e-3 and is reached at -0.580167296300
62879863317e-1
```

In this example, we define a function f, an interval d and we compute the best degree-4 polynomial approximation of f on d with respect to the infinite norm. In other words, $\max_{x \in d} \{|p(x) - f(x)|\}$ is minimal amongst polynomials with degree not greater than 4. Then, we compute the list of the zeros of the derivative of p - f and add the bounds of d to this list. Finally, we evaluate |p - f| for each point in the list and store the maximum and the point where it is reached. We conclude by printing the result in a formatted way.

Note that you do not really need to use such a script for computing infinite norm; as we will see, the command dirtyinfnorm does this for you.

3 General principles

The first goal of Sollya is to help people using numerical functions and numerical algorithm in a safe way. It is first designed to be used interactively but it can also be used in scripts¹.

One of the originalities of Sollya is to work with multi-precision arithmetic (it uses the MPFR library). For safety purposes, Sollya knows how to use interval arithmetic. It uses the interval arithmetic to produce tight and safe results with the precision required by the user.

The general philosophy of Sollya is: When you can make a computation exactly and sufficiently quickly, do it; when you cannot, do not, unless you have been explicitly asked for.

The precision of the tools is set by the global variable prec. It indicates the number of bits used to represent the constants in Sollya. In general, the variable prec determines the precision of the outputs of commands: more precisely, the command will internally determine what precision should be used during the computations in order to ensure that the output is a faithful result with prec bits.

For decidability and efficiency reasons, this general principle cannot be applied everytime, so be careful. Moreover certain commands are known to be unsafe: they give in general excellent results and give almost prec correct bits in output for everyday examples. However they are just heuristic and should not be used when the result must be safe. See the documentation of each command to know precisely how confident you can be with its result.

A second principle (that comes together with the first one) is: When a computation leads to inexact results, inform the user with a warning. This can be quite irritating in some circumstances: in particular if you are using Sollya within other scripts. The global variable verbosity lets you change the level of verbosity of Sollya. When set to 0, Sollya becomes completely silent on stdout and prints only very important messages on stderr. Increase verbosity if you want more informations about what Sollya is doing. Note that when you affect a value to a global variable, a message is always printed even if verbosity is set to 0. In order to silently affect a global variable, use !:

```
> prec=30;
The precision has been set to 30 bits.
> prec=30!;
>
```

For conviviality reasons, values are displayed in decimal by default. This lets a normal human being understand the numbers he or she manipulates. But since constants are internally represented in binary, this causes permanent conversions that are sources of roundings. Thus you are loosing in accuracy and Sollya is always complaining about inexact results. If you just want to store or communicate your results (to another tools for instance) you can use bit-exact representations available in Sollya. The global variable display defines the way constants are displayed. Here is an example of the five available modes:

¹Remark: some of the behaviours of Sollya slightly change when used in scripts. For example, no prompt is printed.

```
> prec=30!;
> a = 17.25;
> display=decimal;
Display mode is decimal numbers.
0.1725e2
> display=binary;
Display mode is binary numbers.
1.000101_2 * 2^{4}
> display=powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
> a;
69 * 2^{(-2)}
> display=dyadic;
Display mode is dyadic numbers.
> a:
69b-2
> display=hexadecimal;
Display mode is hexadecimal numbers.
> a;
0x1.14p4
>
```

As always, the symbol e means $\times 10^{\square}$. The same way the symbol e means $\times 2^{\square}$. The symbol e means $\times 16^{\square}$ and is used only with the following e0x prefix. The prefix e0x indicates that the digits of the following number until a symbol e0 or whitespace are hexadecimal. The suffix e2 indicates to Sollya that the previous number has been written in binary. Sollya can parse these notations even if you are not in the corresponding display mode, so you can always use them.

You can also use memory-dump hexadecimal notation frequently used to represent IEEE 754 double and single precision numbers. Since this notation does not allow for exactly representing numbers with arbitrary precision, there is no corresponding display mode. However, the commands printhexa respectively printfloat round the value to the nearest double respectively single. The number is then printed in hexadecimal as the integer number corresponding to the memory representation of the IEEE 754 double or single number:

```
> printhexa(a);
0x403140000000000
> printfloat(a);
0x418a0000
```

Sollya can parse these memory-dump hexadecimal notation back in any display mode.

4 Data types

Sollya has a (very) basic system of types. If you try to perform an illicit operation (such as adding a number and a string, for instance), you will get a type error. Let us see the available data types.

4.1 Booleans

There are two special values true and false. Boolean expressions can be constructed using the boolean connectors && (and), || (or), ! (not), and comparisons.

The comparison operators <, <=, > and >= can only be used between two numbers or constant expressions.

The comparison operators == and != are polymorphic. You can use it to compare any two objects, like two strings, two intervals, etc. Note that testing the equality between two functions will return true

if and only if the expression trees representing the two functions are exactly the same. See 7.43 for an exception concerning the special object error. Example:

```
> 1+x==1+x;
true
> 1+x==x+1;
false
```

4.2 Numbers

As seen above, Sollya represents numbers as floating-point values with the current precision prec. A number in an expression is rounded to the precision prec when the expression gets evaluated:

```
> prec=12!;
> 4097;
Warning: Rounding occured when converting the constant "4097" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
4096
> 4098;
4098
> 4097+1;
Warning: Rounding occured when converting the constant "4097" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
4096
```

Note that each variable has its own precision that corresponds to the value of prec when the variable was set. Thus you can work with variables having a precision bigger than the current precision.

The same way, if you define a function that refers to some constant, this constant is stored in the function with the current precision and will keep this value in the future, even if prec becomes smaller.

If you define a function that refers to some variable, the precision of the variable is kept, independently of the current precision:

```
> prec=50!;
> a = 4097;
> prec=12!;
> f = x+a;
> g = x+4097;
Warning: Rounding occured when converting the constant "4097" to floating-point with 12 bits.
If safe computation is needed, try to increase the precision.
> prec=50!;
> f;
4097 + x
> g;
4096 + x
```

4.3 Intervals

Intervals are composed of two numbers or constant expressions representing the lower and the upper bound. These values are separated either by commas or semi-colons:

```
> d=[1;2];
> d2=[1,1+1];
> d==d2;
true
```

If bounds are defined by constant expressions, these are evaluated to floating-point numbers using the current precision. Numbers or variables containing numbers keep their precision for the interval bounds. Interval bound evaluation is performed in a way that ensures the inclusion property: all points in the original, unevaluated interval will be contained in the interval with its bounds evaluated to floating-point numbers. Remark that evaluation bounds defined by constant expressions includes π :

```
> prec=30!;
> a=4097;
> prec=12!;
> d=[4096; a];
> prec=30!;
> d;
[4096;4097]
> [-pi;pi];
[-0.31415926591e1;0.31415926591e1]
```

You can get the upper-bound (respectively the lower-bound)) of an interval with the function sup (respectively inf). The middle of the interval is got with the function mid. Note that these functions can also be used on numbers (in that case, the number is interpreted as an interval containing only one single point. Thus the functions inf, mid and sup are just the identity):

```
> d=[1;3];
> inf(d);
1
> mid(d);
2
> sup(4);
4
```

4.4 Functions

Sollya knows only functions with one single variable. The first time in a session that an unbound name is used (without being assigned) it determines the name used to refer to the free variable.

The basic functions available in Sollya are the following:

- +, -, *, /, ^
- sqrt
- abs
- sin, cos, tan, sinh, cosh, tanh
- asin, acos, atan, asinh, atanh
- exp, expm1 (defined as expm1(x) = exp(x) 1)
- log (neperian logarithm), log2 (binary logarithm), log10 (decimal logarithm), log1p (defined as log1p(x) = log(1+x))
- erf, erfc

The constant π is available through the keword pi as a 0-ary function: its behavior is exactly the same as if it were a constant with an infinite precision:

```
> display=binary!;
> prec=12!;
> a=pi;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
1.10010010001_2 * 2^(1)
> prec=30!;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of the true result.
1.1001001000011111101101010101_2 * 2^(1)
```

4.5 Strings

Anything written between quotes is interpreted as a string. The infix operator @ concatenates two strings. To get the length of a string, use the length function. You can access the *i*-th character of a string using brackets (see the example above). There is no character type in Sollya: the *i*-th character of a string is returned as a string itself.

```
> s1 = "Hello "; s2 = "World!";
> s = s1@s2;
> length(s);
12
> s[0];
H
> s[11];
!
```

4.6 Particular values

Sollya knows some particular values. These values do not really have a type but they can be stored in variables and in lists. A (possibly not exhaustive) list of such values is the following:

- on, off (see sections 7.92 and 7.91)
- dyadic, powers, binary, decimal, hexadecimal (see sections 7.39, 7.103, 7.15, 7.25 and 7.62)
- file, postscript, postscriptfile (see sections 7.53, 7.100 and 7.101)
- RU, RD, RN, RZ
- absolute, relative (see sections 7.2 and 7.120)
- ullet double, doubleextended, doubledouble, tripledouble (see sections 7.36, 7.38, 7.37 and 7.149)
- D, DE, DD, TD (see sections 7.36, 7.38, 7.37 and 7.149)
- perturb (see section 7.95)
- honorcoeffprec (see section 7.63)
- default (see section 7.26)
- error (see section 7.43)
- void (see section 7.153)

4.7 Lists

Objects can be grouped into lists. A list can contain elements with different types. As for strings, you can concatenate two lists with Q. The function length gives also the length of a list.

You can prepend an element to a list using .: (in $\mathcal{O}(1)$) and you can append an element to a list using :. (in $\mathcal{O}(n)$). The following example illustrates some features:

```
> list = [| "foo" |];
> list = list:.1;
> list = "bar".:list;
> list;
[|bar, foo, 1|]
> list[1];
foo
> list@list;
[|bar, foo, 1, bar, foo, 1|]
```

Lists can be considered as arrays and elements of lists can be referenced using brackets. Possible indices start at 0. The following example illustrates this point:

```
> 1 = [|1,2,3,4,5|];
> 1;
[|1, 2, 3, 4, 5|]
> 1[3];
4
```

Remark that the complexity for accessing an element of the list using indices is $\mathcal{O}(n)$.

Lists may contain ellipses indicated by ,..., between elements that are constant and evaluate to integers that are incrementally ordered. Sollya translates such ellipses to the full list upon evaluation. Using ellipses between elements that are not constants is not allowed. This feature is provided for ease of programmation; remark that the complexity of expanding such lists is high. For illustration, see the following example:

```
> [|1,...,5|];
[|1, 2, 3, 4, 5|]
> [|-5,...,5|];
[|-5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5|]
> [|3,...,1|];
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error
> [|true,...,false|];
Warning: at least one of the given expressions or a subexpression is not correctly typed
or its evaluation has failed because of some error on a side-effect.
error
```

Lists may be continued to infinity by means of the . . . indicator after the last element given. At least one element must explicitly be given. If the last element given is a constant expression that evaluates to an integer, the list is considered as continued to infinity by all integers greater than that last element. If the last element is another object, the list is considered as continued to infinity by reduplicating this last element. Remark that bracket notation is supported for such end-elliptic lists even for implicitly given elements. However, evaluation complexity is high. Combinations of ellipses inside a list and in its end are possible. The usage of lists described here is best illustrated by the following examples:

```
> 1 = [|1,2,true,3...|];
> 1;
[|1, 2, true, 3...|]
> 1[2];
true
> 1[3];
3
> 1[4];
4
> 1[1200];
1200
> 1 = [|1,...,5,true...|];
> 1;
[|1, 2, 3, 4, 5, true...|]
> 1[1200];
true
```

5 Iterative language elements: assignments, conditional statements and loops

5.1 Assignments

Sollya has two different assignment operators, = and :=. The assignment operator = assigns its right-hand-object "as is", i.e. without evaluating functional expressions. For instance, $\mathbf{i} = \mathbf{i} + \mathbf{1}$; will dereferentiate the identifier \mathbf{i} with some content, notate it y, build up the expression (function) y+1 and assign this expression back to \mathbf{i} . In the example, if \mathbf{i} stood for the value 1000, the statement $\mathbf{i} = \mathbf{i} + \mathbf{1}$; will assign 1000+1 and not 1001 to \mathbf{i} . The assignment operator := evaluates constant functional expressions before assigning them. On other expressions it behaves like =. Still in the example, the statement $\mathbf{i} := \mathbf{i} + \mathbf{1}$; really assigns 1001 to \mathbf{i} .

Both Sollya assignment operators support indexing of lists or strings elements using brackets on the left-hand-side of the assignment operator. The indexed element of the list or string gets replaced by the right-hand-side of the assignment operator. When indexing strings this way, that right-hand side must evaluate to a string of length 1. End-elliptic lists are supported with their usual semantic for this kind of assignment. When referencing and assigning a value in the implicit part of the end-elliptic list, the list gets expanded to the corresponding length. The indexing of lists on left-hand sides of assignments is reduced to the first order. Multiple indexing of lists of lists is not supported for complexity reasons.

The following examples well illustrate the behaviour of assignment statements:

```
> autosimplify = off;
Automatic pure tree simplification has been deactivated.
> i = 1000;
> i = i + 1;
> print(i);
1000 + 1
> i := i + 1;
> print(i);
1002
> 1 = [|1,...,5|];
> print(1);
[|1, 2, 3, 4, 5|]
> 1[3] = 1[3] + 1;
> 1[4] := 1[4] + 1;
> print(1);
[|1, 2, 3, 4 + 1, 6|]
> 1[5] = true;
> 1;
[|1, 2, 3, 5, 6, true|]
> s = "Hello world";
> s;
Hello world
> s[1] = "a";
> s;
Hallo world
> 1 = [|true,1,...,5,9...|];
> 1;
[|true, 1, 2, 3, 4, 5, 9...|]
> 1[13] = "Hello";
> 1;
[|true, 1, 2, 3, 4, 5, 9, 10, 11, 12, 13, 14, 15, Hello...|]
```

5.2 Conditional statements

TODO

If, If-Else, syntaxe bizarre du if-Else.

5.3 Loops

For from by, for in, while TODO

6 Functional language elements: procedures

procedures commes objets, declaration, recursion et identifiants, variables locales, return, procedures en argument, evaluation retardee de constantes etc.

TODO

Sollya also supports external procedures, i.e. procedures written in $\tt C$ (or some other language) and dynamically bound to Sollya identifiers. See 7.51 for details.

7 Commands and functions

7.1 abs

Name: abs

the absolute value.

Description:

• **abs** is the absolute value function. **abs**(x)= $\begin{cases} x & x > 0 \\ -x & x \le 0 \end{cases}$.

7.2 absolute

Name: absolute

indicates an absolute error for **externalplot**

Usage:

absolute: absolute relative

Description:

• The use of **absolute** in the command **externalplot** indicates that during plotting in **externalplot** an absolute error is to be considered. See **externalplot** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l
mpfr");
> externalplot("./externalplotexample",absolute,exp(x),[-1/2;1/2],12,perturb);
```

See also: externalplot (7.50), relative (7.120), bashexecute (7.14)

7.3 accurateinfnorm

Name: accurateinfnorm

computes a faithful rounding of the infinite norm of a function

Usage:

Parameters:

- function represents the function whose infinite norm is to be computed
- range represents the infinite norm is to be considered on
- constant represents the number of bits in the significant of the result
- \bullet exclusion range 1 through exclusion range n represent ranges to be excluded

Description:

• The command accurateinfnorm computes an upper bound to the infinite norm of function function in range. This upper bound is the least floating-point number greater than the value of the infinite norm that lies in the set of dyadic floating point numbers having constant significant mantissa bits. This means the value accurateinfnorm evaluates to is at the time an upper bound and a faithful rounding to constant bits of the infinite norm of function function on range range. If given, the fourth and further arguments of the command accurateinfnorm, exclusion range 1 through exclusion range n the infinite norm of the function function is not to be considered on.

Example 1:

```
> p = remez(exp(x), 5, [-1;1]);
> accurateinfnorm(p - exp(x), [-1;1], 20);
0.45205641072243e-4
> accurateinfnorm(p - exp(x), [-1;1], 30);
0.45205621802324458e-4
> accurateinfnorm(p - exp(x), [-1;1], 40);
0.45205621769406345578e-4
```

Example 2:

```
> p = remez(exp(x), 5, [-1;1]);
> midpointmode = on!;
> infnorm(p - exp(x), [-1;1]);
0.45205~5/7~e-4
> accurateinfnorm(p - exp(x), [-1;1], 40);
0.45205621769406345578e-4
```

See also: infnorm (7.68), dirtyinfnorm (7.32), checkinfnorm (7.19), remez (7.121), diam (7.29)

7.4 acos

Name: acos

the arccosine function.

Description:

- acos is the inverse of the function cos: $a\cos(y)$ is the unique number $x \in [0, \pi]$ such that $\cos(x) = y$.
- It is defined only for $y \in [-1; 1]$.

See also: $\cos (7.23)$

7.5 acosh

Name: acosh

the arg-hyperbolic cosine function.

Description:

- **acosh** is the inverse of the function **cosh**: **acosh**(y) is the unique number $x \in [0; +\infty]$ such that $\mathbf{cosh}(x) = y$.
- It is defined only for $y \in [0; +\infty]$.

See also: $\cosh (7.24)$

7.6 and

Name: &&

boolean AND operator

Usage:

$$expr1 \&\& expr2 : (boolean, boolean) \rightarrow boolean$$

Parameters:

• expr1 and expr2 represent boolean expressions

Description:

• && evaluates to the boolean AND of the two boolean expressions expr1 and expr2. && evaluates to true iff both expr1 and expr2 evaluate to true.

Example 1:

```
> true && false;
false
```

Example 2:

```
> (1 == exp(0)) && (0 == log(1));
true
```

See also: || (7.93), ! (7.89)

7.7 append

Name: :.

add an element at the end of a list.

Usage:

$$L: x : (list, any type) \rightarrow list$$

Parameters:

- L is a list (possibly empty).
- \bullet x is an object of any type.

Description:

- :. adds the element x at the end of the list L.
- Note that since x may be of any type, it can be in particular a list.

Example 1:

```
> [|2,3,4|]:.5;
[|2, 3, 4, 5|]
```

Example 2:

```
> [|1,2,3|]:.[|4,5,6|];
[|1, 2, 3, [|4, 5, 6|]|]
```

Example 3:

```
> [||]:.1;
[|1|]
```

See also: .: (7.106), @ (7.21)

7.8 asciiplot

Name: asciiplot

plots a function in a range using ASCII characters

Usage:

```
\mathbf{asciiplot}(function, range) : (function, range) \rightarrow \mathsf{void}
```

Parameters:

- function represents a function to be plotted
- range represents a range the function is to be plotted in


Description:

• asciiplot plots the function function in range range using ASCII characters. If Sollya is connected to a terminal, the size of the plot is determined by the size of the terminal. If not, the plot is of fixed size. The function is evaluated on a number of points equal to the number of columns available. Its value is rounded to the next integer in the range of lines available. A letter "x" is written at this place. If zero is in the hull of the image domain of the function, a x-axis is displayed. If zero is in range, an y-axis is displayed. If the function is constant or if the range is reduced to one point, the function is evaluated to a constant and the constant is displayed instead of a plot.

Example 1:

```
> asciiplot(exp(x),[1;2]);
                                                                                    хx
                                                                                  XX
                                                                             хx
                                                                           xx
                                                                       XXX
                                                                     XX
                                                                  xxx
                                                               xx
                                                            XXX
                                                         xxx
                                                  XXX
                                              XXX
                                          xxxx
                                      XXX
                                  XXXX
                             XXXX
                         xxxx
                    xxxx
               XXXXX
         XXXXX
   xxxxx
XXX
```

Example 2:



Example 3:

```
> asciiplot(5,[-1;1]);
5
```

Example 4:

```
> asciiplot(exp(x),[1;1]);
0.271828182845904523536028747135266249775724709369998e1
```

See also: \mathbf{plot} (7.97)

7.9 asin

Name: asin

the arcsine function.

Description:

- asin is the inverse of the function sin: asin(y) is the unique number $x \in [-\pi/2; \pi/2]$ such that $\sin(x)=y$.
- It is defined only for $y \in [-1; 1]$.

See also: sin (7.135)

7.10 asinh

Name: asinh

the arg-hyperbolic sine function.

Description:

- **asinh** is the inverse of the function **sinh**: **asinh**(y) is the unique number $x \in [-\infty; +\infty]$ such that $\sinh(x)=y$.
- It is defined for every real number y.

See also: sinh (7.136)

7.11 atan

Name: atan

the arctangent function.

Description:

- atan is the inverse of the function tan: atan(y) is the unique number $x \in [-\pi/2; +\pi/2]$ such that tan(x)=y.
- It is defined for every real number y.

See also: tan (7.144)

7.12 atanh

Name: atanh

the hyperbolic arctangent function.

Description:

- atanh is the inverse of the function tanh: atanh(y) is the unique number $x \in [-\infty; +\infty]$ such that tanh(x)=y.
- It is defined only for $y \in [-1; 1]$.

See also: tanh (7.145)

7.13 autosimplify

Name: autosimplify

activates, deactivates or inspects the value of the automatic simplification state variable

Usage:

```
autosimplify = activation\ value : on|off \rightarrow void

autosimplify = activation\ value ! : on|off \rightarrow void

autosimplify = ? : void \rightarrow on|off
```

Parameters:

• activation value represents on or off, i.e. activation or deactivation

Description:

• An assignment **autosimplify** = activation value, where activation value is one of **on** or **off**, activates respectively deactivates the automatic safe simplification of expressions of functions generated by the evaluation of commands or in argument of other commands. Sollya commands like **remez**, **taylor** or **rationalapprox** sometimes produce expressions that can be simplified. Constant subexpressions can be evaluated to dyadic floating-point numbers, monomials with coefficients 0 can be eliminated. Further, expressions indicated by the user perform better in many commands when simplified before being passed in argument to a commans. When the automatic simplification of expressions is activated, Sollya automatically performs a safe (not value changing) simplification

process on such expression. The automatic generation of subexpressions can be annoying, in particular if it takes too much time for not enough usage. Further the user might want to inspect the structure of the expression tree returned by a command. In this case, the automatic simplification should be deactivated. If the assignment ${\bf autosimplify} = activation\ value$ is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

• The expression **autosimplify** = ? evaluates to a variable of type on|off, indicating whether or not the automatic simplifications of expressions of functions is activated.

Example 1:

```
> autosimplify = on !;
> print(x - x);
0
> autosimplify = off;
Automatic pure tree simplification has been deactivated.
> print(x - x);
x - x
```

Example 2:

```
> autosimplify = on !;
> print(rationalapprox(sin(pi/5.9),7));
0.5
> autosimplify = off !;
> print(rationalapprox(sin(pi/5.9),7));
1 / 2
```

Example 3:

```
> autosimplify = ?;
on
```

See also: print (7.107), prec (7.104), points (7.99), diam (7.29), display (7.34), verbosity (7.152), canonical (7.17), taylorrecursions (7.147), timing (7.148), fullparentheses (7.56), midpointmode (7.84), hopitalrecursions (7.64), remez (7.121), rationalapprox (7.116), taylor (7.146)

7.14 bashexecute

Name: **bashexecute** executes a shell command.

Usage:

 $\mathbf{bashexecute}(command): \mathsf{string} \to \mathsf{void}$

Parameters:

• command is a command to be interpreted by the shell.

Description:

- bashexecute(command) lets the shell interpret command. It is useful to execute some external code within Sollya.
- bashexecute does not return anything. It just executes its argument. However, if *command* produces an output in a file, this result can be imported in Sollya with help of commands like execute, readfile and parse.

Example 1:

```
> bashexecute("ls /");
bin
boot
cdrom
dev
emul
etc
home
initrd
initrd.img
lib
1ib32
lib64
lost+found
media
mnt
opt
proc
root
sbin
srv
sys
tmp
usr
var
vmlinuz
```

See also: execute (7.45), readfile (7.118), parse (7.94)

7.15 binary

Name: hexadecimal

special value for global state display

Description:

• hexadecimal is a special value used for the global state display. If the global state display is equal to hexadecimal, all data will be output in binary notation. As any value it can be affected to a variable and stored in lists.

See also: decimal (7.25), dyadic (7.39), powers (7.103), hexadecimal (7.62)

7.16 boolean

Name: boolean

keyword representing a boolean type

Usage:

boolean: type type

Description:

• boolean represents the boolean type for declarations of external procedures by means of externalproc. Remark that in contrast to other indicators, type indicators like boolean cannot be handled outside the externalproc context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), constant (7.22), function (7.57), integer (7.69), list of (7.76), range (7.115), string (7.139)

7.17 canonical

Name: canonical

brings all polynomial subexpressions of an expression to canonical form or activates, deactivates or checks canonical form printing

Usage:

```
 \begin{array}{l} \mathbf{canonical}(\mathit{function}) : \mathsf{function} \to \mathsf{function} \\ \mathbf{canonical} = \mathit{activation} \ \mathit{value} : \mathsf{on}|\mathsf{off} \to \mathsf{void} \\ \mathbf{canonical} = \mathit{activation} \ \mathit{value} \ ! : \mathsf{on}|\mathsf{off} \to \mathsf{void} \\ \mathbf{canonical} = ? : \mathsf{void} \to \mathsf{on}|\mathsf{off} \\ \end{array}
```

Parameters:

- function represents the expression to be rewritten in canonical form
- activation value represents on or off, i.e. activation or deactivation

Description:

- The command **canonical** rewrites the expression representing the function function in a way such that all polynomial subexpressions (or the whole expression itself, if it is a polynomial) are written in canonical form, i.e. as a sum of monomials in the canonical base. The canonical base is the base of the integer powers of the global free variable. The command **canonical** does not endanger the safety of computations even in Sollya's floating-point environment: the function returned is mathematically equal to the function function.
- An assignment **canonical** = activation value, where activation value is one of **on** or **off**, activates respectively deactivates the automatic printing of polynomial expressions in canonical form, i.e. as a sum of monomials in the canonical base. If automatic printing in canonical form is deactivated, automatic printing yield to displaying polynomial subexpressions in Horner form. If the assignment **canonical** = activation value is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.
- The expression **canonical** = ? evaluates to a variable of type on|off, indicating whether or not the automatic printing of subexpressions in canonical form is activated. If automatic printing in canonical form is deactivated, automatic printing yield to displaying polynomial subexpressions in Horner form.

Example 1:

```
> print(canonical(1 + x * (x + 3 * x^2));
> print(canonical((x + 1)^7));
1 + 7 * x + 21 * x^2 + 35 * x^3 + 35 * x^4 + 21 * x^5 + 7 * x^6 + x^7
```

Example 2:

```
> print(canonical(exp((x + 1)^5) - log(asin(((x + 2) + x)^4 * (x + 1)) + x)));
exp(1 + 5 * x + 10 * x^2 + 10 * x^3 + 5 * x^4 + x^5) - log(asin(16 + 80 * x + 16 0 * x^2 + 160 * x^3 + 80 * x^4 + 16 * x^5) + x)
```

Example 3:

```
> canonical = ?;
off
> (x + 2)^9;
512 + x * (2304 + x * (4608 + x * (5376 + x * (4032 + x * (2016 + x * (672 + x * (2016 +
     (144 + x * (18 + x))))))))
> canonical = on;
Canonical automatic printing output has been activated.
> (x + 2)^9;
512 + 2304 * x + 4608 * x^2 + 5376 * x^3 + 4032 * x^4 + 2016 * x^5 + 672 * x^6 +
    144 * x^7 + 18 * x^8 + x^9
> canonical = ?;
on
> canonical = off!;
> (x + 2)^9;
512 + x * (2304 + x * (4608 + x * (5376 + x * (4032 + x * (2016 + x * (672 + x *
      (144 + x * (18 + x)))))))))
```

See also: horner (7.65), print (7.107)

7.18 ceil

Name: ceil

the usual function ceil.

Description:

- ceil is defined as usual: ceil(x) is the smallest integer y such that $y \ge x$.
- It is defined for every real number x.

See also: floor (7.55)

7.19 checkinfnorm

Name: checkinfnorm

checks whether the infinite norm of a function is bounded by a value

Usage:

 $\mathbf{checkinfnorm}(function, range, constant) : (function, range, constant) \rightarrow \mathsf{boolean}$

Parameters:

- function represents the function whose infinite norm is to be checked
- range represents the infinite norm is to be considered on
- constant represents the upper bound the infinite norm is to be checked to

Description:

• The command **checkinfnorm** checks whether the infinite norm of the given function function in the range range can be proven (by Sollya) to be less than the given bound bound. This means, if **checkinfnorm** evaluates to **true**, the infinite norm has been proven (by Sollya's interval arithmetic) to be less than the bound. If **checkinfnorm** evaluates to **false**, there are two possibilities: either the bound is less than or equal to the infinite norm of the function or the bound is greater than the infinite norm but Sollya could not conclude using its internal interval arithmetic. **checkinfnorm** is sensitive to the global variable **diam**. The smaller **diam**, the more time Sollya will spend on the evaluation of **checkinfnorm** in order to prove the bound before returning **false** although the infinite is bounded by the bound. If **diam** is equal to 0, Sollya will eventually spend

infinite time on instances where the given bound bound is less or equal to the infinite norm of the function function in range range. In contrast, with **diam** being zero, **checkinfnorm** evaluates to **true** iff the infinite norm of the function in the range is bounded by the given bound.

Example 1:

```
> checkinfnorm(sin(x),[0;1.75], 1);
true
> checkinfnorm(sin(x),[0;1.75], 1/2); checkinfnorm(sin(x),[0;20/39],
false
> 1/2);
true
```

Example 2:

```
> p = remez(exp(x), 5, [-1;1]);
> b = dirtyinfnorm(p - exp(x), [-1;1]);
> checkinfnorm(p - exp(x), [-1;1], b);
false
> b1 = round(b, 20, RU);
> checkinfnorm(p - exp(x), [-1;1], b1);
true
> b2 = round(b, 25, RU);
> checkinfnorm(p - exp(x), [-1;1], b2);
false
> diam = 1b-20!;
> checkinfnorm(p - exp(x), [-1;1], b2);
true
```

See also: infnorm (7.68), dirtyinfnorm (7.32), accurateinfnorm (7.3), remez (7.121), diam (7.29)

7.20 coeff

Name: coeff

gives the coefficient of degree n of a polynomial

Usage:

```
\mathbf{coeff}(f,n): (function, integer) \rightarrow constant
```

Parameters:

- f is a function (usually a polynomial).
- \bullet *n* is an integer

Description:

- If f is a polynomial, $\mathbf{coeff}(f, n)$ returns the coefficient of degree n in f.
- If f is a function that is not a polynomial, $\mathbf{coeff}(f, n)$ returns 0.

Example 1:

```
> coeff((1+x)^5,3);
10
```

Example 2:

```
> coeff(sin(x),0);
0
```

See also: **degree** (7.27)

7.21 concat

Name: @

concatenates two lists or strings.

Usage:

```
L1@L2: (list, list) \rightarrow list
string1@string2: (string, string) \rightarrow string
```

Parameters:

- L1 and L2 are two lists.
- string1 and string2 are two strings.

Description:

• @ concatenates two lists or strings.

Example 1:

```
> [|1,...,3|]@[|7,8,9|];
[|1, 2, 3, 7, 8, 9|]
```

Example 2:

```
> "Hello "@"World!";
Hello World!
```

See also: (7.106), (7.7)

7.22 constant

Name: constant

keyword representing a constant type

Usage:

constant: type type

Description:

• **constant** represents the **constant** type for declarations of external procedures by means of **externalproc**. Remark that in contrast to other indicators, type indicators like **constant** cannot be handled outside the **externalproc** context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), boolean (7.16), function (7.57), integer (7.69), list of (7.76), range (7.115), string (7.139)

7.23 \cos

Name: cos

the cosine function.

Description:

- cos is the usual cosine function.
- It is defined for every real number x.

See also: $a\cos(7.4)$, $\sin(7.135)$, $\tan(7.144)$

$7.24 \cosh$

Name: cosh

the hyperbolic cosine function.

Description:

- cosh is the usual hyperbolic function: $\cosh(x) = \frac{e^x + e^{-x}}{2}$.
- It is defined for every real number x.

See also: **acosh** (7.5), **sinh** (7.136), **tanh** (7.145), **exp** (7.46)

7.25 decimal

Name: decimal

special value for global state display

Description:

• **decimal** is a special value used for the global state **display**. If the global state **display** is equal to **decimal**, all data will be output in decimal notation. As any value it can be affected to a variable and stored in lists.

See also: dyadic (7.39), powers (7.103), hexadecimal (7.62), binary (7.15)

7.26 default

Name: default

default value for some commands.

$\label{eq:Description:} Description:$

- **default** is a special value and is replaced by something depending on the context where it is used. It can often be used as a joker, when you want to specify one of the optional parameters of a command and not the others: set the value of uninterresting parameters to **default**.
- Global variables can be reset by affecting them the special value **default**.

Example 1:

```
> p = remez(exp(x),5,[0;1],default,1e-5);
> q = remez(exp(x),5,[0;1],1,1e-5);
> p==q;
true
```

Example 2:

```
> prec=?;
165
> prec=200;
The precision has been set to 200 bits.
> prec=default;
The precision has been set to 165 bits.
```

7.27 degree

Name: degree

gives the degree of a polynomial.

Usage:

```
\mathbf{degree}(f): function \rightarrow integer
```

Parameters:

• f is a function (usually a polynomial).

Description:

- If f is a polynomial, $\mathbf{degree}(f)$ returns the degree of f.
- Contrary to the usage, Sollya considers that the degree of the null polynomial is 0.
- If f is a function that is not a polynomial, $\mathbf{degree}(f)$ returns -1.

Example 1:

```
> degree((1+x)*(2+5*x^2));
3
> degree(0);
0
```

Example 2:

```
> degree(sin(x));
-1
```

See also: coeff (7.20)

7.28 denominator

Name: denominator

gives the denominator of an expression

Usage:

```
\mathbf{denominator}(\mathit{expr}): \mathsf{function} \to \mathsf{function}
```

Parameters:

• expr represents an expression

Description:

• If expr represents a fraction expr1/expr2, **denominator** (expr) returns the denominator of this fraction, i.e. expr2. If expr represents something else, **denominator** (expr) returns 1. Note that for all expressions expr, numerator(expr) / denominator(expr) is equal to expr.

Example 1:

```
> denominator(5/3);
3
```

Example 2:

```
> denominator(exp(x));
1
```

Example 3:

```
> a = 5/3;
> b = numerator(a)/denominator(a);
> print(a);
5 / 3
> print(b);
5 / 3
```

Example 4:

```
> a = exp(x/3);
> b = numerator(a)/denominator(a);
> print(a);
exp(x / 3)
> print(b);
exp(x / 3)
```

See also: numerator (7.90)

7.29 diam

Name: diam

parameter used in safe algorithms of Sollya and controlling the maximal length of the involved intervals.

Description:

- diam is a global variable. Its value represents the maximal length allowed for intervals involved in safe algorithms of Sollya (namely infnorm, checkinfnorm, accurateinfnorm, integral, find-zeros).
- More precisely, **diam** is relative to the diameter of the input interval of the command. For instance, suppose that **diam**=1e-5: if **infnorm** is called on interval [0, 1], the maximal length of an interval will be 1e-5. But if it is called on interval [0, 1e-3], the maximal length will be 1e-8.

See also: infnorm (7.68), checkinfnorm (7.19), accurateinfnorm (7.3), integral (7.70), findzeros (7.54)

7.30 diff

Name: diff

differentiation operator

Usage:

 $\mathbf{diff}(\mathit{function}): \mathsf{function} \to \mathsf{function}$

Parameters:

• function represents a function

Description:

• diff(function) returns the symbolic derivative of the function function by the global free variable. If function represents a function symbol that is externally bound to some code by library, the derivative is performed as a symbolic annotation to the returned expression tree.

Example 1:

```
> diff(sin(x));
cos(x)
```

Example 2:

```
> diff(x);
1
```

Example 3:

```
> diff(x^x);
x^x * (1 + log(x))
```

See also: **library** (7.75)

7.31 dirtyfindzeros

Name: dirtyfindzeros

gives a list of numerical values listing the zeros of a function on an interval.

Usage:

 $\mathbf{dirtyfindzeros}(f,I):(\mathsf{function},\mathsf{range})\to\mathsf{list}$

Parameters:

- \bullet f is a function.
- I is an interval.

Description:

- $\operatorname{dirtyfindzeros}(f,I)$ returns a list containing some zeros of f in the interval I. The values in the list are numerical approximation of the exact zeros. The precision of these approximations is approximately the precision stored in prec . If f does not have two zeros very close to each other, it can be expected that all zeros are listed. However, some zeros may be forgotten. This command should be considered as a numerical algorithm and should not be used if safety is critical.
- More precisely, the algorithm relies on global variables **prec** and **points** and is the following: let n be the value of variable **points** and t be the value of variable **prec**.
 - Evaluate |f| at n evenly distributed points in the interval I. the precision used is automatically chosen in order to ensure that the sign is correct.
 - Whenever f changes its sign for two consecutive points, find an approximation x of its zero with precision t using Newton's algorithm. The number of steps in Newton's iteration depends on t: the precision of the approximation is supposed to be doubled at each step.
 - Add this value to the list.

Example 1:

```
> dirtyfindzeros(sin(x),[-5;5]);
[|-0.31415926535897932384626433832795028841971693993750801e1, 0, 0.3141592653589
7932384626433832795028841971693993750801e1|]
```

Example 2:

```
> L1=dirtyfindzeros(x^2*sin(1/x),[0;1]);
> points=1000!;
> L2=dirtyfindzeros(x^2*sin(1/x),[0;1]);
> length(L1); length(L2);
18
25
```

See also: **prec** (7.104), **points** (7.99), **findzeros** (7.54)

7.32 dirtyinfnorm

Name: dirtyinfnorm

computes a numerical approximation of the infinite norm of a function on an interval.

Usage:

```
\operatorname{\mathbf{dirtyinfnorm}}(f,I):(\operatorname{\mathsf{function}},\operatorname{\mathsf{range}})\to\operatorname{\mathsf{constant}}
```

Parameters:

- \bullet f is a function.
- I is an interval.

Description:

- **dirtyinfnorm**(f,I) computes an approximation of the infinite norm of the given function f on the interval I, e.g. $\max_{x \in I} \{|f(x)|\}$.
- The interval must be bound. If the interval contains one of -Inf or +Inf, the result of **dirtyinfnorm** is NaN.
- The result of this command depends on the global variables **prec** and **points**. Therefore, the returned result is generally a good approximation of the exact infinite norm, with precision **prec**. However, the result is generally underestimated and should not be used when safety is critical. Use **infnorm** instead.
- The following algorithm is used: let *n* be the value of variable **points** and *t* be the value of variable **prec**.
 - Evaluate |f| at n evenly distributed points in the interval I. The evaluation are faithful roundings of the exact results at precision t.
 - Whenever the derivative of f changes its sign for two consecutive points, find an approximation x of its zero with precision t. Then compute a faithful rounding of |f(x)| at precision t.
 - Return the maximum of all computed values.

Example 1:

```
> dirtyinfnorm(sin(x),[-10;10]);
1
```

Example 2:

```
> prec=15!;
> dirtyinfnorm(exp(cos(x))*sin(x),[0;5]);
0.145855712891e1
> prec=40!;
> dirtyinfnorm(exp(cos(x))*sin(x),[0;5]);
0.14585285371358622797e1
> prec=100!;
> dirtyinfnorm(exp(cos(x))*sin(x),[0;5]);
0.1458528537136237644438147455024510015e1
> prec=200!;
> dirtyinfnorm(exp(cos(x))*sin(x),[0;5]);
0.14585285371362376444381474550238417182992140879936823740941534981883e1
```

Example 3:

```
> dirtyinfnorm(x^2, [log(0);log(1)]);
@NaN@
```

See also: prec (7.104), points (7.99), infnorm (7.68), checkinfnorm (7.19)

7.33 dirtyintegral

Name: dirtyintegral

computes a numerical approximation of the integral of a function on an interval.

Usage:

```
\mathbf{dirtyintegral}(f,I):(\mathsf{function},\mathsf{range})\to\mathsf{constant}
```

Parameters:

- \bullet f is a function.
- I is an interval.

Description:

- $\operatorname{dirtyintegral}(f,I)$ computes an approximation of the integral of f on I.
- The interval must be bound. If the interval contains one of -Inf or +Inf, the result of **dirtyintegral** is NaN, even if the integral has a meaning.
- The result of this command depends on the global variables **prec** and **points**. The method used is the trapezium rule applied at n evenly distributed points in the interval, where n is the value of global variable **points**.
- This command computes a numerical approximation of the exact value of the integral. It should not be used if safety is critical. In this case, use command **integral** instead.
- Warning: this command is known to be currently unsatisfactory. If you really need to compute integrals, think of using an other tool or report a feature request to sylvain.chevillard@ens-lyon.fr.

Example 1:

```
> sin(10);
-0.544021110889369813404747661851377281683643012916219
> dirtyintegral(cos(x),[0;10]);
-0.544003049051526298224480588824753820365362983562818375241
> points=2000!;
> dirtyintegral(cos(x),[0;10]);
-0.544019977511583219722226973125831990359958379268927638295
```

See also: **prec** (7.104), **points** (7.99), **integral** (7.70)

7.34 display

Name: display

sets or inspects the global variable specifying number notation

Usage:

```
\begin{aligned} \mathbf{display} &= notation \ value : \ \mathsf{decimal}|\mathsf{binary}|\mathsf{dyadic}|\mathsf{powers}|\mathsf{hexadecimal} \to \mathsf{void} \\ \mathbf{display} &= notation \ value : \ \mathsf{decimal}|\mathsf{binary}|\mathsf{dyadic}|\mathsf{powers}|\mathsf{hexadecimal} \to \mathsf{void} \\ \mathbf{display} &= ? : \mathsf{void} \to \mathsf{decimal}|\mathsf{binary}|\mathsf{dyadic}|\mathsf{powers}|\mathsf{hexadecimal} \end{aligned}
```

Parameters:

• notation value represents a variable of type decimal|binary|dyadic|powers|hexadecimal

Description:

- An assignment display = notation value, where notation value is one of decimal, dyadic, powers, binary or hexadecimal, activates the corresponding notation for output of values in print, write or at the Sollya prompt. If the global notation variable display is decimal, all numbers will be output in scientific decimal notation. If the global notation variable display is dyadic, all numbers will be output as dyadic numbers with Gappa notation. If the global notation variable display is powers, all numbers will be output as dyadic numbers with a notation compatible with Maple and PARI/GP. If the global notation variable display is binary, all numbers will be output in binary notation. If the global notation variable display is hexadecimal, all numbers will be output in C99/IEEE754R notation. All output notations can be reparsed by Sollya, inducing no error if the input and output precisions are the same (see prec). If the assignment display = notation value is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.
- The expression **display** = ? evaluates to a variable of type decimal|binary|dyadic|powers|hexadecimal, indicating the current notation used.

Example 1:

```
> display = decimal;
Display mode is decimal numbers.
> a = evaluate(sin(pi * x), 0.25);
> a;
0.707106781186547524400844362104849039284835937688470740971063\\
> display = binary;
Display mode is binary numbers.
> a;
010001_2 * 2^{-1}
> display = hexadecimal;
Display mode is hexadecimal numbers.
> a;
0xb.504f333f9de6484597d89b3754abe9f1d6f60ba88p-4
> display = dyadic;
Display mode is dyadic numbers.
33070006991101558613323983488220944360067107133265b-165
> display = powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
33070006991101558613323983488220944360067107133265 * 2^(-165)
```

See also: print (7.107), write (7.155), decimal (7.25), dyadic (7.39), powers (7.103), binary (7.15), hexadecimal (7.62), prec (7.104)

7.35 divide

Name: / division function

Usage:

 $function1 / function2 : (function, function) \rightarrow function$

Parameters:

• function1 and function2 represent functions

Description:

• / represents the division (function) on reals. The expression function1 / function2 stands for the function composed of the division function and the two functions function1 and function2, where function1 is the numerator and function2 the denominator.

Example 1:

```
> 5 / 2;
0.25e1
```

Example 2:

```
> x / 2;
x * 0.5
```

Example 3:

```
> x / x;
1
```

Example 4:

```
> 3 / 0;
@Inf@
```

Example 5:

```
> diff(sin(x) / exp(x));
(exp(x) * cos(x) - sin(x) * exp(x)) / exp(x)^2
```

```
See also: +(7.98), -(7.85), *(7.86), ^(7.102)
```

7.36 double

Names: double, D

rounding to the nearest IEEE double.

Description:

- double is both a function and a constant.
- As a function, it rounds its argument to the nearest double precision number. Subnormal numbers are supported as well as standard numbers: it is the real ounding described in the standard.
- As a constant, it symoblizes the double precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **roundcoefficients** and **implementpoly**. See the corresponding help pages for examples.

Example 1:

See also: doubleextended (7.38), doubledouble (7.37), tripledouble (7.149), roundcoefficients (7.128), implementpoly (7.66)

7.37 doubledouble

Names: doubledouble, DD

represents a number as the sum of two IEEE doubles.

Description:

- doubledouble is both a function and a constant.
- As a function, it rounds its argument to the nearest number that can be written as the sum of two double precision numbers.
- The algorithm used to compute $\mathbf{doubledouble}(x)$ is the following: let $xh = \mathbf{double}(x)$ and let $xh = \mathbf{double}(x)$. Return the number xh + xh. Note that if the current precision is not sufficient to represent exactly xh + xh, a rounding will occur and the result of $\mathbf{doubledouble}(x)$ will be useless.
- As a constant, it symbolizes the double-double precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **roundcoefficients** and **implementpoly**. See the corresponding help pages for examples.

Example 1:

See also: double (7.36), doubleextended (7.38), tripledouble (7.149), roundcoefficients (7.128), implementpoly (7.66)

7.38 doubleextended

Names: doubleextended, DE

computes the nearest number with 64 bits of mantissa.

Description:

- doubleextended is a function that computes the nearest floating-point number with 64 bits of mantissa to a given number. Since it is a function, it can be composed with other functions of Sollya such as exp, sin, etc.
- It does not handle subnormal numbers. The range of possible exponents is the range used for all numbers represented in Sollya (e.g. basically the range used in the library MPFR).
- Since it is a function and not a command, its behavior is a bit different from the behavior of **round**(x,64,RN) even if the result is exactly the same. **round**(x,64,RN) is immediately evaluated whereas **doubleextended**(x) can be composed with other functions (and thus be plotted and so on).
- Be aware that **doubleextended** cannot be used as a constant to represent a format in the commands **roundcoefficients** and **implementpoly** (contrary to **D**, **DD**, and **TD**).

Example 1:

Example 2:

```
> D(2^(-2000));
0
> DE(2^(-2000));
0.870980981621721667557619549477887229585910374270538862e-602
```

Example 3:

```
> verbosity=1!;
> f = sin(DE(x));
> f(pi);
Warning: rounding has happened. The value displayed is a faithful rounding of th
e true result.
-0.501655761266833202355732708033075701383156167025495e-19
> g = sin(round(x,64,RN));
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
```

See also: double (7.36), doubledouble (7.37), tripledouble (7.149), round (7.127)

7.39 dyadic

Name: dyadic

special value for global state display

Description:

• dyadic is a special value used for the global state display. If the global state display is equal to dyadic, all data will be output in dyadic notation with numbers displayed in Gappa format. As any value it can be affected to a variable and stored in lists.

See also: decimal (7.25), powers (7.103), hexadecimal (7.62), binary (7.15)

7.40 equal

```
Name: ==
equality test operator
```

Usage:

```
expr1 == expr2 : (any type, any type) \rightarrow boolean
```

Parameters:

 \bullet expr1 and expr2 represent expressions

Description:

• The operator == evaluates to true iff its operands expr1 and expr2 are syntactically equal and different from **error** or constant expressions that evaluate to the same floating-point number with the global precision **prec**. The user should be aware of the fact that because of floating-point evaluation, the operator == is not exactly the same as the mathematical equality.

Example 1:

```
> "Hello" == "Hello";
true
> "Hello" == "Salut";
false
> "Hello" == 5;
false
> 5 + x == 5 + x;
true
```

Example 2:

```
> 1 == exp(0);
true
> asin(1) * 2 == pi;
true
> exp(5) == log(4);
false
```

Example 3:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384 == 16385;
true
```

Example 4:

```
> error == error;
false
```

See also: !=(7.87), > (7.59), >=(7.58), <=(7.73), < (7.81), ! (7.89), && (7.6), || (7.93), error (7.43), prec (7.104)

7.41 erf

Name: **erf**

the error function.

Description:

• **erf** is the error function defined by:

$$\operatorname{erf}(x) = \frac{2}{\sqrt{\pi}} \int_0^x e^{-t^2} dt.$$

• It is defined for every real number x.

See also: \mathbf{erfc} (7.42), \mathbf{exp} (7.46)

7.42 erfc

Name: **erfc**

the complementary error function.

Description:

- **erfc** is the complementary error function defined by $\operatorname{erfc}(x) = 1 \operatorname{erf}(x)$.
- It is defined for every real number x.

See also: \mathbf{erf} (7.41)

7.43 error

Name: error

expression representing an input that is wrongly typed or that cannot be executed

Usage:

error: error

Description:

• The variable **error** represents an input during the evaluation of which a type or execution error has been detected or is to be detected. Inputs that are syntactically correct but wrongly typed evaluate to **error** at some stage. Inputs that are correctly typed but containing commands that depend on side-effects that cannot be performed or inputs that are wrongly typed at meta-level (cf. **parse**), evaluate to **error**. Remark that in contrast to all other elements of the Sollya language, **error** compares neither equal nor unequal to itself. This provides a means of detecting syntax errors inside the Sollya language itself without introducing issues of two different wrongly typed input being equal.

Example 1:

```
> print(5 + "foo");
error
```

Example 2:

```
> error;
error
```

Example 3:

```
> error == error;
false
> error != error;
false
```

Example 4:

```
> correct = 5 + 6;
> incorrect = 5 + "foo";
> (correct == error || correct != error);
true
> (incorrect == error || incorrect != error);
false
```

See also: **void** (7.153), **parse** (7.94)

7.44 evaluate

Name: evaluate

evaluates a function at a constant point or in a range

Usage:

```
 \begin{aligned} \mathbf{evaluate}(function,\ constant) : \ (\mathsf{function},\ \mathsf{constant}) \to \mathsf{constant} & \longrightarrow \mathsf{range} \\ \mathbf{evaluate}(function,\ range) : \ (\mathsf{function},\ \mathsf{range}) \to \mathsf{range} \\ \mathbf{evaluate}(function,\ function2) : \ (\mathsf{function},\ \mathsf{function}) \to \mathsf{function} \end{aligned}
```

Parameters:

- function represents a function
- constant represents a constant point
- range represents a range
- function2 represents a function that is not constant

Description:

- If its second argument is a constant constant, evaluate evaluates its first argument function at the point indicated by constant. This evaluation is performed in a way that the result is a faithful rounding of the real value of the function at constant to the current global precision. If such a faithful rounding is not possible, evaluate returns a range surely encompassing the real value of the function function at constant. If even interval evaluation is not possible because the expression is undefined or numerically unstable, NaN will be produced.
- If its second argument is a range range, **evaluate** evaluates its first argument function by interval evaluation on this range range. This ensures that the image domain of the function function on the pre-image domain range is surely enclosed in the returned range.
- If its second argument is a function function2 that is not a constant, **evaluate** replaces all occurrences of the free variable in function function by function function2.

Example 1:

```
> print(evaluate(sin(pi * x), 2.25));
0.707106781186547524400844362104849039284835937688470741
```

Example 2:

```
> print(evaluate(sin(pi * x), 2));
[-0.172986452514381269516508615031098129542836767991679e-12714;0.759411982011879
631450695643145256617060390843900679e-12715]
```

Example 3:

```
> print(evaluate(sin(pi * x), [2, 2.25]));
[-0.514339027267725463004699891996191240734922416542101e-49;0.707106781186547524
400844362104849039284835937688663]
```

Example 4:

```
> print(evaluate(sin(pi * x), 2 + 0.25 * x));
sin((pi) * (2 + 0.25 * x))
```

Example 5:

```
> print(evaluate(sin(pi * 1/x), 0));
@NaN@
```

See also: **isevaluable** (7.72)

7.45 execute

Name: execute

executes the content of a file

Usage:

```
execute(filename) : string \rightarrow void
```

Parameters:

• filename is a string representing a file name

Description:

- **execute** opens the file indicated by *filename*, and executes the sequence of commands it contains. This command is evaluated at execution time: this way you can modify the file *filename* (for instance using **bashexecute**) and execute it just after.
- If *filename* contains a command **execute**, it will be executed recursively.
- If filename contains a call to **restart**, it will be neglected.
- If filename contains a call to quit, the commands following quit in filename will be neglected.

Example 1:

```
> a=2;
> a;
2
> print("a=1;") > "example.sollya";
> execute("example.sollya");
> a;
1
```

Example 2:

```
> verbosity=1!;
> print("a=1; restart; a=2;") > "example.sollya";
> execute("example.sollya");
Warning: a restart command has been used in a file read into another.
This restart command will be neglected.
> a;
2
```

Example 3:

```
> verbosity=1!;
> print("a=1; quit; a=2;") > "example.sollya";
> execute("example.sollya");
Warning: the execution of a file read by execute demanded stopping the interpret ation but it is not stopped.
> a;
1
```

See also: parse (7.94), readfile (7.118), write (7.155), print (7.107), bashexecute (7.14)

$7.46 \exp$

Name: **exp**

the exponential function.

Description:

- **exp** is the usual exponential function defined as the solution of the ordinary differential equation y' = y with y(0) = 1.
- exp(x) is defined for every real number x.

See also: $\exp (7.46)$, $\log (7.77)$

7.47 expand

Name: expand

expands polynomial subexpressions

Usage:

```
\mathbf{expand}(function) : \mathsf{function} \to \mathsf{function}
```

Parameters:

• function represents a function

Description:

• **expand**(function) expands all polynomial subexpressions in function function is far as possible. Factors of sums are multiplied out, power operators with constant positive integer exponents are replaced by multiplications and divisions are multiplied out, i.e. denomiators are brought at the most interior point of expressions.

Example 1:

```
> print(expand(x^3));
x * x * x
```

Example 2:

```
> print(expand((x + 2)^3 + 2 * x));
8 + 12 * x + 6 * x * x + x * x * x + 2 * x
```

Example 3:

See also: simplify (7.133), simplifysafe (7.134), horner (7.65)

7.48 expm1

Name: expm1

translated exponential function.

$\label{eq:Description:} Description:$

- expm1 is defined by expm1(x) = exp(x) 1.
- It is defined for every real number x.

See also: $\exp(7.46)$

7.49 exponent

Name: exponent

returns the scaled binary exponent of a number.

Usage:

```
exponent(x) : constant \rightarrow integer
```

Parameters:

 \bullet x is a dyadic number.

Description:

- **exponent**(x) is by definition 0 if x equals 0, NaN, or Inf.
- If x is not zero, it can be uniquely written as $x = m \cdot 2^e$ where m is an odd integer and e is an integer. **exponent**(x) returns e.

Example 1:

```
> a=round(Pi,20,RN);
> e=exponent(a);
> e;
-17
> m=mantissa(a);
> a-m*2^e;
0
```

See also: mantissa (7.82), precision (7.105)

7.50 externalplot

Name: externalplot

plots the error of an external code with regard to a function

Usage:

Description:

• The command **externalplot** plots the error of an external function evaluation code sequence implemented in the object file named *filename* with regard to the function *function*. If *mode* evaluates to *absolute*, the difference of both functions is considered as an error function; if *mode* evaluates to *relative*, the difference is quotiented by the function *function*. The resulting error function is plotted on all floating-point numbers with *precision* significant mantissa bits in the range *range*. If the sixth argument of the command **externalplot** is given an evaluates to **perturb**, each of these floating-point numbers is perturbed by a random value that is uniformly distributed in ±1 ulp around the original *precision* bit floating-point variable. If a sixth and seventh argument, respectively a seventh and eighth argument in the presence of **perturb** as a sixth argument, are given that evaluate to a variable of type file|postscript|postscriptfile respectively to a character sequence

of type string, externalplot will plot (additionally) to a file in the same way as the command plot does. See plot for details. The external function evaluation code given in the object file name filename is supposed to define a function name f as follows (here in C syntax): void f(mpfr_t rop, mpfr_ op). This function is supposed to evaluate op with an accuracy corresponding to the precision of rop and assign this value to rop.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l mpfr");
> externalplot("./externalplotexample",relative,exp(x),[-1/2;1/2],12,perturb);
```

See also: plot (7.97), asciiplot (7.8), perturb (7.95), absolute (7.2), relative (7.120), file (7.53), postscript (7.100), postscript (7.101), bashexecute (7.14), externalproc (7.51), library (7.75)

7.51 externalproc

Name: externalproc

binds an external code to a Sollya procedure

Usage:

```
externalproc(identifier, filename, argumenttype -> resulttype) : (identifier type, string, type type, type type) \rightarrow void
```

Parameters:

- identifier represents the identifier the code is to be bound to
- filename of type string represents the name of the object file where the code of procedure can be found
- argumenttype represents a definition of the types of the arguments of the Sollya procedure and the external code
- ullet result type represents a definition of the result type of the external code

Description:

• external proc allows for binding the Sollya identifier identifier to an external code. After this binding, when Sollya encounters identifier applied to a list of actual parameters, it will evaluate these parameters and call the external code with these parameters. If the external code indicated success, it will receive the result produced by the external code, transform it to Sollya's iternal representation and return it. In order to allow correct evaluation and typing of the data in parameter and in result to be passed to and received from the external code, external proc has a third parameter argument type - > result type. Both argument type and result type are one of void, constant, function, range, integer, string, boolean, list of constant, list of function, list of range, list of integer, list of string, list of boolean. If upon a usage of a procedure bound to an external procedure the type of the actual parameters given or its number is not correct, Sollya produces a type error. An external function not applied to arguments represents itself and prints out with its argument and result types. The external function is supposed to return an integer indicating success. It returns its result depending on its Sollya result type as follows. Here, the external procedure is assumed to be implemented as a C function. If the Sollya result type is void, the C function has no pointer argument for the result. If the Sollya result type is constant, the first argument of the C function is of C type mpfr_t *, the result is returned by affecting the MPFR variable. If the Sollya result type is function, the first argument of the C function is of C type node **node **, the result is returned by the node * pointed with a new node *. If the Sollya result type is range, the first argument of the C function is of C type mpfi_t *, the result is returned

by affecting the MPFI variable. If the Sollya result type is integer, the first argument of the C function is of C type int *, the result is returned by affecting the int variable. If the Sollya result type is string, the first argument of the C function is of C type char **, the result is returned by the char * pointed with a new char *. If the Sollya result type is boolean, the first argument of the C function is of C type int *, the result is returned by affecting the int variable with a boolean value. If the Sollya result type is list of type, the first argument of the C function is of C type chain **, the result is returned by the chain * pointed with a new chain *. This chain contains for Sollya type constant pointers mpfr_t * to new MPFR variables, for Sollya type function pointers node * to new nodes, for Sollya type range pointers mpfi_t * to new MPFI variables, for Sollya type integer pointers int * to new int variables for Sollya type string pointers char * to new char * variables and for Sollya type boolean pointers int * to new int variables representing boolean values.

The external procedure affects its possible pointer argument if and only if it succeeds. This means, if the function returns an integer indicating failure, it does not leak any memory to the encompassing environment. The external procedure receives its arguments as follows: If the Sollya argument type is void, no argument array is given. Otherwise the C function receives a C void ** argument representing an array of size equal to the arity of the function where each entry (of C type void *) represents a value with a C type depending on the corresponding Sollya type. If the Sollya type is constant, the C type the void * is to be casted to is mpfr_t *. If the Sollya type is function, the C type the void * is to be casted to is node *. If the Sollya type is range, the C type the void * is to be casted to is mpfi_t *. If the Sollya type is integer, the C type the void * is to be casted to is int *. If the Sollya type is string, the C type the void * is to be casted to is char *. If the Sollya type is boolean, the C type the void * is to be casted to is int *. If the Sollya type is list of type, the C type the void * is to be casted to is chain *. Here depending on type, the values in the chain are to be casted to mpfr_t * for Sollya type constant, node * for Sollya type function, mpfi_t * for Sollya type range, int * for Sollya type integer, char * for Sollya type string and int * for Sollya type boolean. The external procedure is not supposed to alter the memory pointed by its array argument void **. In both directions (argument and result values), empty lists are represented by chain * NULL pointers. In contrast to internal procedures, externally bounded procedures can be considered as objects inside Sollya that can be assigned to other variables, stored in list etc.

Example 1:

```
> bashexecute("gcc -fPIC -Wall -c externalprocexample.c");
> bashexecute("gcc -fPIC -shared -o externalprocexample externalprocexample.o");
> externalproc(foo, "./externalprocexample", (integer, integer) -> integer);
> foo;
foo(integer, integer) -> integer
 > foo(5, 6); 
11
> verbosity = 1!;
> foo();
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
error
> a = foo;
> a(5,6);
11
```

See also: library (7.75), externalplot (7.50), bashexecute (7.14), void (7.153), constant (7.22), function (7.57), range (7.115), integer (7.69), string (7.139), boolean (7.16), list of (7.76)

7.52 false

Name: false

the boolean value representing the false.

Description:

• false is the usual boolean value.

Example 1:

```
> true && false;
false
> 2<1;
false
```

See also: **true** (7.150), && (7.6), || (7.93)

7.53 file

Name: file

special value for commands plot and externalplot

Description:

- file is a special value used in commands plot and externalplot to save the result of the command in a data file.
- As any value it can be affected to a variable and stored in lists.

Example 1:

```
> savemode=file;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);
```

See also: externalplot (7.50), plot (7.97), postscript (7.100), postscriptfile (7.101)

7.54 findzeros

Name: findzeros

gives a list of intervals containing all zeros of a function on an interval.

Usage:

 $\mathbf{findzeros}(f,I):(\mathsf{function},\mathsf{range})\to\mathsf{list}$

Parameters:

- f is a function.
- \bullet I is an interval.

Description:

- findzeros(f,I) returns a list of intervals I1, ... ,In such that, for every zero z of f, there exists some k such that $z \in I_k$.
- The list may contain intervals Ik that do not contain any zero of f. An interval Ik may contain many zeros of f.

- This command is ment for cases when safety is critical. If you want to be sure not to forget any zero, use **findzeros**. However, if you just want to know numerical values for the zeros of *f*, **dirtyfindzeros** should be quite satisfactory and a lot faster.
- If δ denotes the value of global variable diam, the algorithm ensures that for each $k, |I_k| \leq \delta \cdot |I|$.
- The algorithm used is basically a bisection algorithm. It is the same algorithm that the one used for **infnorm**. See the help page of this command for more details. In short, the behavior of the algorithm depends on global variables **prec**, **diam**, **taylorrecursions** and **hopitalrecursions**.

Example 1:

```
> findzeros(sin(x),[-5;5]);
[|[-0.314208984375e1;-0.3140869140625e1], [-0.1220703125e-2;0.1220703125e-2], [0
.3140869140625e1;0.314208984375e1]|]
> diam=1e-10!;
> findzeros(sin(x),[-5;5]);
[|[-0.314159265370108187198638916015625e1;-0.3141592652536928653717041015625e1],
[-0.116415321826934814453125e-8;0.116415321826934814453125e-8], [0.314159265253
6928653717041015625e1;0.314159265370108187198638916015625e1]|]
```

See also: dirtyfindzeros (7.31), infnorm (7.68), prec (7.104), diam (7.29), taylorrecursions (7.147), hopitalrecursions (7.64)

7.55 floor

Name: floor

the usual function floor.

Description:

- floor is defined as usual: floor(x) is the greatest integer y such that $y \leq x$.
- It is defined for every real number x.

See also: ceil (7.18)

7.56 fullparentheses

Name: fullparentheses

activates, deactivates or inspects the state variable controlling output with full parenthesizing

Usage:

```
fullparentheses = activation\ value : on|off \rightarrow void
fullparentheses = activation\ value : on|off \rightarrow void
fullparentheses = ?: void \rightarrow on|off
```

Parameters:

• activation value represents on or off, i.e. activation or deactivation

Description:

• An assignment fullparentheses = activation value, where activation value is one of on or off, activates respectively deactivates the output of expressions with full parenthezing. In full parenthezing mode, Sollya commands like print, write and the implicit command when an expression is given at the prompt will output expressions with parentheses at all places where it is necessary for expressions containing infix operators to be reparsed with the same result. Otherwise parentheses around associative operators are omitted. If the assignment fullparentheses = activation value is followed by an exclamation mark, no message indicating the new state is displayed. Otherwise the user is informed of the new state of the global mode by an indication.

• The expression **fullparentheses** = ? evaluates to a variable of type **on**|**off**, indicating whether or not the full parenthesized output of expressions is activated or not.

Example 1:

```
> autosimplify = off!;
> fullparentheses = off;
Full parentheses mode has been deactivated.
> print(1 + 2 + 3);
1 + 2 + 3
> fullparentheses = on;
Full parentheses mode has been activated.
> print(1 + 2 + 3);
(1 + 2) + 3
```

See also: print (7.107), write (7.155), autosimplify (7.13)

7.57 function

Name: function

keyword representing a function type

Usage:

function: type type

Description:

• function represents the function type for declarations of external procedures by means of externalproc. Remark that in contrast to other indicators, type indicators like function cannot be handled outside the externalproc context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), boolean (7.16), constant (7.22), integer (7.69), list of (7.76), range (7.115), string (7.139)

7.58 ge

Name: >= greater-than-or-equal-to operator

Usage:

```
expr1 >= expr2 : (constant, constant) \rightarrow boolean
```

Parameters:

• expr1 and expr2 represent constant expressions

Description:

• The operator >= evaluates to true iff its operands expr1 and expr2 evaluate to two floating-point numbers a_1 respectively a_2 with the global precision **prec** and a_1 is greater than or equal to a_2 . The user should be aware of the fact that because of floating-point evaluation, the operator >= is not exactly the same as the mathematical operation greater-than-or-equal-to.

Example 1:

```
> 5 >= 4;
true
> 5 >= 5;
true
> 5 >= 6;
false
> exp(2) >= exp(1);
true
> log(1) >= exp(2);
false
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384 >= 16385;
true
```

See also: ==(7.40), !=(7.87), >(7.59), <=(7.73), <(7.81), ! (7.89), && (7.6), || (7.93), prec (7.104)

7.59 gt

Name: >

greater-than operator

Usage:

$$expr1 > expr2 : (constant, constant) \rightarrow boolean$$

Parameters:

• expr1 and expr2 represent constant expressions

Description:

• The operator > evaluates to true iff its operands expr1 and expr2 evaluate to two floating-point numbers a_1 respectively a_2 with the global precision **prec** and a_1 is greater than a_2 . The user should be aware of the fact that because of floating-point evaluation, the operator > is not exactly the same as the mathematical operation greater-than.

Example 1:

```
> 5 > 4;
true
> 5 > 5;
false
> 5 > 6;
false
> exp(2) > exp(1);
true
> log(1) > exp(2);
false
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384 > 16385;
false
```

See also: ==(7.40), !=(7.87), >=(7.58), <=(7.73), <(7.81), !=(7.89), && (7.6), ||=(7.93), prec (7.104)

7.60 guessdegree

Name: guessdegree

returns the minimal degree needed for a polynomial to approximate a function with a certain error on an interval.

Usage:

```
\mathbf{guessdegree}(f, I, eps, w) : (function, range, constant, function) \rightarrow range
```

Parameters:

- \bullet f is the function to be approximated.
- \bullet I is the interval where the function must be approximated.
- \bullet eps is the maximal acceptable error.
- ullet w (optional) is a weight function. Default is 1.

Description:

- guessdegree tries to find the minimal degree needed to approximate f on I by a polynomial with an infinite error not greater than eps. More precisely, it finds n minimal such that there exists a polynomial p of degree n such that $||pw f||_{\infty} < eps$.
- guessdegree returns an interval: for common cases, this interval is reduced to a single number (e.g. the minimal degree). But in certain cases, guessdegree does not succeed in finding the minimal degree. In such cases the returned interval is of the form [n, p] such that:
 - no polynomial of degree n-1 gives an error less than eps.
 - there exists a polynomial of degree p giving an error less than eps.

Example 1:

```
> guessdegree(exp(x),[-1;1],1e-10);
[10;10]
```

Example 2:

```
> guessdegree(1, [-1;1], 1e-8, 1/exp(x));
[8;9]
```

See also: $\operatorname{dirtyinfnorm}$ (7.32), remez (7.121)

7.61 head

Name: head

gives the first element of a list.

Usage:

```
\mathbf{head}(L):\mathsf{list}\to\mathsf{any}\;\mathsf{type}
```

Parameters:

• L is a list.

Description:

- $\mathbf{head}(L)$ returns the first element of the list L. It is equivalent to L[0].
- If L is empty, the command will fail with an error.

Example 1:

```
> head([|1,2,3|]);
1
> head([|1,2...|]);
1
```

See also: **tail** (7.143)

7.62 hexadecimal

Name: hexadecimal

special value for global state display

Description:

• hexadecimal is a special value used for the global state display. If the global state display is equal to hexadecimal, all data will be output in hexadecimal C99/ IEEE 754R notation. As any value it can be affected to a variable and stored in lists.

See also: **decimal** (7.25), **dyadic** (7.39), **powers** (7.103), **binary** (7.15)

7.63 honorcoeffprec

Name: honorcoeffprec

indicates the (forced) honoring the precision of the coefficients in implementpoly

Usage:

honorcoeffprec : honorcoeffprec

Description:

• Used with command **implementpoly**, **honorcoeffprec** makes **implementpoly** honor the precision of the given polynomial. This means if a coefficient needs a double-double or a triple-double to be exactly stored, **implementpoly** will allocate appropriate space and use a double-double or triple-double operation even if the automatic (heuristical) determination implemented in command **implementpoly** indicates that the coefficient could be stored on less precision or, respectively, the operation could be performed with less precision. See **implementpoly** for details.

Example 1:

```
> verbosity = 1!;
> q = implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6))) * x^2,[-1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6))) * x^2,[-1b-10],1b-60,DD,"p","impleme
lementation.c");
Warning: at least one of the coefficients of the given polynomial has been round
ed in a way
that the target precision can be achieved at lower cost. Nevertheless, the imple
mented polynomial
is different from the given one.
> printexpansion(q);
0x3ff00000000000000 + x^2 * 0xbfc55555555555555
> r = implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","imp
lementation.c",honorcoeffprec);
Warning: the infered precision of the 2th coefficient of the polynomial is great
er than
the necessary precision computed for this step. This may make the automatic dete
rmination
of precisions useless.
> printexpansion(r);
55555555)
```

See also: implementpoly (7.66), printexpansion (7.108)

7.64 hopital recursions

Name: hopitalrecursions

controls the number of recursion steps when applying L'Hopital's rule.

Description:

- hopitalrecursions is a global variable. Its value represents the number of steps of recursion that are tried when applying L'Hopital's rule. This rule is applied by the interval evaluator present in the core of Sollya (and particularly visible in commands like infnorm).
- If an expression of the form f/g has to be evaluated by interval arithmetic on an interval I and if f and g have a common zero in I, a direct evaluation leads to NaN. Sollya implements a safe heuristic to avoid this, based on L'Hopital's rule: in such a case, it can be shown that $(f/g)(I) \subseteq (f'/g')(I)$. Since the same problem may hold for f'/g', the rule is applied recursively. The number of step in this recursion process is controlled by **hopitalrecursions**.
- Setting hopitalrecursions to 0 makes Sollya use this rule only one time; setting it to 1 makes Sollya use the rule two times, and so on. In particular: the rule is always applied at least once, if necessary.

Example 1:

```
> hopitalrecursions=0;

The number of recursions for Hopital's rule has been set to 0.

> evaluate(log(1+x)^2/x^2,[-1/2; 1]);

[-@Inf@;@Inf@]

> hopitalrecursions=1;

The number of recursions for Hopital's rule has been set to 1.

> evaluate(log(1+x)^2/x^2,[-1/2; 1]);

[-0.252258872223978123766892848583270627230200053744108e1;0.67725887222397812376

6892848583270627230200053744116e1]
```

7.65 horner

Name: horner

brings all polynomial subexpressions of an expression to Horner form

Usage:

```
\mathbf{horner}(function) : \mathsf{function} \to \mathsf{function}
```

Parameters:

• function represents the expression to be rewritten in Horner form

Description:

• The command **horner** rewrites the expression representing the function function in a way such that all polynomial subexpressions (or the whole expression itself, if it is a polynomial) are written in Horner form. The command **horner** does not endanger the safety of computations even in Sollya's floating-point environment: the function returned is mathematically equal to the function function.

Example 1:

```
> print(horner(1 + 2 * x + 3 * x^2));
1 + x * (2 + x * 3)
> print(horner((x + 1)^7));
1 + x * (7 + x * (21 + x * (35 + x * (21 + x * (7 + x))))))
```

Example 2:

```
> print(horner(exp((x + 1)^5) - log(asin(x + x^3) + x)));
exp(1 + x * (5 + x * (10 + x * (10 + x * (5 + x))))) - log(asin(x * (1 + x^2)) + x)
```

See also: canonical (7.17), print (7.107)

7.66 implementpoly

Name: implementpoly

implements a polynomial using double, double-double and triple-double arithmetic and generates a Gappa proof

Usage:

```
\label{eq:constant} \begin{split} & \textbf{implementpoly}(polynomial,\ range,\ error\ bound,\ format,\ functionname,\ filename): (function,\ range,\ constant,\ D|double|DD|doubledouble,\ string,\ string) \rightarrow \text{function} \\ & \textbf{implementpoly}(polynomial,\ range,\ error\ bound,\ format,\ functionname,\ filename,\ honor\ coefficient\ precisions): (function,\ range,\ constant,\ D|double|DD|doubledouble|TD|tripledouble,\ string,\ string,\ honorcoeffprec) \rightarrow \text{function} \end{split}
```

implementpoly(polynomial, range, error bound, format, functionname, filename, proof filename): (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string, string) \rightarrow function implementpoly(polynomial, range, error bound, format, functionname, filename, honor coefficient precisions, proof filename): (function, range, constant, D|double|DD|doubledouble|TD|tripledouble, string, string, honorcoeffprec, string) \rightarrow function

Description:

• The command **implementpoly** implements the polynomial polynomial in range range as a function called functionname in C code using double, double-double and triple-double arithmetic in a way that the rounding error (estimated at its first order) is bounded by jerror bound_i. The

produced code is output in a file named *filename*. The argument *format* indicates the double, double-double or triple-double format of the variable in which the polynomial varies, influencing also in the signature of the C function. If a seventh or eightth argument proof filename is given and if this argument evaluates to a variable of type string, the command implementpoly will produce a Gappa proof that the rounding error is less than the given bound. This proof will be output in Gappa syntax in a file name proof filename. The command implementpoly returns the polynomial that has been implemented. As the command **implementpoly** tries to adapt the precision needed in each evaluation step to its strict minimum and as it renormalizes double-double and triple-double precision coefficients to a round-to-nearest expansion, the polynomial return may differ from the polynomial polynomial. Nevertheless the difference will be small enough that the rounding error bound with regard to the polynomial polynomial (estimated at its first order) will be less than the given error bound. If a seventh argument honor coefficient precisions is given and evaluates to a variable honorcoeffprec of type honorcoeffprec, implementpoly will honor the precision of the given polynomial polynomials. This means if a coefficient needs a double-double or a triple-double to be exactly stored, **implementpoly** will allocate appropriate space and use a double-double or triple-double operation even if the automatic (heuristical) determination implemented in command implementpoly indicates that the coefficient could be stored on less precision or, respectively, the operation could be performed with less precision. The use of honorcoeffprec has advantages and disadvantages. If the polynomial polynomial given has not been determined by a process considering directly polynomials with floating-point coefficients, honorcoeffprec should not be indicated. The implementpoly command can then determine the needed precision using the same error estimation as used for the determination of the precisions of the operations. Generally, the coefficients will get rounded to double, double-double and triple-double precision in a way that minimizes their number and respects the rounding error bound error bound. Indicating honorcoeffprec may in this case short-circuit most precision estimations leading to sub-optimal code. On the other hand, if the polynomial polynomial has been determined with floating-point precisions in mind, honorcoeffprec should be indicated because such polynomials often are very sensitive in terms of error propgation with regard to their coefficients' values. Indicating honorcoeffprec prevents the implementpoly command from rounding the coefficients and altering by many orders of magnitude approximation error of the polynomial with regard to the function it approximates. The implementer behind the **implementpoly** command makes some assumptions on its input and verifies them. If some assumption cannot be verified, the implementation will not succeed and implementpoly will evaluate to a variable error of type error. The same behaviour is observed if some file is not writeable or some other side-effect fails, e.g. if the implementer runs out of memory. As error estimation is performed only on the first order, the code produced by the implementpoly command should be considered valid iff a Gappa proof has been produced and successfully run in Gappa.

Example 1:

```
> implementpoly(1 - 1/6 * x^2 + 1/120 * x^4, [-1b-10;1b-10], 1b-30, D, "p","impl
ementation.c");
1 + x^2 * ((-0.16666666666666666666674148081281236954964697360992431640625) + x^2 *
0.833333333333333321768510160154619370587170124053955078125e-2)
> readfile("implementation.c");
0000000000000000000000e+00
#define p_coeff_2h -1.6666666666666665741480812812369549646973609924316406250000
0000000000000000000000000e-01
000000000000000000000e-03
void p(double *p_resh, double x) {
double p_x_0_pow2h;
p_x_0_pow2h = x * x;
double p_t_1_0h;
double p_t_2_0h;
double p_t_3_0h;
double p_t_4_0h;
double p_t_5_0h;
p_t_1_0h = p_coeff_4h;
p_t_20h = p_t_10h * p_x_0pow2h;
p_t_3_0h = p_coeff_2h + p_t_2_0h;
p_t_4_0h = p_t_3_0h * p_x_0_pow2h;
p_t_5_0h = p_coeff_0h + p_t_4_0h;
*p_resh = p_t_5_0h;
}
```

Example 2:

```
> implementpoly(1 - 1/6 * x^2 + 1/120 * x^4, [-1b-10;1b-10], 1b-30, D, "p","impl
ementation.c","implementation.gappa");
0.833333333333333321768510160154619370587170124053955078125e-2)
> readfile("implementation.gappa");
# The polynomial to implement is: (1b0 - ((x^(1b1)) / 3b1)) + ((x^(1b2)) / 15b3)
# The polynomial implemented is: 1b0 + (x * (0 + (x * ((-6004799503160661b-55) +
(x * (0 + (x * 4803839602528529b-59))))))))
# The domain is [-1b-10;1b-10]
# The free variable x is a double precision number, the result p_res* is stored
on a double precision number.
# The code produces 10 intermediate and final arithmetical approximations.
# Double precision rounding operator:
@double = float<ieee_64,ne>;
# Disable some annoying warnings:
#@-Wno-dichotomy-failure
# Helper definitions for decomposing the free variable
xh = x;
# Transcription of the C code
0000000000000000000000000e+00);
p_coeff_2h = double(-1.666666666666666574148081281236954964697360992431640625000)
000000000000000000000000e-01);
000000000000000000000e-03);
p_x_0_pow2h = double(xh * xh);
p_t_1_0h = p_coeff_4h;
p_t_2_0h = double(p_t_1_0h * p_x_0_pow2h);
p_t_3_0h = double(p_coeff_2h + p_t_2_0h);
p_t_4_0h = double(p_t_3_0h * p_x_0_pow2h);
p_t_5_0h = double(p_coeff_0h + p_t_4_0h);
p_resh = p_t_5_0h;
# Mathematical equivalents
Mx = x;
Mp_coeff_0 = p_coeff_0h;
Mp_coeff_2 = p_coeff_2h;
Mp_coeff_4 = p_coeff_4h;
Mp_x_0_pow2 = Mx * Mx;
Mp_t_1_0 = Mp_coeff_4;
Mp_t_2_0 = Mp_t_1_0 * Mp_x_0_pow2;
Mp_t_3_0 = Mp_coeff_2 + Mp_t_2_0;
Mp_t_4_0 = Mp_t_3_0 * Mp_x_0_pow2;
Mp_t_5_0 = Mp_coeff_0 + Mp_t_4_0;
Mp_res = Mp_t_5_0;
# Definition of the relative arithmetical error
epsilon = (p_resh - Mp_res) / Mp_res; 55
# Implication to prove
{((
  x in [-1b-10,-1b-410]
) \/ (
  x in [1b-410,1b-10]
```

Example 3:

```
> verbosity = 1!;
> q = implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6))) * x^2,[-1b-10],1b-60,DD,"p","implementpoly(1 - simplify(TD(1/6))) * x^2,[-1b-10],1b-60,DD,"p","impleme
lementation.c");
Warning: at least one of the coefficients of the given polynomial has been round
ed in a way
that the target precision can be achieved at lower cost. Nevertheless, the imple
mented polynomial
is different from the given one.
> printexpansion(q);
0x3ff0000000000000 + x^2 * 0xbfc55555555555555
> r = implementpoly(1 - simplify(TD(1/6)) * x^2,[-1b-10;1b-10],1b-60,DD,"p","imp
lementation.c",honorcoeffprec);
Warning: the infered precision of the 2th coefficient of the polynomial is great
er than
the necessary precision computed for this step. This may make the automatic dete
rmination
of precisions useless.
> printexpansion(r);
5555555)
```

Example 4:

See also: honorcoeffprec (7.63), roundcoefficients (7.128), D (??), DD (??), TD (??), double (7.36), doubledouble (7.37), tripledouble (7.149), readfile (7.118), printexpansion (7.108), error (7.43)

7.67 inf

Name: inf

gives the lower bound of an interval.

Usage:

```
\inf(I): range \rightarrow constant \inf(x): constant \rightarrow constant
```

Parameters:

- *I* is an interval.
- x is a real number.

Description:

- Returns the lower bound of the interval *I*. Each bound of an interval has its own precision, so this command is exact, even if the current precision is too small to represent the bound.
- When called on a real number x, **inf** considers it as an interval formed of a single point: [x, x]. In other words, **inf** behaves like the identity.

Example 1:

```
> inf([1;3]);
1
> inf(0);
0
```

Example 2:

```
> display=binary!;
> I=[0.111110000011111_2; 1];
> inf(I);
1.11110000011111_2 * 2^(-1)
> prec=12!;
> inf(I);
1.11110000011111_2 * 2^(-1)
```

See also: mid(7.83), sup(7.142)

7.68 infnorm

 ${\rm Name:}\ \mathbf{infnorm}$

computes an interval bounding the infinite norm of a function on an interval.

Usage:

 $infnorm(f,I,filename,Ilist): (function, range, string, list) \rightarrow range$

Parameters:

- f is a function.
- \bullet I is an interval.
- filename (optional) is the name of the file into a proof will be saved.
- *IList* (optional) is a list of intervals to be excluded.

Description:

- infnorm(f,range) computes an interval bounding the infinite norm of the given function f on the interval I, e.g. computes an interval J such that $\max_{x\in I}\{|f(x)|\}\subseteq J$.
- If filename is given, a proof in english will be produced (and stored in file called filename) proving that $\max_{x \in I} \{|f(x)|\} \subseteq J$.
- If a list *IList* of intervals $I1, \ldots, In$ is given, the infinite norm will be computed on $I(I_1 \cup \ldots \cup I_n)$.
- The function f is assumed to be at least twice continuous on I. More generally, if f is C^k , global variables **hopitalrecursions** and **taylorrecursions** must have values not greater than k.
- If the interval is reduced to a single point, the result of **infnorm** is an interval containing the exact absolute value of f at this point.
- If the interval is not bound, the result will be $[0, +\infty]$ which is true but perfectly useless. **infnorm** is not ment to be used with infinite intervals.

- The result of this command depends on the global variables **prec**, **diam**, **taylorrecursions** and **hopitalrecursions**. The contribution of each variable is not easy even to analyse.
 - The algorithm uses interval arithmetic with precision **prec**. The precision should thus be set big enough to ensure that no critical cancellation will occur.
 - When an evaluation is performed on an interval [a, b], if the result is considered being too large, the interval is split into $[a, \frac{a+b}{2}]$ and $[\frac{a+b}{2}, b]$ and so on recursively. This recursion step is not performed if the $(b-a) < \delta \cdot |I|$ where δ is the value of variable **diam**. In other words, **diam** controls the minimum length of an interval during the algorithm.
 - To perform the evaluation of a function on an interval, Taylor's rule is applied, e.g. $f([a,b]) \subseteq f(m) + [a-m,b-m] \cdot f'([a,b])$ where $m = \frac{a+b}{2}$. This rule is applied recursively n times where n is the value of variable **taylorrecursions**. Roughly speaking, the evaluations will avoid decorrelation up to order n.
 - When a function of the form $\frac{g}{h}$ has to be evaluated on an interval [a, b] and when g and h vanish at a same point z of the interval, the ratio may be defined even if the expression $\frac{g(z)}{h(z)} = \frac{0}{0}$ does not make any sense. In this case, L'Hopital's rule may be used and $\left(\frac{g}{h}\right)([a, b]) \subseteq \left(\frac{g'}{h'}\right)([a, b])$. Since the same can occur with the ratio $\frac{g'}{h'}$, the rule is applied recursively. Variable **hopitalrecursions** controls the number of recursion steps.
- The algorithm used for this command is quite complex to be explained here. Please find a complete description in the following article: S. Chevillard and C. Lauter A certified infinite norm for the implementation of elementary functions LIP Research Report number RR2007-26 http://prunel.ccsd.cnrs.fr/ensl-00119810

Example 1:

```
> infnorm(exp(x),[-2;3]);
[0.200855369231876677409285296545817178969879078385537e2;0.200855369231876677409
285296545817178969879078385544e2]
```

Example 2:

```
> infnorm(exp(x),[-2;3],"proof.txt");
[0.200855369231876677409285296545817178969879078385537e2;0.200855369231876677409
285296545817178969879078385544e2]
```

Example 3:

```
> infnorm(exp(x),[-2;3],[| [0;1], [2;2.5]|]);
[0.200855369231876677409285296545817178969879078385537e2;0.200855369231876677409
285296545817178969879078385544e2]
```

Example 4:

```
> infnorm(exp(x),[-2;3],"proof.txt", [| [0;1], [2;2.5]|]);
[0.200855369231876677409285296545817178969879078385537e2;0.200855369231876677409
285296545817178969879078385544e2]
```

Example 5:

```
> infnorm(exp(x),[1;1]);
[0.271828182845904523536028747135266249775724709369989e1;0.271828182845904523536
028747135266249775724709369998e1]
```

Example 6:

```
> infnorm(exp(x), [log(0);log(1)]);
[0;@Inf@]
```

See also: **prec** (7.104), **diam** (7.29), **hopitalrecursions** (7.64), **dirtyinfnorm** (7.32), **checkinfnorm** (7.19)

7.69 integer

Name: integer

keyword representing a machine integer type

Usage:

integer : type type

Description:

• integer represents the machine integer type for declarations of external procedures by means of externalproc. Remark that in contrast to other indicators, type indicators like integer cannot be handled outside the externalproc context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), boolean (7.16), constant (7.22), function (7.57), list of (7.76), range (7.115), string (7.139)

7.70 integral

Name: integral

computes an interval bounding the integral of a function on an interval.

Usage:

 $\mathbf{integral}(f,I):(\mathsf{function},\,\mathsf{range})\to\mathsf{range}$

Parameters:

- f is a function.
- \bullet I is an interval.

Description:

- integral (f,I) returns an interval J such that the exact value of the integral of f on I lies in J.
- This command is safe but very unefficient. Use **dirtyintegral** if you just want an approximate value.
- The result of this command depends on the global variable diam. The method used is the following: I is cut into intervals of length not greater then $\delta \cdot |I|$ where δ is the value of global variable diam. On each small interval J, an evaluation of f by interval is performed. The result is multiplied by the length of J. Finally all values are summed.

Example 1:

```
> sin(10);

-0.544021110889369813404747661851377281683643012916219

> integral(cos(x),[0;10]);

[-0.547101979835796902240976371635259430756985992573329;-0.540940151300131838481

505408813733707440537411917285]

> diam=1e-5!;

> integral(cos(x),[0;10]);

[-0.544329156859554271018577802959369567752938763827772;-0.543713064012499695080

396442219274890104258031735553]
```

See also: **points** (7.99), **dirtyintegral** (7.33)

7.71 isbound

Name: isbound

indicates whether a variable is bound or not.

Usage:

isbound(ident): boolean

Parameters:

• *ident* is a name.

Description:

- **isbound**(*ident*) returns a boolean value indicating whether the name *ident* is used or not to represent a variable. It returns true when *ident* is the name used to represent the global variable or if the name is currently used to refer to a (possibly local) variable.
- When a variable is defined in a block and has not been defined outside, **isbound** returns true when called inside the block, and false outside. Note that **isbound** returns true as soon as a variable has been declared with **var**, even if no value is actually stored in it.
- If *ident1* is bound to a variable and if *ident2* refers to the global variable, the command **rename**(*ident2*, *ident1*) hides the value of *ident1* which becomes the global variable. However, if the global variable is again renamed, *ident1* gets its value back. In this case, **isbound**(*ident1*) returns true. If *ident1* was not bound before, **isbound**(*ident1*) returns false after that *ident1* has been renamed.

Example 1:

```
> isbound(x);
false
> isbound(f);
false
> isbound(g);
false
> f=sin(x);
> isbound(x);
true
> isbound(f);
true
> isbound(g);
false
```

Example 2:

```
> isbound(a);
false
> { var a; isbound(a); };
true
> isbound(a);
false
```

Example 3:

```
> f=sin(x);
> isbound(x);
true
> rename(x,y);
> isbound(x);
false
```

Example 4:

```
> x=1;
> f=sin(y);
> rename(y,x);
> f;
sin(x)
> x;
x
> isbound(x);
true
> rename(x,y);
> isbound(x);
true
> x;
1
```

See also: **rename** (7.122)

7.72 isevaluable

Name: isevaluable

tests whether a function can be evaluated at a point

Usage:

 $\mathbf{isevaluable}(\mathit{function},\,\mathit{constant}):\,(\mathsf{function},\,\mathsf{constant})\to\mathsf{boolean}$

Parameters:

- function represents a function
- constant represents a constant point

Description:

• **isevaluable** applied to function function and a constant constant returns a boolean indicating whether or not a subsequent call to **evaluate** on the same function function and constant constant will produce a numerical result or NaN. I.e. **isevaluable** returns false iff **evaluate** will return NaN.

Example 1:

```
> isevaluable(sin(pi * 1/x), 0.75);
true
> print(evaluate(sin(pi * 1/x), 0.75));
-0.866025403784438646763723170752936183471402626905185165
```

Example 2:

```
> isevaluable(sin(pi * 1/x), 0.5);
true
> print(evaluate(sin(pi * 1/x), 0.5));
[-0.172986452514381269516508615031098129542836767991679e-12714;0.759411982011879
631450695643145256617060390843900679e-12715]
```

Example 3:

```
> isevaluable(sin(pi * 1/x), 0);
false
> print(evaluate(sin(pi * 1/x), 0));
@NaN@
```

See also: **evaluate** (7.44)

7.73 le

Name: \leq =

less-than-or-equal-to operator

Usage:

```
expr1 \le expr2 : (constant, constant) \rightarrow boolean
```

Parameters:

• expr1 and expr2 represent constant expressions

Description:

• The operator \leq evaluates to true iff its operands expr1 and expr2 evaluate to two floating-point numbers a_1 respectively a_2 with the global precision **prec** and a_1 is less than or equal to a_2 . The user should be aware of the fact that because of floating-point evaluation, the operator \leq is not exactly the same as the mathematical operation less-than-or-equal-to.

Example 1:

```
> 5 <= 4;
false
> 5 <= 5;
true
> 5 <= 6;
true
> exp(2) <= exp(1);
false
> log(1) <= exp(2);
true</pre>
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16385 <= 16384;
true</pre>
```

See also: ==(7.40), !=(7.87), >=(7.58), >(7.59), <(7.81), ! (7.89), && (7.6), || (7.93), **prec** (7.104)

7.74 length

Name: length

computes the length of a list or string.

Usage:

```
\mathbf{length}(L):\mathsf{list}\to\mathsf{integer}\\ \mathbf{length}(s):\mathsf{string}\to\mathsf{integer}
```

Parameters:

- \bullet L is a list.
- s is a string.

Description:

• length returns the length of a list or a string, e.g. the number of elements or letters.

• The empty list or string has length 0. If L is an end-elliptic list, **length** returns +Inf.

Example 1:

```
> length("Hello World!");
12
```

Example 2:

```
> length([|1,...,5|]);
5
```

Example 3:

```
> length([||]);
0
```

Example 4:

```
> length([|1,2...|]);
@Inf@
```

7.75 library

Name: library

binds an external mathematical function to a variable in Sollya

Usage:

```
\mathbf{library}(\mathit{path}): \mathsf{string} \to \mathsf{function}
```

Description:

- The command **library** lets you extends the set of mathematical functions known by Sollya. By default, Sollya knows the most common mathematical functions such as **exp**, **sin**, **erf**, etc. Within Sollya, these functions may be composed. This way, Sollya should satisfy the needs of a lot of users. However, for particular applications, one may want to manipulates other functions such as Bessel functions, or functions defined by an integral or even a particular solution of an ODE.
- **library** makes it possible to let Sollya know about new functions. In order to let it know, you have to provide an implementation of the function you are interested with. This implementation is a C file containing a function of the form:

```
int my_ident(mpfi_t result, mpfi_t op, int n)
```

The semantic of this function is the following: it is an implementation of the function and its derivatives in interval arithmetic. $my_ident(result, I, n)$ shall store in result an enclosure of the image set of the n-th derivative of the function f over I: $f^{(n)}(I) \subseteq result$.

- The integer returned value has no meaning currently.
- You must not provide a non trivial implementation for any n. Most functions of Sollya needs a relevant implementation of f, f' and f''. For higher derivatives, its is not so critical and the implementation may just store $[-\infty, +\infty]$ in result whenever n > 2.
- Note that you should respect somehow MPFI standards in your implementation: result has its
 own precision and you should perform the intermediate computations so that result is as tighter
 as possible.
- You can include sollya.h in your implementation and use library functionnalities of Sollya for your implementation.

• To bind your function into Sollya, you must use the same identifier as the function name used in your implementation file (my_ident in the previous example).

Example 1:

```
> bashexecute("gcc -fPIC -Wall -c libraryexample.c");
> bashexecute("gcc -shared -o libraryexample libraryexample.o -lgmp -lmpfr");
> myownlog = library("./libraryexample");
> evaluate(log(x), 2);
0.693147180559945309417232121458176568075500134360248314207
> evaluate(myownlog(x), 2);
0.693147180559945309417232121458176568075500134360248314207
```

See also: bashexecute (7.14), externalproc (7.51), externalplot (7.50)

7.76 listof

Name: list of

keyword used in combination with a type keyword

Description:

• list of is used in combination with one of the following keywords for indicating lists of the respective type in declarations of external procedures using external process boolean, constant, function, integer, range and string.

See also: externalproc (7.51), boolean (7.16), constant (7.22), function (7.57), integer (7.69), range (7.115), string (7.139)

$7.77 \log$

Name: \log

neperian logarithm.

Description:

- \log is the neperian logarithm defined as the inverse of the exponential function: $\log(y)$ is the unique real number x such that $\exp(x) = y$.
- It is defined only for $y \in [0; +\infty]$.

See also: $\exp (7.46)$, $\log 2 (7.80)$, $\log 10 (7.78)$

$7.78 \log 10$

Name: log10 decimal logarithm.

Description:

- log10 is the decimal logarithm defined by: $\log 10(x) = \log(x)/\log(10)$.
- It is defined only for $x \in [0; +\infty]$.

See also: $\log (7.77)$, $\log 2 (7.80)$

$7.79 \log 1p$

Name: log1p

translated logarithm.

Description:

- log1p is the function defined by log1p(x) = log(1+x).
- It is defined only for $x \in [-1; +\infty]$.

See also: $\log (7.77)$

$7.80 \log 2$

Name: log2 binary logarithm.

Description:

- $\log 2$ is the binary logarithm defined by: $\log 2(x) = \log(x)/\log(2)$.
- It is defined only for $x \in [0; +\infty]$.

See also: log (7.77), log10 (7.78)

7.81 lt

Name: <

less-than operator

Usage:

```
expr1 < expr2 : (constant, constant) \rightarrow boolean
```

Parameters:

 \bullet expr1 and expr2 represent constant expressions

Description:

• The operator < evaluates to true iff its operands expr1 and expr2 evaluate to two floating-point numbers a_1 respectively a_2 with the global precision **prec** and a_1 is less than a_2 . The user should be aware of the fact that because of floating-point evaluation, the operator < is not exactly the same as the mathematical operation less-than.

Example 1:

```
> 5 < 4;
false
> 5 < 5;
false
> 5 < 6;
true
> exp(2) < exp(1);
false
> log(1) < exp(2);
true</pre>
```

Example 2:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384 < 16385;
false</pre>
```

```
See also: == (7.40), != (7.87), >= (7.58), > (7.59), <= (7.73), ! (7.89), && (7.6), || (7.93), prec (7.104)
```

7.82 mantissa

Name: mantissa

returns the integer mantissa of a number.

Usage:

```
\mathbf{mantissa}(x) : \mathsf{constant} \to \mathsf{integer}
```

Parameters:

 \bullet x is a dyadic number.

Description:

- mantissa(x) is by definition x if x equals 0, NaN, or Inf.
- If x is not zero, it can be uniquely written as $x = m \cdot 2^e$ where m is an odd integer and e is an integer. **mantissa**(x) returns m.

Example 1:

```
> a=round(Pi,20,RN);
> e=exponent(a);
> m=mantissa(a);
> m;
411775
> a-m*2^e;
0
```

See also: **exponent** (7.49), **precision** (7.105)

7.83 mid

Name: mid

gives the middle of an interval.

Usage:

```
\mathbf{mid}(I): \mathsf{range} \to \mathsf{constant}
\mathbf{mid}(x): \mathsf{constant} \to \mathsf{constant}
```

Parameters:

- \bullet I is an interval.
- \bullet x is a real number.

Description:

• Returns the middle of the interval *I*. If the middle is not exactly representable at the current precision, the value is rounded.

• When called on a real number x, **mid** considers it as an interval formed of a single point: [x, x]. In other words, **mid** behaves like the identity.

Example 1:

```
> mid([1;3]);
2
> mid(17);
17
```

See also: inf (7.67), sup (7.142)

7.84 midpointmode

Name: midpointmode

global variable controlling the way intervals are displayed.

Description:

- midpointmode is a global variable. When its value is off, intervals are displayed as usual (with the form [a;b]). When its value is on, and if a and b have the same first significant digits, the interval in displayed in a way that lets one immediately see the common digits of the two bounds.
- This mode is supported only with display set to decimal. In other modes of display, midpoint-mode value is simply ignored.

Example 1:

```
> a = round(Pi,30,RD);
> b = round(Pi,30,RU);
> d = [a,b];
> d;
[0.31415926516056060791015625e1;0.31415926553308963775634765625e1]
> midpointmode=on!;
> d;
0.314159265~1/5~e1
```

See also: **on** (7.92), **off** (7.91)

7.85 minus

Name: –

substraction function

Usage:

 $function1 - function2 : (function, function) \rightarrow function$

Parameters:

• function1 and function2 represent functions

Description:

• - represents the substraction (function) on reals. The expression function 1 - function 2 stands for the function composed of the substraction function and the two functions function 1 and function 2, where function 1 is the subtrahent and function 2 the substractor.

Example 1:

```
> 5 - 2;
3
```

Example 2:

```
> x - 2;
(-2) + x
```

Example 3:

```
> x - x;
0
```

Example 4:

```
> diff(sin(x) - exp(x));
cos(x) - exp(x)
```

See also: +(7.98), *(7.86), /(7.35), $^{(7.102)}$

7.86 mult

Name: *

multiplication function

Usage:

 $function1 * function2 : (function, function) \rightarrow function$

Parameters:

• function1 and function2 represent functions

Description:

• * represents the multiplication (function) on reals. The expression function1 * function2 stands for the function composed of the multiplication function and the two functions function1 and function2.

Example 1:

```
> 5 * 2;
10
```

Example 2:

```
> x * 2;
x * 2
```

Example 3:

```
> x * x;
x^2
```

Example 4:

```
> diff(sin(x) * exp(x));
sin(x) * exp(x) + exp(x) * cos(x)
```

See also: $+(7.98), -(7.85), /(7.35), ^(7.102)$

7.87 neq

Name: !=

negated equality test operator

Usage:

```
expr1 != expr2 : (any type, any type) \rightarrow boolean
```

Parameters:

• expr1 and expr2 represent expressions

Description:

• The operator != evaluates to true iff its operands expr1 and expr2 are syntactically unequal and both different from **error** or constant expressions that evaluate to two different floating-point number with the global precision **prec**. The user should be aware of the fact that because of floating-point evaluation, the operator != is not exactly the same as the negation of the mathematical equality. Note that the expressions !(expr1!= expr2) and expr1 == expr2 do not evaluate to the same boolean value. See **error** for details.

Example 1:

```
> "Hello" != "Hello";
false
> "Hello" != "Salut";
true
> "Hello" != 5;
true
> 5 + x != 5 + x;
false
```

Example 2:

```
> 1 != exp(0);
false
> asin(1) * 2 != pi;
false
> exp(5) != log(4);
true
```

Example 3:

```
> prec = 12;
The precision has been set to 12 bits.
> 16384 != 16385;
false
```

Example 4:

```
> error != error;
false
```

```
See also: ==(7.40), > (7.59), >=(7.58), <=(7.73), < (7.81), ! (7.89), && (7.6), || (7.93), error (7.43), prec (7.104)
```

7.88 nop

Name: **nop** no operation

Usage:

 $\mathbf{nop}:\mathsf{void}\to\mathsf{void}$

Description:

- The command **nop** does nothing. This means it is an explicit parse element in the Sollya language that finally does not produce any result or side-effect.
- The keyword **nop** is implicit in some procedure definitions. Procedures without imperative body get parsed as if they had an imperative body containing one **nop** statement.

Example 1:

```
> nop;
```

Example 2:

```
> succ = proc(n) { return n + 1; };
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
> succ(5);
```

See also: **proc** (7.112)

7.89 not

Name: !

boolean NOT operator

Usage:

 $! expr : boolean \rightarrow boolean$

Parameters:

 \bullet expr represents a boolean expression

Description:

• ! evaluates to the boolean NOT of the boolean expression expr. ! expr evaluates to true iff expr does not evaluate to true.

Example 1:

```
> ! false;
true
```

Example 2:

```
> ! (1 == exp(0));
false
```

See also: && (7.6), || (7.93)

7.90 numerator

Name: numerator

gives the numerator of an expression

Usage:

```
numerator(expr): function \rightarrow function
```

Parameters:

• expr represents an expression

Description:

• If expr represents a fraction expr1/expr2, numerator(expr) returns the numerator of this fraction, i.e. expr1. If expr represents something else, numerator(expr) returns the expression itself, i.e. expr. Note that for all expressions expr, numerator(expr) / denominator(expr) is equal to expr.

Example 1:

```
> numerator(5/3);
5
```

Example 2:

```
> numerator(exp(x));
exp(x)
```

Example 3:

```
> a = 5/3;
> b = numerator(a)/denominator(a);
> print(a);
5 / 3
> print(b);
5 / 3
```

Example 4:

```
> a = exp(x/3);
> b = numerator(a)/denominator(a);
> print(a);
exp(x / 3)
> print(b);
exp(x / 3)
```

See also: **denominator** (7.28)

7.91 off

Name: off

special value for certain global variables.

Description:

- off is a special value used to deactivate certain functionnalities of Sollya (namely canonical, timing, fullparentheses, midpointmode).
- As any value it can be affected to a variable and stored in lists.

Example 1:

```
> canonical=on;
Canonical automatic printing output has been activated.
> p=1+x+x^2;
> mode=off;
> p;
1 + x + x^2
> canonical=mode;
Canonical automatic printing output has been deactivated.
> p;
1 + x * (1 + x)
```

See also: on (7.92), canonical (7.17), timing (7.148), fullparentheses (7.56), midpointmode (7.84)

7.92 on

Name: on

special value for certain global variables.

Description:

- on is a special value used to activate certain functionnalities of Sollya (namely canonical, timing, fullparentheses, midpointmode).
- As any value it can be affected to a variable and stored in lists.

Example 1:

```
> p=1+x+x^2;
> mode=on;
> p;
1 + x * (1 + x)
> canonical=mode;
Canonical automatic printing output has been activated.
> p;
1 + x + x^2
```

See also: off (7.91), canonical (7.17), timing (7.148), fullparentheses (7.56), midpointmode (7.84)

7.93 or

Name: ||

boolean OR operator

Usage:

```
expr1 \mid\mid expr2 : (boolean, boolean) \rightarrow boolean
```

Parameters:

• expr1 and expr2 represent boolean expressions

Description:

• || evaluates to the boolean OR of the two boolean expressions expr1 and expr2. || evaluates to true iff at least one of expr1 or expr2 evaluate to true.

Example 1:

```
> false || false;
false
```

Example 2:

```
> (1 == exp(0)) || (0 == log(1));
true
```

See also: && (7.6), ! (7.89)

7.94 parse

Name: parse

parses an expression contained in a string

Usage:

 $\mathbf{parse}(string) : \mathsf{string} \to \mathsf{function} --- \mathsf{error}$

Parameters:

• string represents a character sequence

Description:

• parse(string) parses the character sequence string containing an expression built on constants and base functions. If the character sequence does not contain a well-defined expression, a warning is displayed indicating a syntax error and parse returns a error of type error.

Example 1:

```
> parse("exp(x)");
exp(x)
```

Example 2:

```
> verbosity = 1!;
> parse("5 + + 3");
Warning: syntax error, unexpected PLUSTOKEN. Will try to continue parsing (expecting ";"). May leak memory.
Warning: the string "5 + + 3" could not be parsed by the miniparser.
Warning: at least one of the given expressions or a subexpression is not correct ly typed
or its evaluation has failed because of some error on a side-effect.
error
```

See also: execute (7.45), readfile (7.118)

7.95 perturb

Name: perturb

indicates random perturbation of sampling points for externalplot

Usage:

perturb: perturb

Description:

• The use of **perturb** in the command **externalplot** enables the addition of some random noise around each sampling point in **externalplot**. See **externalplot** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l
mpfr");
> externalplot("./externalplotexample",relative,exp(x),[-1/2;1/2],12,perturb);
```

See also: externalplot (7.50), absolute (7.2), relative (7.120), bashexecute (7.14)

7.96 pi

Name: **pi** the constant Pi.

$\label{eq:Description:} Description:$

- pi is the constant Pi, defined as half the period of sine and cosine.
- In Sollya, **pi** is considered as a 0-ary function. This way, the constant is not evaluated at the time of its definition but at the time of its use. For instance, when you define a constant or a function relating to Pi, the current precision at the time of the definition does not matter. What is important is the current precision when you evaluate the function or the constant value.
- Remark that when you define an interval, the bounds are first evaluated and then the interval is defined. In this case, **pi** will be evaluated as any other constant value at the definition time of the interval, thus using the current precision at this time.

Example 1:

```
> verbosity=1!; prec=12!;
> a = 2*pi;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of th
e true result.
0.62832e1
> prec=20!;
> a;
Warning: rounding has happened. The value displayed is a faithful rounding of th
e true result.
0.62831879e1
```

Example 2:

```
> prec=12!;
> d = [pi; 5];
> d;
[0.31406e1;5]
> prec=20!;
> d;
[0.31406e1;5]
```

See also: $\cos (7.23)$, $\sin (7.135)$

7.97 plot

Name: **plot**

plots one or several functions

Usage:

```
\mathbf{plot}(f1, \dots, fn, I) : (\mathsf{function}, \dots, \mathsf{function}, \mathsf{range}) \to \mathsf{void} \mathbf{plot}(f1, \dots, fn, I, \mathsf{file}, name) : (\mathsf{function}, \dots, \mathsf{function}, \mathsf{range}, \mathsf{file}, \mathsf{string}) \to \mathsf{void} \mathbf{plot}(f1, \dots, fn, I, \mathsf{postscript}, name) : (\mathsf{function}, \dots, \mathsf{function}, \mathsf{range}, \mathsf{postscript}, \mathsf{string}) \to \mathsf{void} \mathbf{plot}(f1, \dots, fn, I, \mathsf{postscriptfile}, name) : (\mathsf{function}, \dots, \mathsf{function}, \mathsf{range}, \mathsf{postscriptfile}, \mathsf{string}) \to \mathsf{void} \mathbf{plot}(L, I, \mathsf{file}, name) : (\mathsf{list}, \mathsf{range}, \mathsf{file}, \mathsf{string}) \to \mathsf{void} \mathbf{plot}(L, I, \mathsf{postscript}, name) : (\mathsf{list}, \mathsf{range}, \mathsf{postscript}, \mathsf{string}) \to \mathsf{void} \mathbf{plot}(L, I, \mathsf{postscriptfile}, name) : (\mathsf{list}, \mathsf{range}, \mathsf{postscriptfile}, \mathsf{string}) \to \mathsf{void}
```

Parameters:

- f1, ..., fn are functions to be plotted.
- L is a list of functions to be plotted.
- I is the interval where the functions have to be plotted.
- name is a string representing the name of a file.

Description:

- This command plots one or several functions $f1, \ldots, fn$ on an interval I. Functions can be either given as parameters of **plot** or as a list L which elements are functions. Functions are plotted on the same graphic with different colors.
- If L contains an element that is not a function (or a constant), an error occurs.
- **plot** relies on the value of global variable **points**. Let *n* be the value of this variable. The algorithm is the following: each function is evaluated at *n* evenly distributed points in *I*. At each point, the computed value is a faithful rounding of the exact value with a sufficiently big precision. Each point is finally plotted. This avoid numerical artefacts such as critical cancellations.
- You can save the graphic either as a data file or as a postscript file.
- If you use argument **file** with a string *name*, Sollya will save a data file called name.dat and a gnuplot directives file called name.p. Invoking gnuplot on name.p will plots datas stored in name.dat.
- If you use argument **postscript** with a string *name*, Sollya will save a postscript file called name.eps representing your graphic.
- If you use argument **postscriptfile** with a string *name*, Sollya will produce the corresponding name.dat, name.p and name.eps.
- This command uses gnuplot to produce the final graphic. If your terminal is not graphic (typically if you use Sollya by ssh without -X) gnuplot should be able to detect it and produce an ASCII-art version on the standard output. If it is not the case, you can either store the graphic in a postscript file to view it locally, or use **asciiplot** command.
- If every function is constant, **plot** will not plot them but just display their value.
- If the interval is reduced to a single point, **plot** will just display the value of the functions at this point.

Example 1:

```
> plot(sin(x),0,cos(x),[-Pi,Pi]);
```

Example 2:

```
> plot(sin(x),0,cos(x),[-Pi,Pi],postscriptfile,"plotSinCos");
```

Example 3:

```
> plot(exp(0), sin(1), [0;1]);
1
0.841470984807896506652502321630298999622563060798373
```

Example 4:

```
> plot(sin(x), cos(x), [1;1]);
0.841470984807896506652502321630298999622563060798373
0.540302305868139717400936607442976603732310420617923
```

See also: externalplot (7.50), asciiplot (7.8), file (7.53), postscript (7.100), postscriptfile (7.101)

7.98 plus

Name: +

addition function

Usage:

```
function1 + function2 : (function, function) \rightarrow function
```

Parameters:

• function1 and function2 represent functions

Description:

• + represents the addition (function) on reals. The expression function 1 + function 2 stands for the function composed of the addition function and the two functions function 1 and function 2.

Example 1:

```
> 1 + 2;
3
```

Example 2:

```
> x + 2;
2 + x
```

Example 3:

```
> x + x;
x * 2
```

Example 4:

```
> diff(sin(x) + exp(x));
cos(x) + exp(x)
```

```
See also: -(7.85), *(7.86), /(7.35), ^(7.102)
```

7.99 points

Name: points

controls the number of points chosen by Sollya in certain commands.

Description:

• **points** is a global variable. Its value represents the number of points used in numerical algorithms of Sollya (namely **dirtyinfnorm**, **dirtyintegral**, **dirtyfindzeros**, **plot**).

Example 1:

See also: dirtyinfnorm (7.32), dirtyintegral (7.33), dirtyfindzeros (7.31), plot (7.97)

7.100 postscript

Name: postscript

special value for commands plot and externalplot

Description:

- **postscript** is a special value used in commands **plot** and **externalplot** to save the result of the command in a postscript file.
- As any value it can be affected to a variable and stored in lists.

Example 1:

```
> savemode=postscript;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);
```

See also: externalplot (7.50), plot (7.97), file (7.53), postscriptfile (7.101)

7.101 postscriptfile

Name: postscriptfile

special value for commands plot and externalplot

Description:

• postscriptfile is a special value used in commands plot and externalplot to save the result of the command in a data file and a postscript file.

• As any value it can be affected to a variable and stored in lists.

Example 1:

```
> savemode=postscriptfile;
> name="plotSinCos";
> plot(sin(x),0,cos(x),[-Pi,Pi],savemode, name);
```

See also: externalplot (7.50), plot (7.97), file (7.53), postscript (7.100)

7.102 power

Name: ^

power function

Usage:

 $function1 \ \hat{} function2 : (function, function) \rightarrow function$

Parameters:

• function1 and function2 represent functions

Description:

• $\hat{}$ represents the power (function) on reals. The expression function 1 $\hat{}$ function 2 stands for the function composed of the power function and the two functions function 1 and function 2, where function 1 is the base and function 2 the exponent. If function 2 is a constant integer, $\hat{}$ is defined on negative values of function 1. Otherwise $\hat{}$ is defined as $\exp(y * \log(x))$.

Example 1:

```
> 5 ^ 2;
25
```

Example 2:

```
> x ^ 2;
x^2
```

Example 3:

```
> 3 ^ (-5);
0.411522633744855967078189300411522633744855967078186e-2
```

Example 4:

```
> (-3) ^ (-2.5);
@NaN@
```

Example 5:

```
> diff(sin(x) ^ exp(x));
sin(x)^exp(x) * ((cos(x) * exp(x)) / sin(x) + exp(x) * log(sin(x)))
```

See also: +(7.98), -(7.85), *(7.86), /(7.35)

7.103 powers

Name: powers

special value for global state display

Description:

• powers is a special value used for the global state display. If the global state display is equal to powers, all data will be output in dyadic notation with numbers displayed in a Maple and PARI/GP compatible format. As any value it can be affected to a variable and stored in lists.

See also: decimal (7.25), dyadic (7.39), hexadecimal (7.62), binary (7.15)

7.104 prec

Name: **prec**

controls the precision used in numerical computations.

Description:

- **prec** is a global variable. Its value represents the precision of the floating-point format used in numerical computations.
- A lot of commands try to adapt their intern precision in order to have approximately n correct bits in output, where n is the value of **prec**.

Example 1:

7.105 precision

Name: **precision**

returns the precision necessary to represent a number.

Usage:

 $\mathbf{precision}(x) : \mathsf{constant} \to \mathsf{integer}$

Parameters:

 \bullet x is a dyadic number.

Description:

- **precision**(x) is by definition —x— if x equals 0, NaN, or Inf.
- If x is not zero, it can be uniquely written as $x = m \cdot 2^e$ where m is an odd integer and e is an integer. **precision**(x) returns the number of bits necessary to write m (e.g. $\lceil \log_2(m) \rceil$).

Example 1:

```
> a=round(Pi,20,RN);
> precision(a);
19
> m=mantissa(a);
> ceil(log2(m));
19
```

See also: mantissa (7.82), exponent (7.49)

7.106 prepend

Name: .:

add an element at the beginning of a list.

Usage:

$$x:L: (any type, list) \rightarrow list$$

Parameters:

- x is an object of any type.
- L is a list (possibly empty).

Description:

- ullet .: adds the element x at the beginning of the list L.
- Note that since x may be of any type, it can be in particular a list.

Example 1:

```
> 1.:[|2,3,4|];
[|1, 2, 3, 4|]
```

Example 2:

```
> [|1,2,3|].:[|4,5,6|];
[|[|1, 2, 3|], 4, 5, 6|]
```

Example 3:

```
> 1.:[||];
[|1|]
```

See also: (7.7), @ (7.21)

7.107 print

Name: **print**

prints an expression

Usage:

```
\mathbf{print}(expr1,...,exprn): (any type,..., any type) \rightarrow \mathsf{void}

\mathbf{print}(expr1,...,exprn) > filename: (any type,..., any type, string) \rightarrow \mathsf{void}

\mathbf{print}(expr1,...,exprn) >> filename: (any type,...,any type, string) \rightarrow \mathsf{void}
```

Parameters:

 \bullet expr represents an expression

• filename represents a character sequence indicating a file name

Description:

• **print**(expr1,...,exprn) prints the expressions expr1 through exprn separated by spaces and followed by a newline. If a second argument filename is given after a single ">", the displaying is not output on the standard output of Sollya but if in the file filename that get newly created or overwritten. If a double ">>" is given, the output will be appended to the file filename. The global variables **display**, **midpointmode** and **fullparentheses** have some influence on the formatting of the output (see **display**, **midpointmode** and **fullparentheses**). Remark that if one of the expressions expri given in argument is of type string, the character sequence expri evaluates to is displayed. However, if expri is of type list and this list contains a variable of type string, the expression for the list is displayed, i.e. all character sequences get displayed surrounded by quotes (""). Nevertheless, escape sequences used upon defining character sequences are interpreted immediately.

Example 1:

```
> print(x + 2 + exp(sin(x)));
x + 2 + exp(sin(x))
> print("Hello", "world");
Hello world
> print("Hello", "you", 4 + 3, "other persons.");
Hello you 7 other persons.
```

Example 2:

```
> print("Hello");
Hello
> print([|"Hello"|]);
[|"Hello"|]
> s = "Hello";
> print(s,[|s|]);
Hello [|"Hello"|]
> t = "Hello\tyou";
> print(t,[|t|]);
Hello you [|"Hello you"|]
```

Example 3:

```
> print(x + 2 + exp(sin(x))) > "foo.sol";
> readfile("foo.sol");
x + 2 + exp(sin(x))
```

Example 4:

```
> print(x + 2 + exp(sin(x))) >> "foo.sol";
```

Example 5:

```
> display = decimal;
Display mode is decimal numbers.
> a = evaluate(sin(pi * x), 0.25);
> b = evaluate(sin(pi * x), [0.25; 0.25 + 1b-50]);
0.707106781186547524400844362104849039284835937688470740971
> display = binary;
Display mode is binary numbers.
> print(a);
010001_2 * 2^(-1)
> display = hexadecimal;
Display mode is hexadecimal numbers.
> print(a);
0xb.504f333f9de6484597d89b3754abe9f1d6f60ba88p-4
> display = dyadic;
Display mode is dyadic numbers.
> print(a);
33070006991101558613323983488220944360067107133265b-165
> display = powers;
Display mode is dyadic numbers in integer-power-of-2 notation.
> print(a);
33070006991101558613323983488220944360067107133265 * 2^(-165)
> display = decimal;
Display mode is decimal numbers.
> midpointmode = off;
Midpoint mode has been deactivated.
> print(b);
[0.707106781186547524400844362104849039284835937688449; 0.70710678118654949743721]
7825175573477826462744170488]
> midpointmode = on;
Midpoint mode has been activated.
> print(b);
0.70710678118654~7/9~
> display = dyadic;
Display mode is dyadic numbers.
> print(b);
[2066875436943847413332748968013809022504194195829b-161;165350034955508254441962]
37019385936414432675156571b-164]
> display = decimal;
Display mode is decimal numbers.
> autosimplify = off;
Automatic pure tree simplification has been deactivated.
> fullparentheses = off;
Full parentheses mode has been deactivated.
> print(x + x * ((x + 1) + 1));
x + x * (x + 1 + 1)
> fullparentheses = on;
Full parentheses mode has been activated.
> print(x + x * ((x + 1) + 1));
x + (x * ((x + 1) + 1))
```

See also: write (7.155), printexpansion (7.108), printhexa (7.110), printfloat (7.109), printxml (7.111), readfile (7.118), autosimplify (7.13), display (7.34), midpointmode (7.84), fullparentheses (7.56), evaluate (7.44)

7.108 printexpansion

Name: **printexpansion**

prints a polynomial in Horner form with its coefficients written as a expansions of double precision numbers

Usage:

```
\mathbf{printexpansion}(polynomial) : (function) \rightarrow \mathsf{void}
```

Parameters:

• polynomial represents the polynomial to be printed

Description:

• The command **printexpansion** prints the polynomial *polynomial* in Horner form writing its coefficients as expansions of double precision numbers. The double precision numbers themselves are displayed in hexadecimal memory notation (see **printhexa**). If some of the coefficients of the polynomial *polynomial* are not floating-point constants but constant expressions, they are evaluated to floating-point constants using the global precision **prec**. If a rounding occurs in this evaluation, a warning is displayed. If the exponent range of double precision is not sufficient to display all the mantissa bits of a coefficient, the coefficient is displayed rounded and a warning is displayed. If the argument *polynomial* does not a polynomial, nothing but a warning or a newline is displayed. Constants can be displayed using **printexpansion** since they are polynomials of degree 0.

Example 1:

Example 2:

```
> printexpansion(remez(exp(x),5,[-1;1]));
(0x3ff0002eec887685 + 0x3c691f5f67e9bc02 + 0xb8da4eda3533a5a0) + x * ((0x3ff0002
8359b7449 + 0x3c96606599d49aff + 0xb9341c0421ad8f65 + 0x35d8000000000000) + x *
((0x3fdff2d7e6ec686b + 0x3c7bdc5f3e3c6baa + 0xb8f46ddee438d34c) + x * ((0x3fc54d
6732dfead0 + 0xbc4115bd475b2c9f + 0x38d2035ba90e783d) + x * ((0x3fa66c209b813377
+ 0x3c328b7f6cf9b6de + 0x38dcc2cb5ed93c15 + 0xb57800000000000) + x * (0x3f81e5
542ba3ef82 + 0xbc2bd7453be2c03c + 0xb8c7564467912508)))))
```

Example 3:

```
> verbosity = 1!;
> prec = 3500!;
> printexpansion(pi);
(0x400921fb54442d18 + 0x3ca1a62633145c07 + 0xb92f1976b7ed8fbc + 0x35c4cf98e80417
7d + 0x32631d89cd9128a5 + 0x2ec0f31c6809bbdf + 0x2b5519b3cd3a431b + 0x27e8158536
f92f8a + 0x246ba7f09ab6b6a9 + 0xa0eedd0dbd2544cf + 0x1d779fb1bd1310ba + 0x1a1a63
7ed6b0bff6 + 0x96aa485fca40908e + 0x933e501295d98169 + 0x8fd160dbee83b4e0 + 0x8c
59b6d799ae131c + 0x08f6cf70801f2e28 + 0x05963bf0598da483 + 0x023871574e69a459 +
0x8000000005702db3 + 0x800000000000000)
Warning: the expansion is not complete because of the limited exponent range of double precision.
Warning: rounding occurred while printing.
```

See also: **printhexa** (7.110), **horner** (7.65), **print** (7.107), **prec** (7.104), **remez** (7.121), **taylor** (7.146), **roundcoefficients** (7.128)

7.109 printfloat

Name: printfloat

prints a constant value as a hexadecimal single precision number

Usage:

```
\mathbf{printfloat}(constant) : \mathsf{constant} \to \mathsf{void}
```

Parameters:

• constant represents a constant

Description:

• Prints a constant value as a hexadecimal number on 8 hexadecimal digits. The hexadecimal number represents the integer equivalent to the 32 bit memory representation of the constant considered as a single precision number. If the constant value does not hold on a single precision number, it is first rounded to the nearest single precision number before displayed. A warning is displayed in this case.

Example 1:

```
> printfloat(3);
0x40400000
```

Example 2:

```
> prec=100!;
> verbosity = 1!;
> printfloat(exp(5));
Warning: the given expression is not a constant but an expression to evaluate.
Warning: rounding occurred before printing a value as a simple.
0x431469c5
```

See also: printhexa (7.110)

7.110 printhexa

Name: printhexa

prints a constant value as a hexadecimal double precision number

Usage:

```
\mathbf{printhexa}(constant) : \mathsf{constant} \to \mathsf{void}
```

Parameters:

 $\bullet \ constant$ represents a constant

Description:

• Prints a constant value as a hexadecimal number on 16 hexadecimal digits. The hexadecimal number represents the integer equivalent to the 64 bit memory representation of the constant considered as a double precision number. If the constant value does not hold on a double precision number, it is first rounded to the nearest double precision number before displayed. A warning is displayed in this case.

Example 1:

```
> printhexa(3);
0x400800000000000
```

Example 2:

```
> prec=100!;
> verbosity = 1!;
> printhexa(exp(5));
Warning: the given expression is not a constant but an expression to evaluate.
Warning: rounding occurred before printing a value as a double.
0x40628d389970338f
```

See also: printfloat (7.109), printexpansion (7.108)

7.111 printxml

Name: **printxml**

prints an expression as an MathML-Content-Tree

Usage:

Parameters:

- expr represents a functional expression
- filename represents a character sequence indicating a file name

Description:

• **printxml**(*expr*) prints the functional expression *expr* as a tree of MathML Content Definition Markups. This XML tree can be re-read in external tools or by usage of the **readxml** command. If a second argument *filename* is given after a single "¿", the MathML tree is not output on the standard output of Sollya but if in the file *filename* that get newly created or overwritten. If a double "¿;" is given, the output will be appended to the file *filename*.

Example 1:

```
> printxml(x + 2 + exp(sin(x)));
<?xml version="1.0" encoding="UTF-8"?>
<!-- generated by sollya: http://sollya.gforge.inria.fr/ -->
<!-- syntax: printxml(...); exemple: printxml(x^2-2*x+5); -->
<?xml-stylesheet type="text/xsl" href="http://perso.ens-lyon.fr/nicolas.jourdan/</pre>
mathmlc2p-web.xsl"?>
<?xml-stylesheet type="text/xsl" href="mathmlc2p-web.xsl"?>
<!-- This stylesheet allows direct web browsing of MathML-c XML files (http://o
r file://) -->
<math xmlns="http://www.w3.org/1998/Math/MathML">
<semantics>
<annotation-xml encoding="MathML-Content">
<lambda>
<br/>
<br/>
dvar><ci> x </ci></bvar>
<apply>
<apply>
<plus/>
<apply>
<plus/>
<ci> x </ci>
<cn type="integer" base="10"> 2 </cn>
</apply>
<apply>
<exp/>
<apply>
<sin/>
<ci> x </ci>
</apply>
</apply>
</apply>
</apply>
</lambda>
</annotation-xml>
<annotation encoding="sollya/text">(x + 1b1) + exp(sin(x))</annotation>
</semantics>
```

Example 2:

```
> printxml(x + 2 + exp(sin(x))) > "foo.xml";
```

Example 3:

```
> printxml(x + 2 + exp(sin(x))) >> "foo.xml";
```

See also: readxml (7.119), print (7.107), write (7.155)

7.112 proc

Name: **proc**

defines a Sollya procedure

Usage:

```
\mathbf{proc}(formal\ parameter1,\ formal\ parameter2,...,\ formal\ parameter\ n)\ \mathbf{begin}\ procedure\ body\ \mathbf{end}:\ \mathsf{void}\ \to \mathsf{proc}(formal\ parameter1,\ formal\ parameter2,...,\ formal\ parameter\ n)\ \mathbf{begin}\ procedure\ body\ \mathbf{return}
expression;\ \mathbf{end}:\ \mathsf{any}\ \mathsf{type}\to\mathsf{procedure}
```

Parameters:

- \bullet formal parameter1, formal parameter2 through formal parameter n represent identifiers used as formal parameters
- procedure body represents the imperative statements in the body of the procedure
- expression represents the expression **proc** shall evaluate to

Description:

- The **proc** keyword allows for defining procedures in the Sollya language. These procedures are common Sollya objects that can be applied to actual parameters after definition. Upon such an application, the Sollya interpreter applies the actual parameters to the formal parameters formal parameter1 through formal parameter n and executes the procedure body. The procedure applied to actual parameters evaluates then to the expression expression in the **return** statement after the procedure body or to **void**, if no return statement is given (i.e. a **return void** statement is implicitly given).
- Sollya procedures defined by **proc** have no name. They can be bound to an identifier by assigning the procedure object a **proc** expression produces to an identifier. However, it is possible to use procedures without giving them any name. For instance, Sollya procedures, i.e. procedure objects, can be elements of lists. They can even be given as an argument to other internal Sollya procedures. See also **procedure** on this subject.
- Upon definition of a Sollya procedure using **proc**, no type check is performed. More precisely, the statements in *procedure body* are merely parsed but not interpreted upon procedure definition with **proc**. Type checks are performed once the procedure is applied to actual parameters or to **void**. At this time, it is checked whether the number of actual parameters corresponds to the number of formal parameters. Type checks are further performed upon execution of each statement in *procedure body* and upon evaluation of the expression expression to be returned. Procedures defined by **proc** containing a **quit** or **restart** command cannot be executed (i.e. applied). Upon application of a procedure, the Sollya interpreter checks beforehand for such a statement. If one is found, the application of the procedure to its arguments evaluates to **error**. A warning is displayed. Remark that in contrast to other type or semantical correctness checks, this check is really performed before interpreting any other statement in body of the procedure.
- By means provided by the **var** keyword, it is possible to declare local variables and thus to have full support of recursive procedures. This means a procedure defined using **proc** may contain in its ¡procedure body; an application of itself to some actual parameters: it suffices to assign the procedure (object) to an identifier with an appropriate name.
- Sollya procedures defined using **proc** may return other procedures. Further *procedure body* may contain assignments of locally defined procedure objects to identifiers. See **var** for the particular behaviour of local and global variables.
- The expression expression returned by a procedure is evaluated with regard to Sollya commands, procedures and external procedures. Simplification may be performed. However, an application of a procedure defined by **proc** to actual parameters evaluates to the expression expression that may contain the free global variable or that may be composed.

Example 1:

```
> succ = proc(n) { return n + 1; };
> succ(5);
6
> 3 + succ(0);
4
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
```

Example 2:

```
> add = proc(m,n) { var res; res := m + n; return res; };
> add(5,6);
11
> add;
proc(m, n)
begin
var res;
res := (m) + (n);
return res;
> verbosity = 1!;
> add(3);
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
error
> add(true,false);
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
Warning: the given expression or command could not be handled.
Warning: the given expression or command could not be handled.
error
```

Example 3:

```
> succ = proc(n) { return n + 1; };
> succ(5);
6
> succ(x);
1 + x
```

Example 4:

```
> hey = proc() { print("Hello world."); };
> hey();
Hello world.
> print(hey());
Hello world.
void
> hey;
proc()
begin
print("Hello world.");
return void;
end
```

Example 5:

```
> fac = proc(n) { var res; if (n == 0) then res := 1 else res := n * fac(n - 1);
  return res; };
> fac(5);
120
> fac(11);
39916800
> fac;
proc(n)
begin
  var res;
if (n) == (0) then
  res := 1
else
  res := (n) * (fac((n) - (1)));
  return res;
end
```

Example 6:

```
> myprocs = [| proc(m,n) { return m + n; }, proc(m,n) { return m - n; } |];
> (myprocs[0])(5,6);
11
> (myprocs[1])(5,6);
-1
> succ = proc(n) { return n + 1; };
> pred = proc(n) { return n - 1; };
> applier = proc(p,n) { return p(n); };
> applier(succ,5);
6
> applier(pred,5);
4
```

Example 7:

```
> verbosity = 1!;
> myquit = proc(n) { print(n); quit; };
> myquit;
proc(n)
begin
print(n);
quit;
return void;
end
> myquit(5);
Warning: a quit or restart command may not be part of a procedure body.
The procedure will not be executed.
Warning: an error occured while executing a procedure.
Warning: the given expression or command could not be handled.
error
```

Example 8:

```
> printsucc = proc(n) { var succ; succ = proc(n) { return n + 1; }; print("Succe
ssor of",n,"is",succ(n)); };
> printsucc(5);
Successor of 5 is 6
```

Example 9:

```
> makeadd = proc(n) { var add; print("n =",n); add = proc(m,n) { return n + m; }
; return add; };
> makeadd(4);
n = 4
proc(m, n)
begin
nop;
return (n) + (m);
end
> (makeadd(4))(5,6);
n = 4
11
```

See also: return (7.124), externalproc (7.51), void (7.153), quit (7.114), restart (7.123), var (7.151)

7.113 procedure

Name: procedure

defines and assigns a Sollya procedure

Usage:

```
procedure identifier(formal\ parameter1,\ formal\ parameter2,...,\ formal\ parameter\ n) begin procedure\ body\ \mathbf{end}:\ \mathsf{void}\ \to\ \mathsf{void}
procedure identifier(formal\ parameter1,\ formal\ parameter2,...,\ formal\ parameter\ n) begin procedure
```

 $body \ \mathbf{return} \ expression; \ \mathbf{end} : \ \mathsf{any} \ \mathsf{type} \to \mathsf{void}$

Parameters:

- identifier represents the name of the procedure to be defined and assigned
- \bullet formal parameter1, formal parameter2 through formal parameter n represent identifiers used as formal parameters

- procedure body represents the imperative statements in the body of the procedure
- expression represents the expression **procedure** shall evaluate to

Description:

• The **procedure** keyword allows for defining and assigning procedures in the Sollya language. It is an abbreviation to a procedure definition using **proc** with the same formal parameters, procedure body and return-expression followed by an assignment of the procedure (object) to the identifier *identifier*. In particular, all rules concerning local variables declared using the **var** keyword apply for **procedure**.

Example 1:

```
> procedure succ(n) { return n + 1; };
> succ(5);
6
> 3 + succ(0);
4
> succ;
proc(n)
begin
nop;
return (n) + (1);
end
```

See also: \mathbf{proc} (7.112), \mathbf{var} (7.151)

7.114 quit

Name: **quit** quits Sollya

Usage:

 $\mathbf{quit}:\mathsf{void}\to\mathsf{void}$

Description:

• The command quit, when executed abandons the execution of a Sollya script and leaves the Sollya interpreter unless the quit command is executed in a Sollya script read into a main Sollya script by execute or #include. Upon exiting the Sollya interpreter, all state is thrown away, all memory is deallocated, all bound libraries are unbound and the temporary files produced by plot and externalplot are deleted. If the quit command does not lead to the abandon of the Sollya interpreter, a warning is displayed.

Example 1:

```
> quit;
```

See also: restart (7.123), execute (7.45), plot (7.97), externalplot (7.50)

7.115 range

Name: range

keyword representing a range type

Usage:

range: type type

Description:

• range represents the range type for declarations of external procedures by means of externalproc. Remark that in contrast to other indicators, type indicators like range cannot be handled outside the externalproc context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), boolean (7.16), constant (7.22), function (7.57), integer (7.69), list of (7.76), string (7.139)

7.116 rational approx

Name: rationalapprox

returns a fraction close to a given number.

Usage:

```
rational approx(x,n) : (constant, integer) \rightarrow function
```

Parameters:

- \bullet x is a number to approximate.
- *n* is a integer (representing a format).

Description:

- rationalapprox(x,n) returns a constant function of the form a/b where a and b are integers. The value a/b is an approximation of x. The quality of this approximation is determined by the parameter n that indicates the number of correct bits that a/b should have.
- The command is not safe in the sense that it is not ensured that the error between a/b and x is less than 2^{-n} .
- The following algorithm is used: x is first rounded downwards and upwards to a format of n bits, thus obtaining an interval $[x_l, x_u]$. This interval is then developed into a continued fraction as far as the representation is the same for every elements of $[x_l, x_u]$. The corresponding fraction is returned.
- Since rational numbers are not a primitive object of Sollya, the fraction is returned as a constant function. It can be quite amazing, because Sollya immediately simplifies a constant function by evaluating it when the constant has to be displayed. To avoid this, you can use **print** (that displays the expression representing the constant and not the constant itself) or the commands **numerator** and **denominator**.

Example 1:

```
> pi10 = rationalapprox(Pi,10);
> pi50 = rationalapprox(Pi,50);
> pi100 = rationalapprox(Pi,100);
> print( pi10, ": ", simplify(floor(-log2(abs(pi10-Pi)/Pi))), "bits." );
22 / 7 : 11 bits.
> print( pi50, ": ", simplify(floor(-log2(abs(pi50-Pi)/Pi))), "bits." );
90982559 / 28960648 : 50 bits.
> print( pi100, ": ", simplify(floor(-log2(abs(pi100-Pi)/Pi))), "bits." );
4850225745369133 / 1543874804974140 : 101 bits.
```

Example 2:

```
> a=0.1;
> b=rationalapprox(a,4);
> numerator(b); denominator(b);
1
10
> print(simplify(floor(-log2(abs((b-a)/a)))), "bits.");
166 bits.
```

See also: print (7.107), numerator (7.90), denominator (7.28)

7.117 RD

Name: RD

constant representing rounding-downwards mode.

Description:

• RD is used in command round to specify that the value x must be rounded to the greatest floating-point number y such that $y \le x$.

Example 1:

```
> display=binary!;
> round(Pi,20,RD);
1.1001001000011111101_2 * 2^(1)
```

See also: RZ (??), RU (??), RN (??), round (7.127)

7.118 readfile

Name: readfile

reads the content of a file into a string variable

Usage:

```
readfile(filename) : string \rightarrow string
```

Parameters:

• filename represents a character sequence indicating a file name

Description:

• **readfile** opens the file indicated by *filename*, reads it and puts its contents in a character sequence of type **string** that is returned. If the file indicated by *filename* cannot be opened for reading, a warning is displayed and **readfile** evaluates to an **error** variable of type **error**.

Example 1:

```
> print("Hello world") > "myfile.txt";
> t = readfile("myfile.txt");
> t;
Hello world
```

Example 2:

```
> verbosity=1!;
> readfile("afile.txt");
Warning: the file "afile.txt" could not be opened for reading.
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
error
```

See also: **parse** (7.94), **execute** (7.45), **write** (7.155), **print** (7.107)

7.119 readxml

Name: readxml

reads an expression written as a MathML-Content-Tree in a file

Usage:

 $\mathbf{readxml}(filename): \mathsf{string} \to \mathsf{function} -- \mathsf{error}$

Parameters:

• filename represents a character sequence indicating a file name

Description:

• readxml(filename) reads the first occurrence of a lambda application with one bounded variable on applications of the supported basic functions in file filename and returns it as a Sollya functional expression. If the file filename does not contain a valid MathML-Content tree, readxml tries to find an "annotation encoding" markup of type "sollya/text". If this annotation contains a character sequence parseable by parse, readxml returns that expression. Otherwise readxml displays a warning and returns an error variable of type error.

Example 1:

```
> readxml("readxmlexample.xml");
2 + x + exp(sin(x))
```

See also: printxml (7.111), readfile (7.118), parse (7.94)

7.120 relative

Name: perturb

indicates a relative error for externalplot

Usage:

perturb : absolute relative

${\bf Description:}$

• The use of **perturb** in the command **externalplot** indicates that during plotting in **externalplot** a relative error is to be considered. See **externalplot** for details.

Example 1:

```
> bashexecute("gcc -fPIC -c externalplotexample.c");
> bashexecute("gcc -shared -o externalplotexample externalplotexample.o -lgmp -l mpfr");
> externalplot("./externalplotexample",relative,exp(x),[-1/2;1/2],12,perturb);
```

See also: externalplot (7.50), absolute (7.2), bashexecute (7.14)

7.121 remez

Name: remez

computes the minimax of a function on an interval.

Usage:

```
\mathbf{remez}(f,\ n,\ range,\ w,\ quality): (function, integer, range, function, constant) \rightarrow function \mathbf{remez}(f,\ L,\ range,\ w,\ quality): (function, list, range, function, constant) \rightarrow function
```

Parameters:

- f is the function to be approximated
- \bullet n is the degree of the polynomial that must approximate f
- L is a list of monomials that can be used to represent the polynomial that must approximate f
- range is the interval where the function must be approximated
- w (optional) is a weight function. Default is 1.
- quality (optional) is a parameter that controls the quality of the returned polynomial p, with respect to the exact minimax p^* . Default is 1e-5.

Description:

- remez computes an approximation of the function f with respect to the weight function w on the interval range. More precisely, it searches a polynomial p such that $||pw f||_{\infty}$ is (almost minimal) among all polynomials p of a certain form. The norm is the infinite norm, e.g. $||g||_{\infty} = \max\{|g(x)|, x \in \text{range}\}$.
- If w = 1 (the default case), it consists in searching the best polynomial approximation of f with respect to the absolute error. If f = 1 and w is of the form 1/g, it consists in searching the best polynomial approximation of g with respect to the relative error.
- If n is given, the polynomial p is searched among the polynomials with degree not greater than n. If L is given, the polynomial p is searched as a linear combination of monomials X^k where k belongs to L. L may contain ellipses but cannot be end-elliptic.
- The polynomial is obtained by a convergent iteration called Remez' algorithm. The algorithm computes a sequence p_1, \ldots, p_k, \ldots such that $e_k = ||p_k w f||_{\infty}$ converges towards the optimal value e. The algorithm is stopped when the relative error between e_k and e is less than quality.
- Note: the algorithm may not converge in certain cases. Moreover, it may converge towards a polynomial that is not optimal. These cases correspond to the cases when Haar's condition is not fulfilled. See [Cheney Approximation theory] for details.

Example 1:

```
> p = remez(exp(x),5,[0;1]);
> degree(p);
5
> dirtyinfnorm(p-exp(x),[0;1]);
0.112956994145777826976474581166951642161746831108566851946e-5
```

Example 2:

```
> p = remez(1,[|0,2,4,6,8|],[0,Pi/4],1/cos(x));
> canonical=on!;
> p;
0.99999999943937321805410306927692541203518389031685 + (-0.49999999571556857755
9574413581766900435294037106493613) * x^2 + 0.4166661323347363211356520614037568
56904074913853682927e-1 * x^4 + (-0.13886529147145676586432612354047793709150039
0886117905e-2) * x^6 + 0.243726791772232726278486347508569482304048351169722066e
-4 * x^8
```

Example 3:

```
> p1 = remez(exp(x),5,[0;1],default,1e-5);
> p2 = remez(exp(x),5,[0;1],default,1e-10);
> p3 = remez(exp(x),5,[0;1],default,1e-15);
> dirtyinfnorm(p1-exp(x),[0;1]);
0.112956994145777826976474581166951642161746831108566851946e-5
> dirtyinfnorm(p2-exp(x),[0;1]);
0.11295698022747876310115474830183586181755211237038868201e-5
> dirtyinfnorm(p3-exp(x),[0;1]);
0.112956980227478673699869114581631945410176127063383668503e-5
```

See also: dirtyinfnorm (7.32), infnorm (7.68)

7.122 rename

Name: rename

rename the free variable.

Usage:

rename(ident1,ident2): void

Parameters:

- *ident1* is the current name of the free variable.
- *ident2* is a fresh name.

Description:

- **rename** lets one change the name of the free variable. Sollya can handle only one free variable at a time. The first time in a session that an unbound name is used in a context where it can be interpreted as a free variable, the name is used to represent the free variable of Sollya. In the following, this name can be changed using **rename**.
- Be careful: if *ident2* has been set before, its value will be lost. Use the command **isbound** to know if *ident2* is already used or not.
- If *ident1* is not the current name of the free variable, an error occurs.
- If **rename** is used at a time when the name of the free variable has not been defined, *ident1* is just ignored and the name of the free variable is set to *ident2*.

Example 1:

```
> f=sin(x);
> f;
sin(x)
> rename(x,y);
> f;
sin(y)
```

Example 2:

```
> a=1;
> f=sin(x);
> rename(x,a);
> a;
a
> f;
sin(a)
```

Example 3:

```
> verbosity=1!;
> f=sin(x);
> rename(y,z);
Warning: the current free variable is named "x" and not "y". Can only rename the free variable.
The last command will have no effect.
```

Example 4:

```
> rename(x,y);
> isbound(x);
false
> isbound(y);
true
```

See also: **isbound** (7.71)

7.123 restart

Name: restart

brings Sollya back to its initial state

Usage:

 $\mathbf{restart}: \mathsf{void} \to \mathsf{void}$

Description:

• The command **restart** brings Sollya back to its initial state. All current state is abandoned, all libraries unbound and all memory freed. The **restart** command has no effect when executed inside a Sollya script read into a main Sollya script using **execute**. It is executed in a Sollya script included by a #include macro. Using the **restart** command in nested elements of imperative programming like for or while loops is possible. Since in most cases abandoning the current state of Sollya means altering a loop invariant, warnings of the impossibility of continuing a loop may follow unless the state is rebuilt.

Example 1:

```
> print(exp(x));
exp(x)
> a = 3;
> restart;
The tool has been restarted.
> print(x);
x
> a;
Warning: the identifier "a" is neither assigned to, nor bound to a library funct ion nor equal to the current free variable.
Will interpret "a" as "x".
x
```

Example 2:

```
> print(exp(x));
exp(x)
> for i from 1 to 10 do {
>      print(i);
>      if (i == 5) then restart;
> };
1
2
3
4
5
The tool has been restarted.
Warning: the tool has been restarted inside a for loop.
The for loop will no longer be executed.
```

Example 3:

```
> print(exp(x));
exp(x)
> a = 3;
> for i from 1 to 10 do {
      print(i);
      if (i == 5) then {
>
          restart;
>
          i = 7;
      };
> };
1
2
3
4
5
The tool has been restarted.
8
9
10
> print(x);
х
> a;
Warning: the identifier "a" is neither assigned to, nor bound to a library funct
ion nor equal to the current free variable.
Will interpret "a" as "x".
```

See also: quit (7.114), execute (7.45), while (??), for (??)

7.124 return

Name: return

indicates an expression to be returned in a procedure

Usage:

return expression: void

Parameters:

• expression represents the expression to be returned

Description:

• The keyword **return** allows for returning the (evaluated) expression *expression* at the end of a begin-end-block (-block) used as a Sollya procedure body. See **proc** for further details concerning Sollya procedure definitions.

Statements for returning expressions using **return** are only possible at the end of a begin-end-block used as a Sollya procedure body. Only one **return** statement can be given per begin-end-block.

• If at the end of a procedure definition using **proc** no **return** statement is given, a **return void** statement is implicitly added. Procedures, i.e. procedure objects, when printed out in Sollya defined with an implicit **return void** statement are displayed with this statement explicited.

Example 1:

```
> succ = proc(n) { var res; res := n + 1; return res; };
> succ(5);
6
> succ;
proc(n)
begin
var res;
res := (n) + (1);
return res;
end
```

Example 2:

```
> hey = proc(s) { print("Hello",s); };
> hey("world");
Hello world
> hey;
proc(s)
begin
print("Hello", s);
return void;
end
```

See also: **proc** (7.112), **void** (7.153)

7.125 revert

Name: **revert** reverts a list.

Usage:

 $\mathbf{revert}(L) : \mathsf{list} \to \mathsf{list}$

Parameters:

 \bullet L is a list.

Description:

- \bullet $\mathbf{revert}(L)$ returns the same list, but with its elements in reverse order.
- ullet If L is an end-elliptic list, **revert** will fail with an error.

Example 1:

```
> revert([| |]);
[| |]
```

Example 2:

```
> revert([|2,3,5,2,1,4|]);
[|4, 1, 2, 5, 3, 2|]
```

7.126 RN

Name: $\mathbf{R}\mathbf{N}$

constant representing rounding-to-nearest mode.

Description:

• RN is used in command round to specify that the value must be rounded to the nearest representable floating-point number.

Example 1:

```
> display=binary!;
> round(Pi,20,RN);
1.100100100001111111_2 * 2^(1)
```

See also: RD (??), RU (??), RZ (??), round (7.127)

7.127 round

Name: round

rounds a number to a floating-point format.

Usage:

```
\mathbf{round}(x, n, mode) : (\mathsf{constant}, \mathsf{integer}, \mathbf{RD} - \mathbf{RU} - \mathbf{RN} - \mathbf{RZ}) \to \mathsf{constant}
```

Parameters:

- \bullet x is a constant to be rounded.
- \bullet *n* is the precision of the target format.
- *mode* is the desired rounding mode.

Description:

- $\mathbf{round}(x, n, mode)$ rounds x to a floating-point number with precision n, according to rounding-mode mode.
- Subnormal numbers are not handled. The range of possible exponents is the range used for all numbers represented in Sollya (e.g. basically the range used in the library MPFR). Please use the functions double, doubleextended, doubledouble and tripledouble for roundings to classical formats with their range of exponents.

Example 1:

```
> display=binary!;
> round(Pi,20,RN);
1.100100100001111111_2 * 2^(1)
```

Example 2:

```
> display=binary!;
> a=2^(-1100);
> round(a,53,RN);
1._2 * 2^(-1100)
> double(a);
0
```

See also: RN (??), RD (??), RU (??), RZ (??), double (7.36), doubleextended (7.38), double-double (7.37), tripledouble (7.149), roundcoefficients (7.128), roundcorrectly (7.129)

7.128 roundcoefficients

Name: roundcoefficients

rounds the coefficients of a polynomial to classical formats.

Usage:

```
roundcoefficients(p,L): (function, list) \rightarrow function
```

Parameters:

- p is a function. Usually a polynomial.
- L is a list of formats.

Description:

- If p is a polynomial and L a list of floating-point formats, **roundcoefficients**(p,L) rounds each coefficient of p to the corresponding format in L.
- If p is not a polynomial, **roundcoefficients** does not do anything.
- If L contains other elements than **D**, **double**, **DD**, **doubledouble**, **TD** and **tripledouble**, an error occurs.
- The coefficients in p corresponding to X^i is rounded to the format L[i]. If L does not contain enough elements (e.g. if length(L); length(L); length(L); a warning is displayed. However, the coefficients corresponding to an element of L are rounded. The last coefficients (that do not have a corresponding element in L) are kept with their own precision. If L contains too much elements, the last useless elements are ignored. In particular L may be end-elliptic in which case **roundcoefficients** has the natural behavior.

Example 1:

Example 2:

Example 3:

```
> p=exp(1) + x*(exp(2) + x*exp(3));
> verbosity=1!;
> display=binary!;
> roundcoefficients(p,[|DD,D|]);
Warning: the number of the given formats does not correspond to the degree of th
e given polynomial.
Warning: the 0th coefficient of the given polynomial does not evaluate to a floa
ting-point constant without any rounding.
Will evaluate the coefficient in the current precision in floating-point before
rounding to the target format.
Warning: the 1th coefficient of the given polynomial does not evaluate to a floa
ting-point constant without any rounding.
Will evaluate the coefficient in the current precision in floating-point before
rounding to the target format.
Warning: rounding may have happened.
001010110001000000010011101_2 * 2^(1) + x * (1.1101100011100110010010111100011010
10000010110000101100000111001011100101001_2 * 2^(4))
```

See also: double (7.36), doubledouble (7.37), tripledouble (7.149)

7.129 roundcorrectly

Name: roundcorrectly

rounds an approximation range correctly to some precision

Usage:

```
roundcorrectly(\mathit{range}): range \rightarrow constant
```

Parameters:

• range represents a range in which an exact value lies

Description:

• Let range be a range of values, determined by some approximation process, safely bounding an unknown value v. The command $\mathbf{roundcorrectly}(range)$ determines a precision such that for this precision, rounding to the nearest any value in range yields to the same result, i.e. to the correct rounding of v. If no such precision exists, a warning is displayed and $\mathbf{roundcorrectly}$ evaluates to NaN.

Example 1:

```
> printbinary(roundcorrectly([1.010001_2; 1.0101_2]));
1.01_2 * 2^(0)
> printbinary(roundcorrectly([1.00001_2; 1.001_2]));
1._2 * 2^(0)
```

Example 2:

```
> roundcorrectly([-1; 1]);
@NaN@
```

See also: round (7.127)

7.130 RU

Name: RU

constant representing rounding-upwards mode.

Description:

• **RU** is used in command **round** to specify that the value x must be rounded to the smallest floating-point number y such that $x \leq y$.

Example 1:

```
> display=binary!;
> round(Pi,20,RU);
1.100100100001111111_2 * 2^(1)
```

See also: **RZ** (??), **RD** (??), **RN** (??), **round** (7.127)

7.131 RZ

Name: **RZ**

constant representing rounding-to-zero mode.

Description:

• **RZ** is used in command **round** to specify that the value must be rounded to the closest floating-point number towards zero. It just consists in truncate the value to the desired format.

Example 1:

```
> display=binary!;
> round(Pi,20,RZ);
1.1001001000011111101_2 * 2^(1)
```

See also: RD (??), RU (??), RN (??), round (7.127)

7.132 searchgal

Name: searchgal

searches for a preimage of a function such that the rounding the image commits an error smaller than a constant

Usage:

```
\begin{aligned} \mathbf{searchgal}(function,\ start,\ preimage\ precision,\ steps,\ format,\ error\ bound): \ (\mathsf{function},\ \mathsf{constant},\ \mathsf{integer},\ \mathsf{biteger},\ \mathsf{bound}): \ (\mathsf{function},\ \mathsf{constant}) \to \mathsf{list} \\ \mathbf{searchgal}(\mathit{list}\ of\ functions,\ start,\ preimage\ precision,\ steps,\ \mathit{list}\ of\ format,\ \mathit{list}\ of\ error\ bounds): \ (\mathsf{list},\ \mathsf{constant},\ \mathsf{integer},\ \mathsf{integer},\ \mathsf{list},\ \mathsf{list}) \to \mathsf{list} \end{aligned}
```

Parameters:

- \bullet function represents the function to be considered
- start represents a value around which the search is to be performed
- preimage precision represents the precision (discretisation) for the eligible preimage values
- steps represents the log2 of the number of search steps to be performed
- format represents the format the image of the function is to be rounded to

- error bound represents a upper bound on the relative rounding error when rounding the image
- list of functions represents the functions to be considered
- list of formats represents the respective formats the images of the functions are to be rounded to
- list of error bounds represents a upper bound on the relative rounding error when rounding the image

Description:

• The command searchgal searches for a preimage z of a function function or a list of functions list of functions such that z is a floating-point number with preimage precision significant mantissa bits and the image y of the function, respectively each image y_i of the functions, rounds to format format respectively to the corresponding format in ilist of format; with a relative rounding error less than error bound respectively the corresponding value in list of error bounds. During this search, at most 2 raised to steps attempts are made. The search starts with a preimage value equal to start. This value is then increased and decreased by 1 ulp in precision ipreimage precision; until a value is found or the step limit is reached. If the search finds an appropriate preimage z, searchgal evaluates to a list containing this value. Otherwise, searchgal evaluates to an empty list.

Example 1:

```
> searchgal(log(x),2,53,15,DD,1b-112);

[| |]

> searchgal(log(x),2,53,18,DD,1b-112);

[|0.20000000000384972054234822280704975128173828125e1|]
```

Example 2:

```
> f = exp(x);
> s = searchgal(f,2,53,18,DD,1b-112);
> if (s != [||]) then {
> v = s[0];
> print("The rounding error is 2^(",evaluate(log2(abs(DD(f)/f - 1)),v),")");
> } else print("No value found");
The rounding error is 2^( -0.112106878438809380148206984258358542322113874177832 146e3 )
```

Example 3:

```
> searchgal([|sin(x),cos(x)|],1,53,15,[|D,D|],[|1b-62,1b-60|]);
[|0.100000000015949463971764998859725892543792724609375e1|]
```

See also: round (7.127), **D** (???), **DD** (???), **TD** (???), double (7.36), doubledouble (7.37), tripledouble (7.149), evaluate (7.44), worstcase (7.154)

7.133 simplify

Name: simplify

simplifies an expression representing a function

Usage:

 $simplify(function) : function \rightarrow function$

Parameters:

• function represents the expression to be simplified

Description:

• The command **simplify** simplifies constant subexpressions of the expression given in argument representing the function function. Those constant subexpressions are evaluated in using floating-point arithmetic with the global precision **prec**.

Example 1:

```
> print(simplify(sin(pi * x)));
sin(0.31415926535897932384626433832795028841971693993750801e1 * x)
> print(simplify(erf(exp(3) + x * log(4))));
erf(0.200855369231876677409285296545817178969879078385543785e2 + x * 0.138629436
111989061883446424291635313615100026872049663e1)
```

Example 2:

```
> prec = 20!;
> t = erf(0.5);
> s = simplify(erf(0.5));
> prec = 200!;
> t;
0.52049987781304653768274665389196452873645157575796370005880583
> s;
0.52050018310546875
```

See also: simplifysafe (7.134), autosimplify (7.13), prec (7.104), evaluate (7.44)

7.134 simplifysafe

Name: simplifysafe

simplifies an expression representing a function

Usage:

```
simplifysafe(function) : function \rightarrow function
```

Parameters:

• function represents the expression to be simplified

Description:

• The command **simplifysafe** simplifies the expression given in argument representing the function function. The command **simplifysafe** does not endanger the safety of computations even in Sollya's floating-point environment: the function returned is mathematically equal to the function function. Remark that the simplification provided by **simplifysafe** is not perfect: they may exist simpler equivalent expressions for expressions returned by **simplifysafe**.

Example 1:

```
> print(simplifysafe((6 + 2) + (5 + exp(0)) * x));
8 + 6 * x
```

Example 2:

```
> print(simplifysafe((log(x - x + 1) + asin(1))));
(pi) / 2
```

Example 3:

```
> print(simplifysafe((log(x - x + 1) + asin(1)) - (atan(1) * 2)));
(pi) / 2 - (pi) / 4 * 2
```

See also: simplify (7.133), autosimplify (7.13)

$7.135 \sin$

Name: **sin** the sine function.

Description:

- sin is the usual sine function.
- It is defined for every real number x.

See also: asin (7.9), cos (7.23), tan (7.144)

7.136 sinh

Name: sinh

the hyperbolic sine function.

Description:

- sinh is the usual hyperbolic sine function: $\sinh(x) = \frac{e^x e^{-x}}{2}$.
- It is defined for every real number x.

See also: asinh (7.10), cosh (7.24), tanh (7.145)

7.137 sort

Name: sort

sorts a list of real numbers.

Usage:

$$\mathbf{sort}(L): \mathsf{list} \to \mathsf{list}$$

Parameters:

 \bullet L is a list.

Description:

- If L contains only constant values, $\mathbf{sort}(L)$ returns the same list, but sorted increasingly.
- If L contains at least one element that is not a constant, the command fails with a type error.
- If L is an end-elliptic list, **sort** will fail with an error.

Example 1:

```
> sort([| |]);
[| |]
> sort([|2,3,5,2,1,4|]);
[|1, 2, 2, 3, 4, 5|]
```

7.138 sqrt

Name: **sqrt** square root.

Description:

- sqrt is the square root, e.g. the inverse of the function square: $\sqrt(y)$ is the unique positive x such that $x^2 = y$.
- It is defined only for x in $[0; +\infty]$.

7.139 string

Name: string

keyword representing a string type

Usage:

string: type type

Description:

• string represents the string type for declarations of external procedures by means of externalproc. Remark that in contrast to other indicators, type indicators like string cannot be handled outside the externalproc context. In particular, they cannot be assigned to variables.

See also: externalproc (7.51), boolean (7.16), constant (7.22), function (7.57), integer (7.69), list of (7.76), range (7.115)

7.140 subpoly

Name: subpoly

restricts the monomial basis of a polynomial to a list of monomials

Usage:

```
\mathbf{subpoly}(polynomial, list) : (function, list) \rightarrow function
```

Parameters:

- polynomial represents the polynomial the coefficients are taken from
- list represents the list of monomials to be taken

Description:

• **subpoly** extracts the coefficients of a polynomial *polynomial* and builds up a new polynomial out of those coefficients associated to monomial degrees figuring in the list *list*. If *polynomial* represents a function that is not a polynomial, subpoly returns 0. If *list* is a list that is end-elliptic, let be j the last value explicitly specified in the list. All coefficients of the polynomial associated to monomials greater or equal to j are taken.

Example 1:

```
> p = taylor(exp(x),5,0);
> s = subpoly(p,[|1,3,5|]);
> print(p);
1 + x * (1 + x * (0.5 + x * (1 / 6 + x * (1 / 24 + x / 120))))
> print(s);
x * (1 + x^2 * (1 / 6 + x^2 / 120))
```

Example 2:

```
\begin{array}{l} > \text{p = remez(atan(x),10,[-1,1]);} \\ > \text{subpoly(p,[|1,3,5...|]);} \\ \text{x * (0.999866329465927392192206568432088436991654470572188 + x^2 * ((-0.33030478504971132950658277728545438994810895546443) + x^2 * (0.180159294636895327241868940582645835027165398881204 + x * (0.184005739836594527600896857815217464340110063213703e-46 + x * ((-0.851563508341582145150897325769046842604973435036432e-1) + x * ((-0.204304173639774340598827812338387968056700475880779e-46) + x * (0.208451141756196733464162256100630144371647287176463e-1 + x * 0.788341690807029292048116493729741056214183302078353e-47))))))))\\ \end{array}
```

```
> subpoly(exp(x),[|1,2,3|]);
0
```

See also: roundcoefficients (7.128), taylor (7.146), remez (7.121)

7.141 substitute

Name: substitute

replace the occurences of the free variable in an expression.

Usage:

```
\mathbf{substitute}(f,g): (\mathsf{function}, \mathsf{function}) \to \mathsf{function}
\mathbf{substitute}(f,t): (\mathsf{function}, \mathsf{constant}) \to \mathsf{constant}
```

Parameters:

- \bullet f is a function.
- \bullet g is a function.
- \bullet t is a real number.

Description:

- substitute(f, g) produces the function ($f \circ g$) : $x \mapsto f(g(x))$.
- substitute (f, t) is the constant f(t). Note that the constant is represented by its expression until it has been evaluated (exactly the same way as if you type the expression f replacing instances of the free variable by t).
- If f is stored in a variable F. It is absolutely equivalent to writing F(g) or F(t).

Example 1:

```
> f=sin(x);
> g=cos(x);
> substitute(f,g);
sin(cos(x))
> f(g);
sin(cos(x))
```

Example 2:

```
> a=1;
> f=sin(x);
> substitute(f,a);
0.841470984807896506652502321630298999622563060798373
> f(a);
0.841470984807896506652502321630298999622563060798373
```

$7.142 \quad \sup$

Name: sup

gives the upper bound of an interval.

Usage:

```
\mathbf{sup}(I): \mathsf{range} \to \mathsf{constant}
\mathbf{sup}(x): \mathsf{constant} \to \mathsf{constant}
```

Parameters:

- \bullet I is an interval.
- \bullet x is a real number.

Description:

- Returns the upper bound of the interval *I*. Each bound of an interval has its own precision, so this command is exact, even if the current precision is too small to represent the bound.
- When called on a real number x, \sup considers it as an interval formed of a single point: [x, x]. In other words, \sup behaves like the identity.

Example 1:

```
> sup([1;3]);
3
> sup(5);
5
```

Example 2:

```
> display=binary!;
> I=[0; 0.111110000011111_2];
> sup(I);
1.11110000011111_2 * 2^(-1)
> prec=12!;
> sup(I);
1.11110000011111_2 * 2^(-1)
```

See also: inf (7.67), mid (7.83)

7.143 tail

Name: tail

gives the tail of a list.

Usage:

 $\mathbf{tail}(L): \mathsf{list} \to \mathsf{any} \; \mathsf{type}$

Parameters:

 \bullet L is a list.

Description:

- tail(L) returns the list L without its first element.
- \bullet If L is empty, the command will fail with an error.
- tail can also be used with end-elliptic lists. In this case, the result of tail is also an end-elliptic list.

Example 1:

```
> tail([|1,2,3|]);
[|2, 3|]
> tail([|1,2...|]);
[|2...|]
```

See also: head (7.61)

7.144 tan

Name: tan

the tangent function.

Description:

- tan is the tangent function, defined by tan(x) = sin(x)/cos(x).
- It is defined for every real number x that is not of the form $n\pi + \pi/2$ where n is an integer.

See also: atan (7.11), cos (7.23), sin (7.135)

7.145 tanh

Name: tanh

the hyperbolic tangent function.

Description:

- tanh is the hyperbolic tangent function, defined by tanh(x) = sinh(x)/cosh(x).
- It is defined for every real number x.

See also: atanh (7.12), cosh (7.24), sinh (7.136)

7.146 taylor

Name: taylor

computes a Taylor expansion of a function in a point

Usage:

 $\mathbf{taylor}(function, degree, point) : (function, integer, constant) \rightarrow function$

Parameters:

- function represents the function to be expanded
- degree represents the degree of the expansion to be delivered
- point represents the point in which the function is to be developed

Description:

• The command **taylor** returns an expression that is a Taylor expansion of function function in point point having the degree degree. Let f be the function function, t be the point point and n be the degree degree. Then, **taylor**(function, degree, point) evaluates to an expression mathematically equal to

$$\sum_{i=0}^{n} \frac{f^{(i)}}{i!} (x-t)^{i}$$

Remark that **taylor** evaluates to 0 if the degree degree is negative.

Example 1:

```
> print(taylor(exp(x),5,0));
1 + x * (1 + x * (0.5 + x * (1 / 6 + x * (1 / 24 + x / 120))))
```

Example 2:

```
> print(taylor(asin(x),7,0));
x * (1 + x^2 * (1 / 6 + x^2 * (9 / 120 + x^2 * 225 / 5040)))
```

```
> print(taylor(erf(x),6,0));
x * (1 / sqrt((pi) / 4) + x^2 * ((sqrt((pi) / 4) * 4 / (pi) * (-2)) / 6 + x^2 * (sqrt((pi) / 4) * 4 / (pi) * 12) / 120))
```

See also: \mathbf{remez} (7.121)

7.147 taylorrecursions

Name: taylorrecursions

controls the number of recursion steps when applying Taylor's rule.

Description:

- taylorrecursions is a global variable. Its value represents the number of steps of recursion that are used when applying Taylor's rule. This rule is applied by the interval evaluator present in the core of Sollya (and particularly visible in commands like infnorm).
- To improve the quality of an interval evaluation of a function f, in particular when there are problems of decorrelation), the evaluator of Sollya uses Taylor's rule: $f([a,b]) \subseteq f(m) + [a-m, b-m] \cdot f'([a,b])$ where $m = \frac{a+b}{2}$. This rule can be applied recursively. The number of step in this recursion process is controlled by **taylorrecursions**.
- Setting taylorrecursions to 0 makes Sollya use this rule only one time; setting it to 1 makes Sollya use the rule two times, and so on. In particular: the rule is always applied at least once.

Example 1:

```
> f=exp(x);
> p=remez(f,3,[0;1]);
> taylorrecursions=0;
The number of recursions for Taylor evaluation has been set to 0.
> evaluate(f-p, [0;1]);
[-0.468393649043687401639739552931015355858675716815424;0.4694778180126977487653
76148435266259461851121717393]
> taylorrecursions=1;
The number of recursions for Taylor evaluation has been set to 1.
> evaluate(f-p, [0;1]);
[-0.138131114682060169766463621512015307211349976570287;0.1392152836510705168921
00217016266210814525381472255]
```

7.148 timing

Name: timing

global variable controlling timing measures in Sollya.

Description:

• **timing** is a global variable. When its value is **on**, the time spent in each command is measured and displayed (for **verbosity** levels higher than 1).

Example 1:

```
> verbosity=1!;
> timing=on;
Timing has been activated.
> p=remez(sin(x),10,[-1;1]);
Information: Remez: computing the matrix spent 1 ms
Information: Remez: computing the quality of approximation spent 12 ms
Information: computing a minimax approximation spent 16 ms
Information: assignment spent 16 ms
Information: full execution of the last parse chunk spent 16 ms
```

See also: on (7.92), off (7.91)

7.149 tripledouble

Names: tripledouble, TD

represents a number as the sum of three IEEE doubles.

Description:

- tripledouble is both a function and a constant.
- As a function, it rounds its argument to the nearest number that can be written as the sum of three double precision numbers.
- The algorithm used to compute **tripledouble**(x) is the following: let xh = **double**(x) and let xl = **doubledouble**(x-xh). Return the number xh+xl. Note that if the current precision is not sufficient to represent exactly xh+xl, a rounding will occur and the result of **tripledouble**(x) will be useless.
- As a constant, it symbolizes the triple-double precision format. It is used in contexts when a precision format is necessary, e.g. in the commands **roundcoefficients** and **implementpoly**. See the corresponding help pages for examples.

Example 1:

See also: double (7.36), doubleextended (7.38), doubledouble (7.37), roundcoefficients (7.128), implementpoly (7.66)

7.150 true

Name: true

the boolean value representing the truth.

Description:

• true is the usual boolean value.

Example 1:

```
> true && false;
false
> 2>1;
true
```

See also: **false** (7.52), && (7.6), || (7.93)

7.151 var

Name: var

declaration of a local variable in a scope

Usage:

var identifier1, identifier2,..., identifiern: void

Parameters:

• identifier1, identifier2,..., identifiern represent variable identifiers

Description:

• The keyword **var** allows for the declaration of local variables *identifier1* through *identifiern* in a begin-end-block (-block). Once declared as a local variable, an identifier will shadow identifiers declared in higher scopes and undeclared identifiers available at top-level. Variable declarations using **var** are only possible in the beginning of a begin-end-block. Several **var** statements can be given. Once another statement is given in a begin-end-block, no more **var** statements can be given. Variables declared by **var** statements are dereferenced as **error** until they are assigned a value.

Example 1:

```
> exp(x);
exp(x)
> a = 3;
> {var a, b; a=5; b=3; {var a; var b; b = true; a = 1; a; b;}; a; b; };
1
true
5
3
> a;
3
```

See also: **error** (7.43)

7.152 verbosity

Name: verbosity

global variable controlling the quantity of information displayed by commands.

Description:

- **verbosity** accepts any integer value. At level 0, commands do not display anything on standard out. Note that very critical information may however be displayed on standard err.
- Default level is 1. It displays important informations such as warnings when roundings happen.
- For higher levels more informations are displayed depending on the command.

Example 1:

```
> verbosity=0!;
> 1.2+"toto";
error
> verbosity=1!;
> 1.2+"toto";
Warning: Rounding occurred when converting the constant "1.2" to floating-point
with 165 bits.
If safe computation is needed, try to increase the precision.
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
error
> verbosity=2!;
> 1.2+"toto";
Warning: Rounding occurred when converting the constant "1.2" to floating-point
with 165 bits.
If safe computation is needed, try to increase the precision.
Warning: at least one of the given expressions or a subexpression is not correct
ly typed
or its evaluation has failed because of some error on a side-effect.
Information: the expression or a partial evaluation of it has been the following
error
```

7.153 void

Name: void

the functional result of a side-effect or empty argument resp. the correponding type

Usage:

void : void — type type

Description:

- The variable **void** represents the functional result of a side-effect or an empty argument. It is used only in combination with the applications of procedures or identifiers bound through **external proc** to external procedures. The **void** result produced by a procedure or an external procedure is not printed at the prompt. However, it is possible to print it out in a print statement or in complex data types such as lists. The **void** argument is implicit when giving not argument to a procedure or an external procedure when applied. It can be explicited nevertheless. For example, suppose that foo is a procedure or an external procedure with a void argument. Then foo() and foo(void) are correct calls to foo.
- **void** is used also as a type identifier for **externalproc**. Typically, an external procedure taking **void** as an argument or returning **void** is bound with a signature **void** -; some type or some type -; **void**. See **externalproc** for more details.

Example 1:

```
> print(void);
void
> void;
```

Example 2:

```
> hey = proc() { print("Hello world."); };
> hey;
proc()
begin
print("Hello world.");
return void;
end
> hey();
Hello world.
> hey(void);
Hello world.
> print(hey());
Hello world.
void
```

```
> bashexecute("gcc -fPIC -Wall -c externalprocvoidexample.c");
> bashexecute("gcc -fPIC -shared -o externalprocvoidexample externalprocvoidexam
ple.o");
> externalproc(foo, "./externalprocvoidexample", void -> void);
> foo;
foo(void) -> void
> foo();
Hello from the external world.
> foo(void);
Hello from the external world.
> print(foo());
Hello from the external world.
void
```

See also: error (7.43), proc (7.112), externalproc (7.51)

7.154 worstcase

Name: worstcase searches for hard-to-round

Usage:

 $\mathbf{worstcase}(function, preimage\ precision,\ preimage\ exponent\ range,\ image\ precision,\ error\ bound):$ $(function,\ integer,\ range,\ integer,\ constant) \to \mathsf{void}$ $\mathbf{worstcase}(function,\ preimage\ precision,\ preimage\ exponent\ range,\ image\ precision,\ error\ bound,$ $filename):\ (function,\ integer,\ range,\ integer,\ constant,\ string) \to \mathsf{void}$

Parameters:

- \bullet function represents the function to be considered
- preimage precision represents the precision of the preimages
- ullet preimage exponent range represents the exponents in the preimage format
- \bullet image precision represents the precision of the format the images are to be rounded to
- error bound represents the upper bound for the search w.r.t. the relative rounding error
- filename represents a character sequence containing a filename

Description:

• The worstcase command is deprecated. It searches hard-to-round cases of a function. The command searchgal has a comparable functionality.

Example 1:

See also: round (7.127), searchgal (7.132), evaluate (7.44)

7.155 write

Name: write prints an expression without separators

Usage:

```
\mathbf{write}(expr1,...,exprn): (\mathsf{any}\ \mathsf{type},...,\ \mathsf{any}\ \mathsf{type}) \to \mathsf{void} \mathbf{write}(expr1,...,exprn) > \mathit{filename}: (\mathsf{any}\ \mathsf{type},...,\ \mathsf{any}\ \mathsf{type},\ \mathsf{string}) \to \mathsf{void} \mathbf{write}(expr1,...,exprn) >> \mathit{filename}: (\mathsf{any}\ \mathsf{type},...,\mathsf{any}\ \mathsf{type},\ \mathsf{string}) \to \mathsf{void}
```

Parameters:

- *expr* represents an expression
- filename represents a character sequence indicating a file name

Description:

• write(expr1,...,exprn) prints the expressions expr1 through exprn. The character sequences corresponding to the expressions are concatenated without any separator. No newline is displayed at the end. In contrast to print, write expects the user to give all separators and newlines explicitely. If a second argument filename is given after a single ">", the displaying is not output on the standard output of Sollya but if in the file filename that get newly created or overwritten. If a double ">>" is given, the output will be appended to the file filename. The global variables display, midpointmode and fullparentheses have some influence on the formatting of the output (see display, midpointmode and fullparentheses). Remark that if one of the expressions expri given in argument is of type string, the character sequence expri evaluates to is displayed. However, if expri is of type list and this list contains a variable of type string, the expression for the list is displayed, i.e. all character sequences get displayed surrounded by quotes (""). Nevertheless, escape sequences used upon defining character sequences are interpreted immediately.

Example 1:

```
> write(x + 2 + exp(sin(x)));
> write("Hello\n");
x + 2 + exp(sin(x))Hello
> write("Hello","world\n");
Helloworld
> write("Hello","you", 4 + 3, "other persons.\n");
Helloyou7other persons.
```

Example 2:

```
> write("Hello","\n");
Hello
> write([|"Hello"|],"\n");
[|"Hello"|]
> s = "Hello";
> write(s,[|s|],"\n");
Hello[|"Hello"|]
> t = "Hello\tyou";
> write(t,[|t|],"\n");
Hello you[|"Hello you"|]
```

```
> write(x + 2 + exp(sin(x))) > "foo.sol";
> readfile("foo.sol");
x + 2 + exp(sin(x))
```

Example 4:

```
> write(x + 2 + exp(sin(x))) >> "foo.sol";
```

See also: print (7.107), printexpansion (7.108), printhexa (7.110), printfloat (7.109), printxml (7.111), readfile (7.118), autosimplify (7.13), display (7.34), midpointmode (7.84), fullparentheses (7.56), evaluate (7.44)

8 Grammar of the Sollya language

```
program
                             statement
                             program statement
statement
                             command;
command
                             simplecommand
                             { commandlist }
                             { variabledeclarationlist commandlist }
                             { variabledeclarationlist }
                             { }
                             if ifcommand
                             while thing do command
                             for forcommand
ifcommand
                             thing then command
                             thing then command else command
forcommand
                             identifier from thing to thing do command
                             identifier from thing to thing by thing do command
                             identifier in thing do command
commandlist
                             command;
                             command; commandlist
variable declaration list
                             variabledeclaration;
                             variable declaration \ ; \ variable declaration list
variabledeclaration
                             var identifierlist
identifierlist
                             identifier
                             identifier, identifierlist
                             () { commandlist }
procbody
                             () { variabledeclarationlist commandlist }
                             () { variabledeclarationlist }
                             (){}
                             () { commandlist return thing; }
                              ) { variabledeclarationlist commandlist return thing ; }
                             () { variable declaration list return thing; }
                             () { return thing; }
                             (identifierlist) { commandlist }
                             (identifierlist) { variabledeclarationlist commandlist }
                             (identifierlist) { variabledeclarationlist }
                             (identifierlist) { }
                             (identifierlist) { commandlist return thing;}
                             (identifierlist) { variabledeclarationlist commandlist return thing; }
                              identifierlist ) { variabledeclarationlist return thing ; }
                             (identifierlist) { return thing; }
```

```
simplecommand
                        quit
                        restart
                        nop
                        print ( thinglist )
                        print ( thinglist ) > thing
                        print ( thinglist ) > > thing
                        plot ( thing , thinglist )
                        printhexa ( thing )
                        printfloat ( thing )
                        printbinary ( thing )
                        printexpansion ( thing )
                        {\bf bashexecute}\ (\ {\rm thing}\ )
                        external plot (thing, thing, thing, thing, thinglist)
                        write (thinglist)
                        \mathbf{write} ( thinglist ) > thing
                        \mathbf{write} ( thinglist ) > > thing
                        asciiplot (thing, thing)
                        printxml ( thing )
                        execute ( thing )
                        printxml ( thing ) > thing
                        printxml ( thing ) >> thing
                        worstcase (thing, thing, thing, thing ist)
                        rename (identifier, identifier)
                        external
proc ( identifier , thing , external
proctypelist - >
                        extended external proctype )
                        assignment
                        thinglist
                        procedure identifier procbody
assignment
                        stateassignment
                        stillstateassignment!
                        simpleassignment
                        simpleassignment!
simpleassignment
                        identifier = thing
                        identifier := thing
                        identifier = library ( thing )
                        indexing = thing
                        indexing := thing
stateassignment
                        prec = thing
                        points = thing
                        diam = thing
                        display = thing
                        verbosity = thing
                        canonical = thing
                        autosimplify = thing
                        taylorrecursions = thing
                        timing = thing
                        full parentheses = thing
                        midpointmode = thing
                        hopitalrecursions = thing
```

```
stillstateassignment
                           prec = thing
                           points = thing
                           diam = thing
                           display = thing
                           verbosity = thing
                           canonical = thing
                           autosimplify = thing
                           taylorrecursions = thing
                           timing = thing
                           fullparentheses = thing
                           midpointmode = thing
                           hopitalrecursions = thing
thinglist
                           thing
                           thing , thing
list
thing
                           megaterm
                           thing && megaterm
                           thing || megaterm
                           ! megaterm
indexing
                           basicthing [thing]
megaterm
                           hyperterm
                           megaterm == hyperterm
                           {\rm megaterm} < {\rm hyperterm}
                           {\rm megaterm} > {\rm hyperterm}
                           megaterm < = hyperterm
                           megaterm > = hyperterm
                           megaterm != hyperterm
hyperterm
                           \operatorname{term}
                           {\rm hyperterm}\,+\,{\rm term}
                           {\rm hyperterm} - {\rm term}
                           hyperterm @ term
                           hyperterm :: term
                           hyperterm :: term
                           hyperterm:. term
_{\text{term}}
                           \operatorname{subterm}
                           - subterm
                           term * subterm
                           term / subterm
subterm
                           basicthing
                           subterm ^ basicthing
```

```
basicthing \rightarrow
                   on
                   off
                   dyadic
                   powers
                   binary
                   hexadecimal
                   file
                   postscript
                   postscriptfile
                   perturb
                   RD
                   \mathbf{R}\mathbf{U}
                   \mathbf{R}\mathbf{Z}
                   RN
                   {\bf honor coeffprec}
                   \mathbf{true}
                   void
                   false
                   default
                   decimal
                   absolute
                   relative
                   error
                   double
                   {\bf double extended}
                   double double
                   {f triple double}
                   string
                   constant
                   identifier
                   isbound ( identifier )
                   identifier (thinglist)
                   identifier ()
                   list
                   range
                   debound
                   {\it head function}
                   (thing)
                   state de reference
                   indexing
                   (thing) (thinglist)
                   \mathbf{proc} procbody
constant
                   constant
                   dyadicconstant
                   hexconstant
                   hexa decimal constant\\
                   binaryconstant
                   pi
list
                   [ || ]
                   [ | simplelist | ]
                   [ | simplelist ... | ]
```

```
simplelist
                  thing
                   thing, simplelist
                   thing, ..., simplelist
                  [thing, thing]
range
                  [thing; thing]
                   *< thing >*
debound
                   *< thing >.
                   * < thing >_{-}
                  sup (thing)
                  mid (thing)
                  inf (thing)
headfunction
                  diff (thing)
                  simplify ( thing )
                  remez (thing, thing, thinglist)
                  horner (thing)
                  canonical (thing)
                  expand (thing)
                  simplifysafe (thing)
                  taylor (thing, thing, thing)
                   degree (thing)
                  numerator ( thing )
                   denominator (thing)
                  substitute ( thing , thing )
                  coeff (thing, thing)
                  subpoly (thing, thing)
                  roundcoefficients (thing, thing)
                  rational approx (thing, thing)
                  accurateinfnorm (thing, thing thinglist)
                  roundtoformat (thing, thing, thing)
                   evaluate (thing, thing)
                  parse (thing)
                  readxml (thing)
                  infnorm ( thing , thinglist )
                  findzeros (thing, thing)
                  fpfindzeros (thing, thing)
                   dirtyinfnorm (thing, thing)
                  integral (thing, thing)
                   dirtyintegral (thing, thing)
                  implementpoly ( thing , thing , thing , thing , thing , thing ist )
                  {\bf checkinfnorm} ( {\bf thing} , {\bf thing} , {\bf thing} )
                  zerodenominators (thing, thing)
                  isevaluable (thing, thing)
                  searchgal (thinglist)
                  guessdegree (thing, thing, thinglist)
                   dirtyfindzeros (thing, thing)
                  head (thing)
                  roundcorrectly (thing)
                  readfile (thing)
                  revert (thing)
                  sort (thing)
                  mantissa (thing)
                  exponent (thing)
```

```
precision ( thing )
                     tail (thing)
                     sqrt (thing)
                     \mathbf{exp} ( thing )
                     log (thing)
                     log2 (thing)
                     log10 (thing)
                     sin (thing)
                     cos (thing)
                     tan (thing)
                     asin (thing)
                     acos (thing)
                     atan (thing)
                     sinh (thing)
                     cosh (thing)
                     tanh (thing)
                     asinh (thing)
                     acosh (thing)
                     atanh (thing)
                     abs (thing)
                     erf (thing)
                     erfc (thing)
                     log1p (thing)
                     expm1 (thing)
                     double (thing)
                     doubledouble (thing)
                     tripledouble ( thing )
                     doubleextended (thing)
                     ceil (thing)
                     floor (thing)
                     length ( thing )
statedereference
                     prec = ?
                     points = ?
                     diam = ?
                     display = ?
                     verbosity = ?
                     canonical = ?
                     autosimplify = ?
                     taylorrecursions = ?
                     timing = ?
                     full parentheses = ?
                     midpointmode = ?
                     hopital recursions = ?
```

 ${\bf external proctype}$ constantfunction range ${\bf integer}$ string boolean list of constant list of function list of range list of integer list of string list of boolean ${\bf extended external proctype}$ void ${\it external proctype}$ ${\bf external proctype simple list}$ external proctype ${\bf external proctype} \ , \ {\bf external proctype simple list}$ ${\bf external proctype list}$ ${\bf extended external proctype}$

(external proctypesimplelist)